OCaml Programming: Correct + Efficient + Beautiful

Michael R. Clarkson et al.

CONTENTS

I	Intr	roduction	3
1	Bette	er Programming Through OCaml The Past of OCaml	5
	1.1	The Present of OCaml	6
	1.3	Look to Your Future	8
	1.4	About This Book	8
	1.5	A Brief History of CS 3110	10
	1.6	Summary	11
	1.0	outmany	- 11
2	The l	Basics of OCaml	13
	2.1	The OCaml Toplevel	14
	2.2	Compiling OCaml Programs	17
	2.3	Expressions	18
	2.4	Functions	24
	2.5	Documentation	38
	2.6	Printing	40
	2.7	Debugging	44
	2.8	Summary	48
	2.9	Exercises	49
TT	0.0		5 2
II	OC	Caml Programming	53
3	Data	and Types	55
	3.1	Lists	55
	3.2	Variants	66
	3.3	Unit Testing with OUnit	68
	3.4	Records and Tuples	75
	3.5	Advanced Pattern Matching	79
	3.6	Type Synonyms	82
	3.7	Options	82
	3.8	Association Lists	84
	3.9	Algebraic Data Types	85
	3.10	Exceptions	93
	3.11	Example: Trees	97
	3.12	1	100
	3.13	2	102
	3.14	Exercises	104

5	5.1 Module Systems				114 120 122 130 132 134 135 136 141 142 143 158
	5.5 Functional Data Structures 5.6 Module Type Constraints 5.7 Compilation Units 5.8 Includes 5.9 Functors 5.10 Summary 5.11 Exercises	 		 	183 189 193 201 214
П	III Correctness and Efficiency				223
6	6.1 Specifications 6.2 Function Documentation 6.3 Module Documentation 6.4 Testing and Debugging 6.5 Black-box and Glass-box Testing 6.6 Randomized Testing with QCheck 6.7 Proving Correctness 6.8 Structural Induction 6.9 Algebraic Specification 6.10 Summary 6.11 Exercises			 	227 230 238 240 246 253 263 273 282
7 IX	7 Mutability 7.1 Refs 7.2 Mutable Fields 7.3 Arrays and Loops 7.4 Hash Tables IV Appendix				291 305 309 310
	1 1 Appendix				<i>32</i> 3
8					325

8.7	Algorithms and Effici	iency, Attempt 3	3	329

A textbook on functional programming and data structures in OCaml, with an emphasis on semantics and software engineering. This book is the textbook for CS 3110 Data Structures and Functional Programming at Cornell University. A past title of this book was "Functional Programming in OCaml".

THIS EDITION IS IN PROGRESS AND INCOMPLETE. If you have stumbled across this during Summer 2021, please know that I'm in the middle of converting from an old version that used Gitbook to this new version based on Jupyter Book. I'm porting each chapter in order, so it will become more complete as the summer goes on.

Authors. This book is based on courses taught by Michael R. Clarkson, Robert L. Constable, Nate Foster, Michael D. George, Dan Grossman, Daniel P. Huttenlocher, Dexter Kozen, Greg Morrisett, Andrew C. Myers, Radu Rugina, and Ramin Zabih. Together they have created over 20 years worth of course notes and intellectual contributions. Teasing out who contributed what is, by now, not an easy task. The primary compiler and author of this work in its form as a unified textbook is Michael R. Clarkson, who as of this edition is the author of about 40% of the words and code tokens.

Copyright 2021 Michael R. Clarkson. Released under the Creative Commons Attribution-NonCommercial-NoDerivatives 4.0 International License.

CONTENTS 1

2 CONTENTS

Part I

Introduction

BETTER PROGRAMMING THROUGH OCAML

Do you already know how to program in a mainstream language like Python or Java? Good. This book is for you. It's time to learn how to program better. It's time to learn a functional language, OCaml.

Functional programming provides a different perspective on programming than what you have experienced so far. Adapating to that perspective requires letting go of old ideas: assignment statements, loops, classes and objects, among others. That won't be easy.

Nan-in, a Japanese master during the Meiji era (1868-1912), received a university professor who came to inquire about Zen. Nan-in served tea. He poured his visitor's cup full, and then kept on pouring. The professor watched the overflow until he no longer could restrain himself. "It is overfull. No more will go in!" "Like this cup," Nan-in said, "you are full of your own opinions and speculations. How can I show you Zen unless you first empty your cup?"

I believe that learning OCaml will make you a better programmer. Here's why:

- You will experience the freedom of *immutability*, in which the values of so-called "variables" cannot change. Goodbye, debugging.
- You will improve at abstraction, which is the practice of avoiding repetition by factoring out commonality. Goodbye, bloated code.
- You will be exposed to a *type system* that you will at first hate because it rejects programs you think are correct. But you will come to love it, because you will humbly realize it was right and your programs were wrong. Goodbye, failing tests.
- You will be exposed to some of the *theory and implementation of programming languages*, helping you to understand the foundations of what you are saying to the computer when you write code. Goodbye, mysterious and magic incantations.

All of those ideas can be learned in other contexts and languages. But OCaml provides an incredible opportunity to bundle them all together. **OCaml will change the way you think about programming.**

"A language that doesn't affect the way you think about programming is not worth knowing."

---Alan J. Perlis (1922-1990), first recipient of the Turing Award

Moreover, OCaml is beautiful. OCaml is elegant, simple, and graceful. Aesthetics do matter. Code isn't written just to be executed by machines. It's also written to communicate to humans. Elegant code is easier to read and maintain. It isn't necessarily easier to write.

The OCaml code you write can be stylish and tasteful. At first, this might not be apparent. You are learning a new language after all—you wouldn't expect to appreciate Sanskrit poetry on day 1 of Introductory Sanskrit. In fact, you'll likely feel frustrated for awhile as you struggle to express yourself in a new language. So give it some time. After you've mastered OCaml, you might be surprised at how ugly those other languages you already know end up feeling when you return to them.

1.1 The Past of OCaml

Genealogically, OCaml comes from the line of programming languages whose grandfather is Lisp and includes other modern languages such as Clojure, F#, Haskell, and Racket.

OCaml originates from work done by Robin Milner and others at the Edinburgh Laboratory for Computer Science in Scotland. They were working on theorem provers in the late 1970s and early 1980s. Traditionally, theorem provers were implemented in languages such as Lisp. Milner kept running into the problem that the theorem provers would sometimes put incorrect "proofs" (i.e., non-proofs) together and claim that they were valid. So he tried to develop a language that only allowed you to construct valid proofs. ML, which stands for "Meta Language", was the result of that work. The type system of ML was carefully constructed so that you could only construct valid proofs in the language. A theorem prover was then written as a program that constructed a proof. Eventually, this "Classic ML" evolved into a full-fledged programming language.

In the early '80s, there was a schism in the ML community with the French on one side and the British and US on another. The French went on to develop CAML and later Objective CAML (OCaml) while the Brits and Americans developed Standard ML. The two dialects are quite similar. Microsoft introduced its own variant of OCaml called F# in 2005.

Milner received the Turing Award in 1991 in large part for his work on ML. The ACM website for his award includes this praise:

ML was way ahead of its time. It is built on clean and well-articulated mathematical ideas, teased apart so that they can be studied independently and relatively easily remixed and reused. ML has influenced many practical languages, including Java, Scala, and Microsoft's F#. Indeed, no serious language designer should ignore this example of good design.

1.2 The Present of OCaml

OCaml is a functional programming language. The key linguistic abstraction of functional languages is the mathematical function. A function maps an input to an output; for the same input, it always produces the same output. That is, mathematical functions are *stateless*: they do not maintain any extra information or *state* that persists between usages of the function. Functions are *first-class*: you can use them as input to other functions, and produce functions as output. Expressing everything in terms of functions enables a uniform and simple programming model that is easier to reason about than the procedures and methods found in other families of languages.

Imperative programming languages such as C and Java involve *mutable* state that changes throughout execution. *Commands* specify how to compute by destructively changing that state. Procedures (or methods) can have *side effects* that update state in addition to producing a return value.

The **fantasy of mutability** is that it's easy to reason about: the machine does this, then this, etc.

The **reality of mutability** is that whereas machines are good at complicated manipulation of state, humans are not good at understanding it. The essence of why that's true is that mutability breaks *referential transparency*: the ability to replace an expression with its value without affecting the result of a computation. In math, if f(x) = y, then you can substitute y anywhere you see f(x). In imperative languages, you cannot: f might have side effects, so computing f(x) at time t might result in a different value than at time t'.

It's tempting to believe that there's a single state that the machine manipulates, and that the machine does one thing at a time. Computer systems go to great lengths in attempting to provide that illusion. But it's just that: an illusion. In reality, there are many states, spread across threads, cores, processors, and networked computers. And the machine does many things concurrently. Mutability makes reasoning about distributed state and concurrent execution immensely difficult.

Immutability, however, frees the programmer from these concerns. It provides powerful ways to build correct and concurrent programs. OCaml is primarily an immutable language, like most functional languages. It does support imperative programming with mutable state, but we won't use those features until many chapters into the book—in part because we

simply won't need them, and in part to get you to quit "cold turkey" from a dependence you might not have known that you had. This freedom from mutability is one of the biggest changes in perspective that OCaml can give you.

1.2.1 The Features of OCaml

OCaml is a *statically-typed* and *type-safe* programming language. A statically-typed language detects type errors at compile time, so that programs with type errors cannot be executed. A type-safe language limits which kinds of operations can be performed on which kinds of data. In practice, this prevents a lot of silly errors (e.g., treating an integer as a function) and also prevents a lot of security problems: over half of the reported break-ins at the Computer Emergency Response Team (CERT, a US government agency tasked with cybersecurity) were due to buffer overflows, something that's impossible in a type-safe language.

Some functional languages, like Python and Racket, are type-safe but *dynamically typed*. That is, type errors are caught only at run time. Other languages, like C and C++, are statically typed but not type safe. There's no guarantee that a type error won't occur an run time. And still other languages, like Java, use a combination of static and dynamic typing to achieve type safety.

OCaml supports a number of advanced features, some of which you will have encountered before, and some of which are likely to be new:

- Algebraic datatypes: You can build sophisticated data structures in OCaml easily, without fussing with pointers
 and memory management. Pattern matching—a feature we'll soon learn about that enables examining the shape of
 a data structure—makes them even more convenient.
- **Type inference:** You do not have to write type information down everywhere. The compiler automatically figures out most types. This can make the code easier to read and maintain.
- **Parametric polymorphism:** Functions and data structures can be parameterized over types. This is crucial for being able to re-use code.
- Garbage collection: Automatic memory management relieves you from the burden of memory allocation and deallocation, a common source of bugs in languages such as C.
- **Modules:** OCaml makes it easy to structure large systems through the use of modules. Modules are used to encapsulate implementations behind interfaces. OCaml goes well beyond the functionality of most languages with modules by providing functions (called *functors*) that manipulate modules.

1.2.2 OCaml in Industry

OCaml and other functional languages are nowhere near as popular as Python, C, or Java. OCaml's real strength lies in language manipulation (i.e., compilers, analyzers, verifiers, provers, etc.). This is not surprising, because OCaml evolved from the domain of theorem proving.

That's not to say that functional languages aren't used in industry. There are many industry projects using OCaml and Haskell, among other languages. Yaron Minsky (Cornell PhD '02) even wrote a paper about using OCaml in the financial industry. It explains how the features of OCaml make it a good choice for quickly building complex software that works.

1.3 Look to Your Future

General-purpose languages come and go. In your life you'll likely learn a handful. Today, it's Python and Java. Yesterday, it was Pascal and Cobol. Before that, it was Fortran and Lisp. Who knows what it will be tomorrow? In this fast-changing field you need to be able to rapidly adapt. A good programmer has to learn the principles behind programming that transcend the specifics of any specific language. There's no better way to get at these principles than to approach programming from a functional perspective. Learning a new language from scratch affords the opportunity to reflect along the way about the difference between *programming* and programming *in a language*.

If after OCaml you want to learn more about functional programming, you'll be well prepared. OCaml does a great job of clarifying and simplifying the essence of functional programming in a way that other languages that blend functional and imperative programming (like Scala) or take functional programming to the extreme (like Haskell) do not.

And even if you never code in OCaml again after learning it, you'll still be better prepared for the future. Advanced features of functional languages have a surprising tendency to predict new features of more mainstream languages. Java brought garbage collection into the mainstream in 1995; Lisp had it in 1958. Java didn't have generics until version 5 in 2004; the ML family had it in 1990. First-class functions and type inference have been incorporated into mainstream languages like Java, C#, and C++ over the last 10 years, long after functional languages introduced them.

News Flash!

Python just announced plans to support pattern matching in February 2021.

1.4 About This Book

Reporting Errors. If you find an error, please report it! Or if you have a suggestion about how to rewrite some part of the book, let us know. Just go to the page of the book for which you'd like to make a suggestion, click on the Github icon (it looks like a cat) near the top right of the page, and click "open issue" or "suggest edit". The latter is a little heavier weight, because it requires you to fork the textbook repository with Github. But for minor edits that will be appreciated and lead to much quicker uptake of suggestions.

Videos. You will find about 200 YouTube videos embedded throughout this book. The videos usually provide an introduction to material, upon which the textbook then expands. These videos were produced during pandemic when the Cornell course that uses this textbook, CS 3110, had to be asynchronous. The student response to them was overwhelmingly positive, so they are now being made public as part of the textbook. But just so you know, they were not produced by a professional A/V team—just a guy in his basement who was learning as he went.

The videos mostly use the versions of OCaml and its ecosystem that were current in Fall 2020. Current versions you are using are likely to look different from the videos, but don't be alarmed: the underlying ideas are the same. The most visible difference is likely to be the VS Code plugin for OCaml. In Fall 2020 the badly-aging "OCaml and Reason IDE" plugin was still being used. It has since been superseded by the "OCaml Platform" plugin.

The order that the textbook covers topics sometimes differs from the order that the videos cover the topics, simply because the videos originate from lectures. The videos are placed in the textbook nearest to the topic they cover, but that does mean sometimes the videos are not in chronological order.

Collaborative Annotations. At the right margin of each page, you will find an annotation feature provided by hypothes.is. You can use this to highlight and make private notes as you study the text. You can form study groups to share your annotations, or share them publicly. Check out these tips for how to annotate effectively.

Executable Code. Many pages of this book have OCaml code embedded in them. The output of that code is already shown in the book. Here's an example:

```
print_endline "Hello world!"
```

Hello world!

```
- : unit = ()
```

You can also edit and re-run the code yourself to experiment and check your understanding. Look for the icon near the top right of the page that looks like a rocket ship. In the drop-down menu you'll find two ways to interact with the code:

- *Binder* will launch the site mybinder.org, which is a free cloud-based service for "reproducible, interactive, sharable environments for science at scale." All the computation happens in their cloud servers, but the UI is provided through your browser. It will take a little while for the textbook page to open in Binder. Once it does, you can edit and run the code in a *Jupyter notebook*. Jupyter notebooks are documents (usually ending in the .ipynb extension) that can be viewed in web browsers and used to write narrative content as well as code. They became popular in data science communities (especially Python, R, and Julia) as a way of sharing analyses. Now many languages can run in Jupyter notebooks, including OCaml. Code and text are written in *cells* in a Jupyter notebook. Look at the "Cell" menu in it for commands to run cells. Note that Shift-Enter is usually a hotkey for running the cell that has focus.
- Live code will actually do about the same thing, except that instead of leaving the current textbook page and taking you off to Binder, it will modify the code cells on the page to be editable. It takes some time for the connection to be made behind the scenes, during which you will see "Waiting for kernel". After the connection has been made, you can edit all the code cells on the page and re-run them.

Try interacting with the cell above now to make it print a string of your choice. How about: "Camels are bae."

Tip: When you write "real" OCaml code, this is not the interface you'll be using. You'll write code in an editor such as Visual Studio Code or Emacs, and you'll compile it from a terminal. Binder and Live Code are just for interacting in a seamless with the textbook.

Downloadable Pages. Each page of this book is downloadable in a variety of formats. The download icon is at the top right of each page. You'll always find the original source code of the page, which is usually Markdown—or more precisely MyST Markdown, which is an extension of Markdown for technical writing. Each page is also individually available as PDF, which simply prints from your browser. For the entire book as a PDF, see the paragraph about that below.

Pages with OCaml code cells embedded in them can also be downloaded as Jupyter notebooks. To run those locally on your own machine (instead of in the cloud on Binder), you'll need to install Jupyter. The easiest way of doing that is typically to install Anaconda. Then you'll need to install OCaml Jupyter, which requires that you already have OCaml installed. To be clear, there's no need to install Jupyter or to use notebooks. It's just another way to interact with this textbook beyond reading it.

Exercises and Solutions. At the end of each chapter except the first, you will find a section of exercises. The exercises are annotated with a difficulty rating:

- One star [★]: easy exercises that should take only a minute or two.
- Two stars [★★]: straightforward exercises that should take a few minutes.
- Three stars $[\star\star\star]$: exercises that might require anywhere from five to twenty minutes or so.
- Four [****] or more stars: challenging or time-consuming exercises provided for students who want to dig deeper into the material.

It's possible we've misjudged the difficulty of a problem from time to time. Let us know if you think an annotation is off.

Please do not post your solutions to the exercises anywhere, especially not in public repositories where they could be found by search engines. Solutions to exercises are available to students in Cornell's CS 3110. Instructors at other institutions are welcome to contact Michael Clarkson for access.

1.4. About This Book 9

PDF. A full PDF version of this book is available. It does not contain the embedded videos, annotations, or other features that the HTML version has. It might also have typesetting errors. At this time, no tablet (ePub, etc.) version is available, but most tablets will let you import PDFs.

1.5 A Brief History of CS 3110

This book is the primary textbook for CS 3110 at Cornell University. The course has existed for over two decades and has always taught functional programming, but it has not always used OCaml.

Once upon a time, there was a course at MIT known as 6.001 *Structure and Interpretation of Computer Programs* (SICP). It had a textbook by the same name, and it used Scheme, a functional programming language. Tim Teitelbaum taught a version of the course at Cornell in Fall 1988, following the book rather closely and using Scheme.

CS 212. Dan Huttenlocher had been a TA for 6.001 at MIT; he later became faculty at Cornell. In Fall 1989, he inaugurated CS 212 Modes of Algorithm Expression. Basing the course on SICP, he infused a more rigorous approach to the material. Huttenlocher continued to develop CS 212 through the mid 1990s, using various homegrown dialects of Scheme.

Other faculty began teaching the course regularly. Ramin Zabih had taken 6.001 as a first-year student at MIT. In Spring 1994, having become faculty at Cornell, he taught CS 212. Dexter Kozen (Cornell PhD 1977) first taught the course in Spring 1996. The earliest surviving online record of the course seems to be Spring 1998, which was taught by Greg Morrisett in Dylan; the name of the course had become Structure and Interpretation of Computer Programs.

By Fall 1999, CS 212 had its own lecture notes. As CS 3110 still does, that instance of CS 212 covered functional programming, the substitution and environment models, some data structures and algorithms, and programming language implementation.

CS 312. At that time, the CS curriculum had two introductory programming courses, CS 211 Computers and Programming, and CS 212. Students took one or the other, similar to how students today take either CS 2110 or CS 2112. Then they took CS 410 Data Structures. The earliest surviving online record of CS 410 seems to be from Spring 1998. It covered many data structures and algorithms not covered by CS 212, including balanced trees and graphs, and it used Java as the programming language.

Depending on which course they took, CS 211 or 212, students were entering upper-level courses with different skill sets. After extensive discussions, the faculty chose to make CS 211 required, to rename CS 212 into CS 312 Data Structures and Functional Programming, and to make CS 211 a prerequisite for CS 312. At the same time, CS 410 was eliminated from the curriculum and its contents parceled out to CS 312 and CS 482 Introduction to Analysis of Algorithms. Dexter Kozen taught the final offering of CS 410 in Fall 1999.

Greg Morrisett inaugurated the new CS 312 in Spring 2001. He switched from Scheme to Standard ML. Kozen first taught it in Fall 2001, and Andrew Myers in Fall 2002. Myers began to incorporate material on modular programming from another MIT textbook, *Program Development in Java: Abstraction, Specification, and Object-Oriented Design* by Barbara Liskov and John Guttag. Huttenlocher first taught the course in Spring 2006.

CS 3110. In Fall 2008 two big changes came: the language switched to OCaml, and the university switched to four-digit course numbers. CS 312 became CS 3110. Myers, Huttenlocher, Kozen, and Zabih first taught the revised course in Fall 2008, Spring 2009, Fall 2009, and Fall 2010, respectively. Nate Foster first taught the course in Spring 2012; and Bob Constable and Michael George co-taught for the first time in Fall 2013.

Michael Clarkson (Cornell PhD 2010) first taught the course in Fall 2014, after having first TA'd the course as a PhD student back in Spring 2008. He began to revise the presentation of the OCaml programming material to incorporate ideas by Dan Grossman (Cornell PhD 2003) about a principled approach to learning a programming language by decomposing it into syntax, dynamic, and static semantics. Grossman uses that approach in CSE 341 Programming Languages at the University of Washington and in his popular Programming Languages MOOC.

In Fall 2018 the compilation of this textbook began. It synthesizes the work of over two decades of functional programming instruction at Cornell. In the words of the Cornell Evening Song,

'Tis an echo from the walls Of our own, our fair Cornell.

1.6 Summary

This book is about becoming a better programmer. Studying functional programming will help with that. The biggest obstacle in our way is the frustration of speaking a new language, particularly letting go of mutable state. But the benefits will be great: a discovery that programming transcends programming in any particular language or family of languages, an exposure to advanced language features, and an appreciation of beauty.

1.6.1 Terms and Concepts

- · dynamic typing
- · first-class functions
- · functional programming languages
- · immutability
- Lisp
- ML
- OCaml
- · referential transparency
- · side effects
- state
- static typing
- · type safety

1.6.2 Further Reading

- Introduction to Objective Caml, chapters 1 and 2, a freely available textbook that is recommended for this course
- OCaml from the Very Beginning, chapter 1, a relatively inexpensive PDF textbook that is very gentle and recommended for this course
- A guided tour [of OCaml]: chapter 1 of *Real World OCaml*, a book written by some Cornellians that some students might enjoy reading
- The history of Standard ML: though it focuses on the SML variant of the ML language, it's relevant to OCaml
- The value of values: a lecture by the designer of Clojure (a modern dialect of Lisp) on how the time of imperative programming has passed
- The perils of JavaSchools: an essay by the CEO of Stack Overflow on why (my words here) CS 2110 is not enough, and why you need both CS 3110 and CS 3410.
- Teach yourself programming in 10 years: an essay by a Director of Research at Google that puts the time required to become an educated programmer into perspective

1.6. Summary 11

THE BASICS OF OCAML

This chapter will cover some of the basic features of OCaml. But before we dive in to learning OCaml, let's first talk about a bigger idea: learning languages in general.

One of the secondary goals of this course is not just for you to learn a new programming language, but to improve your skills at learning *how to learn* new languages.

There are five essential components to learning a language: syntax, semantics, idioms, libraries, and tools.

Syntax. By *syntax*, we mean the rules that define what constitutes a textually well-formed program in the language, including the keywords, restrictions on whitespace and formatting, punctuation, operators, etc. One of the more annoying aspects of learning a new language can be that the syntax feels odd compared to languages you already know. But the more languages you learn, the more you'll become used to accepting the syntax of the language for what it is, rather than wishing it were different. (If you want to see some languages with really unusual syntax, take a look at APL, which needs its own extended keyboard, and Whitespace, in which programs consist entirely of spaces, tabs, and newlines.) You need to understand syntax just to be able to speak to the computer at all.

Semantics. By *semantics*, we mean the rules that define the behavior of programs. In other words, semantics is about the meaning of a program—what computation a particular piece of syntax represents. Note that although "semantics" is plural in form, we use it as singular. That's similar to "mathematics" or "physics".

There are two pieces to semantics, the *dynamic* semantics of a language and the *static* semantics of a language. The dynamic semantics define the run-time behavior of a program as it is executed or evaluated. The static semantics define the compile-time checking that is done to ensure that a program is legal, beyond any syntactic requirements. The most important kind of static semantics is probably *type checking*: the rules that define whether a program is well typed or not. Learning the semantics of a new language is usually the real challenge, even though the syntax might be the first hurdle you have to overcome. You need to understand semantics to say what you mean to the computer, and you need to say what you mean so that your program performs the right computation.

Idioms. By *idioms*, we mean the common approaches to using language features to express computations. Given that you might express one computation in many ways inside a language, which one do you choose? Some will be more natural than others. Programmers who are fluent in the language will prefer certain modes of expression over others. We could think of this in terms of using the dominant paradigms in the language effectively, whether they are imperative, functional, object oriented, etc. You need to understand idioms to say what you mean not just to the computer, but to other programmers. When you write code idiomatically, other programmers will understand your code better.

Libraries. *Libraries* are bundles of code that have already been written for you and can make you a more productive programmer, since you won't have to write the code yourself. (It's been said that laziness is a virtue for a programmer.) Part of learning a new language is discovering what libraries are available and how to make use of them. A language usually provides a *standard library* that gives you access to a core set of functionality, much of which you would be unable to code up in the language yourself, such as file I/O.

Tools. At the very least any language implementation provides either a compiler or interpreter as a tool for interacting with the computer using the language. But there are other kinds of tools: debuggers; integrated development environments (IDE); and analysis tools for things like performance, memory usage, and correctness. Learning to use tools that are associated with a language can also make you a more productive programmer. Sometimes it's easy to confuse the tool

itself for the language; if you've only ever used Eclipse and Java together for example, it might not be apparent that Eclipse is an IDE that works with many languages, and that Java can be used without Eclipse.

When it comes to learning OCaml in this book, our focus is primarily on semantics and idioms. We'll have to learn syntax along the way, of course, but it's not the interesting part of our studies. We'll get some exposure to the OCaml standard library and a couple other libraries, notably OUnit (a unit testing framework similar to JUnit, HUnit, etc.). Besides the OCaml compiler and build system, the main tool we'll use is the toplevel, which provides the ability to interactively experiment with code.

2.1 The OCaml Toplevel

The *toplevel* is like a calculator or command-line interface to OCaml. It's similar to JShell for Java, or the interactive Python interpreter. The toplevel is handy for trying out small pieces of code without going to the trouble of launching the OCaml compiler. But don't get too reliant on it, because creating, compiling, and testing large programs will require more powerful tools. Some other languages would call the toplevel a *REPL*, which stands for read-eval-print-loop: it reads programmer input, evaluates it, prints the result, and then repeats.

In a terminal window, type utop to start the toplevel. Press Control-D to exit the toplevel. You can also enter #quit;; and press return. Note that you must type the # there: it is in addition to the # prompt you already see.

2.1.1 Types and values

You can enter expressions into the OCaml toplevel. End an expression with a double semi-colon;; and press the return key. OCaml will then evaluate the expression, tell you the resulting value, and the value's type. For example:

```
# 42;;
-: int = 42
```

Let's dissect that response from utop, reading right to left:

- 42 is the value.
- int is the type of the value.
- The value was not given a name, hence the symbol -.

That utop interaction was "hardcoded" as part of this book. We had to type in all the characters: the #, the -, etc. But the infrastructure used to write this book actually enables us to write code that is evaluated by OCaml at the time the book is translated into HTML or PDF. From now on, that's usually what we will do. It looks like this:

```
42
```

```
- : int = 42
```

The first code block with the 42 in it is the code we asked OCaml to run. If you want to enter that into utop, you can copy and paste it. There's an icon in the top right of the block to do that easily. Just remember to add the double semicolon at the end. The second code block, which is indented a little, is the output from OCaml as the book was being translated.

Tip: If you're viewing this in a web browser, look to the top right for a download icon. Choose the .md option, and you'll see the original MyST Markdown source code for this page of the book. You'll see that the output from the second example above is not actually present in the source code. That's good! It means that the output stays consistent with whatever current version of the OCaml compiler we use to build the book. It also means that any compilation errors can be detected as part of building the book, instead of lurking for you, dear reader, to find them.

You can bind values to names with a let definition, as follows:

```
let x = 42
```

```
val x : int = 42
```

Again, let's dissect that response, this time reading left to right:

- A value was bound to a name, hence the val keyword.
- x is the name to which the value was bound.
- int is the type of the value.
- 42 is the value.

You can pronounce the entire output as "x has type int and equals 42."

2.1.2 Functions

A function can be defined at the toplevel using syntax like this:

```
let increment x = x + 1
```

```
val increment : int -> int = <fun>
```

Let's dissect that response:

- increment is the identifier to which the value was bound.
- int -> int is the type of the value. This is the type of functions that take an int as input and produce an int as output. Think of the arrow -> as a kind of visual metaphor for the transformation of one value into another value—which is what functions do.
- The value is a function, which the toplevel chooses not to print (because it has now been compiled and has a representation in memory that isn't easily amenable to pretty printing). Instead, the toplevel prints <fun>, which is just a placeholder.

Note: <fun> itself is not a value. It just indicates an unprintable function value.

You can "call" functions with syntax like this:

```
increment 0
```

```
- : int = 1
```

increment(21)

```
- : int = 22
```

increment (increment 5)

```
- : int = 7
```

But in OCaml the usual vocabulary is that we "apply" the function rather than "call" it.

Note how OCaml is flexible about whether you write the parentheses or not, and whether you write whitespace or not. One of the challenges of first learning OCaml can be figuring out when parentheses are actually required. So if you find yourself having problems with syntax errors, one strategy is to try adding some parentheses. The preferred style, though, is usually to omit parentheses when they are not needed. So, increment 21 is better than increment (21).

2.1.3 Loading code in the toplevel

In addition to allowing you to define functions, the toplevel will also accept *directives* that are not OCaml code but rather tell the toplevel itself to do something. All directives begin with the # character. Perhaps the most common directive is #use, which loads all the code from a file into the toplevel, just as if you had typed the code from that file into the toplevel.

For example, suppose you create a file named mycode.ml. In that file put the following code:

```
let inc x = x + 1
```

Start the toplevel. Try entering the following expression, and observe the error:

```
inc 3
```

The error occurs because the toplevel does not yet know anything about a function named inc. Now issue the following directive to the toplevel:

```
# #use "mycode.ml";;
```

Note that the first # character above indicates the toplevel prompt to you. The second # character is one that you type to tell the toplevel that you are issuing a directive. Without that character, the toplevel would think that you are trying to apply a function named use.

Now try again:

```
inc 3
```

```
- : int = 4
```

2.1.4 Workflow in the toplevel

The best workflow when using the toplevel with code stored in files is:

- Edit the code in the file.
- Load the code in the toplevel with #use.
- Interactively test the code.
- Exit the toplevel. Warning: do not skip this step.

Tip: Suppose you wanted to fix a bug in your code. It's tempting to not exit the toplevel, edit the file, and re-issue the #use directive into the same toplevel session. Resist that temptation. The "stale code" that was loaded from an earlier #use directive in the same session can cause surprising things to happen—surprising when you're first learning the language, anyway. So always exit the toplevel before re-using a file.

2.2 Compiling OCaml Programs

Using OCaml as a kind of interactive calculator can be fun, but we won't get very far with writing large programs that way. We instead need to store code in files and compile them.

2.2.1 Storing code in files

Open a terminal and use a text editor to create a file called hello.ml. Enter the following code into the file:

```
let _ = print_endline "Hello world!"
```

Note: There is no double semicolon;; at the end of that line of code. The double semicolon is intended for interactive sessions in the toplevel, so that the toplevel knows you are done entering a piece of code. There's usually no reason to write it in a .ml file.

The let _ = above means that we don't care to give a name (hence the "blank" or underscore) to code on the right-hand side of the =.

Save the file and return to the command line. Compile the code:

```
$ ocamlc -o hello.byte hello.ml
```

The compiler is named ocamlc. The -o hello.byte option says to name the output executable hello.byte. The executable contains compiled OCaml bytecode. In addition, two other files are produced, hello.cmi and hello.cmo. We don't need to be concerned with those files for now. Run the executable:

```
$ ./hello.byte
```

It should print Hello world! and terminate.

Now change the string that is printed to something of your choice. Save the file, recompile, and rerun. Try making the code print multiple lines.

This edit-compile-run cycle between the editor and the command line is something that might feel unfamiliar if you're used to working inside IDEs like Eclipse. Don't worry; it will soon become second nature.

Running the compiler directly is good to know how to do, but in larger projects, we want to use the OCaml build system to automatically find and link in libraries. Let's try using it:

```
$ ocamlbuild hello.byte
```

You will get an error from that command. Don't worry; just keep reading this exercise.

The build system is named ocamlbuild. The file we are asking it to build is the compiled bytecode hello.byte. The build system will automatically figure out that hello.ml is the source code for that desired bytecode.

However, the build system likes to be in charge of the whole compilation process. When it sees leftover files generated by a direct call to the compiler, as we did in the previous exercise, it rightly gets nervous and refuses to proceed. If you look at the error message, it says that a script has been generated to clean up from the old compilation. Run that script, and also remove the compiled file:

```
$ _build/sanitize.sh
$ rm hello.byte
```

After that, try building again:

```
$ ocamlbuild hello.byte
```

That should now succeed. There will be a directory _build that is created; it contains all the compiled code. That's one benefit of the build system over directly running the compiler: instead of polluting your source directory with a bunch of generated files, they get cleanly created in a separate directory. There's also a file hello.byte that is created, and it is actually just a link to "real" file of that name, which is in the _build directory.

Now run the executable:

```
$ ./hello.byte
```

You can now easily clean up all the compiled code:

```
$ ocamlbuild -clean
```

That removes the _build directory and hello.byte link, leaving just your source code.

2.2.2 What about Main?

Unlike C or Java, OCaml programs do not need to have a special function named main that is invoked to start the program. The usual idiom is just to have the very last definition in a file serve as the main function that kicks off whatever computation is to be done.

2.3 Expressions

The primary piece of OCaml syntax is the *expression*. Just like programs in imperative languages are primarily built out of *commands*, programs in functional languages are primarily built out of expressions. Examples of expressions include 2+2 and increment 21.

The OCaml manual has a complete definition of all the expressions in the language. Though that page starts with a rather cryptic overview, if you scroll down, you'll come to some English explanations. Don't worry about studying that page now; just know that it's available for reference.

The primary task of computation in a functional language is to *evaluate* an expression to a *value*. A value is an expression for which there is no computation remaining to be performed. So, all values are expressions, but not all expressions are values. Examples of values include 2, true, and "yay!".

The OCaml manual also has a definition of all the values, though again, that page is mostly useful for reference rather than study.

Sometimes an expression might fail to evaluate to a value. There are two reasons that might happen:

- 1. Evaluation of the expression raises an exception.
- 2. Evaluation of the expression never terminates (e.g., it enters an "infinite loop").

2.3.1 Assertions

The expression assert e evaluates e. If the result is true, nothing more happens, and the entire expression evaluates to a special value called *unit*. The unit value is written () and its type is unit. But if the result is false, an exception is raised.

2.3.2 Operators

Operators can be used to form expressions. OCaml has more or less all the usual operators you would expect in a language from the C or Java family of languages. See the table of all operators in the OCaml manual for details.

Here are two things to watch out for as you begin:

- OCaml deliberately does not support operator overloading. As a consequence, the integer and floating-point operators are distinct. E.g., to add integers, use +. To add floating-point numbers, use +...
- There are two equality operators in OCaml, = and ==, with corresponding inequality operators <> and !=. Operators = and <> examine *structural* equality whereas == and != examine *physical* equality. Until we've studied the imperative features of OCaml, the difference between them will be tricky to explain. See the documentation of Stdlib. (==) if you're curious now.

Important: Start training yourself now to use = and not to use ==. This will be difficult if you're coming from a language like Java where == is the usual equality operator.

2.3.3 If Expressions

The expression if e1 then e2 else e3 evaluates to e2 if e1 evaluates to true, and to e3 otherwise. We call e1 the *guard* of the if expression.

```
if 3 + 5 > 2 then "yay!" else "boo!"
```

```
- : string = "yay!"
```

Unlike if-then-else *statements* that you may have used in imperative languages, if-then-else *expressions* in OCaml are just like any other expression; they can be put anywhere an expression can go. That makes them similar to the ternary operator? : that you might have used in other languages.

```
4 + (if 'a' = 'b' then 1 else 2)
```

```
- : int = 6
```

If expressions can be nested in a pleasant way:

```
if e1 then e2
else if e3 then e4
else if e5 then e6
...
else en
```

You should regard the final else as mandatory, regardless of whether you are writing a single if expression or a highly nested if expression. If you omit it you'll likely get an error message that, for now, is inscrutable:

2.3. Expressions 19

if 2 > 3 **then** 5

Syntax. The syntax of an if expression:

```
if e1 then e2 else e3
```

The letter e is used here to represent any other OCaml expression; it's an example of a *syntactic variable* aka *metavariable*, which is not actually a variable in the OCaml language itself, but instead a name for a certain syntactic construct. The numbers after the letter e are being used to distinguish the three different occurrences of it.

Dynamic semantics. The dynamic semantics of an if expression:

- If el evaluates to true, and if el evaluates to a value v, then if el then el else else else else valuates to v
- If e1 evaluates to false, and if e3 evaluates to a value v, then if e1 then e2 else e3 evaluates to v.

We call these *evaluation rules*: they define how to evaluate expressions. Note how it takes two rules to describe the evaluation of an if expression, one for when the guard is true, and one for when the guard is false. The letter v is used here to represent any OCaml value; it's another example of a metavariable. Later we will develop a more mathematical way of expressing dynamic semantics, but for now we'll stick with this more informal style of explanation.

Static semantics. The static semantics of an if expression:

• If e1 has type bool and e2 has type t and e3 has type t then if e1 then e2 else e3 has type t

We call this a *typing rule*: it describes how to type check an expression. Note how it only takes one rule to describe the type checking of an if expression. At compile time, when type checking is done, it makes no difference whether the guard is true or false; in fact, there's no way for the compiler to know what value the guard will have at run time. The letter t here is used to represent any OCaml type; the OCaml manual also has definition of all types (which curiously does not name the base types of the language like int and bool).

We're going to be writing "has type" a lot, so let's introduce a more compact notation for it. Whenever we would write "e has type t", let's instead write e : t. The colon is pronounced "has type". This usage of colon is consistent with how the toplevel responds after it evaluates an expression that you enter:

```
let x = 42
```

```
val x : int = 42
```

In the above example, variable x has type int, which is what the colon indicates.

2.3.4 Let Expressions

In our use of the word let thus far, we've been making definitions in the toplevel and in .ml files. For example,

```
let x = 42;;
```

```
val x : int = 42
```

defines x to be 42, after which we can use x in future definitions at the toplevel. We'll call this use of let a let definition.

There's another use of let which is as an expression:

```
let x = 42 in x + 1
```

```
- : int = 43
```

Here we're binding a value to the name x then using that binding inside another expression, x+1. We'll call this use of let a let expression. Since it's an expression it evaluates to a value. That's different than definitions, which themselves do not evaluate to any value. You can see that if you try putting a let definition in place of where an expression is expected:

```
(let x = 42) + 1
```

Syntactically, a let definition is not permitted on the left-hand side of the + operator, because a value is needed there, and definitions do not evaluate to values. On the other hand, a let expression would work fine:

```
(let x = 42 in <math>x) + 1
```

```
- : int = 43
```

Another way to understand let definitions at the toplevel is that they are like let expression where we just haven't provided the body expression yet. Implicitly, that body expression is whatever else we type in the future. For example,

```
# let a = "big";;
# let b = "red";;
# let c = a ^ b;;
# ...
```

is understand by OCaml in the same way as

```
let a = "big" in
let b = "red" in
let c = a ^ b in
...
```

That latter series of let bindings is idiomatically how several variables can be bound inside a given block of code.

Syntax.

```
let x = e1 in e2
```

2.3. Expressions 21

As usual, x is an identifier. These identifiers must begin with lower-case, not upper, and idiomatically are written with snake_case not camelCase. We call e1 the *binding expression*, because it's what's being bound to x; and we call e2 the *body expression*, because that's the body of code in which the binding will be in scope.

Dynamic semantics.

To evaluate let x = e1 in e2:

- Evaluate e1 to a value v1.
- Substitute v1 for x in e2, yielding a new expression e2'.
- Evaluate e2 ' to a value v2.
- The result of evaluating the let expression is v2.

Here's an example:

```
let x = 1 + 4 in x * 3
--> (evaluate e1 to a value v1)
  let x = 5 in x * 3
--> (substitute v1 for x in e2, yielding e2')
  5 * 3
--> (evaluate e2' to v2)
  15
    (result of evaluation is v2)
```

Static semantics.

• If e1: t1 and if under the assumption that x: t1 it holds that e2: t2, then (let x = e1 in e2): t2.

We use the parentheses above just for clarity. As usual, the compiler's type inferencer determines what the type of the variable is, or the programmer could explicitly annotate it with this syntax:

```
let x : t = e1 in e2
```

2.3.5 Scope

Let bindings are in effect only in the block of code in which they occur. This is exactly what you're used to from nearly any modern programming language. For example:

The *scope* of a variable is where its name is meaningful. Variable y is in scope only inside of the let expression that binds it above.

It's possible to have overlapping bindings of the same name. For example:

```
let x = 5 in ((let x = 6 in x) + x)
```

But this is darn confusing, and for that reason, it is strongly discouraged style—much like ambiguous pronouns are discouraged in natural language. Nonetheless, let's consider what that code means.

To what value does that code evaluate? The answer comes down to how x is replaced by a value each time it occurs. Here are a few possibilities for such *substitution*:

```
(* possibility 1 *)
let x = 5 in
   ((let x = 6 in 6) + 5)

(* possibility 2 *)
let x = 5 in
   ((let x = 6 in 5) + 5)

(* possibility 3 *)
let x = 5 in
   ((let x = 6 in 6) + 6)
```

The first one is what nearly any reasonable language would do. And most likely it's what you would guess But, why?

The answer is something we'll call the *Principle of Name Irrelevance*: the name of a variable shouldn't intrinsically matter. You're used to this from math. For example, the following two functions are the same:

$$f(x) = x^2$$
$$f(y) = y^2$$

It doesn't intrinsically matter whether we call the argument to the function x or y; either way, it's still the squaring function. Therefore, in programs, these two functions should be identical:

```
let f x = x * x
let f y = y * y
```

This principle is more commonly known as *alpha equivalence*: the two functions are equivalent up to renaming of variables, which is also called *alpha conversion* for historical reasons that are unimportant here.

According to the Principle of Name Irrelevance, these two expressions should be identical:

```
let x = 6 in x
let y = 6 in y
```

Therefore, the following two expressions, which have the above expressions embedded in them, should also be identical:

```
let x = 5 in (let x = 6 in x) + x
let x = 5 in (let y = 6 in y) + x
```

But for those to be identical, we **must** choose the first of the three possibilities above. It is the only one that makes the name of the variable be irrelevant.

There is a term commonly used for this phenomenon: a new binding of a variable *shadows* any old binding of the variable name. Metaphorically, it's as if the new binding temporarily casts a shadow over the old binding. But eventually the old binding could reappear as the shadow recedes.

Shadowing is not mutable assignment. For example, both of the following expressions evaluate to 11:

```
let x = 5 in ((let x = 6 in x) + x)
let x = 5 in (x + (let x = 6 in x))
```

Likewise, the following utop transcript is not mutable assignment, though at first it could seem like it is:

```
# let x = 42;;
val x : int = 42
# let x = 22;;
val x : int = 22
```

2.3. Expressions 23

Recall that every let definition in the toplevel is effectively a nested let expression. So the above is effectively the following:

```
let x = 42 in
let x = 22 in
... (* whatever else is typed in the toplevel *)
```

The right way to think about this is that the second let binds an entirely new variable that just happens to have the same name as the first let.

Here is another utop transcript that is well worth studying:

```
# let x = 42;;
val x : int = 42
# let f y = x + y;;
val f : int -> int = <fun>
# f 0;;
: int = 42
# let x = 22;;
val x : int = 22
# f 0;;
- : int = 42 (* x did not mutate! *)
```

To summarize, each let definition binds an entirely new variable. If that new variable happens to have the same name as an old variable, the new variable temporarily shadows the old one. But the old variable is still around, and its value is immutable: it never, ever changes. So even though let expressions might superficially look like assignment statements from imperative languages, they are actually quite different.

2.4 Functions

Since OCaml is a functional language, there's a lot to cover about functions. Let's get started.

Important: Methods and functions are not the same idea. A method is a component of an object, and it implicitly has a receiver that is usually accessed with a keyword like this or self. OCaml functions are not methods: they are not components of objects, and they do not have a receiver.

Some might say that all methods are functions, but not all functions are methods. Some might even quibble with that, making a distinction between functions and procedures. The latter would be functions that do not return any meaningful value, such as a void return type in Java or None return value in Python.

So if you're coming from an object-oriented background, be careful about the terminology. **Everything here is a strictly a function, not a method.**

2.4.1 Function Definitions

The following code

```
let x = 42
```

has an expression in it (42) but is not itself an expression. Rather, it is a *definition*. Definitions bind values to names, in this case the value 42 being bound to the name x. The OCaml manual describes definitions (see the third major grouping titled "definition" on that page), but that manual page is again primarily for reference not for study. Definitions are not

expressions, nor are expressions definitions—they are distinct syntactic classes. But definitions can have expressions nested inside them, and vice-versa.

For now, let's focus on one particular kind of definition, a *function definition*. Non-recursive functions are defined like this:

```
let f x = ...
```

Recursive functions are defined like this:

```
let rec f x = ...
```

The difference is just the rec keyword. It's probably a bit surprising that you explicitly have to add a keyword to make a function recursive, because most languages assume by default that they are. OCaml doesn't make that assumption, though. (Nor does the Scheme family of languages.)

One of the best known recursive functions is the factorial function. In OCaml, it can be written as follows:

```
(** [fact n] is [n]!.
    Requires: [n >= 0]. *)
let rec fact n = if n = 0 then 1 else n * fact (n - 1)
```

```
val fact : int -> int = <fun>
```

We provided a specification comment above the function to document the precondition (Requires) and postcondition (is) of the function.

Note that, as in many languages, OCaml integers are not the "mathematical" integers but are limited to a fixed number of bits. The manual specifies that (signed) integers are at least 31 bits, but they could be wider. As architectures have grown, so has that size. In current implementations, OCaml integers are 63 bits. So if you test on large enough inputs, you might begin to see strange results. The problem is machine arithmetic, not OCaml. (For interested readers: why 31 or 63 instead of 32 or 64? The OCaml garbage collector needs to distinguish between integers and pointers. The runtime representation of these therefore steals one bit to flag whether a word is an integer or a pointer.)

Here's another recursive function:

```
val pow : int -> int = <fun>
```

Note how we didn't have to write any types in either of our functions: the OCaml compiler infers them for us automatically. The compiler solves this *type inference* problem algorithmically, but we could do it ourselves, too. It's like a mystery that can be solved by our mental power of deduction:

- Since the if expression can return 1 in the then branch, we know by the typing rule for if that the entire if expression has type int.
- Since the if expression has type int, the function's return type must be int.
- Since y is compared to 0 with the equality operator, y must be an int.
- Since x is multiplied with another expression using the * operator, x must be an int.

If we wanted to write down the types for some reason, we could do that:

```
let rec pow (x : int) (y : int = ...
```

2.4. Functions 25

The parentheses are mandatory when we write the *type annotations* for x and y. We will generally leave out these annotations, because it's simpler to let the compiler infer them. There are other times when you'll want to explicitly write down types. One particularly useful time is when you get a type error from the compiler that you don't understand. Explicitly annotating the types can help with debugging such an error message.

Syntax. The syntax for function definitions:

```
let rec f x1 x2 ... xn = e
```

The f is a metavariable indicating an identifier being used as a function name. These identifiers must begin with a lowercase letter. The remaining rules for lowercase identifiers can be found in the manual. The names $\times 1$ through $\times n$ are metavariables indicating argument identifiers. These follow the same rules as function identifiers. The keyword rec is required if f is to be a recursive function; otherwise it may be omitted.

Note that syntax for function definitions is actually simplified compared to what OCaml really allows. We will learn more about some augmented syntax for function definition in the next couple weeks. But for now, this simplified version will help us focus.

Mutually recursive functions can be defined with the and keyword:

```
let rec f x1 ... xn = e1
and g y1 ... yn = e2
```

For example:

```
(** [even n] is whether [n] is even.
    Requires: [n >= 0]. *)

let rec even n =
    n = 0 || odd (n - 1)

(** [odd n] is whether [n] is odd.
    Requires: [n >= 0]. *)

and odd n =
    n <> 0 && even (n - 1);;
```

```
val even : int -> bool = <fun>
val odd : int -> bool = <fun>
```

The syntax for function types is:

```
t -> u
t1 -> t2 -> u
t1 -> ... -> tn -> u
```

The t and u are metavariables indicating types. Type $t \to u$ is the type of a function that takes an input of type t and returns an output of type t. We can think of t1 t2 t2 t3 t4 t5 t7 t8 as the type of a function that takes two inputs, the first of type t1 and the second of type t2, and returns an output of type t8. Likewise for a function that takes t8 arguments.

Dynamic semantics. There is no dynamic semantics of function definitions. There is nothing to be evaluated. OCaml just records that the name f is bound to a function with the given arguments x1..xn and the given body e. Only later, when the function is applied, will there be some evaluation to do.

Static semantics. The static semantics of function definitions:

- For non-recursive functions: if by assuming that x1 : t1 and x2 : t2 and ... and xn : tn, we can conclude that e : u, then f : t1 -> t2 -> ... -> tn -> u.
- For recursive functions: if by assuming that x1:t1 and x2:t2 and ... and xn:tn and f:t1 -> t2 -> ... -> tn -> u, we can conclude that e:u, then f:t1 -> t2 -> ... -> tn -> u.

Note how the type checking rule for recursive functions assumes that the function identifier f has a particular type, then checks to see whether the body of the function is well-typed under that assumption. This is because f is in scope inside the function body itself (just like the arguments are in scope).

2.4.2 Anonymous Functions

We already know that we can have values that are not bound to names. The integer 42, for example, can be entered at the toplevel without giving it a name:

```
42
```

```
- : int = 42
```

Or we can bind it to a name:

```
let x = 42
```

```
val x : int = 42
```

Similarly, OCaml functions do not have to have names; they may be *anonymous*. For example, here is an anonymous function that increments its input: $fun \times -> \times + 1$. Here, fun is a keyword indicating an anonymous function, x is the argument, and -> separates the argument from the body.

We now have two ways we could write an increment function:

```
let inc x = x + 1
let inc = fun x -> x + 1
```

```
val inc : int -> int = <fun>
```

```
val inc : int -> int = <fun>
```

They are syntactically different but semantically equivalent. That is, even though they involve different keywords and put some identifiers in different places, they mean the same thing.

Anonymous functions are also called *lambda expressions*, a term that comes from the *lambda calculus*, which is a mathematical model of computation in the same sense that Turing machines are a model of computation. In the lambda calculus, fun \times -> e would be written $\lambda x.e.$ The λ denotes an anonymous function.

It might seem a little mysterious right now why we would want functions that have no names. Don't worry; we'll see good uses for them later in the course, especially when we study so-called "higher-order programming". In particular, we will often create anonymous functions and pass them as input to other functions.

Syntax.

```
fun x1 ... xn -> e
```

Static semantics.

• If by assuming that x1: t1 and x2: t2 and ... and xn: tn, we can conclude that e: u, then fun x1 ... xn -> e: t1 -> t2 -> ... -> tn -> u.

Dynamic semantics. An anonymous function is already a value. There is no computation to be performed.

2.4. Functions 27

2.4.3 Function Application

Here we cover a somewhat simplified syntax of function application compared to what OCaml actually allows.

Syntax.

```
e0 e1 e2 ... en
```

The first expression e0 is the function, and it is applied to arguments e1 through en. Note that parentheses are not required around the arguments to indicate function application, as they are in languages in the C family, including Java.

Static semantics.

```
• If e0 : t1 -> ... -> tn -> u and e1 : t1 and ... and en : tn then e0 e1 ... en : u.
```

Dynamic semantics.

To evaluate e0 e1 ... en:

- 1. Evaluate e0 to a function. Also evaluate the argument expressions e1 through en to values v1 through vn.
 - For e0, the result might be an anonymous function fun $x1 \dots xn \rightarrow e$ or a name f. In the latter case, we need to find the definition of f, which we can assume to be of the form $let rec f x1 \dots xn = e$. Either way, we now know the argument names x1 through xn and the body e.
- 2. Substitute each value \forall i for the corresponding argument name x i in the body e of the function. That substitution results in a new expression e.
- 3. Evaluate e' to a value v, which is the result of evaluating e0 e1 ... en.

If you compare these evaluation rules to the rules for let expressions, you will notice they both involve substitution. This is not an accident. In fact, anywhere let x = e1 in e2 appears in a program, we could replace it with (fun x -> e2) e1. They are syntactically different but semantically equivalent. In essence, let expressions are just syntactic sugar for anonymous function application.

2.4.4 Pipeline

There is a built-in infix operator in OCaml for function application called the *pipeline* operator, written |>. Imagine that as depicting a triangle pointing to the right. The metaphor is that values are sent through the pipeline from left to right. For example, suppose we have the increment function inc from above as well as a function square that squares its input:

```
let square x = x * x
```

```
val square : int -> int = <fun>
```

Here are two equivalent ways of squaring 6:

```
square (inc 5);;
5 |> inc |> square;;
```

```
-: int = 36
```

```
- : int = 36
```

The latter uses the pipeline operator to send 5 through the inc function, then send the result of that through the square function. This is a nice, idiomatic way of expressing the computation in OCaml. The former way is arguably not as elegant: it involves writing extra parentheses and requires the reader's eyes to jump around, rather than move linearly

from left to right. The latter way scales up nicely when the number of functions being applied grows, where as the former way requires more and more parentheses:

```
5 |> inc |> square |> inc |> inc |> square;;
square (inc (square (inc 5))));;
```

```
- : int = 1444
```

```
- : int = 1444
```

It might feel weird at first, but try using the pipeline operator in your own code the next time you find yourself writing a big chain of function applications.

Since $e1 \mid > e2$ is just another way of writing $e2 \mid e1$, we don't need to state the semantics for $\mid >$: it's just the same as function application. These two programs are another example of expressions that are syntactically different but semantically equivalent.

2.4.5 Polymorphic Functions

The *identity* function is the function that simply returns its input:

```
let id x = x
```

```
val id : 'a -> 'a = <fun>
```

Or equivalently as an anonymous function:

```
let id = fun x -> x
```

```
val id : 'a -> 'a = <fun>
```

The 'a is a *type variable*: it stands for an unknown type, just like a regular variable stands for an unknown value. Type variables always begin with a single quote. Commonly used type variables include 'a, 'b, and 'c, which OCaml programmers typically pronounce in Greek: alpha, beta, and gamma.

We can apply the identity function to any type of value we like:

```
id 42;;
id true;;
id "bigred";;
```

```
- : int = 42
```

```
- : bool = true
```

```
- : string = "bigred"
```

Because you can apply id to many types of values, it is a *polymorphic* function: it can be applied to many (*poly*) forms (*morph*).

With manual type annotations, it's possible to give a more restrictive type to a polymorphic function than the type the compiler automatically infers. For example:

2.4. Functions 29

```
let id_int (x : int) : int = x
```

```
val id_int : int -> int = <fun>
```

That's the same function as id, except for the two manual type annotations. Because of those, we cannot apply id_int to a bool like we did id:

```
id_int true
```

Another way of writing id_int would be in terms of id:

```
let id_int' : int -> int = id
```

```
val id_int' : int -> int = <fun>
```

In effect we took a value of type 'a -> 'a, and we bound it to a name whose type was manually specified as being int -> int. You might ask, why does that work? They aren't the same types, after all.

One way to think about this is in terms of **behavior**. The type of id_int specifies one aspect of its behavior: given an int as input, it promises to produce an int as output. It turns out that id also makes the same promise: given an int as input, it too will return an int as output. Now id also makes many more promises, such as: given a bool as input, it will return a bool as output. So by binding id to a more restrictive type int -> int, we have thrown away all those additional promises as irrelevant. Sure, that's information lost, but at least no promises will be broken. It's always going to be safe to use a function of type 'a -> 'a when what we needed was a function of type int -> int.

The converse is not true. If we needed a function of type 'a -> 'a but tried to use a function of type int -> int, we'd be in trouble as soon as someone passed an input of e.g. type bool. Here are two concrete examples, which of course do not compile:

```
let id' : 'a -> 'a = fun x -> x + 1;;
let id'' : 'a -> 'a = id_int;;
```

```
val id' : int -> int = <fun>
```

```
val id'' : int -> int = <fun>
```

The first is maybe more obviously wrong than the second. Function id' is actually the increment function, not the identify function. So passing it a bool or string or some complicated data structure is not safe; the only data + can safely manipulate are integers.

Function id' is maybe less obviously wrong. If follow the chain of definition backwards, we discover that id' is really going to be just $fun \times -> \times$. There's nothing wrong with that code as the implementation of id': it's guaranteed to return an output of the same type as its input, because it *does* return its input. But the compiler doesn't go back and peek "under the covers" like that. Function id_int has type int -> int at the point id' is being defined, and that's all that matters. The compiler has to assume it could be *any* function of that type, including the increment function.

That leads us to another, more mechanical, way to think about all of this in terms of **application**. By that we mean the very same notion of how a function is applied to arguments: when evaluating the application id 5, the argument x is instantiated with value 5. Likewise, the 'a in the type of id is being instantiated with type int at that application. So if we write

```
let id_int' : int -> int = id
```

```
val id_int' : int -> int = <fun>
```

we are in fact instantiating the 'a in the type of id with the type int. And just as there is no way to "unapply" a function—for example, given 6 we can't compute backwards to id 5—we can't unapply that type instantiation and change int back to 'a.

To make that precise, suppose we have a let definition [or expression]:

```
let x = e [in e']
```

and that OCaml infers x has a type t that includes some type variables 'a, 'b, etc. Then we are permitted to instantiate those type variables. We can do that by applying the function to arguments that reveal what the type instantiations should be (as in id 5) or by a type annotation (as in id_int'), among other ways. But we have to be consistent with the instantiation. For example, we cannot take a function of type 'a -> 'b -> 'a and instantiate it at type int -> 'b -> string, because the instantiation of 'a is not the same type in each of the two places it occurred:

```
let first x y = x;;
let first_int : int -> 'b -> int = first;;
let bad_first : int -> 'b -> string = first;;
```

```
val first : 'a -> 'b -> 'a = <fun>
```

```
val first_int : int -> 'b -> int = <fun>
```

2.4.6 Labeled and Optional Arguments

The type and name of a function usually give you a pretty good idea of what the arguments should be. However, for functions with many arguments (especially arguments of the same type), it can be useful to label them. For example, you might guess that the function String.sub returns a substring of the given string (and you would be correct). You could type in String.sub to find its type:

```
String.sub;;
```

```
- : string -> int -> int -> string = <fun>
```

But it's not clear from the type how to use it—you're forced to consult the documentation.

OCaml supports labeled arguments to functions. You can declare this kind of function using the following syntax:

```
let f ~name1:arg1 ~name2:arg2 = arg1 + arg2;;
```

```
val f : name1:int -> name2:int -> int = <fun>
```

This function can be called by passing the labeled arguments in either order:

2.4. Functions 31

```
f ~name2:3 ~name1:4
```

Labels for arguments are often the same as the variable names for them. OCaml provides a shorthand for this case. The following are equivalent:

```
let f ~name1:name1 ~name2:name2 = name1 + name2
let f ~name1 ~name2 = name1 + name2
```

Use of labeled arguments is largely a matter of taste. They convey extra information, but they can also add clutter to types.

The syntax to write both a labeled argument and an explicit type annotation for it is:

```
let f ~name1:(arg1 : int) ~name2:(arg2 : int) = arg1 + arg2
```

It is also possible to make some arguments optional. When called without an optional argument, a default value will be provided. To declare such a function, use the following syntax:

```
let f ?name: (arg1=8) arg2 = arg1 + arg2
```

```
val f : ?name:int -> int -> int = <fun>
```

You can then call a function with or without the argument:

```
f ~name:2 7
```

```
- : int = 9
```

```
f 7
```

```
-: int = 15
```

2.4.7 Partial Application

We could define an addition function as follows:

```
let add x y = x + y
```

```
val add : int -> int -> int = <fun>
```

Here's a rather similar function:

```
let addx x = fun y -> x + y
```

```
val addx : int -> int -> int = <fun>
```

Function addx takes an integer x as input and returns a function of type int \rightarrow int that will add x to whatever is passed to it.

The type of addx is int -> int -> int. The type of add is also int -> int -> int. So from the perspective of their types, they are the same function. But the form of addx suggests something interesting: we can apply it to just a single argument.

```
let add5 = addx 5
```

```
val add5 : int -> int = <fun>
```

```
add5 2
```

```
- : int = 7
```

It turns out the same can be done with add:

```
let add5 = add 5
```

```
val add5 : int -> int = <fun>
```

```
add5 2;;
```

```
- : int = 7
```

What we just did is called *partial application*: we partially applied the function add to one argument, even though you would normally think of it as a multi-argument function. This works because the following three functions are *syntactically different* but *semantically equivalent*. That is, they are different ways of expressing the same computation:

```
let add x y = x + y
let add x = fun y -> x + y
let add = fun x -> (fun y -> x + y)
```

So add is really a function that takes an argument x and returns a function (fun $y \rightarrow x + y$). Which leads us to a deep truth...

2.4.8 Function Associativity

Are you ready for the truth? Take a deep breath. Here goes...

Every OCaml function takes exactly one argument.

Why? Consider add: although we can write it as let add x y = x + y, we know that's semantically equivalent to let add = fun $x \rightarrow (\text{fun } y \rightarrow x + y)$. And in general,

```
let f x1 x2 ... xn = e
```

is semantically equivalent to

So even though you think of f as a function that takes n arguments, in reality it is a function that takes 1 argument and returns a function.

The type of such a function

2.4. Functions 33

```
t1 -> t2 -> t3 -> t4
```

really means the same as

```
t1 -> (t2 -> (t3 -> t4))
```

That is, function types are *right associative*: there are implicit parentheses around function types, from right to left. The intuition here is that a function takes a single argument and returns a new function that expects the remaining arguments.

Function application, on the other hand, is *left associative*: there are implicit parenthesis around function applications, from left to right. So

```
e1 e2 e3 e4
```

really means the same as

```
((e1 e2) e3) e4
```

The intuition here is that the left-most expression grabs the next expression to its right as its single argument.

2.4.9 Operators as Functions

The addition operator + has type int -> int -> int. It is normally written *infix*, e.g., 3 + 4. By putting parentheses around it, we can make it a *prefix* operator:

```
( + )
```

```
- : int -> int -> int = <fun>
```

```
( + ) 3 4;;
```

```
- : int = 7
```

```
let add3 = ( + ) 3
```

```
val add3 : int -> int = <fun>
```

```
add3 2
```

```
- : int = 5
```

The same technique works for any built-in operator.

Normally the spaces are unnecessary. We could write (+) or (-+), but it is best to include them. Beware of multiplication, which *must* be written as (-*), because (*) would be parsed as beginning a comment.

We can even define our own new infix operators, for example:

```
let ( ^^ ) x y = max x y
```

And now 2 ^^ 3 evaluates to 3.

The rules for which punctuation can be used to create infix operators are not necessarily intuitive. Nor is the relative precedence with which such operators will be parsed. So be careful with this usage.

2.4.10 Tail Recursion

Consider the following seemingly uninteresting function, which counts from 1 to n:

```
(** [count n] is [n], computed by adding 1 to itself [n] times.
    this function counts up from 1 to [n]. *)
let rec count n =
 if n = 0 then 0 else 1 + count (n - 1)
```

```
val count : int -> int = <fun>
```

Counting to 10 is no problem:

```
count 10
```

```
-: int = 10
```

Counting to 100,000 is no problem either:

```
count 100_000
```

```
-: int = 100000
```

But try counting to 1,000,000 and you'll get the following error:

val count_forever : 'a -> int = <fun>

```
Stack overflow during evaluation (looping recursion?).
```

What's going on here?

The Call Stack. The issue is that the call stack has a limited size. You probably learned in one of your introductory programming classes that most languages implement function calls with a stack. That stack contains one element for each function call that has been started but has not yet completed. Each element stores information like the values of local variables and which instruction in the function is currently being executed. When the evaluation of one function body calls another function, a new element is pushed on the call stack, and it is popped off when the called function completes.

The size of the stack is usually limited by the operating system. So if the stack runs out of space, it becomes impossible to make another function call. Normally this doesn't happen, because there's no reason to make that many successive function calls before returning. In cases where it does happen, there's good reason for the operating system to make that program stop: it might be in the process of eating up all the memory available on the entire computer, thus harming other programs running on the same computer. The count function isn't likely to do that, but this function would:

```
let rec count_forever n = 1 + count_forever n
```

```
So the operating system for safety's sake limits the call stack size. That means eventually count will run out of stack
space on a large enough input. Notice how that choice is really independent of the programming language. So this same
```

issue can and does occur in languages other than OCaml, including Python and Java. You're just less likely to have seen it manifest there, because you probably never wrote quite as many recursive functions in those languages.

Tail Recursion. There is a solution to this issue that was described in a 1977 paper about LISP by Guy Steele. The solution, tail-call optimization, requires some cooperation between the programmer and the compiler. The programmer does a little rewriting of the function, which the compiler then notices and applies an optimization. Let's see how it works.

Suppose that a recursive function f calls itself then returns the result of that recursive call. Our count function does not do that:

2.4. Functions 35

```
let rec count n =
  if n = 0 then 0 else 1 + count (n - 1)
```

```
val count : int -> int = <fun>
```

Rather, after the recursive call count (n - 1), there is computation remaining: the computer still needs to add 1 to the result of that call.

But we as programmers could rewrite the count function so that it does *not* need to do any additional computation after the recursive call. The trick is to create a helper function with an extra parameter:

```
let rec count_aux n acc =
  if n = 0 then acc else count_aux (n - 1) (acc + 1)
let count_tr n = count_aux n 0
```

```
val count_aux : int -> int -> int = <fun>
```

```
val count_tr : int -> int = <fun>
```

Function count_aux is almost the same as our original count, but it adds an extra parameter named acc, which is idiomatic and stands for "accumulator". The idea is that the value we want to return from the function is slowly, with each recursive call, being accumulated in it. The "remaining computation" —the addition of 1— now happens before the recursive call not after. When the base case of the recursion finally arrives, the function now returns acc, where the answer has been accumulated.

But the original base case of 0 still needs to exist in the code somwhere. And it does, as the original value of acc that is passed to count_aux. Now count_tr (we'll get to why the name is "tr" in just a minute) works as a replacement for our original count.

At this point we've completed the programmer's responsibility, but it's probably not clear why we went through this effort. After all count_aux will still call itself recursively too many times as count did, and eventually overflow the stack.

That's where the compiler's responsibility kicks in. A good compiler (and the OCaml compiler is good this way) can notice when a recursive call is in *tail position*, which is a technical way of saying "there's no more computation to be done after it returns". The recursive call to count_aux is in tail position; the recursive call to count is not. Here they are again so you can compare them:

```
let rec count n =
  if n = 0 then 0 else 1 + count (n - 1)

let rec count_aux n acc =
  if n = 0 then acc else count_aux (n - 1) (acc + 1)
```

Here's why tail position matters: A recursive call in tail position does not need a new stack frame. It can just reuse the existing stack frame. That's because there's nothing left of use in the existing stack frame! There's no computation left to be done, so none of the local variables, or next instruction to execute, etc. matter any more. None of that memory ever needs to be read again, because that call is effectively already finished. So instead of wasting space by allocating another stack frame, the compiler "recycles" the space used by the previous frame.

This is the *tail-call optimization*. It can even be applied in cases beyond recursive functions if the calling function's stack frame is suitably compatible with the callee. And, it's a big deal. The tail-call optimization reduces the stack space requirements from linear to constant. Whereas count needed O(n) stack frames, count_aux needs only O(1), because the same frame gets reused over and over again for each recursive call. And that means count_tr actually can count to 1,000,000:

```
count_tr 1_000_000
```

```
- : int = 1000000
```

Finally, why did we name this function count_tr? The "tr" stands for *tail recursive*. A tail recursive function is a recursive function whose recursive calls are all in tail position. In other words, it's a function that (unless there are other pathologies) will not exhaust the stack.

The Importance of Tail Recursion. Sometimes beginning functional programmers fixate a bit too much upon it. If all you care about is writing the first draft of a function, you probably don't need to worry about tail recursion. It's pretty easy to make it tail recursive later if you need to, just by adding an accumulator argument. Or maybe you should rethink how you have designed the function. Take count, for example: it's kind of dumb. But later we'll see examples that aren't dumb, such as iterating over lists with thousands of elements.

It is important that the compiler support the optimization. Otherwise, the transformation you do to the code as a programmer makes no difference. Indeed, most compilers do support it, at least as an option. Java is a notable exception.

The Recipe for Tail Recursion. In a nutshell, here's how we made a function be tail recursive:

- 1. Change the function into a helper function. Add an extra argument: the accumulator, often named acc.
- 2. Write a new "main" version of the function that calls the helper. It passes the original base case's return value as the initial value of the accumulator.
- 3. Change the helper function to return the accumulator in the base case.
- 4. Change the helper function's recursive case. It now needs to do the extra work on the accumulator argument, before the recursive call. This is the only step that requires much ingenuity.

An Example: Factorial. Let's tranform this factorial function to be tail recursive:

```
(* [fact n] is [n] factorial *)
let rec fact n =
  if n = 0 then 1 else n * fact (n - 1)
```

```
val fact : int -> int = <fun>
```

First, we change its name and add an accumulator argument:

```
let rec fact_aux n acc = ...
```

Second, we write a new "main" function that calls the helper with the original base case as the accumulator:

```
let rec fact_tr n = fact_aux n 1
```

Third, we change the helper function to return the accumulator in the base case:

```
if n = 0 then acc ...
```

Finally, we change the recursive case:

```
else fact (n - 1) (n * acc)
```

Putting it all together, we have:

```
let rec fact_aux n acc =
  if n = 0 then acc else fact_aux (n - 1) (n * acc)
let fact_tr n = fact_aux n 1
```

2.4. Functions 37

```
val fact_aux : int -> int -> int = <fun>
```

```
val fact_tr : int -> int = <fun>
```

It was a nice exercise, but maybe not worthwhile. Even before we exhaust the stack space, the computation suffers from integer overflow:

```
fact 50
```

```
-: int = -3258495067890909184
```

To solve that problem, we turn to OCaml's big integer library, Zarith. Here we use a few OCaml features that are beyond anything we've seen so far, but hopefully nothing terribly surprising.

```
#require "zarith.top";;

let rec zfact_aux n acc =
   if Z.equal n Z.zero then acc else zfact_aux (Z.pred n) (Z.mul acc n);;

let zfact_tr n = zfact_aux n Z.one;;

zfact_tr (Z.of_int 50)
```

```
val zfact_aux : Z.t -> Z.t -> Z.t = <fun>
```

```
val zfact_tr : Z.t -> Z.t = <fun>
```

```
- : Z.t = 3041409320171337804361260816606476884437764156896051200000000000
```

If you want you can use that code to compute <code>zfact_tr 1_000_000</code> without stack or integer overflow, though it will take several minutes.

The chapter on modules will explain the OCaml features we used above in detail, but for now:

- #require loads the library, which provides a module named Z. Recall that Z is the symbol used in mathematics to denote the integers.
- Z.n means the name n defined inside of Z.
- The type ${\tt Z.t}$ is the library's name for the type of big integers.
- We use library values Z.equal for equality comparison, Z.zero for 0, Z.pred for predecessor (i.e., subtracting 1), Z.mul for multiplication, Z.one for 1, and Z.of_int to convert a primitive integer to a big integer.

2.5 Documentation

OCaml provides a tool called OCamldoc that works a lot like Java's Javadoc tool: it extracts specially formatted comments from source code and renders them as HTML, making it easy for programmers to read documentation.

2.5.1 How to Document

Here's an example of an OCamldoc comment:

```
(** [sum 1st] is the sum of the elements of [1st]. *)
let rec sum 1st = ...
```

- The double asterisk is what causes the comment to be recognized as an OCamldoc comment.
- The square brackets around parts of the comment mean that those parts should be rendered in HTML as typewriter font rather than the regular font.

Also like Javadoc, OCamldoc supports *documentation tags*, such as @author, @deprecated, @param, @return, etc. For example, in the first line of most programming assignments, we ask you to complete a comment like this:

```
(** @author Your Name (your netid) *)
```

For the full range of possible markup inside a OCamldoc comment, see the OCamldoc manual. But what we've covered here is good enough for most documentation that you'll need to write.

2.5.2 What to Document

The documentation style we favor in this book resembles that of the OCaml standard library: concise and declarative. As an example, let's revisit the documentation of sum:

```
(** [sum 1st] is the sum of the elements of [1st]. *)
let rec sum 1st = ...
```

That comment starts with sum lst, which is an example application of the function to an argument. The comment continues with the word "is", thus declaratively describing the result of the application. (The word "returns" could be used instead, but "is" emphasizes the mathematical nature of the function.) That description uses the name of the argument, lst, to explain the result.

Note how there is no need to add tags to redundantly describe parameters or return values, as is often done with Javadoc. Everything that needs to be said has already been said. We strongly discourage documentation like the following:

```
(** Sum a list.
    @param lst The list to be summed.
    @return The sum of the list. *)
let rec sum lst = ...
```

That poor documentation takes three needlessly hard-to-read lines to say the same thing as the limpid one-line version.

There is one way we might improve the documentation we have so far, which is to explicitly state what happens with empty lists:

```
(** [sum lst] is the sum of the elements of [lst].
   The sum of an empty list is 0. *)
let rec sum lst = ...
```

2.5. Documentation 39

2.5.3 Preconditions and Postconditions

Here are a few more examples of comments written in the style we favor.

```
(** [lowercase_ascii c] is the lowercase ASCII equivalent of
    character [c]. *)

(** [index s c] is the index of the first occurrence of
    character [c] in string [s]. Raises: [Not_found]
    if [c] does not occur in [s]. *)

(** [random_int bound] is a random integer between 0 (inclusive)
    and [bound] (exclusive). Requires: [bound] is greater than 0
    and less than 2^30. *)
```

The documentation of index specifies that the function raises an exception, as well as what that exception is and the condition under which it is raised. (We will cover exceptions in more detail in the next chapter.) The documentation of random_int specifies that the function's argument must satisfy a condition.

In previous courses, you were exposed to the ideas of *preconditions* and *postconditions*. A precondition is something that must be true before some section of code; and a postcondition, after.

The "Requires" clause above in the documentation of random_int is a kind of precondition. It says that the client of the random_int function is responsible for guaranteeing something about the value of bound. Likewise, the first sentence of that same documentation is a kind of postcondition. It guarantees something about the value returned by the function.

The "Raises" clause in the documentation of index is another kind of postcondition. It guarantees that the function raises an exception. Note that the clause is not a precondition, even though it states a condition in terms of an input.

Note that none of these examples has a "Requires" clause that says something about the type of an input. If you're coming from a dynamically-typed language, like Python, this could be a surprise. Python programmers frequently document preconditions regarding the types of function inputs. OCaml programmers, however, do not. That's because the compiler itself does the type checking to ensure that you never pass a value of the wrong type to a function. Consider <code>lower-case_ascii</code> again: although the English comment helpfully identifies the type of <code>c</code> to the reader, the comment does not state a "Requires" clause like this:

```
(** [lowercase_ascii c] is the lowercase ASCII equivalent of [c].
Requires: [c] is a character. *)
```

Such a comment reads as highly unidiomatic to an OCaml programmer, who would read that comment and be puzzled, perhaps thinking: "Well of course c is a character; the compiler will guarantee that. What did the person who wrote that really mean? Is there something they or I am missing?"

2.6 Printing

OCaml has built-in printing functions for a few of the built-in primitive types: print_char, print_string, print_int, and print_float. There's also a print_endline function, which is like print_string, but also outputs a newline.

```
print_endline "Camels are bae"
```

```
Camels are bae
```

```
- : unit = ()
```

2.6.1 Unit

Let's look at the types of a couple of those functions:

```
print_endline

- : string -> unit = <fun>
print_string
```

They both take a string as input and return a value of type unit, which we haven't seen before. There is only one value of this type, which is written () and is also pronounced "unit". So unit is like bool, except there is one fewer value of type unit than there is of bool.

Unit is used when you need to take an argument or return a value, but there's no interesting value to pass or return. It is the equivalent of void in Java, and is similar to None in Python. Unit is often used when you're writing or using code that has side effects. Printing is an example of a side effect: it changes the world and can't be undone.

2.6.2 Semicolon

- : string -> unit = <fun>

If you want to print one thing after another, you could sequence some print functions using nested let expressions:

```
let _ = print_endline "Camels" in
let _ = print_endline "are" in
print_endline "bae"
```

```
Camels
```

```
are
```

bae

```
- : unit = ()
```

The let _ = e syntax above is a way of evaluating e but not binding its value to any name. Indeed, we know the value each of those print_endline functions will return: it will always be (), the unit value. So there's no good reason to bind it to a variable name. We could also write let () = e to indicate we know it's just a unit value that we don't care about:

```
let () = print_endline "Camels" in
let () = print_endline "are" in
print_endline "bae"
```

```
Camels
```

```
are
```

2.6. Printing 41

hae

```
- : unit = ()
```

But either way the boilerplate of all the let..in is annoying to have to write! So there's a special syntax that can be used to chain together multiple functions that return unit. The expression e1; e2 first evaluates e1, which should evaluate to (), then discards that value, and evaluates e2. So we could rewrite the above code as:

```
print_endline "Camels";
print_endline "are";
print_endline "bae";
```

Camels

```
are
```

bae

```
- : unit = ()
```

That is more idiomatic OCaml code, and it also looks more natural to imperative programmers.

Warning: There is no semicolon after the final print_endline in that example. A common mistake is to put a semicolon *after* each print statement. Instead, the semicolons go strictly *between* statements. That is, semicolon is a statement *separator* not a statement *terminator*. If you were to add a semicolon at the end, you could get a syntax error depending on the surrounding code.

2.6.3 Ignore

If e1 does not have type unit, then e1; e2 will give a warning, because you are discarding a potentially useful value. If that is truly your intent, you can call the built-in function ignore: 'a -> unit to convert any value to ():

```
(ignore 3); 5
```

```
- : int = 5
```

Actually ignore is easy to implement yourself:

```
let ignore x = ()
```

```
val ignore : 'a -> unit = <fun>
```

Or you can even write underscore to indicate the function takes in a value but does not bind that value to a name. That means the function can never use that value in its body. But that's okay: we want to ignore it.

```
let ignore _ = ()
```

```
val ignore : 'a -> unit = <fun>
```

2.6.4 Printf

For complicated text outputs, using the built-in functions for primitive type printing quickly becomes tedious. For example, suppose you wanted to write a function to print a statistic:

```
(** [print_stat name num] prints [name: num]. *)
let print_stat name num =
  print_string name;
  print_string ": ";
  print_float num;
  print_newline ()
```

```
val print_stat : string -> float -> unit = <fun>
```

```
print_stat "mean" 84.39
```

```
mean: 84.39
```

```
- : unit = ()
```

How could we shorten print_stat? In Java you might use the overloaded + operator to turn all objects into strings:

```
void print_stat(String name, double num) {
   System.out.println(name + ": " + num);
}
```

But OCaml values are not objects, and they do not have a toString() method they inherit from some root Object class. Nor does OCaml permit overloading of operators.

Long ago though, FORTRAN invented a different solution that other languages like C and Java and even Python support. The idea is to use a *format specifier* to —as the name suggest—specify how to format output. The name this idea is best known under is probably "printf", which refers to the name of the C library function that implemented it. Many other languages and libraries still use that name, including OCaml's Printf module.

Here's how we'd use printf to re-implement print_stat:

```
let print_stat name num =
    Printf.printf "%s: %F\n%!" name num
```

```
val print_stat : string -> float -> unit = <fun>
```

```
print_stat "mean" 84.39
```

```
mean: 84.39
```

```
- : unit = ()
```

The first argument to function Printf.printf is the format specifier. It *looks* like a string, but there's more to it than that. It's actually understood by the OCaml compiler in quite a deep way. Inside the format specifier there are:

- · plain characters, and
- conversion specifiers, which begin with %.

There are about two dozen conversion specifiers available, which you can read about in the documentation of Printf. Let's pick apart the format specifier above as an example.

2.6. Printing 43

- It starts with "%s", which is the conversion specifier for strings. That means the next argument to printf must be a string, and the contents of that string will be output.
- It continues with ": ", which are just plain characters. Those are inserted into the output.
- It then has another conversion specifier, %F. That means the next argument of printf must have type float, and will be output in the same format that OCaml uses to print floats.
- The newline "\n" after that is another plain character sequence.
- Finally the conversion specifier "%!" means to *flush the output buffer*. As you might have learned in earlier programming classes, output is often *buffered*, meaning that it doesn't all happen at once or right away. Flushing the buffer ensures that anything still sitting in the buffer gets output immediately. This specifier is special in that it doesn't actually need another argument to printf.

If the type of an argument is incorrect with respect to the conversion specifier, OCaml will detect that. Let's add a type annotation to force num to be an int, and see what happens with the float conversion specifier %F:

```
let print_stat name (num : int) =
    Printf.printf "%s: %F\n%!" name num
```

```
File "[14]", line 2, characters 34-37:
2 | Printf.printf "%s: %F\n%!" name num

^^^

Error: This expression has type int but an expression was expected of type

float
```

To fix that, we can change to the conversion specifier for int, which is %i:

```
let print_stat name num =
    Printf.printf "%s: %i\n%!" name num
```

```
val print_stat : string -> int -> unit = <fun>
```

Another very useful variant of printf is sprintf, which collects the output in string instead of printing it:

```
let string_of_stat name num =
    Printf.sprintf "%s: %F" name num
```

```
val string_of_stat : string -> float -> string = <fun>
```

```
string_of_stat "mean" 84.39
```

```
- : string = "mean: 84.39"
```

2.7 Debugging

Debugging is a last resort when everything else has failed. Let's take a step back and think about everything that comes *before* debugging.

2.7.1 Defenses against Bugs

According to Rob Miller, there are four defenses against bugs:

1. The first defense against bugs is to make them impossible.

Entire classes of bugs can be eradicated by choosing to program in languages that guarantee *memory safety* (that no part of memory can be accessed except through a *pointer* (or reference) that is valid for that region of memory) and *type safety* (that no value can be used in a way inconsistent with its type). The OCaml type system, for example, prevents programs from buffer overflows and meaningless operations (like adding a boolean to a float), whereas the C type system does not.

2. The second defense against bugs is to use tools that find them.

There are automated source-code analysis tools, like FindBugs, which can find many common kinds of bugs in Java programs, and SLAM, which is used to find bugs in device drivers. The subfield of CS known as *formal methods* studies how to use mathematics to specify and verify programs, that is, how to prove that programs have no bugs. We'll study verification later in this course.

Social methods such as code reviews and pair programming are also useful tools for finding bugs. Studies at IBM in the 1970s-1990s suggested that code reviews can be remarkably effective. In one study (Jones, 1991), code inspection found 65% of the known coding errors and 25% of the known documentation errors, whereas testing found only 20% of the coding errors and none of the documentation errors.

3. The third defense against bugs is to make them immediately visible.

The earlier a bug appears, the easier it is to diagnose and fix. If computation instead proceeds past the point of the bug, then that further computation might obscure where the failure really occurred. *Assertions* in the source code make programs "fail fast" and "fail loudly", so that bugs appear immediately, and the programmer knows exactly where in the source code to look.

4. The fourth defense against bugs is extensive testing.

How can you know whether a piece of code has a particular bug? Write tests that would expose the bug, then confirm that your code doesn't fail those tests. *Unit tests* for a relatively small piece of code, such as an individual function or module, are especially important to write at the same time as you develop that code. Running of those tests should be automated, so that if you ever break the code, you find out as soon as possible. (That's really Defense 3 again.)

After all those defenses have failed, a programmer is forced to resort to debugging.

2.7.2 How to Debug

So you've discovered a bug. What next?

- 1. **Distill the bug into a small test case.** Debugging is hard work, but the smaller the test case, the more likely you are to focus your attention on the piece of code where the bug lurks. Time spent on this distillation can therefore be time saved, because you won't have to re-read lots of code. Don't continue debugging until you have a small test case!
- 2. **Employ the scientific method.** Formulate a hypothesis as to why the bug is occurring. You might even write down that hypothesis in a notebook, as if you were in a Chemistry lab, to clarify it in your own mind and keep track of what hypotheses you've already considered. Next, design an experiment to affirm or deny that hypothesis. Run your experiment and record the result. Based on what you've learned, reformulate your hypothesis. Continue until you have rationally, scientifically determined the cause of the bug.
- 3. **Fix the bug.** The fix might be a simple correction of a typo. Or it might reveal a design flaw that causes you to make major changes. Consider whether you might need to apply the fix to other locations in your code base—for example, was it a copy and paste error? If so, do you need to refactor your code?

2.7. Debugging 45

4. **Permanently add the small test case to your test suite.** You wouldn't want the bug to creep back into your code base. So keep track of that small test case by keeping it as part of your unit tests. That way, any time you make future changes, you will automatically be guarding against that same bug. Repeatedly running tests distilled from previous bugs is called *regression testing*.

2.7.3 Debugging in OCaml

Here are a couple tips on how to debug—if you are forced into it—in OCaml.

• **Print statements.** Insert a print statement to ascertain the value of a variable. Suppose you want to know what the value of x is in the following function:

```
let inc x = x + 1
```

Just add the line below to print that value:

```
let inc x =
  let () = print_int x in
  x + 1
```

• **Function traces.** Suppose you want to see the *trace* of recursive calls and returns for a function. Use the #trace directive:

```
# let rec fib x = if x \le 1 then 1 else fib (x - 1) + fib (x - 2);;
# #trace fib;;
```

If you evaluate fib 2, you will now see the following output:

```
fib <-- 2
fib <-- 0
fib --> 1
fib <-- 1
fib --> 1
fib --> 2
```

To stop tracing, use the #untrace directive.

• **Debugger.** OCaml has a debugging tool ocamldebug. You can find a tutorial on the OCaml website. Unless you are using Emacs as your editor, you will probably find this tool to be harder to use than just inserting print statements.

2.7.4 Defensive Programming

As we discussed earlier in the section on debugging, one defense against bugs is to make any bugs (or errors) immediately visible. That idea connects with idea of preconditions.

Consider this specification of random_int:

```
(** [random_int bound] is a random integer between 0 (inclusive)
    and [bound] (exclusive). Requires: [bound] is greater than 0
    and less than 2^30. *)
```

If the client of random_int passes a value of bound that violates the "Requires" clause, such as -1, the implementation of random_int is free to do anything whatsoever. All bets are off when the client violates the precondition.

But the most helpful thing for random_int to do is to immediately expose the fact that the precondition was violated. After all, chances are that the client didn't *mean* to violate it.

So the implementor of random_int would do well to check whether the precondition is violated, and if so, raise an exception. Here are three possibilities of that kind of *defensive programming*:

```
(* possibility 1 *)
let random_int bound =
    assert (bound > 0 && bound < 1 lsl 30);
    (* proceed with the implementation of the function *)

(* possibility 2 *)
let random_int bound =
    if not (bound > 0 && bound < 1 lsl 30)
    then invalid_arg "bound";
    (* proceed with the implementation of the function *)

(* possibility 3 *)
let random_int bound =
    if not (bound > 0 && bound < 1 lsl 30)
    then failwith "bound";
    (* proceed with the implementation of the function *)</pre>
```

The second possibility is probably the most informative to the client, because it uses the built-in function invalid_arg to raise the well-named exception Invalid_argument. In fact, that's exactly what the standard library implementation of this function does.

The first possibility is probably most useful when you are trying to debug your own code, rather than choosing to expose a failed assertion to a client.

The third possibility differs from the second only in the name (Failure) of the exception that is raised. It might be useful in situations where the precondition involves more than just a single invalid argument.

In this example, checking the precondition is computationally cheap. In other cases, it might require a lot of computation, so the implementer of the function might prefer not to check the precondition, or only to check some inexpensive approximation to it.

Sometimes programmers worry unnecessarily that defensive programming will be too expensive—either in terms of the time it costs them to implement the checks initially, or in the run-time costs that will be paid in checking assertions. These concerns are far too often misplaced. The time and money it costs society to repair faults in software suggests that we could all afford to have programs that run a little more slowly.

Finally, the implementer might even choose to eliminate the precondition and restate it as a postcondition:

```
(** [random_int bound] is a random integer between 0 (inclusive)
    and [bound] (exclusive). Raises: [Invalid_argument "bound"]
    unless [bound] is greater than 0 and less than 2^30. *)
```

Now instead of being free to do whatever when bound is too big or too small, random_int must raise an exception. For this function, that's probably the best choice.

In this course, we're not going to force you to program defensively. But if you're savvy, you'll start (or continue) doing it anyway. The small amount of time you spend coding up such defenses will save you hours of time in debugging, making you a more productive programmer.

2.7. Debugging 47

2.8 Summary

Syntax and semantics are a powerful paradigm for learning a programming language. As we learn the features of OCaml, we're being careful to write down their syntax and semantics. We've seen that there can be multiple syntaxes for expressing the same semantic idea, that is, the same computation.

The semantics of function application is the very heart of OCaml and of functional programming, and it's something we will come back to several times throughout the course to deepen our understanding.

2.8.1 Terms and Concepts

- · anonymous functions
- · assertions
- binding
- · binding expression
- · body expression
- debugging
- · defensive programming
- · definitions
- · documentation
- · dynamic semantics
- · evaluation
- · expressions
- · function application
- · function definitions
- identifiers
- idioms
- · if expressions
- · lambda expressions
- let definition
- let expression
- libraries
- · metavariables
- · mutual recursion
- · pipeline operator
- postcondition
- precondition
- printing
- · recursion

- · semantics
- static semantics
- · substitution
- syntax
- · tools
- · type checking
- · type inference
- values

2.8.2 Further Reading

- Introduction to Objective Caml, chapter 3
- OCaml from the Very Beginning, chapter 2
- Real World OCaml, chapter 2
- *Tail Recursion, The Musical*. Tail-call optimization explained in the context of JavaScript with cute 8-bit animations, and Disney songs!

2.9 Exercises

Solutions to exercises are available to students in Cornell's CS 3110. Instructors at other institutions are welcome to contact Michael Clarkson for access.

Exercise: values [★]

What is the type and value of each of the following OCaml expressions?

```
• 7 * (1 + 2 + 3)
• "CS " ^ string_of_int 3110
```

Hint: type each expression into the toplevel and it will tell you the answer. Note: ^ is not exponentiation.

Exercise: operators $[\star \star]$

Examine the table of all operators in the OCaml manual (you will have to scroll down to find it on that page).

- Write an expression that multiplies 42 by 10.
- Write an expression that divides 3.14 by 2.0. *Hint: integer and floating-point operators are written differently in OCaml.*
- Write an expression that computes 4 . 2 raised to the seventh power. *Note: there is no built-in integer exponentiation operator in OCaml (nor is there in C, by the way), in part because it is not an operation provided by most CPUs.*

Exercise: equality [★]

• Write an expression that compares 42 to 42 using structural equality.

2.9. Exercises 49

OCaml Programming: Correct + Efficient + Beautiful

- Write an expression that compares "hi" to "hi" using structural equality. What is the result?
- Write an expression that compares "hi" to "hi" using physical equality. What is the result?

Exercise: assert [★]

- Enter assert true;; into utop and see what happens.
- Enter assert false;; into utop and see what happens.
- Write an expression that asserts 2110 is not (structurally) equal to 3110.

Exercise: if [★]

Write an if expression that evaluates to 42 if 2 is greater than 1 and otherwise evaluates to 7.

Exercise: double fun [★]

Using the increment function from above as a guide, define a function double that multiplies its input by 2. For example, double 7 would be 14. Test your function by applying it to a few inputs. Turn those test cases into assertions.

Exercise: more fun $[\star \star]$

- Define a function that computes the cube of a floating-point number. Test your function by applying it to a few inputs.
- Define a function that computes the sign (1, 0, or -1) of an integer. Use a nested if expression. Test your function by applying it to a few inputs.
- · Define a function that computes the area of a circle given its radius. Test your function with assert.

For the latter, bear in mind that floating-point arithmetic is not exact. Instead of asserting an exact value, you should assert that the result is "close enough", e.g., within 1e-5. If that's unfamiliar to you, it would be worthwhile to read up on floating-point arithmetic.

A function that take multiple inputs can be defined just by providing additional names for those inputs as part of the let definition. For example, the following function computes the average of three arguments:

```
let avg3 x y z = (x + . y + . z) / . 3.
```

Exercise: RMS [★★]

Define a function that computes the *root mean square* of two numbers—i.e., $\sqrt{(x^2+y^2)/2}$. Test your function with assert.

Exercise: date fun $[\star \star \star]$

Define a function that takes an integer d and string m as input and returns true just when d and m form a *valid date*. Here, a valid date has a month that is one of the following abbreviations: Jan, Feb, Mar, Apr, May, Jun, Jul, Aug, Sept, Oct, Nov, Dec. And the day must be a number that is between 1 and the minimum number of days in that month, inclusive. For example, if the month is Jan, then the day is between 1 and 31, inclusive, whereas if the month is Feb, then the day is between 1 and 28, inclusive.

How terse (i.e., few and short lines of code) can you make your function? You can definitely do this in fewer than 12 lines.

Exercise: fib [★★]

Define a recursive function fib: int -> int, such that fib n is the nth number in the Fibonacci sequence, which is 1, 1, 2, 3, 5, 8, 13, ... That is:

```
fib 1 = 1,
fib 2 = 1, and
fib n = fib (n-1) + fib (n-2) for any n > 2.
```

Test your function in the toplevel.

Exercise: fib fast $[\star \star \star]$

How quickly does your implementation of fib compute the 50th Fibonacci number? If it computes nearly instantaneously, congratulations! But the recursive solution most people come up with at first will seem to hang indefinitely. The problem is that the obvious solution computes subproblems repeatedly. For example, computing fib 5 requires computing both fib 3 and fib 4, and if those are computed separately, a lot of work (an exponential amount, in fact) is being redone.

Create a function fib_fast that requires only a linear amount of work. *Hint:* write a recursive helper function h : int -> int -> int -> int, where h n pp p is defined as follows:

```
h 1 pp p = p, and
h n pp p = h (n-1) p (pp+p) for any n > 1.
```

The idea of h is that it assumes the previous two Fibonacci numbers were pp and p, then computes forward n more numbers. Hence, fib n = h + n + 0 + 1 for any n > 0.

What is the first value of n for which fib_fast n is negative, indicating that integer overflow occurred?

Exercise: poly types $[\star\star\star]$

What is the type of each of the functions below? You can ask the toplevel to check your answers

```
let f x = if x then x else x
let g x y = if y then x else x
let h x y z = if x then y else z
let i x y z = if x then y else y
```

Exercise: divide [★★]

Write a function divide: numerator:float -> denominator:float -> float. Apply your function.

Exercise: associativity [★★]

Suppose that we have defined let add x y = x + y. Which of the following produces an integer, which produces a function, and which produces an error? Decide on an answer, then check your answer in the toplevel.

• add 5 1

2.9. Exercises 51

OCaml Programming: Correct + Efficient + Beautiful

- add 5
- (add 5) 1
- add (5 1)

Exercise: average [★★]

Define an infix operator +/. to compute the average of two floating-point numbers. For example,

- 1.0 +/. 2.0 = 1.5
- 0. +/. 0. = 0.

Exercise: hello world [★]

Type the following in utop:

- print_endline "Hello world!";;
- print_string "Hello world!";;

Notice the difference in output from each.

Part II OCaml Programming

THREE

DATA AND TYPES

In this chapter, we'll examine some of OCaml's built-in data types, including lists, variants, records, tuples, and options. Many of those are likely to feel familiar from other programming languages. In particular,

- · lists and tuples, might feel similar to Python; and
- records and variants, might feel similar to struct and enum types from C or Java.

Because of that familiarity, we call these *standard* data types. We'll learn about *pattern matching*, which is a feature that's less likely to be familiar.

Almost immediately after we learn about lists, we'll pause our study of standard data types to learn about unit testing in OCaml with OUnit, a unit testing framework similar to those you might have used in other languages. OUnit relies on lists, which is why we couldn't cover it before now.

Later in the chapter, we study some OCaml data types that are unlikely to be as familiar from other languages. They include:

- options, which are loosely related to null in Java;
- association lists, which are an amazingly simple implementation of maps (aka dictionaries) based on lists and tuples;
- **algebraic data types**, which are arguably the most important kind of type in OCaml, and indeed are the power behind many of the other built-in types; and
- exceptions, which are a special kind of algebraic data type.

3.1 Lists

An OCaml list is a sequence of values all of which have the same type. They are implemented as singly-linked lists. These lists enjoy a first-class status in the language: there is special support for easily creating and working with lists. That's a characteristic that OCaml shares with many other functional languages. Mainstream imperative languages, like Python, have such support these days too. Maybe that's because programmers find it so pleasant to work directly with lists as a first-class part of the language, rather than having to go through a library (as in C and Java).

3.1.1 Building Lists

Syntax. There are three syntactic forms for building lists:

```
[]
e1 :: e2
[e1; e2; ...; en]
```

The empty list is written [] and is pronounced "nil", a name that comes from Lisp. Given a list lst and element elt, we can prepend elt to lst by writing elt :: lst. The double-colon operator is pronounced "cons", a name that comes from an operator in Lisp that constructs objects in memory. "Cons" can also be used as a verb, as in "I will cons an element onto the list." The first element of a list is usually called its *head* and the rest of the elements (if any) are called its *tail*.

The square bracket syntax is convenient but unnecessary. Any list [e1; e2; ...; en] could instead be written with the more primitive nil and cons syntax: e1 :: e2 :: ... :: en :: []. When a pleasant syntax can be defined in terms of a more primitive syntax within the language, we call the pleasant syntax *syntactic sugar*: it makes the language "sweeter". Transforming the sweet syntax into the more primitive syntax is called *desugaring*.

Because the elements of the list can be arbitrary expressions, lists can be nested as deeply as we like, e.g., [[[]]; [[1; 2; 3]]].

Dynamic semantics.

- [] is already a value.
- If e1 evaluates to v1, and if e2 evaluates to v2, then e1 :: e2 evaluates to v1 :: v2.

As a consequence of those rules and how to desugar the square-bracket notation for lists, we have the following derived rule:

• If ei evaluates to vi for all i in 1..., then [e1; ...; en] evaluates to [v1; ...; vn].

It's starting to get tedious to write "evaluates to" in all our evaluation rules. So let's introduce a shorter notation for it. We'll write e ==> v to mean that e evaluates to v. Note that e is not a piece of OCaml syntax. Rather, it's a notation we use in our description of the language, kind of like metavariables. Using that notation, we can rewrite the latter two rules above:

```
• If e1 ==> v1, and if e2 ==> v2, then e1 :: e2 ==> v1 :: v2.
```

```
• If ei ==> vi for all i in 1..., then [e1; ...; en] ==> [v1; ...; vn].
```

Static semantics.

All the elements of a list must have the same type. If that element type is t, then the type of the list is t list. You should read such types from right to left: t list is a list of t's, t list list is a list of list of t's, etc. The word list itself here is not a type: there is no way to build an OCaml value that has type simply list. Rather, list is a type constructor: given a type, it produces a new type. For example, given int, it produces the type int list. You could think of type constructors as being like functions that operate on types, instead of functions that operate on values.

The type-checking rules:

```
• [] : 'a list
```

• If e1 : t and e2 : t list then e1 :: e2 : t list. In case the colons and their precedence is confusing, the latter means (e1 :: e2) : t list.

In the rule for [], recall that 'a is a type variable: it stands for an unknown type. So the empty list is a list whose elements have an unknown type. If we cons an int onto it, say 2 :: [], then the compiler infers that for that particular list, 'a must be int. But if in another place we cons a bool onto it, say true :: [], then the compiler infers that for that particular list, 'a must be bool.

3.1.2 Accessing Lists

Note: The video linked above also uses records and tuples as examples. Those are covered in the *next section* of this book.

There are really only two ways to build a list, with nil and cons. So if we want to take apart a list into its component pieces, we have to say what to do with the list if it's empty, and what to do if it's non-empty (that is, a cons of one element onto some other list). We do that with a language feature called *pattern matching*.

Here's an example of using pattern matching to compute the sum of a list:

```
let rec sum lst =
  match lst with
  | [] -> 0
  | h :: t -> h + sum t
```

```
val sum : int list -> int = <fun>
```

This function says to take the input lst and see whether it has the same shape as the empty list. If so, return 0. Otherwise, if it has the same shape as the list h::t, then let h be the first element of lst, and let t be the rest of the elements of lst, and return h+sumt. The choice of variable names here is meant to suggest "head" and "tail" and is a common idiom, but we could use other names if we wanted. Another common idiom is:

```
let rec sum xs =
  match xs with
  | [] -> 0
  | x :: xs' -> x + sum xs'
```

```
val sum : int list -> int = <fun>
```

That is, the input list is a list of xs (pronounced EX-uhs), the head element is an x, and the tail is xs' (pronounced EX-uhs prime).

Syntactically it isn't necessary to use so many lines to define sum. We could do it all on one line:

```
let rec sum xs = match xs with | [] <math>\rightarrow 0 | x :: xs' \rightarrow x + sum xs'
```

```
val sum : int list -> int = <fun>
```

Or, noting that the first | after with is optional regardless of how many lines we use, we could also write:

```
let rec sum xs = match xs with [] -> 0 | x :: xs' -> x + sum xs'
```

```
val sum : int list -> int = <fun>
```

The multi-line format is what we'll usually use in this book, because it helps the human eye understand the syntax a bit better. OCaml code formatting tools, though, are moving toward the single-line format whenever the code is short enough to fit on just one line.

Here's another example of using pattern matching to compute the length of a list:

```
let rec length lst =
  match lst with
  | [] -> 0
  | h :: t -> 1 + length t
```

3.1. Lists 57

```
val length : 'a list -> int = <fun>
```

Note how we didn't actually need the variable h in the right-hand side of the pattern match. When we want to indicate the presence of some value in a pattern without actually giving it a name, we can write _ (the underscore character):

```
let rec length lst =
  match lst with
  | [] -> 0
  | _ :: t -> 1 + length t
```

```
val length : 'a list -> int = <fun>
```

That function is actually built-in as part of the OCaml standard library List module. Its name there is List.length. That "dot" notation indicates the function named length inside the module named Length, much like the dot notation used in many other languages.

And here's a third example that appends one list onto the beginning of another list:

```
let rec append lst1 lst2 =
  match lst1 with
  | [] -> lst2
  | h :: t -> h :: append t lst2
```

```
val append : 'a list -> 'a list = <fun>
```

For example, append [1; 2] [3; 4] is [1; 2; 3; 4]. That function is actually available as a built-in operator @, so we could instead write [1; 2] @ [3; 4].

As a final example, we could write a function to determine whether a list is empty:

```
let empty lst =
   match lst with
   | [] -> true
   | h :: t -> false
```

```
val empty : 'a list -> bool = <fun>
```

But there a much better way to write the same function without pattern matching:

```
let empty lst =
lst = []
```

```
val empty : 'a list -> bool = <fun>
```

Note how all the recursive functions above are similar to doing proofs by induction on the natural numbers: every natural number is either 0 or is 1 greater than some other natural number n, and so a proof by induction has a base case for 0 and an inductive case for n+1. Likewise all our functions have a base case for the empty list and a recursive case for the list that has one more element than another list. This similarity is no accident. There is a deep relationship between induction and recursion; we'll explore that relationship in more detail later in the book.

By the way, there are two library functions List.hd and List.tl that return the head and tail of a list. It is not good, idiomatic OCaml to apply these directly to a list. The problem is that they will raise an exception when applied to the empty list, and you will have to remember to handle that exception. Instead, you should use pattern matching: you'll then be forced to match against both the empty list and the non-empty list (at least), which will prevent exceptions from being raised, thus making your program more robust.

3.1.3 (Not) Mutating Lists

Lists are immutable. There's no way to change an element of a list from one value to another. Instead, OCaml programmers create new lists out of old lists. For example, suppose we wanted to write a function that returned the same list as its input list, but with the first element (if there is one) incremented by one. We could do that:

```
let inc_first lst =
  match lst with
  | [] -> []
  | h :: t -> h + 1 :: t
```

Now you might be concerned about whether we're being wasteful of space. After all, there are at least two ways the compiler could implement the above code:

- 1. Copy the entire tail list t when the new list is created in the pattern match with cons, such that the amount of memory in use just increased by an amount proportionate to the length of t.
- 2. Share the tail list t between the old list and the new list, such that the amount of memory in use does not increase—beyond the one extra piece of memory needed to store h + 1.

In fact, the compiler does the latter. So there's no need for concern. The reason that it's quite safe for the compiler to implement sharing is exactly that list elements are immutable. If they were instead mutable, then we'd start having to worry about whether the list I have is shared with the list you have, and whether changes I make will be visible in your list. So immutability makes it easier to reason about the code, and makes it safe for the compiler to perform an optimization.

3.1.4 Pattern Matching with Lists

We saw above how to access lists using pattern matching. Let's look more carefully at this feature.

Syntax.

```
match e with
| p1 -> e1
| p2 -> e2
| ...
| pn -> en
```

Each of the clauses pi -> ei is called a *branch* or a *case* of the pattern match. The first vertical bar in the entire pattern match is optional.

The p's here are a new syntactic form called a *pattern*. For now, a pattern may be:

- a variable name, e.g. x
- the underscore character _, which is called the wildcard
- the empty list []
- p1 :: p2
- [p1; ...; pn]

No variable name may appear more than once in a pattern. For example, the pattern x : x = x is illegal. The wildcard may occur any number of times.

As we learn more of data structures available in OCaml, we'll expand the possibilities for what a pattern may be.

Dynamic semantics.

Pattern matching involves two inter-related tasks: determining whether a pattern matches a value, and determining what parts of the value should be associated with which variable names in the pattern. The former task is intuitively about

3.1. Lists 59

determining whether a pattern and a value have the same *shape*. The latter task is about determining the *variable bindings* introduced by the pattern. For example, consider the following code:

```
match 1 :: [] with
| [] -> false
| h :: t -> h >= 1 && List.length t = 0
```

```
- : bool = true
```

When evaluating the right-hand side of the second branch, h is bound to 1 and t is bound to []. Let's write h->1 to mean the variable binding saying that h has value 1; this is not a piece of OCaml syntax, but rather a notation we use to reason about the language. So the variable bindings produced by the second branch would be h->1, t->[].

Using that notation, here is a definition of when a pattern matches a value and the bindings that match produces:

- The pattern x matches any value v and produces the variable binding x->v.
- The pattern _ matches any value and produces no bindings.
- The pattern [] matches the value [] and produces no bindings.
- If p1 matches v1 and produces a set b₁ of bindings, and if p2 matches v2 and produces a set b₂ of bindings, then p1 :: p2 matches v1 :: v2 and produces the set b₁ ∪ b₂ of bindings. Note that v2 must be a list (since it's on the right-hand side of ::) and could have any length: 0 elements, 1 element, or many elements. Note that the union b₁ ∪ b₂ of bindings will never have a problem where the same variable is bound separately in both b₁ and b₂ because of the syntactic restriction that no variable name may appear more than once in a pattern.
- If for all i in 1..., it holds that pi matches vi and produces the set b_i of bindings, then [p1; ...; pn] matches [v1; ...; vn] and produces the set $\bigcup_i b_i$ of bindings. Note that this pattern specifies the exact length the list must be.

Now we can say how to evaluate match e with p1 -> e1 | ... | pn -> en:

- Evaluate e to a value v.
- Match v against p1, then against p2, and so on, in the order they appear in the match expression.
- If v does not match against any of the patterns, then evaluation of the match expression raises a Match_failure exception. We haven't yet discussed exceptions in OCaml, but you're surely familiar with them from other languages. We'll come back to exceptions near the end of this chapter, after we've covered some of the other built-in data structures in OCaml.
- Otherwise, stop trying to match at the first time a match succeeds against a pattern. Let pi be that pattern and let b be the variable bindings produced by matching v against pi.
- Substitute those bindings inside ei, producing a new expression e'.
- Evaluate e' to a value v'.
- The result of the entire match expression is v'.

For example, here's how this match expression would be evaluated:

```
match 1 :: [] with
| [] -> false
| h :: t -> h = 1 && t = []
```

```
- : bool = true
```

- 1 :: [] is already a value.
- [] does not match 1 :: [].

- h :: t does match 1 :: [] and produces variable bindings {h->1,t->[]}, because:
 - h matches 1 and produces the variable binding h->1.
 - t matches [] and produces the variable binding $t \rightarrow$ [].
- Substituting $\{h->1,t->[]\}$ inside h=1 && t=[] produces a new expression 1=1 && []=[].
- Evaluating 1 = 1 && [] = [] yields the value true. We omit the justification for that fact here, but it follows from other evaluation rules for built-in operators and function application.
- So the result of the entire match expression is true.

Static semantics.

```
• If e: ta and for all i, it holds that pi: ta and ei: tb, then (match e with p1 \rightarrow e1 | ... | pn \rightarrow en): tb.
```

That rule relies on being able to judge whether a pattern has a particular type. As usual, type inference comes into play here. The OCaml compiler infers the types of any pattern variables as well as all occurrences of the wildcard pattern. As for the list patterns, they have the same type-checking rules as list expressions.

Additional Static Checking.

In addition to that type-checking rule, there are two other checks the compiler does for each match expression.

First, **exhaustiveness:** the compiler checks to make sure that there are enough patterns to guarantee that at least one of them matches the expression e, no matter what the value of that expression is at run time. This ensures that the programmer did not forget any branches. For example, the function below will cause the compiler to emit a warning:

By presenting that warning to the programmer, the compiler is helping the programmer to defend against the possibility of Match_failure exceptions at runtime.

Note: Sorry about how the output from the cell above gets split into many lines in the HTML. That is currently an open issue with JupyterBook, the framework used to build this book.

Second, **unused branches:** the compiler checks to see whether any of the branches could never be matched against because one of the previous branches is guaranteed to succeed. For example, the function below will cause the compiler to emit a warning:

3.1. Lists 61

```
let rec sum lst =
  match lst with
  | h :: t -> h + sum t
  | [ h ] -> h
  | [] -> 0
```

```
File "[13]", line 4, characters 4-9:
```

```
4 | | [ h ] -> h
```

```
^^^^
```

```
Warning 11: this match case is unused.
```

```
val sum : int list -> int = <fun>
```

The second branch is unused because the first branch will match anything the second branch matches.

Unused match cases are usually a sign that the programmer wrote something other than what they intended. So by presenting that warning, the compiler is helping the programmer to detect latent bugs in their code.

Here's an example of one of the most common bugs that causes an unused match case warning. Understanding it is also a good way to check your understanding of the dynamic semantics of match expressions:

```
let length_is lst n =
  match List.length lst with
  | n -> true
  | _ -> false
```

```
File "[14]", line 4, characters 4-5:
```

```
^
```

```
Warning 11: this match case is unused.
```

```
val length_is : 'a list -> 'b -> bool = <fun>
```

The programmer was thinking that if the length of lst is equal to n, then this function will return true, and otherwise will return false. But in fact this function *always* returns true. Why? Because the pattern variable n is distinct from the function argument n. Suppose that the length of lst is 5. Then the pattern match becomes: match 5 with n \rightarrow true | _ \rightarrow false. Does n match 5? Yes, according to the rules above: a variable pattern matches any value and here produces the binding n \rightarrow 5. Then evaluation applies that binding to true, substituting all occurrences of n inside of true with 5. Well, there are no such occurrences. So we're done, and the result of evaluation is just true.

What the programmer really meant to write was:

```
let length_is lst n =
  match List.length lst with
  | m -> m = n
```

```
val length_is : 'a list -> int -> bool = <fun>
```

or better yet:

```
let length_is lst n =
  List.length lst = n
```

```
val length_is : 'a list -> int -> bool = <fun>
```

3.1.5 Deep Pattern Matching

Patterns can be nested. Doing so can allow your code to look deeply into the structure of a list. For example:

```
• _ :: [] matches all lists with exactly one element
```

• _ :: _ matches all lists with at least one element

• _ :: _ :: [] matches all lists with exactly two elements

• _ :: _ :: _ matches all lists with at least three elements

3.1.6 Immediate Matches

When you have a function that immediately pattern-matches against its final argument, there's a nice piece of syntactic sugar you can use to avoid writing extra code. Here's an example: instead of

```
let rec sum lst =
  match lst with
  | [] -> 0
  | h :: t -> h + sum t
```

```
val sum : int list -> int = <fun>
```

you can write

```
val sum : int list -> int = <fun>
```

The word function is a keyword. Notice that we're able to leave out the line containing match as well as the name of the argument, which was never used anywhere else but that line. In such cases, though, it's especially important in the specification comment for the function to document what that argument is supposed to be, since the code no longer gives it a descriptive name.

3.1.7 OCamidoc and List Syntax

OCamldoc is a documentation generator similar to Javadoc. It extracts comments from source code and produces HTML (as well as other output formats). The standard library web documentation for the List module is generated by OCamldoc from the standard library source code for that module, for example.

3.1. Lists 63

Warning: There is a syntactic convention with square brackets in OCamldoc that can be confusing with respect to lists

In an OCamldoc comment, source code is surrounded by square brackets. That code will be rendered in typewriter face and syntax-highlighted in the output HTML. The square brackets in this case do not indicate a list.

For example, here is the comment for List.hd in the standard library source code:

```
(** Return the first element of the given list. Raise [Failure "hd"] if the list is empty. *)
```

The [Failure "hd"] does not mean a list containing the exception Failure "hd". Rather it means to typeset the expression Failure "hd" as source code, as you can see here.

This can get especially confusing when you want to talk about lists as part of the documentation. For example, here is a way we could rewrite that comment:

```
(** [hd lst] returns the first element of [lst].
Raises [Failure "hd"] if [lst = []]. *)
```

In [lst = []], the outer square brackets indicate source code as part of a comment, whereas the inner square brackets indicate the empty list.

3.1.8 List Comprehensions

Some languages, including Python and Haskell, have a syntax called *comprehension* that allows lists to be written somewhat like set comprehensions from mathematics. The earliest example of comprehensions seems to be the functional language NPL, which was designed in 1977.

OCaml doesn't have built-in syntactic support for comprehensions. Though some extensions were developed, none seem to be supported any longer. The primary tasks accomplished by comprehensions (filtering out some elements, and transforming others) are actually well-supported already by *higher-order programming*, which we'll study in a later chapter, and the pipeline operator, which we've already learned. So an additional syntax for comprehensions was never really needed.

3.1.9 Tail Recursion

Recall that a function is *tail recursive* if it calls itself recursively but does not perform any computation after the recursive call returns, and immediately returns to its caller the value of its recursive call. Consider these two implementations, sum and sum_tr of summing a list:

```
val sum : int list -> int = <fun>
```

```
val sum_plus_acc : int -> int list -> int = <fun>
```

```
val sum_tr : int list -> int = <fun>
```

Observe the following difference between the sum and sum_tr functions above: In the sum function, which is not tail recursive, after the recursive call returned its value, we add x to it. In the tail recursive sum_tr, or rather in sum_plus_acc, after the recursive call returns, we immediately return the value without further computation.

If you're going to write functions on really long lists, tail recursion becomes important for performance. So when you have a choice between using a tail-recursive vs. non-tail-recursive function, you are likely better off using the tail-recursive function on really long lists to achieve space efficiency. For that reason, the List module documents which functions are tail recursive and which are not.

But that doesn't mean that a tail-recursive implementation is strictly better. For example, the tail-recursive function might be harder to read. (Consider sum_plus_acc.) Also, there are cases where implementing a tail-recursive function entails having to do a pre- or post-processing pass to reverse the list. On small to medium sized lists, the overhead of reversing the list (both in time and in allocating memory for the reversed list) can make the tail-recursive version less time efficient. What constitutes "small" vs. "big" here? That's hard to say, but maybe 10,000 is a good estimate, according to the standard library documentation of the List module.

Here is a useful tail-recursive function to produce a long list:

```
(** [from i j l] is the list containing the integers from [i] to [j],
    inclusive, followed by the list [l].
    Example: [from 1 3 [0] = [1; 2; 3; 0]] *)
let rec from i j l = if i > j then l else from i (j - 1) (j :: l)

(** [i -- j] is the list containing the integers from [i] to [j], inclusive. *)
let (-- ) i j = from i j []

let long_list = 0 -- 1_000_000
```

```
val from : int -> int list -> int list = <fun>
```

```
val ( -- ) : int -> int list = <fun>
```

```
val long_list : int list =
 [0; 1; 2; 3; 4; 5; 6; 7; 8; 9; 10; 11; 12; 13; 14; 15; 16; 17; 18; 19; 20;
  21; 22; 23; 24; 25; 26; 27; 28; 29; 30; 31; 32; 33; 34; 35; 36; 37; 38;
  39; 40; 41; 42; 43; 44; 45; 46; 47; 48; 49; 50; 51; 52; 53; 54; 55; 56;
  57; 58; 59; 60; 61; 62; 63; 64; 65; 66; 67; 68; 69; 70; 71; 72; 73; 74;
  75; 76; 77; 78; 79; 80; 81; 82; 83; 84; 85; 86; 87; 88; 89; 90; 91; 92;
  93; 94; 95; 96; 97; 98; 99; 100; 101; 102; 103; 104; 105; 106; 107; 108;
  109; 110; 111; 112; 113; 114; 115; 116; 117; 118; 119; 120; 121; 122; 123;
  124; 125; 126; 127; 128; 129; 130; 131; 132; 133; 134; 135; 136; 137; 138;
  139; 140; 141; 142; 143; 144; 145; 146; 147; 148; 149; 150; 151; 152; 153;
  154; 155; 156; 157; 158; 159; 160; 161; 162; 163; 164; 165; 166; 167; 168;
  169; 170; 171; 172; 173; 174; 175; 176; 177; 178; 179; 180; 181; 182; 183;
  184; 185; 186; 187; 188; 189; 190; 191; 192; 193; 194; 195; 196; 197; 198;
  199; 200; 201; 202; 203; 204; 205; 206; 207; 208; 209; 210; 211; 212; 213;
  214; 215; 216; 217; 218; 219; 220; 221; 222; 223; 224; 225; 226; 227; 228;
  229; 230; 231; 232; 233; 234; 235; 236; 237; 238; 239; 240; 241; 242; 243;
  244; 245; 246; 247; 248; 249; 250; 251; 252; 253; 254; 255; 256; 257; 258;
```

(continues on next page)

3.1. Lists 65

(continued from previous page)

```
259; 260; 261; 262; 263; 264; 265; 266; 267; 268; 269; 270; 271; 272; 273; 274; 275; 276; 277; 278; 279; 280; 281; 282; 283; 284; 285; 286; 287; 288; 289; 290; 291; 292; 293; 294; 295; 296; 297; 298; ...]
```

It would be worthwhile to study the definition of — to convince yourself that you understand (i) how it works and (ii) why it is tail recursive.

You might in the future decide you want to create such a list again. Rather than having to remember where this definition is, and having to copy it into your code, here's an easy way to create the same list using a built-in library function:

```
List.init 1_000_000 Fun.id
```

```
- : int list =
[0; 1; 2; 3; 4; 5; 6; 7; 8; 9; 10; 11; 12; 13; 14; 15; 16; 17; 18; 19; 20;
21; 22; 23; 24; 25; 26; 27; 28; 29; 30; 31; 32; 33; 34; 35; 36; 37; 38; 39;
40; 41; 42; 43; 44; 45; 46; 47; 48; 49; 50; 51; 52; 53; 54; 55; 56; 57; 58;
59; 60; 61; 62; 63; 64; 65; 66; 67; 68; 69; 70; 71; 72; 73; 74; 75; 76; 77;
78; 79; 80; 81; 82; 83; 84; 85; 86; 87; 88; 89; 90; 91; 92; 93; 94; 95; 96;
97; 98; 99; 100; 101; 102; 103; 104; 105; 106; 107; 108; 109; 110; 111; 112;
113; 114; 115; 116; 117; 118; 119; 120; 121; 122; 123; 124; 125; 126; 127;
128; 129; 130; 131; 132; 133; 134; 135; 136; 137; 138; 139; 140; 141; 142;
143; 144; 145; 146; 147; 148; 149; 150; 151; 152; 153; 154; 155; 156; 157;
158; 159; 160; 161; 162; 163; 164; 165; 166; 167; 168; 169; 170; 171; 172;
173; 174; 175; 176; 177; 178; 179; 180; 181; 182; 183; 184; 185; 186; 187;
188; 189; 190; 191; 192; 193; 194; 195; 196; 197; 198; 199; 200; 201; 202;
203; 204; 205; 206; 207; 208; 209; 210; 211; 212; 213; 214; 215; 216; 217;
218; 219; 220; 221; 222; 223; 224; 225; 226; 227; 228; 229; 230; 231; 232;
233; 234; 235; 236; 237; 238; 239; 240; 241; 242; 243; 244; 245; 246; 247;
248; 249; 250; 251; 252; 253; 254; 255; 256; 257; 258; 259; 260; 261; 262;
263; 264; 265; 266; 267; 268; 269; 270; 271; 272; 273; 274; 275; 276; 277;
278; 279; 280; 281; 282; 283; 284; 285; 286; 287; 288; 289; 290; 291; 292;
293; 294; 295; 296; 297; 298; ...]
```

Expression List.init len f creates the list [f 0; f 1; ...; f (len - 1)], and it does so tail recursively if len is bigger than 10,000. Function Fun.id is simply the identify function fun $x \rightarrow x$.

3.2 Variants

A *variant* is a data type representing a value that is one of several possibilities. At their simplest, variants are like enums from C or Java:

```
type day = Sun | Mon | Tue | Wed | Thu | Fri | Sat
let d = Tue
```

```
type day = Sun | Mon | Tue | Wed | Thu | Fri | Sat
```

```
val d : day = Tue
```

The individual names of the values of a variant are called *constructors* in OCaml. In the example above, the constructors are Sun, Mon, etc. This is a somewhat different use of the word constructor than in C++ or Java.

For each kind of data type in OCaml, we've been discussing how to build and access it. For variants, building is easy: just write the name of the constructor. For accessing, we use pattern matching. For example:

```
let int_of_day d =
  match d with
  | Sun -> 1
  | Mon -> 2
  | Tue -> 3
  | Wed -> 4
  | Thu -> 5
  | Fri -> 6
  | Sat -> 7
```

```
val int_of_day : day -> int = <fun>
```

There isn't any kind of automatic way of mapping a constructor name to an int, like you might expect from languages with enums.

Syntax.

Defining a variant type:

```
type t = C1 | ... | Cn
```

The constructor names must begin with an uppercase letter. OCaml uses that to distinguish constructors from variable identifiers.

The syntax for writing a constructor value is simply its name, e.g., C.

Dynamic semantics.

• A constructor is already a value. There is no computation to perform.

Static semantics.

• If t is a type defined as type t = ... | C | ..., then C : t.

3.2.1 Scope

Suppose there are two types defined with overlapping constructor names, for example,

```
type t1 = C | D
type t2 = D | E
let x = D
```

```
type t1 = C | D
```

```
type t2 = D | E
```

```
val x : t2 = D
```

When D appears after these definitions, to which type does it refer? That is, what is the type of x above? The answer is that the type defined later wins. So x: t2. That is potentially surprising to programmers, so within any given scope (e.g., a file or a module, though we haven't covered modules yet) it's idiomatic whenever overlapping constructor names might occur to prefix them with some distinguishing character. For example, suppose we're defining types to represent Pokémon:

3.2. Variants 67

```
type ptype =
  TNormal | TFire | TWater

type peff =
  ENormal | ENotVery | ESuper
```

```
type ptype = TNormal | TFire | TWater
```

```
type peff = ENormal | ENotVery | ESuper
```

Because "Normal" would naturally be a constructor name for both the type of a Pokémon and the effectiveness of a Pokémon attack, we add an extra character in front of each constructor name to indicate whether it's a type or an effectiveness.

3.2.2 Pattern Matching

Each time we introduced a new kind of data type, we need to introduce the new patterns associated with it. For variants, this is easy. We add the following new pattern form to the list of legal patterns:

• a constructor name C

And we extend the definition of when a pattern matches a value and produces a binding as follows:

• The pattern C matches the value C and produces no bindings.

Note: Variants are considerably more powerful than what we have seen here. We'll return to them again soon.

3.3 Unit Testing with OUnit

Note: This section is a bit of a detour from our study of data types, but it's a good place to take the detour: we now know just enough to understand how unit testing can be done in OCaml, and there's no good reason to wait any longer to learn about it.

Using the toplevel to test functions will only work for very small programs. Larger programs need *test suites* that contain many *unit tests* and can be re-run every time we update our code base. A unit test is a test of one small piece of functionality in a program, such as an individual function.

We've now learned enough features of OCaml to see how to do unit testing with a library called OUnit. It is a unit testing framework similar to JUnit in Java, HUnit in Haskell, etc. The basic workflow for using OUnit is as follows:

- Write a function in a file f.ml. There could be many other functions in that file too.
- Write unit tests for that function in a separate file f_test.ml. That exact name, with an underscore and "test" is not actually essential, but is a convention we'll often follow.
- Build and run f_test to execute the unit tests.

The OUnit documentation is available on Github.

3.3.1 An Example of OUnit

The following example shows you how to create an OUnit test suite. There are some things in the example that might at first seem mysterious; they are discussed in the next section.

Create a file named sum.ml, and put the following code into it:

```
let rec sum = function
    | [] -> 0
    | x :: xs -> x + sum xs
```

Now create a second file named sum_test.ml, and put this code into it:

```
open OUnit2
open Sum

let tests = "test suite for sum" >::: [
   "empty" >:: (fun _ -> assert_equal 0 (sum []));
   "singleton" >:: (fun _ -> assert_equal 1 (sum [1]));
   "two_elements" >:: (fun _ -> assert_equal 3 (sum [1; 2]));
]

let _ = run_test_tt_main tests
```

Depending on your editor and its configuration, you probably now see some "Unbound module" errors about OUnit2 and Sum. Your code is actually correct, but your editor doesn't know how to understand it yet. To fix that, create a file in the same directory and name it .merlin. (Note the leading dot in that filename.) In that file put the following:

```
B _build
PKG ounit2
```

The first line tells Merlin to look for compiled code from other source files inside the _build directory, which is where OCamlbuild places compiled code. The second line tells Merlin to look for the OUnit2 package.

After you save the .merlin file, the error about OUnit2 should be gone, though you might need to cause your editor to re-check the sum_test.ml file to get rid of it. You can close and re-open the window, or make a trivial change in the file (e.g., add then delete a space) to make that tappen.

But the error about Sum will still be there, because you haven't yet compiled that file. To do that, run this command:

```
$ ocamlbuild -pkgs ounit2 sum_test.byte
```

Now the error about Sum should be gone, though again you might need to convince your editor to re-check the file.

Finally, you can run the test suite:

```
$ ./sum_test.byte
```

You will get a response something like this:

```
Ran: 3 tests in: 0.12 seconds.
OK
```

Now suppose we modify sum.ml to introduce a bug by changing the code in it to the following:

```
let rec sum = function
    | [] -> 1 (* bug *)
    | x :: xs -> x + sum xs
```

If rebuild and re-execute sum_test.byte, all test cases now fail. The output tells us the names of the failing cases. Here's the beginning of the output, in which we've replaced some strings that will be dependent on your own local computer with . . . :

The first line of that output

```
FFF
```

tells us that OUnit ran three test cases and all three failed.

The next interesting line

```
Error: test suite for sum:2:two_elements.
```

tells us that in the test suite named test suite for sum the test case at index 2 named two_elements failed. The rest of the output for that test case is not particularly interesting; let's ignore it for now.

3.3.2 Explanation of the OUnit Example

Let's study more carefully what we just did in the previous section. In the test file, open OUnit2 brings into scope the many definitions in OUnit2, which is version 2 of the OUnit framework. And open Sum brings into scope the definitions from sum.ml. We'll learn more about scope and the open keyword later in a later chapter.

Then we created a list of test cases:

```
[
  "empty" >:: (fun _ -> assert_equal 0 (sum []));
  "one" >:: (fun _ -> assert_equal 1 (sum [1]));
  "onetwo" >:: (fun _ -> assert_equal 3 (sum [1; 2]));
]
```

Each line of code is a separate test case. A test case has a string giving it a descriptive name, and a function to run as the test case. In between the name and the function we write >::, which is a custom operator defined by the OUnit framework. Let's look at the first function from above:

```
fun _ -> assert_equal 0 (sum [])
```

Every test case function receives as input a parameter that OUnit calls a *test context*. Here (and in many of the test cases we write) we don't actually need to worry about the context, so we use the underscore to indicate that the function ignores its input. The function then calls assert_equal, which is a function provided by OUnit that checks to see whether its two arguments are equal. If so the test case succeeds. If not, the test case fails.

Then we created a test suite:

```
let tests = "test suite for sum" >::: [
  "empty" >:: (fun _ -> assert_equal 0 (sum []));
  "singleton" >:: (fun _ -> assert_equal 1 (sum [1]));
  "two_elements" >:: (fun _ -> assert_equal 3 (sum [1; 2]));
]
```

The >::: operator is another custom OUnit operator. It goes between the name of the test suite and the list of test cases in that suite.

Then we ran the test suite:

```
let _ = run_test_tt_main tests
```

The function run_test_tt_main is provided by OUnit. It runs a test suite and prints the results of which test cases passed vs. which failed to standard output. The use of let _ = here indicates that we don't care what value the function returns; it just gets discarded.

Finally, when we compiled the test file, we linked in the OUnit2 package:

```
$ ocamlbuild -pkgs ounit2 sum_test.byte
```

If you get tired of typing the pkgs ounit2 part of that, you can instead create a file named _tags (note the underscore) in the same directory and put the following into it:

```
true: package(ounit2)
```

Now Ocambuild will automatically link in OUnit2 everytime you compile in this directory, without you having to give the pkgs flag. The tradeoff is that you now have to pass a different flag to Ocambuild:

```
$ ocambuild -use-ocamlfind sum_test.byte
```

And you will continue having to pass that flag as long as the _tags file exists. Why is this any better? If there are many packages you want to link, with the tags file you end up having to pass only one option on the command line, instead of many.

3.3.3 Improving OUnit Output

In our example with the buggy implementation of sum, we got the following output:

```
Error: test suite for sum:2:two_elements.

File ".../_build/oUnit-test suite for sum-...#01.log", line 8, characters 1-1:
Error: test suite for sum:2:two_elements (in the log).

Called from unknown location

not equal
```

Let's see how to improve that output to be a little more informative.

Stack Traces

The Called from an unknown location indicates OCaml was unable to provide a stack trace. That happened because, by default, stack traces are disabled. We can enable them by compiling the code with the debug tag:

```
$ ocamlbuild -pkgs ounit2 -tag debug sum_test.byte
$ ./sum_test.byte

Error: test suite for sum:2:two_elements.

File "/Users/clarkson/tmp/sum/_build/oUnit-test suite for sum-...#01.log", line 9,...

characters 1-1:

Error: test suite for sum:2:two_elements (in the log).

Raised at file "src/oUnitAssert.ml", line 45, characters 8-27

Called from file "src/oUnitRunner.ml", line 46, characters 13-26

not equal
```

Now we see the stack trace that resulted from assert_equal raising an exception. You'll probably agree that stack trace isn't very informative though: what matters is which test case fails, not which files in the implementation of OUnit were involved in raising the exception. And we could already identify the failing test case from the first line of output. (It's the test case named two elements, which at position 2 in the test suite named test suite for sum.)

So we don't usually bother enabling stack traces for OUnit test suites. Nonetheless, it could occasionally be useful if your *own* code is raising exceptions that you want to track down.

Output Values

The not equal in the OUnit output means that assert_equal discovered the two values passed to it in that test case were not equal. That's not so informative: we'd like to know why they're not equal. In particular, we'd like to know what the actual output produced by sum was for that test case. To find out, we need to pass an additional argument to assert_equal. That argument, whose label is printer, should be a function that can transform the outputs to strings. In this case, the outputs are integers, so string_of_int from the Stdlib module will suffice. We modify the test suite as follows:

```
let tests = "test suite for sum" >::: [
  "empty" >:: (fun _ -> assert_equal 0 (sum []) ~printer:string_of_int);
  "singleton" >:: (fun _ -> assert_equal 1 (sum [1]) ~printer:string_of_int);
  "two_elements" >:: (fun _ -> assert_equal 3 (sum [1; 2]) ~printer:string_of_int);
]
```

And now we get more informative output:

(continues on next page)

(continued from previous page)

```
expected: 3 but got: 4
```

That output means that the test named two_elements asserted the equality of 3 and 4. The expected output was 3 because that was the first input to assert_equal, and that function's specification says that in assert_equal x y, the output you (as the tester) are expecting to get should be x, and the output the function being tested actually produces should be y.

Notice how our test suite is accumulating a lot of redundant code. In particular, we had to add the printer argument to several lines. Let's improve that code by factoring out a function that constructs test cases:

```
let make_sum_test name expected_output input =
  name >:: (fun _ -> assert_equal expected_output (sum input) ~printer:string_of_int)

let tests = "test suite for sum" >::: [
  make_sum_test "empty" 0 [];
  make_sum_test "singleton" 1 [1];
  make_sum_test "two_elements" 3 [1; 2];
]
```

For output types that are more complicated than integers, you will end up needing to write your own functions to pass to printer. This is similar to writing toString() methods in Java: for complicated types you invent yourself, the language doesn't know how to render them as strings. You have to provide the code that does it.

3.3.4 Testing for Exceptions

We have a little more of OCaml to learn before we can see how to test for exceptions. You can peek ahead to *the section* on exceptions if you want to know now.

3.3.5 Test-Driven Development

Testing doesn't have to happen strictly after you write code. In *test-driven development* (TDD), testing comes first! It emphasizes *incremental* development of code: there is always something that can be tested. Testing is not something that happens after implementation; instead, *continuous testing* is used to catch errors early. Thus, it is important to develop unit tests immediately when the code is written. Automating test suites is crucial so that continuous testing requires essentially no effort.

Here's an example of TDD. We deliberately choose an exceedingly simple function to implement, so that the process is clear. Suppose we are working with a datatype for days:

```
type day = Sunday | Monday | Tuesday | Wednesday | Thursday | Friday | Saturday
```

And we want to write a function next_weekday : day -> day that returns the next weekday after a given day. We start by writing the most basic, broken version of that function we can:

```
let next_weekday d = failwith "Unimplemented"
```

Note: The built-in function failwith raises an exception along with the error message passed to the function.

Then we write the simplest unit test we can imagine. For example, we know that the next weekday after Monday is Tuesday. So we add a test:

```
let tests = "test suite for next_weekday" >::: [
   "tue_after_mon" >:: (fun _ -> assert_equal (next_weekday Monday) Tuesday);
]
```

Then we run the OUnit test suite. It fails, as expected. That's good! Now we have a concrete goal, to make that unit test pass. We revise next weekday to make that happen:

```
let next_weekday d =
  match d with
  | Monday -> Tuesday
  | _ -> failwith "Unimplemented"
```

We compile and run the test; it passes. Time to add some more tests. The simplest remaining possibilities are tests involving just weekdays, rather than weekends. So let's add tests for weekdays.

```
let tests = "test suite for next_weekday" >::: [
  "tue_after_mon" >:: (fun _ -> assert_equal (next_weekday Monday) Tuesday);
  "wed_after_tue" >:: (fun _ -> assert_equal (next_weekday Tuesday) Wednesday);
  "thu_after_wed" >:: (fun _ -> assert_equal (next_weekday Wednesday) Thursday);
  "fri_after_thu" >:: (fun _ -> assert_equal (next_weekday Thursday) Friday);
]
```

We compile and run the tests; many fail. That's good! We add new functionality:

```
let next_weekday d =
   match d with
   | Monday -> Tuesday
   | Tuesday -> Wednesday
   | Wednesday -> Thursday
   | Thursday -> Friday
   | _ -> failwith "Unimplemented"
```

We compile and run the tests; they pass. At this point we could move on to handling weekends, but we should first notice something about the tests we've written: they involve repeating a lot of code. In fact, we probably wrote them by copying-and-pasting the first test, then modifying it for the next three. That's a sign that we should *refactor* the code. (As we did before with the sum function we were testing.)

Let's abstract a function that creates test cases for next_weekday:

```
let make_next_weekday_test name expected_output input=
  name >:: (fun _ -> assert_equal expected_output (next_weekday input))

let tests = "test suite for next_weekday" >::: [
  make_next_weekday_test "tue_after_mon" Tuesday Monday;
  make_next_weekday_test "wed_after_tue" Wednesday Tuesday;
  make_next_weekday_test "thu_after_wed" Thursday Wednesday;
  make_next_weekday_test "fri_after_thu" Friday Thursday;
]
```

Now we finish the testing and implementation by handling weekends. First we add some test cases:

```
...
make_next_weekday_test "mon_after_fri" Monday Friday;
make_next_weekday_test "mon_after_sat" Monday Saturday;
make_next_weekday_test "mon_after_sun" Monday Sunday;
...
```

Then we finish the function:

```
let next_weekday d =
  match d with
  | Monday -> Tuesday
  | Tuesday -> Wednesday
  | Wednesday -> Thursday
  | Thursday -> Friday
  | Friday -> Monday
  | Saturday -> Monday
  | Sunday -> Monday
```

Of course, most people could write that function without errors even if they didn't use TDD. But rarely do we implement functions that are so simple.

Process. Let's review the process of TDD:

- Write a failing unit test case. Run the test suite to prove that the test case fails.
- Implement just enough functionality to make the test case pass. Run the test suite to prove that the test case passes.
- Improve code as needed. In the example above we refactored the test suite, but often we'll need to refactor the functionality being implemented.
- Repeat until you are satisfied that the test suite provides evidence that your implementation is correct.

3.4 Records and Tuples

Singly-linked lists are a great data structure, but what if you want a fixed number of elements, instead of an unbounded number? Or what if you want the elements to have distinct types? Or what if you want to access the elements by name instead of by number? Lists don't make any of those possibilities easy. Instead, OCaml programmers use records and tuples.

3.4.1 Records

A *record* is a composite of other types of data, each of which is named. OCaml records are much like structs in C. Here's an example of a record type definition mon for a Pokémon, re-using the ptype definition from the *variants* section:

```
type ptype = TNormal | TFire | TWater
type mon = {name : string; hp : int; ptype : ptype}
```

```
type ptype = TNormal | TFire | TWater
```

```
type mon = { name : string; hp : int; ptype : ptype; }
```

This type defines a record with three *fields* named name, hp (hit points), and ptype. The type of each of those fields is also given. Note that ptype can be used as both a type name and a field name; the *namespace* for those is distinct in OCaml.

To build a value of a record type, we write a record expression, which looks like this:

```
{name = "Charmander"; hp = 39; ptype = TFire}

- : mon = {name = "Charmander"; hp = 39; ptype = TFire}
```

So in a type definition we write a colon between the name and the type of a field, but in an expression we write an equals sign.

To access a record and get a field from it, we use the dot notation that you would expect from many other languages. For example:

```
let c = {name = "Charmander"; hp = 39; ptype = TFire};;
c.hp
```

```
val c : mon = {name = "Charmander"; hp = 39; ptype = TFire}
```

```
- : int = 39
```

It's also possible to use pattern matching to access record fields:

```
match c with {name = n; hp = h; ptype = t} -> h
```

```
- : int = 39
```

The n, h, and t here are pattern variables. There is a syntactic sugar provided if you want to use the same name for both the field and a pattern variable:

```
match c with {name; hp; ptype} -> hp
```

```
- : int = 39
```

Here, the pattern {name; hp; ptype} is sugar for {name = name; hp = hp; ptype = ptype}. In each of those subexpressions, the identifier appearing on the left-hand side of the equals is a field name, and the identifier appearing on the right-hand side is a pattern variable.

Syntax.

A record expression is written:

```
{f1 = e1; ...; fn = en}
```

The order of the fi=ei inside a record expression is irrelevant. For example, $\{f = e1; g = e2\}$ is entirely equivalent to $\{g = e2; f = e1\}$.

A field access is written:

```
e.f
```

where f must be an identifier of a field name, not an expression. That restriction is the same as in any other language with similar features—for example, Java field names. If you really do want to *compute* which identifier to access, then actually you want a different data structure: a *map* (also known by many other names: a *dictionary* or *association list* or *hash table* etc., though there are subtle differences implied by each of those terms.)

Dynamic semantics.

- If for all i in 1..., it holds that ei ==> vi, then {f1 = e1; ...; fn = en} ==> {f1 = v1; ...; fn = vn}.
- If $e ==> \{...; f = v; ...\}$ then e.f ==> v.

Static semantics.

A record type is written:

```
{f1 : t1; ...; fn : tn}
```

The order of the fi:ti inside a record type is irrelevant. For example, $\{f : t1; g : t2\}$ is entirely equivalent to $\{g:t2;f:t1\}$.

Note that record types must be defined before they can be used. This enables OCaml to do better type inference than would be possible if record types could be used without definition.

The type checking rules are:

- If for all i in 1..., it holds that ei: ti, and if t is defined to be {f1: t1; ...; fn: tn}, then {f1 = e1; ...; fn = en}: t. Note that the set of fields provided in a record expression must be the full set of fields defined as part of the record's type (but see below regarding *record copy*).
- If e: t1 and if t1 is defined to be $\{\ldots; f: t2; \ldots\}$, then e.f: t2.

Record copy.

Another syntax is also provided to construct a new record out of an old record:

```
{e with f1 = e1; ...; fn = en}
```

This doesn't mutate the old record. Rather, it constructs a new record with new values. The set of fields provided after the with does not have to be the full set of fields defined as part of the record's type. In the newly-copied record, any field not provided as part of the with is copied from the old record.

The dynamic and static semantics of this are what you might expect, though they are tedious to write down mathematically.

Pattern matching.

We add the following new pattern form to the list of legal patterns:

```
• \{f1 = p1; ...; fn = pn\}
```

And we extend the definition of when a pattern matches a value and produces a binding as follows:

If for all i in 1...n, it holds that pi matches vi and produces bindings b_i, then the record pattern {f1 = p1; ...; fn = pn} matches the record value {f1 = v1; ...; fn = vn; ...} and produces the set ∪_i b_i of bindings. Note that the record value may have more fields than the record pattern does.

As a syntactic sugar, another form of record pattern is provided: $\{f1; \ldots; fn\}$. It is desugared to $\{f1 = f1; \ldots; fn = fn\}$.

3.4.2 Tuples

Like records, *tuples* are a composite of other types of data. But instead of naming the *components*, they are identified by position. Here are some examples of tuples:

```
(1, 2, 10)
(true, "Hello")
([1; 2; 3], (0.5, 'X'))
```

A tuple with two components is called a *pair*. A tuple with three components is called a *triple*. Beyond that, we usually just use the word *tuple* instead of continuing a naming scheme based on numbers.

Tip: Beyond about three components, it's arguably better to use records instead of tuples, because it becomes hard for a programmer to remember which component was supposed to represent what information.

Building of tuples is easy: just write the tuple, as above. Accessing again involves pattern matching, for example:

```
match (1, 2, 3) with (x, y, z) \rightarrow x + y + z
```

```
- : int = 6
```

Syntax.

A tuple is written

```
(e1, e2, ..., en)
```

The parentheses are not entirely mandatory —often your code can successfully parse without them— but they are usually considered to be good style to include.

Dynamic semantics.

• If for all i in 1... n it holds that ei ==> vi, then (e1, ..., en) ==> (v1, ..., vn).

Static semantics.

Tuple types are written using a new type constructor *, which is different than the multiplication operator. The type t1 * ... * tn is the type of tuples whose first component has type t1, ..., and nth component has type tn.

```
• If for all i in 1...n it holds that ei : ti, then (e1, ..., en) : t1 * ... * tn.
```

Pattern matching.

We add the following new pattern form to the list of legal patterns:

• (p1, ..., pn)

The parentheses are again not entirely mandatory but usually are idimoatic to include.

And we extend the definition of when a pattern matches a value and produces a binding as follows:

• If for all i in 1..n, it holds that pi matches vi and produces bindings b_i , then the tuple pattern (p1, ..., pn) matches the tuple value (v1, ..., vn) and produces the set $\bigcup_i b_i$ of bindings. Note that the tuple value must have exactly the same number of components as the tuple pattern does.

3.4.3 Variants vs. Tuples and Records

Note: The second video above uses more advanced examples of variants that will be studied in a *later section*.

The big difference between variants and the types we just learned (records and tuples) is that a value of a variant type is one of a set of possibilities, whereas a value of a tuple or record type provides each of a set of possibilities. Going back to our examples, a value of type day is one of Sun or Mon or etc. But a value of type mon provides each of a string and an int and ptype. Note how, in those previous two sentences, the word "or" is associated with variant types, and the word "and" is associated with tuple and record types. That's a good clue if you're ever trying to decide whether you want to use a variant, or a tuple or record: if you need one piece of data or another, you want a variant; if you need one piece of data and another, you want a tuple or record.

One-of types are more commonly known as *sum types*, and each-of types as *product types*. Those names come from set theory. Variants are like disjoint union, because each value of a variant comes from one of many underlying sets (and thus far each of those sets is just a single constructor hence has cardinality one). Disjoint union is indeed sometimes written with a summation operator Σ . Tuples/records are like Cartesian product, because each value of a tuple or record contains a value from each of many underlying sets. Cartesian product is usually written with a product operator, \times or Π .

3.5 Advanced Pattern Matching

Here are some additional pattern forms that are useful:

- p1 | ... | pn: an "or" pattern; matching against it succeeds if a match succeeds against any of the individual patterns p1, which are tried in order from left to right. All the patterns must bind the same variables.
- (p: t): a pattern with an explicit type annotation.
- c: here, c means any constant, such as integer literals, string literals, and booleans.
- 'ch1'..'ch2': here, ch means a character literal. For example, 'A'..'Z' matches any uppercase letter.
- p when e: matches p but only if e evaluates to true.

You can read about all the pattern forms in the manual.

3.5.1 Pattern Matching with Let

The syntax we've been using so far for let expressions is, in fact, a special case of the full syntax that OCaml permits. That syntax is:

```
let p = e1 in e2
```

That is, the left-hand side of the binding may in fact be a pattern, not just an identifier. Of course, variable identifiers are on our list of valid patterns, so that's why the syntax we've studied so far is just a special case.

Given this syntax, we revisit the semantics of let expressions.

Dynamic semantics.

To evaluate let p = e1 in e2:

- 1. Evaluate e1 to a value v1.
- 2. Match v1 against pattern p. If it doesn't match, raise the exception Match_failure. Otherwise, if it does match, it produces a set b of bindings.
- 3. Substitute those bindings b in e2, yielding a new expression e2.
- 4. Evaluate e2 ' to a value v2.
- 5. The result of evaluating the let expression is v2.

Static semantics.

- If all the following hold then (let p = e1 in e2): t2:
 - e1 : t1
 - the pattern variables in p are x1..xn
 - e2 : t2 under the assumption that for all i in 1..n it holds that xi : ti,

Let definitions.

As before, a let definition can be understood as a let expression whose body has not yet been given. So their syntax can be generalized to

```
let p = e
```

and their semantics follow from the semantics of let expressions, as before.

3.5.2 Pattern Matching with Functions

The syntax we've been using so far for functions is also a special case of the full syntax that OCaml permits. That syntax is:

The truly primitive syntactic form we need to care about is $fun p \rightarrow e$. Let's revisit the semantics of anonymous functions and their application with that form; the changes to the other forms follow from those below:

Static semantics.

- Let x1..xn be the pattern variables appearing in p. If by assuming that x1:t1 and x2:t2 and ... and xn:tn, we can conclude that p:t and e:u, then fun $p \to e:t \to u$.
- The type checking rule for application is unchanged.

Dynamic semantics.

- The evaluation rule for anonymous functions is unchanged.
- To evaluate e0 e1:
 - 1. Evaluate e0 to an anonymous function fun $p \rightarrow e$, and evaluate e1 to value v1.
 - 2. Match v1 against pattern p. If it doesn't match, raise the exception Match_failure. Otherwise, if it does match, it produces a set *b* of bindings.
 - 3. Substitute those bindings b in e, yielding a new expression e'.
 - 4. Evaluate e' to a value v, which is the result of evaluating e0 e1.

3.5.3 Pattern Matching Examples

Here are several ways to get a Pokemon's hit points:

```
(* Pokemon types *)
type ptype = TNormal | TFire | TWater

(* A record to represent Pokemon *)
type mon = { name : string; hp : int; ptype : ptype }

(* OK *)
let get_hp m = match m with { name = n; hp = h; ptype = t } -> h

(* better *)
let get_hp m = match m with { name = _; hp = h; ptype = _ } -> h

(* better *)
let get_hp m = match m with { name; hp; ptype } -> hp

(* better *)
let get_hp m = match m with { hp } -> hp

(* better *)
let get_hp m = match m with { hp } -> hp
```

```
type ptype = TNormal | TFire | TWater

type mon = { name : string; hp : int; ptype : ptype; }

val get_hp : mon -> int = <fun>

val get_hp : mon -> int = <fun>
```

Here's how to get the first and second components of a pair:

```
let fst (x, _) = x
let snd (_, y) = y
```

```
val fst : 'a * 'b -> 'a = <fun>
```

```
val snd : 'a * 'b -> 'b = <fun>
```

Both fst and snd are actually already defined for you in the standard library.

Finally, here are several ways to get the 3rd component of a triple:

```
(* OK *)
let thrd t = match t with x, y, z -> z

(* good *)
let thrd t =
    let x, y, z = t in
    z

(* better *)
let thrd t =
    let _, _, z = t in
    z

(* best *)
let thrd (_, _, z) = z
```

```
val thrd : 'a * 'b * 'c -> 'c = <fun>
```

```
val thrd : 'a * 'b * 'c -> 'c = <fun>
```

```
val thrd : 'a * 'b * 'c -> 'c = <fun>
```

```
val thrd : 'a * 'b * 'c -> 'c = <fun>
```

The standard library does not define any functions for triples, quadruples, etc.

3.6 Type Synonyms

A *type synonym* is a new name for an already existing type. For example, here are some type synonyms that might be useful in representing some types from linear algebra:

```
type point = float * float
type vector = float list
type matrix = float list list
```

```
type point = float * float
```

```
type vector = float list
```

```
type matrix = float list list
```

Anywhere that a float * float is expected, you could use point, and vice-versa. The two are completely exchangeable for one another. In the following code, get_x doesn't care whether you pass it a value that is annotated as one vs. the other:

```
let get_x = fun (x, _) -> x

let p1 : point = (1., 2.)
let p2 : float * float = (1., 3.)

let a = get_x p1
let b = get_x p2
```

```
val get_x : 'a * 'b -> 'a = <fun>
```

```
val p1 : point = (1., 2.)
```

Type synonyms are useful because they let us give descriptive names to complex types. They are a way of making code more self-documenting.

3.7 Options

Suppose you want to write a function that *usually* returns a value of type t, but *sometimes* returns nothing. For example, you might want to define a function list_max that returns the maximum value in a list, but there's not a sensible thing to return on an empty list:

There are a couple possibilities to consider:

- Return min_int? But then list_max will only work for integers—not floats or other types.
- Raise an exception? But then the user of the function has to remember to catch the exception.
- Return null? That works in Java, but by design OCaml does not have a null value. That's actually a good thing: null pointer bugs are not fun to debug.

Note: Sir Tony Hoare calls his invention of null a "billion-dollar mistake".

In addition to those possibilities, OCaml provides something even better called an *option*. (Haskellers will recognize options as the Maybe monad.)

You can think of an option as being like a closed box. Maybe there's something inside the box, or maybe box is empty. We don't know which until we open the box. If there turns out to be something inside the box when we open it, we can take that thing out and use it. Thus, options provide a kind of "maybe type," which ultimately is a kind of one-of type: the box is in one of two states, full or empty.

In list_max above, we'd like to metaphorically return a box that's empty if the list is empty, or a box that contains the maximum element of the list if the list is non empty.

Here's how we create an option that is like a box with 42 inside it:

```
Some 42
```

```
- : int option = Some 42
```

And here's how we create an option that is like an empty box:

```
None
```

```
- : 'a option = None
```

The Some means there's something inside the box, and it's 42. The None means there's nothing inside the box.

As for the types we see above, to ption is a type for every type t, much like to list is a type for every type t. Values of type to option might contain a value of type to, or they might contain nothing. None has type 'a option because it's unknown what the type is of the thing inside, as there isn't anything inside.

You can access the contents of an option value e using pattern matching. Here's a function that extracts an int from an option, if there is one inside, and converts it to a string:

```
let extract o =
  match o with
  | Some i -> string_of_int i
   | None -> "";;
```

```
val extract : int option -> string = <fun>
```

And here are a couple of example usages of that function:

```
extract (Some 42);;
extract None;;
```

```
- : string = "42"
```

```
- : string = ""
```

Here's how we can write list_max with options:

(continues on next page)

3.7. Options 83

(continued from previous page)

```
val list_max : 'a list -> 'a option = <fun>
```

Tip: The begin..end wrapping the nested pattern match above is not strictly required here but is not a bad habit, as it will head off potential syntax errors in more complicated code. The keywords begin and end are equivalent to (and).

In Java, every object reference is implicitly an option. Either there is an object inside the reference, or there is nothing there. That "nothing" is represented by the value null. Java does not force programmers to explicitly check for the null case, which leads to null pointer exceptions. OCaml options force the programmer to include a branch in the pattern match for None, thus guaranteeing that the programmer thinks about the right thing to do when there's nothing there. So we can think of options as a principled way of eliminating null from the language. Using options is usually considered better coding practice than raising exceptions, because it forces the caller to do something sensible in the None case.

Syntax and semantics of options.

- t option is a type for every type t.
- None is a value of type 'a option.
- Some e is an expression of type t option if e: t. If e ==> v then Some e ==> Some v

3.8 Association Lists

A *map* is a data structure that maps *keys* to *values*. Maps are also known as *dictionaries*. One easy implementation of a map is an *association list*, which is a list of pairs. Here, for example, is an association list that maps some shape names to the number of sides they have:

```
let d = [("rectangle", 4); ("nonagon", 9); ("icosagon", 20)]
```

```
val d : (string * int) list =
  [("rectangle", 4); ("nonagon", 9); ("icosagon", 20)]
```

Note that an association list isn't so much a built-in data type in OCaml as a combination of two other types: lists and pairs.

Here are two functions that implement insertion and lookup in an association list:

```
(** [insert k v lst] is an association list that binds key [k] to value [v]
    and otherwise is the same as [lst] *)
let insert k v lst = (k, v) :: lst

(** [find k lst] is [Some v] if association list [lst] binds key [k] to
    value [v]; and is [None] if [lst] does not bind [k]. *)
let rec lookup k = function
| [] -> None
| (k', v) :: t -> if k = k' then Some v else lookup k t
```

```
val insert : 'a -> 'b -> ('a * 'b) list -> ('a * 'b) list = <fun>
```

```
val lookup : 'a -> ('a * 'b) list -> 'b option = <fun>
```

The insert function simply adds a new map from a key to a value at the front of the list. It doesn't bother to check whether the key is already in the list. The lookup function looks through the list from left to right. So if there did happen to be multiple maps for a given key in the list, only the most recently inserted one would be returned.

Insertion in an association list is therefore constant time, and lookup is linear time. Although there are certainly more efficient implementations of dictionaries—and we'll study some later in this course—association lists are a very easy and useful implementation for small dictionaries that aren't performance critical. The OCaml standard library has functions for association lists in the List module; look for List.assoc and the functions below it in the documentation. What we just wrote as lookup is actually already defined as List.assoc_opt. There is no pre-defined insert function in the library because it's so trivial just to cons a pair on.

3.9 Algebraic Data Types

Thus far, we have seen variants simply as enumerating a set of constant values, such as:

```
type day = Sun | Mon | Tue | Wed | Thu | Fri | Sat

type ptype = TNormal | TFire | TWater

type peff = ENormal | ENotVery | Esuper
```

But variants are far more powerful that this.

3.9.1 Variants that Carry Data

As a running example, here is a variant type shape that does more than just enumerate values:

```
type point = float * float
type shape =
    | Point of point
    | Circle of point * float (* center and radius *)
    | Rect of point * point (* lower-left and upper-right corners *)
```

```
type point = float * float
```

```
type shape = Point of point | Circle of point * float | Rect of point * point
```

This type, shape, represents a shape that is either a point, a circle, or a rectangle. A point is represented by a constructor Point that *carries* some additional data, which is a value of type point. A circle is represented by a constructor Circle that carries a pair of type point * float, which according to the comment represents the center of the circle and its radius. A rectangle is represented by a constructor Rect that carries a pair of type point *point.

Here are a couple functions that use the shape type:

```
let area = function
    | Point _ -> 0.0
    | Circle (_, r) -> Float.pi *. (r ** 2.0)
    | Rect ((x1, y1), (x2, y2)) ->
```

(continues on next page)

(continued from previous page)

```
let w = x2 -. x1 in
let h = y2 -. y1 in
w *. h

let center = function
| Point p -> p
| Circle (p, _) -> p
| Rect ((x1, y1), (x2, y2)) -> ((x2 +. x1) /. 2.0, (y2 +. y1) /. 2.0)
```

```
val area : shape -> float = <fun>
```

```
val center : shape -> point = <fun>
```

The shape variant type is the same as those we've seen before in that it is defined in terms of a collection of constructors. What's different than before is that those constructors carry additional data along with them. Every value of type shape is formed from exactly one of those constructors. Sometimes we call the constructor a *tag*, because it tags the data it carries as being from that particular constructor.

Variant types are sometimes called *tagged unions*. Every value of the type is from the set of values that is the union of all values from the underlying types that the constructor carries. For the shape type, every value is tagged with either Point or Circle or Rect and carries a value from the set of all point valued unioned with the set of all point * float values unioned with the set of all point * point values.

Another name for these variant types is an *algebraic data type*. "Algebra" here refers to the fact that variant types contain both sum and product types, as defined in the previous lecture. The sum types come from the fact that a value of a variant is formed by *one of* the constructors. The product types come from that fact that a constructor can carry tuples or records, whose values have a sub-value from *each of* their component types.

Using variants, we can express a type that represents the union of several other types, but in a type-safe way. Here, for example, is a type that represents either a string or an int:

```
type string_or_int =
    | String of string
    | Int of int
```

```
type string_or_int = String of string | Int of int
```

If we wanted to, we could use this type to code up lists (e.g.) that contain either strings or ints:

```
type string_or_int_list = string_or_int list
```

```
val sum : string_or_int list -> int = <fun>
```

```
val lst_sum : int = 3
```

Variants thus provide a type-safe way of doing something that might before have seemed impossible.

Variants also make it possible to discriminate which tag a value was constructed with, even if multiple constructors carry the same type. For example:

```
type t = Left of int | Right of int
```

```
val x : t = Left 1
```

```
val double_right : t -> int = <fun>
```

3.9.2 Syntax and Semantics

Syntax.

To define a variant type:

```
type t = C1 [of t1] | ... | Cn [of tn]
```

The square brackets above denote that of ti is optional. Every constructor may individually either carry no data or carry data. We call constructors that carry no data *constant*; and those that carry data, *non-constant*.

To write an expression that is a variant:

```
Се
```

Or:

```
С
```

depending on whether the constructor name C is non-constant or constant.

Dynamic semantics.

- If e==>v then C = ==> C v, assuming C is non-constant.
- C is already a value, assuming C is constant.

Static semantics.

```
• If t = \ldots \mid C \mid \ldots then C : t.
• If t = \ldots \mid C of t' \mid \ldots and if e : t' then C e : t.
```

Pattern matching.

We add the following new pattern form to the list of legal patterns:

• Ср

And we extend the definition of when a pattern matches a value and produces a binding as follows:

• If p matches v and produces bindings b, then C p matches C v and produces bindings b.

3.9.3 Catch-all Cases

One thing to beware of when pattern matching against variants is what *Real World OCaml* calls "catch-all cases". Here's a simple example of what can go wrong. Let's suppose you write this variant and function:

```
type color = Blue | Red
```

```
val string_of_color : color -> string = <fun>
```

Seems fine, right? But then one day you realize there are more colors in the world. You need to represent green. So you go back and add green to your variant:

```
type color = Blue | Red | Green
```

```
val string_of_color : color -> string = <fun>
```

But because of the thousand lines of code in between, you forget that string_of_color needs updating. And now, all the sudden, you are red-green color blind:

```
string_of_color Green
```

```
- : string = "red"
```

The problem is the *catch-all* case in the pattern match inside string_of_color: the final case that uses the wildcard pattern to match anything. Such code is not robust against future changes to the variant type.

If, instead, you had originally coded the function as follows, life would be better:

```
let string_of_color = function
    | Blue -> "blue"
    | Red -> "red"
```

```
File "[9]", lines 1-3, characters 22-17:
```

```
1 | .....function
```

```
2 | | Blue -> "blue"
```

```
3 | | Red -> "red"
```

```
Warning 8: this pattern-matching is not exhaustive.
```

```
Here is an example of a case that is not matched:
```

```
Green
```

```
val string_of_color : color -> string = <fun>
```

The OCaml type checker now alerts you that you haven't yet updated string_of_color to account for the new constructor

The moral of the story is: catch-all cases lead to buggy code. Avoid using them.

3.9.4 Recursive Variants

Variant types may mention their own name inside their own body. For example, here is a variant type that could be used to represent something similar to int list:

```
type intlist = Nil | Cons of int * intlist
```

```
val lst3 : intlist = Cons (3, Nil)
```

```
val lst123 : intlist = Cons (1, Cons (2, Cons (3, Nil)))
```

```
val sum : intlist -> int = <fun>
```

```
val length : intlist -> int = <fun>
```

```
val empty : intlist -> bool = <fun>
```

Notice that in the definition of intlist, we define the Cons constructor to carry a value that contains an intlist. This makes the type intlist be *recursive*: it is defined in terms of itself.

Types may be mutually recursive if you use the and keyword:

```
type node = {value : int; next : mylist}
and mylist = Nil | Node of node
```

```
type node = { value : int; next : mylist; }
and mylist = Nil | Node of node
```

Any such mutual recursion must involve at least one variant or record type that the recursion "goes through". For example, the following is not allowed:

```
type t = u and u = t
```

But this is:

```
type t = U of u and u = T of t
```

```
\begin{array}{l} \text{type t} = \text{U of u} \\ \text{and u} = \text{T of t} \end{array}
```

Record types may also be recursive:

```
type node = {value : int; next : node}
```

```
type node = { value : int; next : node; }
```

But plain old type synonyms may not be:

```
type t = t * t
```

Although node is a legal type definition, there is no way to construct a value of that type because of the circularity involved: to construct the very first node value in existence, you would already need a value of type node to exist. Later, when we cover imperative features, we'll see a similar idea used (but successfully) for mutable linked lists.

3.9.5 Parameterized Variants

Variant types may be *parameterized* on other types. For example, the intlist type above could be generalized to provide lists (coded up ourselves) over any type:

```
type 'a mylist = Nil | Cons of 'a * 'a mylist

let lst3 = Cons (3, Nil) (* similar to [3] *)
let lst_hi = Cons ("hi", Nil) (* similar to ["hi"] *)
```

```
type 'a mylist = Nil | Cons of 'a * 'a mylist
```

```
val lst3 : int mylist = Cons (3, Nil)
```

```
val lst_hi : string mylist = Cons ("hi", Nil)
```

Here, mylist is a *type constructor* but not a type: there is no way to write a value of type mylist. But we can write value of type int mylist (e.g., lst3) and string mylist (e.g., lst_hi). Think of a type constructor as being like a function, but one that maps types to types, rather than values to value.

Here are some functions over 'a mylist:

```
let rec length : 'a mylist -> int = function
    | Nil -> 0
    | Cons (_, t) -> 1 + length t

let empty : 'a mylist -> bool = function
    | Nil -> true
    | Cons _ -> false
```

```
val length : 'a mylist -> int = <fun>
```

```
val empty : 'a mylist -> bool = <fun>
```

Notice that the body of each function is unchanged from its previous definition for intlist. All that we changed was the type annotation. And that could even be omitted safely:

```
let rec length = function
    | Nil -> 0
    | Cons (_, t) -> 1 + length t

let empty = function
    | Nil -> true
    | Cons _ -> false
```

```
val length : 'a mylist -> int = <fun>
```

```
val empty : 'a mylist -> bool = <fun>
```

The functions we just wrote are an example of a language feature called **parametric polymorphism**. The functions don't care what the 'a is in 'a mylist, hence they are perfectly happy to work on int mylist or string mylist or any other (whatever) mylist. The word "polymorphism" is based on the Greek roots "poly" (many) and "morph" (form). A value of type 'a mylist could have many forms, depending on the actual type 'a.

As soon, though, as you place a constraint on what the type 'a might be, you give up some polymorphism. For example,

```
let rec sum = function
   | Nil -> 0
   | Cons (h, t) -> h + sum t
```

```
val sum : int mylist -> int = <fun>
```

The fact that we use the (+) operator with the head of the list constrains that head element to be an int, hence all elements must be int. That means sum must take in an int mylist, not any other kind of 'a mylist.

It is also possible to have multiple type parameters for a parameterized type, in which case parentheses are needed:

```
type ('a, 'b) pair = {first : 'a; second : 'b}
let x = {first = 2; second = "hello"}
```

```
type ('a, 'b) pair = { first : 'a; second : 'b; }
```

```
val x : (int, string) pair = {first = 2; second = "hello"}
```

3.9.6 Polymorphic Variants

Thus far, whenever you've wanted to define a variant type, you have had to give it a name, such as day, shape, or 'a mylist:

```
type day = Sun | Mon | Tue | Wed | Thu | Fri | Sat
```

```
type shape = Point of point | Circle of point * float | Rect of point * point
```

```
type 'a mylist = Nil | Cons of 'a * 'a mylist
```

Occasionally, you might need a variant type only for the return value of a single function. For example, here's a function f that can either return an int or ∞ ; you are forced to define a variant type to represent that result:

```
type fin_or_inf = Finite of int | Infinity
```

```
val f : int -> fin_or_inf = <fun>
```

The downside of this definition is that you were forced to defined fin_or_inf even though it won't be used throughout much of your program.

There's another kind of variant in OCaml that supports this kind of programming: *polymorphic variants*. Polymorphic variants are just like variants, except:

- 1. You don't have declare their type or constructors before using them.
- 2. There is no name for a polymorphic variant type. (So another name for this feature could have been "anonymous variants".)
- 3. The constructors of a polymorphic variant start with a backquote character.

Using polymorphic variants, we can rewrite f:

```
val f : int -> [> `Finite of int | `Infinity ] = <fun>
```

This type says that f either returns `Finite n for some n: int or `Infinity. The square brackets do not denote a list, but rather a set of possible constructors. The > sign means that any code that pattern matches against a value of that type must *at least* handle the constructors `Finite and `Infinity, and possibly more. For example, we could write:

```
- : string = "finite"
```

It's perfectly fine for the pattern match to include constructors other than `Finite or `Infinity, because f is guaranteed never to return any constructors other than those.

There are other, more compelling uses for polymorphic variants that we'll see later in the course. They are particularly useful in libraries. For now, we generally will steer you away from extensive use of polymorphic variants, because their types can become difficult to manage.

3.9.7 Built-in Variants

OCami's built-in list data type is really a recursive, parameterized variant. It is defined as follows:

```
type 'a list = [] | ( :: ) of 'a * 'a list
```

So list is really just a type constructor, with (value) constructors [] (which we pronounce "nil") and :: (which we pronounce "cons").

OCaml's built-in option data type is also really a parameterized variant. It's defined as follows:

```
type 'a option = None | Some of 'a
```

So option is really just a type constructor, with (value) constructors None and Some.

You can see both list and option defined in the core OCaml library.

3.10 Exceptions

OCaml has an exception mechanism similar to many other programming languages. A new type of OCaml exception is defined with this syntax:

```
exception E of t
```

where E is a constructor name and t is a type. The of t is optional. Notice how this is similar to defining a constructor of a variant type. For example:

3.10. Exceptions 93

```
exception A
exception B
exception Code of int
exception Details of string
```

```
exception A
```

```
exception B
```

```
exception Code of int
```

```
exception Details of string
```

To create an exception value, use the same syntax you would for creating a variant value. Here, for example, is an exception value whose constructor is Failure, which carries a string:

```
Failure "something went wrong"
```

```
- : exn = Failure "something went wrong"
```

This constructor is pre-defined in the standard library and is one of the more common exceptions that OCaml programmers use.

To raise an exception value e, simply write

```
raise e
```

There is a convenient function failwith: string -> 'a in the standard library that raises Failure. That is, failwith s is equivalent to raise (Failure s).

To catch an exception, use this syntax:

```
try e with
| p1 -> e1
| ...
| pn -> en
```

The expression e is what might raise an exception. If it does not, the entire try expression evaluates to whatever e does. If e does raise an exception value v, that value v is that matched against the provide patterns, exactly like match expression.

3.10.1 Exceptions are Extensible Variants

All exception values have type exn, which is a variant defined in the core. It's an unusual kind of variant, though, called an *extensible* variant, which allows new constructors of the variant to be defined after the variant type itself is defined. See the OCaml manual for more information about extensible variants if you're interested.

3.10.2 Exception Semantics

Since they are just variants, the syntax and semantics of exceptions is already covered by the syntax and semantics of variants—with one exception (pun intended), which is the dynamic semantics of how exceptions are raised and handled.

Dynamic semantics. As we originally said, every OCaml expression either

- · evaluates to a value
- · raises an exception
- or fails to terminate (i.e., an "infinite loop").

So far we've only presented the part of the dynamic semantics that handles the first of those three cases. What happens when we add exceptions? Now, evaluation of an expression either produces a value or produces an *exception packet*. Packets are not normal OCaml values; the only pieces of the language that recognizes them are raise and try. The exception value produced by (e.g.) Failure "oops" is part of the exception packet produced by raise (Failure "oops"), but the packet contains more than just the exception value; there can also be a stack trace, for example.

For any expression e other than try, if evaluation of a subexpression of e produces an exception packet P, then evaluation of e produces packet P.

But now we run into a problem for the first time: what order are subexpressions evaluated in? Sometimes the answer to that question is provided by the semantics we have already developed. For example, with let expressions, we know that the binding expression must be evaluated before the body expression. So the following code raises A:

```
let _ = raise A in raise B;;
```

```
Exception: A. Called from Toploop.load_lambda in file "toplevel/toploop.ml", line 212, characters_ +17-27
```

And with functions, the argument must be evaluated before the function. So the following code also raises A, in addition to producing some compiler warnings that the first expression will never actually be applied as a function to an argument:

```
(raise B) (raise A)
```

It makes sense that both those pieces of code would raise the same exception, given that we know let x = e1 in e2 is syntactic sugar for (fun $x \rightarrow e2$) e1.

But what does the following code raise as an exception?

```
(raise A, raise B)
```

The answer is nuanced. The language specification does not stipulate what order the components of pairs should be evaluated in. Nor did our semantics exactly determine the order. (Though you would be forgiven if you thought it was left to right.) So programmers actually cannot rely on that order. The current implementation of OCaml, as it turns out, evaluates right to left. So the code above actually raises B. If you really want to force the evaluation order, you need to use let expressions:

```
let a = raise A in
let b = raise B in
(a, b)
```

```
Exception: A.

Called from Toploop.load_lambda in file "toplevel/toploop.ml", line 212, characters_ 417-27
```

3.10. Exceptions 95

That code is guaranteed to raise A rather than B.

One interesting corner case is what happens when a raise expression itself has a subexpression that raises:

```
exception C of string;;
exception D of string;;
raise (C (raise (D "oops")))
```

```
exception C of string
```

```
exception D of string
```

```
Exception: D "oops".

Called from Toploop.load_lambda in file "toplevel/toploop.ml", line 212, characters_417-27
```

That code ends up raising D, because the first thing that has to happen is to evaluate C (raise (D "oops")) to a value. Doing that requires evaluating raise (D "oops") to a value. Doing that causes a packet containing D "oops" to be produced, and that packet then propagates and becomes the result of evaluating C (raise (D "oops")), hence the result of evaluating raise (C (raise (D "oops"))).

Once evaluation of an expression produces an exception packet P, that packet propagates until it reaches a try expression:

```
try e with
| p1 -> e1
| ...
| pn -> en
```

The exception value inside P is matched against the provided patterns using the usual evaluation rules for pattern matching—with one exception (again, pun intended). If none of the patterns matches, then instead of producing Match_failure inside a new exception packet, the original exception packet P continues propagating until the next try expression is reached.

3.10.3 Pattern Matching

There is a pattern form for exceptions. Here's an example of its usage:

```
- : string = "hd"
```

Note that the code is above is just a standard match expression, not a try expression. It matches the value of List. hd [] against the three provided patterns. As we know, List.hd [] will raise an exception containing the value Failure "hd". The exception pattern exception (Failure s) matches that value. So the above code will evaluate to "hd".

In general, exception patterns are a kind of syntactic sugar. Consider this code:

Some of the patterns p1..pn could be exception patterns of the form exception q. Let q1..qn be that subsequence of patterns (without the exception keyword), and let r1..rm be the subsequence of non-exception patterns. Then we can rewrite the code as:

```
match
    try e with
        | q1 -> e1
        | ...
        | qn -> en

with
        | r1 -> e1
        | ...
        | rm -> em
```

Which is to say: try evaluating e. If it produces an exception packet, use the exception patterns from the original match expression to handle that packet. If it doesn't produce an exception packet but instead produces a non-exception value, use the non-exception patterns from the original match expression to match that value.

3.10.4 Exceptions and OUnit

If it is part of a function's specification that it raises an exception, you might want to write OUnit tests that check whether the function correctly does so. Here's how to do that:

```
open OUnit2
let tests = "suite" >::: [
    "empty" >:: (fun _ -> assert_raises (Failure "hd") (fun () -> List.hd []));
    ]
let _ = run_test_tt_main tests
```

The expression assert_raises exc (fun () -> e) checks to see whether expression e raises exception exc. If so, the OUnit test case succeeds, otherwise it fails.

Tip: A common error is to forget the (fun () -> ...) around e. If you do, the OUnit test case will fail, and you will likely be confused as to why. The reason is that, without the extra anonymous function, the exception is raised before assert_raises ever gets a chance to handle it.

3.11 Example: Trees

Trees are a very useful data structure. A *binary tree*, as you'll recall from CS 2110, is a node containing a value and two children that are trees. A binary tree can also be an empty tree, which we also use to represent the absence of a child node.

3.11.1 Representation with Tuples

Here is a definition for a binary tree data type:

```
type 'a tree =
| Leaf
| Node of 'a * 'a tree * 'a tree
```

```
type 'a tree = Leaf | Node of 'a * 'a tree * 'a tree
```

A node carries a data item of type 'a and has a left and right subtree. A leaf is empty. Compare this definition to the definition of a list and notice how similar their structure is:

```
      type 'a tree =
      type 'a mylist =

      | Leaf
      | Nil

      | Node of 'a * 'a tree * 'a tree
      | Cons of 'a * 'a mylist
```

The only essential difference is that Cons carries one sublist, whereas Node carries two subtrees.

Here is code that constructs a small tree:

```
val t : int tree =
  Node (4, Node (2, Node (1, Leaf, Leaf), Node (3, Leaf, Leaf)),
  Node (5, Node (6, Leaf, Leaf), Node (7, Leaf, Leaf)))
```

The *size* of a tree is the number of nodes in it (that is, Nodes, not Leafs). For example, the size of tree t above is 7. Here is a function function size: 'a tree -> int that returns the number of nodes in a tree:

3.11.2 Representation with Records

Next, let's revise our tree type to use use a record type to represent a tree node. In OCaml we have to define two mutually recursive types, one to represent a tree node, and one to represent a (possibly empty) tree:

```
type 'a tree =
    | Leaf
    | Node of 'a node

and 'a node = {
    value: 'a;
    left: 'a tree;
    right: 'a tree
}
```

```
type 'a tree = Leaf | Node of 'a node
and 'a node = { value : 'a; left : 'a tree; right : 'a tree; }
```

Here's an example tree:

We can use pattern matching to write the usual algorithms for recursively traversing trees. For example, here is a recursive search over the tree:

```
(** [mem x t] is whether [x] is a value at some node in tree [t]. *)
let rec mem x = function
    | Leaf -> false
    | Node {value; left; right} -> value = x || mem x left || mem x right
```

```
val mem : 'a -> 'a tree -> bool = <fun>
```

The function name mem is short for "member"; the standard library often uses a function of this name to implement a search through a collection data structure to determine whether some element is a member of that collection.

Here's a function that computes the *preorder* traversal of a tree, in which each node is visited before any of its children, by constructing a list in which the values occur in the order in which they would be visited:

```
val preorder : 'a tree -> 'a list = <fun>
```

```
preorder t
```

```
- : int list = [2; 1; 3]
```

Although the algorithm is beautifully clear from the code above, it takes quadratic time on unbalanced trees because of the @ operator. That problem can be solved by introducing an extra argument acc to accumulate the values at each node, though at the expense of making the code less clear:

```
val preorder_lin : 'a tree -> 'a list = <fun>
```

The version above uses exactly one :: operation per Node in the tree, making it linear time.

3.12 Example: Natural Numbers

We can define a recursive variant that acts like numbers, demonstrating that we don't really have to have numbers built into OCaml! (For sake of efficiency, though, it's a good thing they are.)

A *natural number* is either *zero* or the *successor* of some other natural number. This is how you might define the natural numbers in a mathematical logic course, and it leads naturally to the following OCaml type nat:

```
type nat = Zero | Succ of nat
```

```
type nat = Zero | Succ of nat
```

We have defined a new type nat, and Zero and Succ are constructors for values of this type. This allows us to build expressions that have an arbitrary number of nested Succ constructors. Such values act like natural numbers:

```
let zero = Zero
let one = Succ zero
let two = Succ one
let three = Succ two
let four = Succ three
```

```
val zero : nat = Zero
```

```
val one : nat = Succ Zero
```

```
val two : nat = Succ (Succ Zero)
```

```
val three : nat = Succ (Succ (Succ Zero))
```

Now we can write functions to manipulate values of this type. We'll write a lot of type annotations in the code below to help the reader keep track of which values are nat versus int; the compiler, of course, doesn't need our help.

```
let iszero = function
    | Zero -> true
    | Succ _ -> false

let pred = function
    | Zero -> failwith "pred Zero is undefined"
    | Succ m -> m
```

```
val iszero : nat -> bool = <fun>
```

```
val pred : nat -> nat = <fun>
```

Similarly we can define a function to add two numbers:

```
let rec add n1 n2 =
  match n1 with
  | Zero -> n2
  | Succ pred_n -> add pred_n (Succ n2)
```

```
val add : nat -> nat -> nat = <fun>
```

We can convert nat values to type int and vice-versa:

```
let rec int_of_nat = function
  | Zero -> 0
  | Succ m -> 1 + int_of_nat m

let rec nat_of_int = function
  | i when i = 0 -> Zero
  | i when i > 0 -> Succ (nat_of_int (i - 1))
  | _ -> failwith "nat_of_int is undefined on negative ints"
```

```
val int_of_nat : nat -> int = <fun>
```

```
val nat_of_int : int -> nat = <fun>
```

To determine whether a natural number is even or odd, we can write a pair of mutually recursive functions:

```
let rec even = function Zero -> true | Succ m -> odd m
and odd = function Zero -> false | Succ m -> even m
```

```
val even : nat -> bool = <fun>
val odd : nat -> bool = <fun>
```

3.13 Summary

Lists are a highly useful built-in data structure in OCaml. The language provides a lightweight syntax for building them, rather than requiring you to use a library. Accessing parts of a list makes use of pattern matching, a very powerful feature (as you might expect from its rather lengthy semantics). We'll see more uses for pattern matching as the course proceeds.

These built-in lists are implemented as singly-linked lists. That's important to keep in mind when your needs go beyond small to medium sized lists. Recursive functions on long lists will take up a lot of stack space, so tail recursion becomes important. And if you're attempting to process really huge lists, you probably don't want linked lists at all, but instead a data structure that will do a better job of exploiting memory locality.

OCaml provides data types for variants (one-of types), tuples and products (each-of types), and options (maybe types). Pattern matching can be used to access values of each of those data types. And pattern matching can be used in let expressions and functions.

Association lists combine lists and tuples to create a lightweight implementation of dictionaries.

Variants are a powerful language feature. They are the workhorse of representing data in a functional language. OCaml variants actually combine several theoretically independent language features into one: sum types, product types, recursive types, and parameterized (polymorphic) types. The result is an ability to express many kinds of data, including lists, options, trees, and even exceptions.

3.13.1 Terms and Concepts

- algebraic data type
- append
- · association list
- · binary trees as variants
- binding
- branch
- · carried data
- · catch-all cases
- cons
- · constant constructor
- constructor
- copying
- desugaring
- · each-of type
- · exception
- exception as variants
- exception packet
- · exception pattern
- · exception value
- exhaustiveness

- field
- head
- induction
- leaf
- list
- lists as variants
- maybe type
- mutually recursive functions
- natural numbers as variants
- nil
- node
- non-constant constructor
- one-of type
- options
- · options as variants
- order of evaluation
- pair
- parameterized variant
- parametric polymorphism
- · pattern matching
- prepend
- product type
- record
- recursion
- · recursive variant
- sharing
- · stack frame
- sum type
- syntactic sugar
- tag
- tail
- tail call
- tail recursion
- test-driven development (TDD)
- triple
- tuple

3.13. Summary 103

- · type constructor
- · type constructor
- type synonym
- · variant
- · wildcard

3.13.2 Further Reading

- Introduction to Objective Caml, chapters 4, 5.2, 5.3, 5.4, 6, 7, 8.1
- OCaml from the Very Beginning, chapters 3, 4, 5, 7, 8, 10, 11
- Real World OCaml, chapter 3, 5, 6, 7

3.14 Exercises

Solutions to exercises are available to students in Cornell's CS 3110. Instructors at other institutions are welcome to contact Michael Clarkson for access.

Exercise: list expressions [★]

- Construct a list that has the integers 1 through 5 in it. Use the square bracket notation for lists.
- Construct the same list, but do not use the square bracket notation. Instead use :: and [].
- Construct the same list again. This time, the following expression must appear in your answer: [2; 3; 4]. Use the @ operator, and do not use ::.

Exercise: product [★★]

Write a function that returns the product of all the elements in a list. The product of all the elements of an empty list is 1.

Exercise: concat $[\star \star]$

Write a function that concatenates all the strings in a list. The concatenation of all the strings in an empty list is the empty string "".

Exercise: product test [★★]

Unit test the function product that you wrote in an exercise above.

Exercise: patterns $[\star \star \star]$

Using pattern matching, write three functions, one for each of the following properties. Your functions should return true if the input list has the property and false otherwise.

- the list's first element is "bigred"
- the list has exactly two or four elements; do not use the length function

• the first two elements of the list are equal

Exercise: library $[\star \star \star]$

Consult the List standard library to solve these exercises:

- Write a function that takes an int list and returns the fifth element of that list, if such an element exists. If the list has fewer than five elements, return 0. *Hint:* List.length and List.nth.
- Write a function that takes an int list and returns the list sorted in descending order. *Hint:* List.sort with Stdlib.compare as its first argument, and List.rev.

Exercise: library test $[\star \star \star]$

Write a couple OUnit unit tests for each of the functions you wrote in the previous exercise.

Exercise: library puzzle $[\star\star\star]$

- Write a function that returns the last element of a list. Your function may assume that the list is non-empty. *Hint: Use two library functions, and do not write any pattern matching code of your own.*
- Write a function any_zeroes : int list -> bool that returns true if and only if the input list contains at least one 0. *Hint: use one library function, and do not write any pattern matching code of your own.*

Your solutions will be only one or two lines of code each.

Exercise: take drop $[\star \star \star]$

- Write a function take : int -> 'a list -> 'a list such that take n lst returns the first n elements of lst. If lst has fewer than n elements, return all of them.
- Write a function drop: int -> 'a list -> 'a list such that drop n lst returns all but the first n elements of lst. If lst has fewer than n elements, return the empty list.

Exercise: take drop tail $[\star \star \star \star]$

Revise your solutions for take and drop to be tail recursive, if they aren't already. Test them on long lists with large values of n to see whether they run out of stack space. To construct long lists, use the -- operator from the *lists* section.

Exercise: unimodal $[\star\star\star]$

Write a function <code>is_unimodal</code>: <code>int list -> bool</code> that takes an integer list and returns whether that list is unimodal. A *unimodal list* is a list that monotonically increases to some maximum value then monotonically decreases after that value. Either or both segments (increasing or decreasing) may be empty. A constant list is unimodal, as is the empty list.

Exercise: powerset $[\star \star \star]$

Write a function powerset: int list -> int list list that takes a set *S* represented as a list and returns the set of all subsets of *S*. The order of subsets in the powerset and the order of elements in the subsets do not matter.

3.14. Exercises 105

Hint: Consider the recursive structure of this problem. Suppose you already have p, such that p = powerset s. How could you use p to compute powerset (x :: s)?

Exercise: print int list rec $[\star \star]$

Write a function print_int_list : int list -> unit that prints its input list, one number per line. For example, print_int_list [1; 2; 3] should result in this output:

```
1
2
3
```

Here is some code to get you started:

```
let rec print_int_list = function
| [] -> ()
| h :: t -> (* fill in here *); print_int_list t
```

Exercise: print int list iter $[\star \star]$

Write a function print_int_list': int list -> unit whose specification is the same as print_int_list. Do not use the keyword rec in your solution, but instead to use the List module function List. iter. Here is some code to get you started:

```
let print_int_list' lst =
   List.iter (fun x -> (* fill in here *)) lst
```

Exercise: student $[\star \star]$

Assume the following type definition:

```
type student = {first_name : string; last_name : string; gpa : float}
```

Give OCaml expressions that have the following types:

- student
- student -> string * string (a function that extracts the student's name)
- string -> string -> float -> student (a function that creates a student record)

Exercise: pokerecord [★★]

Here is a variant that represents a few Pokémon types:

```
type poketype = Normal | Fire | Water
```

- Define the type pokemon to be a record with fields name (a string), hp (an integer), and ptype (a poketype).
- Create a record named charizard of type pokemon that represents a Pokémon with 78 HP and Fire type.
- Create a record named squirtle of type pokemon that represents a Pokémon with 44 HP and Water type.

Exercise: safe hd and tl $[\star \star]$

Write a function $safe_hd$: 'a list -> 'a option that returns Some x if the head of the input list is x, and None if the input list is empty.

Also write a function safe_tl : 'a list -> 'a list option that returns the tail of the list, or None if the list is empty.

Exercise: pokefun $[\star \star \star]$

Write a function max_hp : pokemon list -> pokemon option that, given a list of pokemon, finds the Pokémon with the highest HP.

Exercise: date before [★★]

Define a *date-like triple* to be a value of type int * int * int. Examples of date-like triples include (2013, 2, 1) and (0, 0, 1000). A *date* is a date-like triple whose first part is a positive year (i.e., a year in the common era), second part is a month between 1 and 12, and third part is a day between 1 and 31 (or 30, 29, or 28, depending on the month and year). (2013, 2, 1) is a date; (0, 0, 1000) is not.

Write a function is_before that takes two dates as input and evaluates to true or false. It evaluates to true if the first argument is a date that comes before the second argument. (If the two dates are the same, the result is false.)

Your function needs to work correctly only for dates, not for arbitrary date-like triples. However, you will probably find it easier to write your solution if you think about making it work for arbitrary date-like triples. For example, it's easier to forget about whether the input is truly a date, and simply write a function that claims (for example) that January 100, 2013 comes before February 34, 2013—because any date in January comes before any date in February, but a function that says that January 100, 2013 comes after February 34, 2013 is also valid. You may ignore leap years.

Exercise: earliest date $[\star\star\star]$

Write a function earliest: (int*int*int) list -> (int * int * int) option. It evaluates to None if the input list is empty, and to Some d if date d is the earliest date in the list. *Hint: use is_before*.

As in the previous exercise, your function needs to work correctly only for dates, not for arbitrary date-like triples.

Exercise: assoc list [★]

Use the functions insert and lookup from the *section on association lists* to construct an association list that maps the integer 1 to the string "one", 2 to "two", and 3 to "three". Lookup the key 2. Lookup the key 4.

Exercise: cards [★★]

- Define a variant type suit that represents the four suits, ♣ ♦ ♥ ♠, in a standard 52-card deck. All the constructors
 of your type should be constant.
- Define a type rank that represents the possible ranks of a card: 2, 3, ..., 10, Jack, Queen, King, or Ace. There are many possible solutions; you are free to choose whatever works for you. One is to make rank be a synonym of int, and to assume that Jack=11, Queen=12, King=13, and Ace=1 or 14. Another is to use variants.
- Define a type card that represents the suit and rank of a single card. Make it a record with two fields.

3.14. Exercises 107

• Define a few values of type card: the Ace of Clubs, the Queen of Hearts, the Two of Diamonds, the Seven of Spades.

Exercise: matching [★]

For each pattern in the list below, give a value of type int option list that does *not* match the pattern and is not the empty list, or explain why that's impossible.

```
Some x :: tl
[Some 3110; None]
[Some x; _]
h1 :: h2 :: tl
h :: tl
```

Exercise: quadrant [★★]

Complete the quadrant function below, which should return the quadrant of the given x, y point according to the diagram on the right (borrowed from Wikipedia). Points that lie on an axis do not belong to any quandrant. *Hints:* (a) define a helper function for the sign of an integer, (b) match against a pair.

```
type quad = I | II | III | IV
type sign = Neg | Zero | Pos

let sign (x:int) : sign =
    ...

let quadrant : int*int -> quad option = fun (x,y) ->
    match ... with
    | ... -> Some I
    | ... -> Some III
    | ... -> Some III
    | ... -> Some IV
    | ... -> None
```

Exercise: quadrant when $[\star \star]$

Rewrite the quadrant function to use the when syntax. You won't need your helper function from before.

Exercise: depth [★★]

Write a function depth: 'a tree \rightarrow int that returns the number of nodes in any longest path from the root to a leaf. For example, the depth of an empty tree (simply Leaf) is 0, and the depth of tree t above is 3. *Hint: there is a library function max*: 'a \rightarrow 'a that returns the maximum of any two values of the same type.

Exercise: shape $[\star\star\star]$

Write a function same_shape: 'a tree -> 'b tree -> bool that determines whether two trees have the same shape, regardless of whether the values they carry at each node are the same. *Hint: use a pattern match with three branches, where the expression being matched is a pair of trees.*

Exercise: list max exn $[\star \star]$

Write a function list_max: int list -> int that returns the maximum integer in a list, or raises Failure "list_max" if the list is empty.

Exercise: list max exn string $[\star \star]$

Write a function list_max_string: int list -> string that returns a string containing the maximum integer in a list, or the string "empty" (note, not the exception Failure "empty" but just the string "empty") if the list is empty. Hint: string_of_int in the standard library will do what its name suggests.

Exercise: list max exn ounit [★]

Write two OUnit tests to determine whether your solution to **list max exn**, above, correctly raises an exception when its input is the empty list, and whether it correctly returns the max value of the input list when that list is nonempty.

Exercise: is_bst [$\star\star\star\star$]

Write a function is_bst: ('a*'b) tree -> bool that returns true if and only if the given tree satisfies the binary search tree invariant. An efficient version of this function that visits each node at most once is somewhat tricky to write. Hint: write a recursive helper function that takes a tree and either gives you (i) the minimum and maximum value in the tree, or (ii) tells you that the tree is empty, or (iii) tells you that the tree does not satisfy the invariant. Your is_bst function will not be recursive, but will call your helper function and pattern match on the result. You will need to define a new variant type for the return type of your helper function.

Exercise: quadrant poly [★★]

Modify your definition of quadrant to use polymorphic variants. The types of your functions should become these:

```
val sign : int -> [> `Neg | `Pos | `Zero ]
val quadrant : int * int -> [> `I | `II | `IV ] option
```

3.14. Exercises 109

CHAPTER

FOUR

HIGHER-ORDER PROGRAMMING

Functions are values just like any other value in OCaml. What does that mean exactly? This means that we can pass functions around as arguments to other functions, that we can store functions in data structures, that we can return functions as a result from other functions, and so forth.

Higher-order functions either take other functions as input or return other functions as output (or both). Higher-order functions are also known as *functionals*, and programming with them could therefore be called *functional programming*—indicating what the heart of programming in languages like OCaml is all about.

Higher-order functions were one of the more recent adoptions from functional languages into mainstream languages. The Java 8 Streams library and Python 2.3's itertools modules are examples of that; C++ has also been increasing its support since at least 2011.

Note: C wizards might object the adoption isn't so recent. After all, C has long had the ability to do higher-order programming through function pointers. But that ability also depends on the programming pattern of passing an additional *environment* parameter to provide the values of variables in the function to be called through the pointer. As we'll see in our later chapter on interpreters, the essence of (higher-order) functions in a functional language is that they are really something called a *closure* that obviates the need for that extra parameter. Bear in mind that the issue is not what is *possible* to compute in a language—after all everything is eventually compiled down to machine code, so we could just write in that exclusively—but what is *pleasant* to compute.

In this chapter we will see what all the fuss is about. Higher-order functions enable beautiful, general, reusable code.

4.1 Higher-Order Functions

Consider these functions double and square on integers:

```
let double x = 2 * x
let square x = x * x
```

```
val double : int -> int = <fun>
```

```
val square : int -> int = <fun>
```

Let's use these functions to write other functions that quadruple and raise a number to the fourth power:

```
let quad x = double (double x)
let fourth x = square (square x)
```

```
val quad : int -> int = <fun>
```

```
val fourth : int -> int = <fun>
```

There is an obvious similarity between these two functions: what they do is apply a given function twice to a value. By passing in the function to another function twice as an argument, we can abstract this functionality:

```
let twice f x = f (f x)
```

```
val twice : ('a -> 'a) -> 'a -> 'a = <fun>
```

The function twice is higher-order: its input f is a function. And—recalling that all OCaml functions really take only a single argument—its output is technically fun $x \rightarrow f$ (f x), so twice returns a function hence is also higher-order in that way.

Using twice, we can implement quad and fourth in a uniform way:

```
let quad x = twice double x
let fourth x = twice square x
```

```
val quad : int -> int = <fun>
```

```
val fourth : int -> int = <fun>
```

4.1.1 The Abstraction Principle

Above, we have exploited the structural similarity between quad and fourth to save work. Admittedly, in this toy example it might not seem like much work. But imagine that twice were actually some much more complicated function. Then if someone comes up with a more efficient version of it, every function written in terms of it (like quad and fourth) could benefit from that improvement in efficiency, without needing to be recoded.

Part of being an excellent programmer is recognizing such similarities and *abstracting* them by creating functions (or other units of code) that implement them. Bruce MacLennan names this the **Abstraction Principle** in his textbook *Functional Programming: Theory and Practice* (1990). The Abstraction Principle says to avoid requiring something to be stated more than once; instead, *factor out* the recurring pattern. Higher-order functions enable such refactoring, because they allow us to factor out functions and parameterize functions on other functions.

Besides twice, here are some more relatively simple examples, indebted also to MacLennan:

Apply. We can write a function that applies its first input to its second input:

```
let apply f x = f x
```

```
val apply : ('a -> 'b) -> 'a -> 'b = <fun>
```

Of course, writing apply f is a lot more work than just writing f.

Pipeline. The pipeline operator, which we've previously seen, is a higher-order function:

```
let pipeline x f = f x
let (|>) = pipeline
let x = 5 |> double
```

```
val pipeline : 'a -> ('a -> 'b) -> 'b = <fun>
```

```
val ( |> ) : 'a -> ('a -> 'b) -> 'b = <fun>
```

```
val x : int = 10
```

Compose. We can write a function that composes two other functions:

```
let compose f g x = f (g x)
```

```
val compose : ('a -> 'b) -> ('c -> 'a) -> 'c -> 'b = <fun>
```

This function would let us create a new function that can be applied many times, such as the following:

```
let square_then_double = compose double square
let x = square_then_double 1
let y = square_then_double 2
```

```
val square_then_double : int -> int = <fun>
```

```
val x : int = 2
```

```
val y : int = 8
```

Both. We can write a function that applies two functions to the same argument and returns a pair of the result:

```
let both f g x = (f x, g x)
let ds = both double square
let p = ds 3
```

```
val both : ('a -> 'b) -> ('a -> 'c) -> 'a -> 'b * 'c = <fun>
```

```
val ds : int -> int * int = <fun>
```

```
val p : int * int = (6, 9)
```

Cond. We can write a function that conditionally chooses which of two functions to apply based on a predicate:

```
let cond p f g x =
  if p x then f x else g x
```

```
val cond : ('a -> bool) -> ('a -> 'b) -> ('a -> 'b) -> 'a -> 'b = <fun>
```

4.1.2 The Meaning of "Higher Order"

The phrase "higher order" is used throughout logic and computer science, though not necessarily with a precise or consistent meaning in all cases.

In logic, first-order quantification refers primarily to the universal and existential (\forall and \exists) quantifiers. These let you quantify over some domain of interest, such as the natural numbers. But for any given quantification, say $\forall x$, the variable being quantified represents an individual element of that domain, say the natural number 42.

Second-order quantification lets you do something strictly more powerful, which is to quantify over properties of the domain. Properties are assertions about individual elements, for example, that a natural number is even, or that it is prime. In some logics we can equate properties with sets of individual, for example the set of all even naturals. So second-order quantification is often thought of as quantification over sets. You can also think of properties as being functions that take in an element and return a Boolean indicating whether the element satisfies the property; this is called the *characteristic function* of the property.

Third-order logic would allow quantification over properties of properties, and *fourth-order* over properties of properties of properties, and so forth. *Higher-order logic* refers to all these logics that are more powerful than first-order logic; though one interesting result in this area is that all higher-order logics can be expressed in second-order logic.

In programming languages, *first-order functions* similarly refer to functions that operate on individual data elements (e.g., strings, ints, records, variants, etc.). Whereas *higher-order function* can operate on functions, much like higher-order logics can quantify over over properties (which are like functions).

4.1.3 Famous Higher-order Functions

In the next few sections we'll dive into three of the most famous higher-order functions: map, filter, and fold. These are functions that can be defined for many data structures, including lists and trees. The basic idea of each is that:

- map transforms elements,
- filter eliminates elements, and
- fold combines elements.

4.2 Map

Here are two functions we might want to write:

```
(** [add1 lst] adds 1 to each element of [lst] *)
let rec add1 = function
   | [] -> []
   | h :: t -> (h + 1) :: add1 t

let lst1 = add1 [1; 2; 3]
```

```
val add1 : int list -> int list = <fun>
```

```
val lst1 : int list = [2; 3; 4]
```

(continues on next page)

(continued from previous page)

```
let lst2 = concat_bang ["sweet"; "salty"]
```

```
val concat_bang : string list -> string list = <fun>
```

```
val lst2 : string list = ["sweet!"; "salty!"]
```

There's a lot of similarity between those two functions:

- They both pattern match against a list.
- They both return the same value for the base case of the empty list.
- They both recurse on the tail in the case of a non-empty list.

In fact the only difference (other than their names) is what they do for the head element: add versus concatenate. Let's rewrite the two functions to make that difference even more explicit:

```
(** [add1 lst] adds 1 to each element of [lst] *)
let rec add1 = function
    | [] -> []
    | h :: t ->
        let f = fun x -> x + 1 in
        f h :: add1 t

(** [concat_bang lst] concatenates "!" to each element of [lst] *)
let rec concat_bang = function
    | [] -> []
    | h :: t ->
        let f = fun x -> x ^ "!" in
        f h :: concat_bang t
```

```
val add1 : int list -> int list = <fun>
```

```
val concat_bang : string list -> string list = <fun>
```

Now the only difference between the two functions (again, other than their names) is the body of helper function f. Why repeat all that code when there's such a small difference between the functions? We might as well *abstract* that one helper function out from each main function and make it an argument:

```
let rec add1' f = function
    | [] -> []
    | h :: t -> f h :: add1' f t

(** [add1 lst] adds 1 to each element of [lst] *)
let add1 = add1' (fun x -> x + 1)

let rec concat_bang' f = function
    | [] -> []
    | h :: t -> f h :: concat_bang' f t

(** [concat_bang lst] concatenates "!" to each element of [lst] *)
let concat_bang = concat_bang' (fun x -> x ^ "!")
```

```
val add1' : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

4.2. Map 115

```
val add1 : int list -> int list = <fun>
```

```
val concat_bang' : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val concat_bang : string list -> string list = <fun>
```

But now there really is no difference at all between add1' and concat_bang' except for their names. They are totally duplicated code. Even their types are now the same, because nothing about them mentions integers or strings. We might as well just keep only one of them and come up with a good new name for it. One possibility could be transform, because they tranform a list by applying a function to each element of the list:

```
let rec transform f = function
    | [] -> []
    | h :: t -> f h :: transform f t

(** [add1 lst] adds 1 to each element of [lst] *)
let add1 = transform (fun x -> x + 1)

(** [concat_bang lst] concatenates "!" to each element of [lst] *)
let concat_bang = transform (fun x -> x ^ "!")
```

```
val transform : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val add1 : int list -> int list = <fun>
```

```
val concat_bang : string list -> string list = <fun>
```

Note: Instead of

```
let add1 lst = transform (fun x -> x + 1) lst
```

above we wrote

```
let add1 = transform (fun x -> x + 1)
```

This is another way of being higher order, but it's one we already learned about under the guise of partial application. The latter way of writing the function partially applies transform to just one of its two arguments, thus returning a function. That function is bound to the name add1.

Indeed, the C++ library does call the equivalent function transform. But OCaml and many other languages (including Java and Python) use the shorter word *map*, in the mathematical sense of how a function maps an input to an output. So let's make one final change to that name:

```
let rec map f = function
    | [] -> []
    | h :: t -> f h :: map f t

    (** [add1 lst] adds 1 to each element of [lst] *)
let add1 = map (fun x -> x + 1)

    (** [concat_bang lst] concatenates "!" to each element of [lst] *)
let concat_bang = map (fun x -> x ^ "!")
```

```
val map : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val add1 : int list -> int list = <fun>
```

```
val concat_bang : string list -> string list = <fun>
```

We have now successfully applied the Abstraction Principle: the common structure has been factored out. What's left clearly expresses the computation, at least to the reader who is familiar with map, in a way that the original versions do not as quickly make apparent.

4.2.1 Side Effects

The map function exists already in OCaml's standard library as List.map, but with one small difference from the implementation we discovered above. First, let's see what's potentially wrong with our own implementation, then we'll look at the standard library's implementation.

We've seen before in our discussion of *exceptions* that the OCaml language specification does not generally specify evaluation order of subexpressions, and that the current language implementation generally evaluates right-to-left. Because of that, the following (rather contrived) code actually causes the list elements to be printed in what might seem like reverse order:

```
let p x = print_int x; print_newline(); x + 1
let lst = map p [1; 2]
```

```
val p : int -> int = <fun>
```

```
2
```

```
val lst : int list = [2; 3]
```

Here's why:

- Expression map p [1; 2] evaluates to p 1 :: map p [2].
- The right-hand side of that expression is then evaluated to p 1 :: (p 2 :: map p []). The application of p to 1 has not yet occurred.
- The right-hand side of :: is again evaluated next, yielding p 1 :: (p 2 :: []).
- Then p is applied to 2, and finally to 1.

That is likely surprising to anyone who is predisposed to thinking that evaluation would occur left-to-right. The solution is to use a let expression to cause the evaluation of the function application to occur before the recursive call:

```
val map : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

4.2. Map 117

```
2
```

```
val 1st2 : int list = [2; 3]
```

Here's why that works:

- Expression map p [1; 2] evaluates to let h' = p 1 in h'' :: map p [2].
- The binding expression p 1 is evaluated, causing 1 to be printed and h' to be bound to 2.
- The body expression h'':: map p [2] is then evaluated, which leads to 2 being printed next.

So that's how the standard library defines List.map. We should use it instead of re-defining the function ourselves from now on. But it's good that we have discovered the function "from scratch" as it were, and that if needed we could quickly re-code it.

The bigger lesson to take away from this discussion is that when evaluation order matters, we need to use let to ensure it. When does it matter? Only when there are side effects. Printing and exceptions are the two we've seen so far. Later we'll add mutability.

4.2.2 Map and Tail Recursion

Astute readers will have noticed that the implementation of map is not tail recursive. That is to some extent unavoidable. Here's a tempting but awful way to create a tail-recursive version of it:

```
let rec map_tr_aux f acc = function
    | [] -> acc
    | h :: t -> map_tr_aux f (acc @ [f h]) t

let map_tr f = map_tr_aux f []

let lst = map_tr (fun x -> x + 1) [1; 2; 3]
```

```
val map_tr_aux : ('a -> 'b) -> 'b list -> 'a list -> 'b list = <fun>
```

```
val map_tr : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val lst : int list = [2; 3; 4]
```

To some extent that works: the output is correct, and map_tr_aux is tail recursive. The subtle flaw is the subexpression $acc \in [fh]$. Recall that append is a linear-time operation on singly-linked lists. That is, if there are n list elements then append takes time O(n). So at each recursive call we perform a O(n) operation. And there will be n recursive calls, one for each element of the list. That's a total of $n \cdot O(n)$ work, which is $O(n^2)$. So we achieved tail recursion, but at a high cost: what ought to be a linear-time operation became quadratic time.

In an attempt to fix that, we could use the constant-time cons operation instead of the linear-time append operation:

```
val map_tr_aux : ('a -> 'b) -> 'b list -> 'a list -> 'b list = <fun>
```

```
val map_tr : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val lst : int list = [4; 3; 2]
```

And to some extent that works: it's tail recursive and linear time. The not-so-subtle flaw this time is that the output is backwards. As we take each element off the front of the input list, we put it on the front of the output list, but that reverses their order.

Note: To understand why the reversal occurs, it might help to think of the input and output lists as people standing in a queue:

• Input: Alice, Bob.

· Output: empty.

Then we remove Alice from the input and add her to the output:

• Input: Bob.

· Output: Alice.

Then we remove Bob from the input and add him to the output:

• Input: empty.

· Output: Bob, Alice.

The point is that with singly-linked lists, we can only operate on the head of the list and still be constant time. We can't move Bob to the back of the output without making him walk past Alice—and anyone else who might be standing in the output.

For that reason, the standard library calls this function List.rev_map, that is, a (tail-recursive) map function that returns its output in reverse order.

```
let rec rev_map_aux f acc = function
    | [] -> acc
    | h :: t -> rev_map_aux f (f h :: acc) t

let rev_map f = rev_map_aux f []

let lst = rev_map (fun x -> x + 1) [1; 2; 3]
```

```
val rev_map_aux : ('a -> 'b) -> 'b list -> 'a list -> 'b list = <fun>
```

```
val rev_map : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val lst : int list = [4; 3; 2]
```

If you want the output in the "right" order, that's easy: just apply List.rev to it:

```
let lst = List.rev (List.rev_map (fun x -> x + 1) [1; 2; 3])
```

```
val lst : int list = [2; 3; 4]
```

4.2. Map 119

Since List.rev is both linear time and tail recursive, that yields a complete solution. We get a linear-time and tail-recursive map computation. The expense is that it requires two passes through the list: one to transform, the other to reverse. We're not going to do better than this efficiency with a singly-linked list. Of course, there are other data structures that implement lists, and we'll come to those eventually. Meanwhile, recall that we generally don't have to worry about tail recursion (which is to say, about stack space) until lists have 10,000 or more elements.

Why doesn't the standard library provide this all-in-one function? Maybe it will someday if there's good enough reason. But you might discover in your own programming there's not a lot of need for it. In many cases, we can either do without the tail recursion, or be content with a reversed list.

The bigger lesson to take away from this discussion is that there can be a tradeoff between time and space efficiency for recursive functions. By attempting to make a function more space efficient (i.e., tail recursive), we can accidentally make it asymptotically less time efficient (i.e., quadratic instead of linear), or if we're clever keep the asymptotic time efficiency the same (i.e., linear) at the cost of a constant factor (i.e., processing twice).

4.2.3 Map in Other Languages

We mentioned above that the idea of map exists in many programming languages. Here's an example from Python:

```
>>> print(list(map(lambda x: x + 1, [1, 2, 3])))
[2, 3, 4]
```

We have to use the list function to convert the result of the map back to a list, because Python for sake of efficiency produces each element of the map output as needed. Here again we see the theme of "when does it get evaluated?" returning.

In Java, map is part of the Stream abstraction that was added in Java 8. Since there isn't a built-in Java syntax for lists or streams, it's a little more verbose to give an example. Here we use a factory method Stream.of to create a stream:

```
jshell> Stream.of(1, 2, 3).map(x \rightarrow x + 1).collect(Collectors.toList()) $1 ==> [2, 3, 4]
```

Like in the Python example, we have to use something to convert the stream back into a list. In this case it's the collect method.

4.3 Filter

Suppose we wanted to filter out only the even numbers from a list, or the odd numbers. Here are some functions to do that:

```
(** [even n] is whether [n] is even. *)
let even n =
  n mod 2 = 0

(** [evens lst] is the sublist of [lst] containing only even numbers. *)
let rec evens = function
  | [] -> []
  | h :: t -> if even h then h :: evens t else evens t

let lst1 = evens [1; 2; 3; 4]
```

```
val even : int -> bool = <fun>
```

```
val evens : int list -> int list = <fun>
```

```
val lst1 : int list = [2; 4]
```

```
(** [odd n] is whether [n] is odd. *)
let odd n =
  n mod 2 <> 0

(** [odds lst] is the sublist of [lst] containing only odd numbers. *)
let rec odds = function
  | [] -> []
  | h :: t -> if odd h then h :: odds t else odds t

let lst2 = odds [1; 2; 3; 4]
```

```
val odd : int -> bool = <fun>
```

```
val odds : int list -> int list = <fun>
```

```
val lst2 : int list = [1; 3]
```

Functions evens and odds are nearly the same code: the only essential difference is the test they apply to the head element. So as we did with map in the previous section, let's factor out that test as a function. Let's name the function p as short for "predicate", which is a fancy way of saying that it tests whether something is true or false:

```
val filter : ('a -> bool) -> 'a list -> 'a list = <fun>
```

And now we can reimplement our original two functions:

```
let evens = filter even
let odds = filter odd
```

```
val evens : int list -> int list = <fun>
```

```
val odds : int list -> int list = <fun>
```

How simple these are! How clear! (At least to the reader who is familiar with filter.)

4.3.1 Filter and Tail Recursion

As we did with map, we can create a tail-recursive version of filter:

```
let rec filter_aux p acc = function
    | [] -> acc
    | h :: t -> if p h then filter_aux p (h :: acc) t else filter_aux p acc t
let filter p = filter_aux p []
let lst = filter even [1; 2; 3; 4]
```

4.3. Filter 121

```
val filter_aux : ('a -> bool) -> 'a list -> 'a list -> 'a list = <fun>
```

```
val filter : ('a -> bool) -> 'a list -> 'a list = <fun>
```

```
val lst : int list = [4; 2]
```

And again we discover the output is backwards. Here, the standard library makes a different choice than it did with map. It builds in the reversal to List.filter, which is implemented like this:

```
let rec filter_aux p acc = function
    | [] -> List.rev acc (* note the built-in reversal *)
    | h :: t -> if p h then filter_aux p (h :: acc) t else filter_aux p acc t
let filter p = filter_aux p []
```

```
val filter_aux : ('a -> bool) -> 'a list -> 'a list -> 'a list = <fun>
```

```
val filter : ('a -> bool) -> 'a list -> 'a list = <fun>
```

Why does the standard library treat map and filter differently on this point? Good question. Perhaps there has simply never been a demand for a filter function whose time efficiency is a constant factor better. Or perhaps it is just historical accident.

4.3.2 Filter in Other Languages

Again, the idea of filter exists in many programming languages. Here it is in Python:

```
>>> print(list(filter(lambda x: x % 2 == 0, [1, 2, 3, 4])))
[2, 4]
```

And in Java:

```
jshell> Stream.of(1, 2, 3, 4).filter(x \rightarrow x % 2 == 0).collect(Collectors.toList()) $1 ==> [2, 4]
```

4.4 Fold

The map functional gives us a way to individually transform each element of a list. The filter functional gives us a way to individually decide whether to keep or throw away each element of a list. But both of those are really just looking at a single element at a time. What if we wanted to somehow combine all the elements of a list? That's what the *fold* functional is for. It turns out that there are two versions of it, which we'll study in this section. But to start, let's look at a related function—not actually in the standard library—that we call *combine*.

4.4.1 Combine

Once more, let's write two functions:

```
(** [sum lst] is the sum of all the elements of [lst]. *)
let rec sum = function
    | [] -> 0
    | h :: t -> h + sum t

let s = sum [1; 2; 3]
```

```
val sum : int list -> int = <fun>
```

```
val s : int = 6
```

```
(** [concat lst] is the concatenation of all the elements of [lst]. *)
let rec concat = function
    | [] -> ""
    | h :: t -> h ^ concat t

let c = concat ["a"; "b"; "c"]
```

```
val concat : string list -> string = <fun>
```

```
val c : string = "abc"
```

As when we went through similar exercises with map and filter, the functions share a great deal of common structure. The differences here are:

- the case for the empty list returns a different initial value, 0 vs ""
- the case of a non-empty list uses a different operator to combine the head element with the result of the recursive call, + vs ^.

So can we apply the Abstraction Principle again? Sure! But this time we need to factor out *two* arguments: one for each of those two differences.

To start, let's factor out only the initial value:

```
let rec sum' init = function
    | [] -> init
    | h :: t -> h + sum' init t

let sum = sum' 0

let rec concat' init = function
    | [] -> init
    | h :: t -> h ^ concat' init t

let concat = concat' ""
```

```
val sum' : int -> int list -> int = <fun>
```

```
val sum : int list -> int = <fun>
```

4.4. Fold 123

```
val concat' : string -> string list -> string = <fun>
```

```
val concat : string list -> string = <fun>
```

Now the only real difference left between sum' and concat' is the operator used to combine the head with the recursive call on the tail. That operator can also become an argument to a unified function we call combine:

```
let rec combine op init = function
    | [] -> init
    | h :: t -> op h (combine op init t)

let sum = combine ( + ) 0
let concat = combine ( ^ ) ""
```

```
val combine : ('a -> 'b -> 'b) -> 'a list -> 'b = <fun>
```

```
val sum : int list -> int = <fun>
```

```
val concat : string list -> string = <fun>
```

One way to think of combine would be that:

- the [] value in the list gets replaced by init, and
- each :: constructor gets replaced by op.

For example, [a; b; c] is just syntactic sugar for a :: (b :: (c :: [])). So if we replace [] with 0 and :: with (+), we get a + (b + (c + 0)). And that would be the sum of the list.

Once more, the Abstraction Principle has led us to an amazingly simple and succinct expression of the computation.

4.4.2 Fold Right

The combine function is the idea underlying an actual OCaml library function. To get there, we need to make a couple of changes to the implementation we have so far.

First, let's rename some of the arguments: we'll change op to f to emphasize that really we could pass in any function, not just a built-in operator like +. And we'll change init to acc, which as usual stands for "accumulator". That yields:

```
val combine : ('a -> 'b -> 'b) -> 'a list -> 'b = <fun>
```

Second, let's make an admittedly less well-motivated change. We'll swap the implicit list argument to combine with the init argument:

```
val combine' : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b = <fun>
```

```
val sum : int list -> int = <fun>
```

```
val concat : string list -> string = <fun>
```

It's a little less convenient to code the function this way, because we no longer get to take advantage of the function keyword, nor of partial application in defining sum and concat. But there's no algorithmic change.

What we now have is the actual implementation of the standard library function List.fold_right. All we have left to do is change the function name:

```
val fold_right : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b = <fun>
```

Why is this function called "fold right"? The intuition is that the way it works is to "fold in" elements of the list from the right to the left, combining each new element using the operator. For example, fold_right (+) [a; b; c] 0 results in evaluation of the expression a + (b + (c + 0)). The parentheses associate from the right-most subexpression to the left.

4.4.3 Tail Recursion and Combine

Neither fold_right nor combine are tail recursive: after the recursive call returns, there is still work to be done in applying the function argument f or op. Let's go back to combine and rewrite it to be tail recursive. All that requires is to change the cons branch:

```
val combine_tr : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a = <fun>
```

(Careful readers will notice that the type of combine_tr is different than the type of combine. We will address that soon.)

Now the function f is applied to the head element h and the accumulator acc *before* the recursive call is made, thus ensuring there's no work remaining to be done after the call returns. If that seems a little mysterious, here's a rewriting of the two functions that might help:

```
let rec combine f acc = function
    | [] -> acc
    | h :: t ->
        let acc' = combine f acc t in
        f h acc'

let rec combine_tr f acc = function
    | [] -> acc
    | h :: t ->
        let acc' = f acc h in
        combine_tr f acc' t
```

4.4. Fold 125

```
val combine : ('a -> 'b -> 'b) -> 'a list -> 'b = <fun>
```

```
val combine_tr : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a = <fun>
```

Pay close attention to the definition of acc', the new accumulator, in each of those version:

- In the original version, we procrastinate using the head element h. First, we combine all the remaining tail elements to get acc'. Only then do we use f to fold in the head. So the value passed as the initial value of acc turns out to be the same for every recursive invocation of combine: it's passed all the way down to where it's needed, at the right-most element of the list, then used there exactly once.
- But in the tail recursive version, we "pre-crastinate" by immediately folding h in with the old accumulator acc. Then we fold that in with all the tail elements. So at each recursive invocation, the value passed as the argument acc can be different.

The tail recursive version of combine works just fine for summation (and concatenation, which we elide):

```
let sum = combine_tr ( + ) 0
let s = sum [1; 2; 3]
```

```
val sum : int list -> int = <fun>
```

```
val s : int = 6
```

But something possibly surprising happens with subtraction:

```
let sub = combine ( - ) 0
let s = sub [3; 2; 1]

let sub_tr = combine_tr ( - ) 0
let s' = sub_tr [3; 2; 1]
```

```
val sub : int list -> int = <fun>
```

```
val s : int = 2
```

```
val sub_tr : int list -> int = <fun>
```

```
val s' : int = -6
```

The two results are different!

- With combine we compute 3 (2 (1 0)). First we fold in 1, then 2, then 3. We are processing the list from right to left, putting the initial accumulator at the far right.
- But with combine_tr we compute (((0 3) 2) 1). We are processing the list from left to right, putting the initial accumulator at the far left.

With addition it didn't matter which order we processed the list, because addition is associative and commutative. But subtraction is not, so the two directions result in different answers.

Actually this shouldn't be too surprising if we think back to when we made map be tail recursive. Then, we discovered that tail recursion can cause us to process the list in reverse order from the non-tail recursive version of the same function. That's what happened here.

4.4.4 Fold Left

Our combine_tr function is also in the standard library under the name List.fold_left:

```
let rec fold_left f acc = function
    | [] -> acc
    | h :: t -> fold_left f (f acc h) t

let sum = fold_left ( + ) 0
let concat = fold_left ( ^ ) ""
```

```
val fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a = <fun>
```

```
val sum : int list -> int = <fun>
```

```
val concat : string list -> string = <fun>
```

We have once more succeeded in applying the Abstraction Principle.

4.4.5 Fold Left vs. Fold Right

Let's review the differences between fold_right and fold_left:

- They combine list elements in opposite orders, as indicated by their names. Function fold_right combines from the right to the left, whereas fold_left proceeds from the left to the right.
- Function fold_left is tail recursive whereas fold_right is not.
- The types of the functions are different.

Regarding that final point, it can be hard to remember what those types are! Luckily we can always ask the toplevel:

```
List.fold_left;;
List.fold_right;;
```

```
- : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a = <fun>
```

```
- : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b = <fun>
```

To understand those types, look for the list argument in each one of them. That tells you the type of the values in the list. Then look for the type of the return value; that tells you the type of the accumulator. From there you can work out everything else.

- In fold_left, the list argument is of type 'b list, so the list contains values of type 'b. The return type is 'a, so the accumulator has type 'a. Knowing that, we can figure out that the second argument is the initial value of the accumulator (because it has type 'a). And we can figure out that the first argument, the combining operator, takes as its own first argument an accumulator value (because it has type 'a), as its own second argument a list element (because it has type 'b), and returns a new accumulator value.
- In fold_right, the list argument is of type 'a list, so the list contains values of type 'a. The return type is 'b, so the accumulator has type 'b. Knowing that, we can figure out that the third argument is the initial value of the accumulator (because it has type 'b). And we can figure out that the first argument, the combining operator, takes as its own second argument an accumulator value (because it has type 'b), as its own first argument a list element (because it has type 'a), and returns a new accumulator value.

4.4. Fold 127

Tip: You might wonder why the argument orders are different between the two fold functions. Good question. Other libraries do in fact use different argument orders. One way to remember it for OCaml is that in fold_X the accumulator argument goes to the X of the list argument.

If you find it hard to keep track of all these argument orders, the ListLabels module in the standard library can help. It uses labeled arguments to give names to the combining operator (which it calls f) and the initial accumulator value (which it calls init). Internally, the implementation is actually identical to the List module.

```
ListLabels.fold_left;;
ListLabels.fold_left ~f:(fun x y -> x - y) ~init:0 [1;2;3];;
```

```
- : f:('a -> 'b -> 'a) -> init:'a -> 'b list -> 'a = <fun>
```

```
-: int = -6
```

```
ListLabels.fold_right;;
ListLabels.fold_right ~f:(fun y x -> x - y) ~init:0 [1;2;3];;
```

```
- : f:('a -> 'b -> 'b) -> 'a list -> init:'b -> 'b = <fun>
```

```
-: int = -6
```

Notice how in the two applications of fold above, we are able to write the arguments in a uniform order thanks to their labels. However, we still have to be careful about which argument to the combining operator is the list element vs. the accumulator value.

4.4.6 A Digression on Labeled Arguments and Fold

It's possible to write our own version of the fold functions that would label the arguments to the combining operator, so we don't even have to remember their order:

```
let rec fold_left ~op:(f: acc:'a -> elt:'b -> 'a) ~init:acc lst =
   match lst with
   | [] -> acc
   | h :: t -> fold_left ~op:f ~init:(f ~acc:acc ~elt:h) t

let rec fold_right ~op:(f: elt:'a -> acc:'b -> 'b) lst ~init:acc =
   match lst with
   | [] -> acc
   | h :: t -> f ~elt:h ~acc:(fold_right ~op:f t ~init:acc)
```

But those functions aren't as useful as they might seem:

```
let s = fold_left ~op:( + ) ~init:0 [1;2;3]
```

The problem is that the built-in + operator doesn't have labeled arguments, so we can't pass it in as the combining operator to our labeled functions. We'd have to define our own labeled version of it:

```
let add ~acc ~elt = acc + elt
let s = fold_left ~op:add ~init:0 [1; 2; 3]
```

But now we have to remember that the $\neg acc$ parameter to add will become the left-hand argument to (+). That's not really much of an improvement over what we had to remember to begin with.

4.4.7 Using Fold to Implement Other Functions

Folding is so powerful that we can write many other list functions in terms of fold_left or fold_right. For example,

```
let length lst =
   List.fold_left (fun acc _ -> acc + 1) 0 lst

let rev lst =
   List.fold_left (fun acc x -> x :: acc) [] lst

let map f lst =
   List.fold_right (fun x acc -> f x :: acc) lst []

let filter f lst =
   List.fold_right (fun x acc -> if f x then x :: acc else acc) lst []
```

```
val length : 'a list -> int = <fun>
```

```
val rev : 'a list -> 'a list = <fun>
```

```
val map : ('a -> 'b) -> 'a list -> 'b list = <fun>
```

```
val filter : ('a -> bool) -> 'a list -> 'a list = <fun>
```

At this point it begins to become debatable whether it's better to express the computations above using folding or using the ways we have already seen. Even for an experienced functional programmer, understanding what a fold does can take longer than reading the naive recursive implementation. If you peruse the source code of the standard library, you'll see that none of the List module internally is implemented in terms of folding, which is perhaps one comment on the readability of fold. On the other hand, using fold ensures that the programmer doesn't accidentally program the recursive traversal incorrectly. And for a data structure that's more complicated than lists, that robustness might be a win.

4.4. Fold 129

4.4.8 Fold vs. Recursive vs. Library

We've now seen three different ways for writing functions that manipulate lists:

- directly as a recursive function that pattern matches against the empty list and against cons,
- using fold functions, and
- using other library functions.

Let's try using each of those ways to solve a problem, so that we can appreciate them better.

Consider writing a function lst_and: bool list -> bool, such that lst_and [a1; ...; an] returns whether all elements of the list are true. That is, it evaluates the same as a1 && a2 && ... && an. When applied to an empty list, it evaluates to true.

Here are three possible ways of writing such a function. We give each way a slightly different function name for clarity.

```
val lst_and_rec : bool list -> bool = <fun>
```

```
val lst_and_fold : bool list -> bool = <fun>
```

```
val lst_and_lib : bool list -> bool = <fun>
```

The worst-case running time of all three functions is linear in the length of the list. But:

- The first function, lst_and_rec has the advantage that it need not process the entire list. It will immediately return false the first time they discover a false element in the list.
- The second function, lst_and_fold, will always process every element of the list.
- As for the third function lst_and_lib, according to the documentation of List.for_all, it returns (p a1) && (p a2) && ... && (p an). So like lst_and_rec it need not process every element.

4.5 Beyond Lists

Functionals like map and fold are not restricted to lists. They make sense for nearly any kind of data collection. For example, recall this tree representation:

```
type 'a tree = Leaf | Node of 'a * 'a tree * 'a tree
```

4.5.1 Map on Trees

This one is easy. All we have to do is apply the function f to the value v at each node:

```
val map_tree : ('a -> 'b) -> 'a tree -> 'b tree = <fun>
```

4.5.2 Fold on Trees

This one is only a little harder. Let's develop a fold functional for 'a tree similar to our fold_right over 'a list. One way to think of List.fold_right would be that the [] value in the list gets replaced by the acc argument, and each :: constructor gets replaced by an application of the f argument. For example, [a; b; c] is syntactic sugar for a :: (b :: (c :: [])). So if we replace [] with 0 and :: with (+), we get a + (b + (c + 0)). Along those lines, here's a way we could rewrite fold_right that will help us think a little more clearly:

```
type 'a mylist =
    | Nil
    | Cons of 'a * 'a mylist

let rec fold_mylist f acc = function
    | Nil -> acc
    | Cons (h, t) -> f h (fold_mylist f acc t)
```

```
type 'a mylist = Nil | Cons of 'a * 'a mylist
```

```
val fold_mylist : ('a -> 'b -> 'b) -> 'b -> 'a mylist -> 'b = <fun>
```

The algorithm is the same. All we've done is to change the definition of lists to use constructors written with alphabetic characters instead of punctuation, and to change the argument order of the fold function.

For trees, we'll want the initial value of acc to replace each Leaf constructor, just like it replaced [] in lists. And we'll want each Node constructor to be replaced by the operator. But now the operator will need to be *ternary* instead of *binary*—that is, it will need to take three arguments instead of two—because a tree node has a value, a left child, and a right child, whereas a list cons had only a head and a tail.

Inspired by those observations, here is the fold function on trees:

```
val fold_tree : ('a -> 'b -> 'b -> 'b -> 'a tree -> 'b = <fun>
```

If you compare that function to fold_mylist, you'll note it very nearly identical. There's just one more recursive call in the second pattern-matching branch, corresponding to the one more occurrence of 'a tree in the definition of that type.

We can then use fold_tree to implement some of the tree functions we've previously seen:

```
let size t = fold_tree (fun _ l r -> 1 + l + r) 0 t
let depth t = fold_tree (fun _ l r -> 1 + max l r) 0 t
let preorder t = fold_tree (fun x l r -> [x] @ l @ r) [] t
```

4.5. Beyond Lists

```
val size : 'a tree -> int = <fun>
```

```
val depth : 'a tree -> int = <fun>
```

```
val preorder : 'a tree -> 'a list = <fun>
```

Why did we pick <code>fold_right</code> and not <code>fold_left</code> for this development? Because <code>fold_left</code> is tail recursive, which is something we're never going to achieve on binary trees. Suppose we process the left branch first; then we still have to process the right branch before we can return. So there will always be work left to do after a recursive call on one branch. Thus on trees an equivalent to <code>fold right</code> is the best which we can hope for.

The technique we used to derive fold_tree works for any OCaml variant type t:

- Write a recursive fold function that takes in one argument for each constructor of t.
- That fold function matches against the constructors, calling itself recursively on any value of type t that it encounters.
- Use the appropriate argument of fold to combine the results of all recursive calls as well as all data not of type t
 at each constructor.

This technique constructs something called a *catamorphism*, aka a *generalized fold operation*. To learn more about catamorphisms, take a course on category theory.

4.5.3 Filter on Trees

This one is perhaps the hardest to design. The problem is: if we decide to filter a node, what should we do with its children?

- We could recurse on the children. If after filtering them only one child remains, we could promote it in place of its parent. But what if both children remain, or neither? Then we'd somehow have to reshape the tree. Without knowing more about how the tree is intended to be used—that is, what kind of data it represents—we are stuck.
- Instead, we could just eliminate the children entirely. So the decision to filter a node means pruning the entire subtree rooted at that node.

The latter is easy to implement:

```
val filter_tree : ('a -> bool) -> 'a tree -> 'a tree = <fun>
```

4.6 Pipelining

Suppose we wanted to compute the sum of squares of the numbers from 0 up to n. How might we go about it? Of course (math being the best form of optimization), the most efficient way would be a closed-form formula:

$$\frac{n(n+1)(2n+1)}{6}$$

But let's imagine you've forgotten that formula. In an imperative language you might use a for loop:

```
# Python
def sum_sq(n):
        sum = 0
        for i in range(0, n):
            sum += i * i
        return sum
```

The equivalent (tail) recursive code in OCaml would be:

```
let sum_sq n =
  let rec loop i sum =
   if i > n then sum
    else loop (i + 1) (sum + i * i)
  in loop 0 0
```

```
val sum_sq : int -> int = <fun>
```

Another, clearer way of producing the same result in OCaml uses higher-order functions and the pipeline operator:

```
val ( -- ) : int -> int -> int list = <fun>
```

```
val square : int -> int = <fun>
```

```
val sum : int list -> int = <fun>
```

```
val sum_sq : int -> int = <fun>
```

The function <code>sum_sq</code> first constructs a list containing all the numbers <code>0..n</code>. Then it uses the pipeline operator <code>|></code> to pass that list through <code>List.map square</code>, which squares every element. Then the resulting list is pipelined through <code>sum</code>, which adds all the elements together.

The other alternatives that you might consider are somewhat uglier:

```
(* Maybe worse: a lot of extra [let..in] syntax and unnecessary names to
    for intermediate values we don't care about. *)
let sum_sq n =
    let l = 0 -- n in
    let sq_l = List.map square l in
    sum sq_l

(* Maybe worse: have to read the function applications from right to left
    rather than top to bottom, and extra parentheses. *)
let sum_sq n =
    sum (List.map square (0--n))
```

4.6. Pipelining 133

```
val sum_sq : int -> int = <fun>
```

```
val sum_sq : int -> int = <fun>
```

The downside of all of these compared to the original tail recursive version is that they are wasteful of space—linear instead of constant—and take a constant factor more time. So as is so often the case in programming, there is a tradeoff between clarity and efficiency of code.

Note that the inefficiency is *not* from the pipeline operator itself, but from having to construct all those unnecessary intermediate lists. So don't get the idea that pipelining is intrinsically bad. In fact it can be quite useful. When we get to the chapter on modules, we'll use it quite often with some of the data structures we study there.

4.7 Currying

We've already seen that an OCaml function that takes two arguments of types t1 and t2 and returns a value of type t3 has the type $t1 \rightarrow t2 \rightarrow t3$. We use two variables after the function name in the let expression:

```
let add x y = x + y
```

```
val add : int -> int -> int = <fun>
```

Another way to define a function that takes two arguments is to write a function that takes a tuple:

```
let add' t = fst t + snd t
```

```
val add' : int * int -> int = <fun>
```

Instead of using fst and snd, we could use a tuple pattern in the definition of the function, leading to a third implementation:

```
let add'' (x, y) = x + y
```

```
val add'' : int * int -> int = <fun>
```

Functions written using the first style (with type $t1 \rightarrow t2 \rightarrow t3$) are called *curried* functions, and functions using the second style (with type $t1 + t2 \rightarrow t3$) are called *uncurried*. Metaphorically, curried functions are "spicier" because you can partially apply them (something you can't do with uncurried functions: you can't pass in half of a pair). Actually, the term curry does not refer to spices, but to a logician named Haskell Curry (one of a very small set of people with programming languages named after both their first and last names).

Sometimes you will come across libraries that offer an uncurried version of a function, but you want a curried version of it to use in your own code; or vice versa. So it is useful to know how to convert between the two kinds of functions, as we did with add above.

You could even write a couple of higher-order functions to do the conversion for you:

```
let curry f x y = f (x, y)
let uncurry f (x, y) = f x y
```

```
val curry : ('a * 'b -> 'c) -> 'a -> 'b -> 'c = <fun>
```

```
val uncurry : ('a -> 'b -> 'c) -> 'a * 'b -> 'c = <fun>
```

```
let uncurried_add = uncurry add
let curried_add = curry add''
```

```
val uncurried_add : int * int -> int = <fun>
```

```
val curried_add : int -> int -> int = <fun>
```

4.8 Summary

This chapter is one of the most important in the book. It didn't cover any new language features. Instead, we learned how to use some of the existing features in ways that might be new, surprising, or challenging. Higher-order programming and the Abstraction Principle are two ideas that will help make you a better programmer in any language, not just OCaml. Of course, languages do vary in the extent to which they support these ideas, with some providing significantly less assistance in writing higher-order code—which is one reason we use OCaml in this course.

Map, filter, fold and other functionals are becoming widely recognized as excellent ways to structure computation. Part of the reason for that is they factor out the *iteration* over a data structure from the *computation* done at each element. Languages such as Python, Ruby, and Java 8 now have support for this kind of iteration.

4.8.1 Terms and concepts

- · Abstraction Principle
- · accumulator
- apply
- · associative
- compose
- factor
- filter
- first-order function
- fold
- · functional
- · generalized fold operation
- higher-order function
- map
- pipeline
- · pipelining

4.8. Summary 135

4.8.2 Further reading

- Introduction to Objective Caml, chapters 3.1.3, 5.3
- OCaml from the Very Beginning, chapter 6
- More OCaml: Algorithms, Methods, and Diversions, chapter 1, by John Whitington. This book is a sequel to OCaml from the Very Beginning.
- *Real World OCaml*, chapter 3 (beware that this book's Core library has a different List module than the standard library's List module, with different types for map and fold than those we saw here)
- "Higher Order Functions", chapter 6 of Functional Programming: Practice and Theory. Bruce J. MacLennan, Addison-Wesley, 1990. Our discussion of higher-order functions and the Abstraction Principle is indebted to this chapter.
- "Can Programming be Liberated from the von Neumann Style? A Functional Style and Its Algebra of Programs." John Backus' 1977 Turing Award lecture in its elaborated form as a published article.
- "Second-order and Higher-order Logic" in The Stanford Encyclopedia of Philosophy.

4.9 Exercises

Solutions to exercises are available to students in Cornell's CS 3110. Instructors at other institutions are welcome to contact Michael Clarkson for access.

Exercise: twice, no arguments [★]

Consider the following definitions:

```
let double x = 2*x
let square x = x*x
let twice f x = f (f x)
let quad = twice double
let fourth = twice square
```

Use the toplevel to determine what the types of quad and fourth are. Explain how it can be that quad is not syntactically written as a function that takes an argument, and yet its type shows that it is in fact a function.

Exercise: mystery operator $1 \left[\bigstar \star \right]$

What does the following operator do?

```
let ( $ ) f x = f x
```

Hint: investigate square \$ 2 + 2 vs. square 2 + 2.

Exercise: mystery operator $2 \left[\bigstar \right]$

What does the following operator do?

```
let ( @@ ) f g x = x |> g |> f
```

Hint: investigate String.length @@ string_of_int applied to 1, 10, 100, etc.

Exercise: repeat $[\star\star]$

Generalize twice to a function repeat, such that repeat f n x applies f to x a total of n times. That is,

```
repeat f 0 x yields x
repeat f 1 x yields f x
repeat f 2 x yields f (f x) (which is the same as twice f x)
repeat f 3 x yields f (f (f x))
```

Exercise: product [★]

Use fold_left to write a function product_left that computes the product of a list of floats. The product of the empty list is 1.0. *Hint: recall how we implemented sum in just one line of code in lecture.*

Use fold_right to write a function product_right that computes the product of a list of floats. Same hint applies.

Exercise: terse product $[\star \star]$

How terse can you make your solutions to the **product** exercise? *Hints: you need only one line of code for each, and you do not need the fun keyword. For fold_left, your function definition does not even need to explicitly take a list argument. If you use ListLabels, the same is true for fold_right.*

Exercise: sum_cube_odd [★★]

Write a function sum_cube_odd n that computes the sum of the cubes of all the odd numbers between 0 and n inclusive. Do not write any new recursive functions. Instead, use the functionals map, fold, and filter, and the (--) operator (defined in the discussion of pipelining).

Exercise: sum cube odd pipeline $[\star \star]$

Rewrite the function sum_cube_odd to use the pipeline operator |>.

Exercise: exists $[\star \star]$

Consider writing a function exists: ('a \rightarrow bool) \rightarrow 'a list \rightarrow bool, such that exists p [a1; ...; an] returns whether at least one element of the list satisfies the predicate p. That is, it evaluates the same as (p a1) || (p a2) || ... || (p an). When applied to an empty list, it evaluates to false.

Write three solutions to this problem, as we did above:

- exists_rec, which must be a recursive function that does not use the List module,
- exists_fold, which uses either List.fold_left or List.fold_right, but not any other List module functions nor the rec keyword, and
- exists_lib, which uses any combination of List module functions other than fold_left or fold_right, and does not use the rec keyword.

4.9. Exercises 137

Exercise: account balance $[\star\star\star]$

Write a function which, given a list of numbers representing debits, deducts them from an account balance, and finally returns the remaining amount in the balance. Write three versions: fold_left, fold_right, and a direct recursive implementation.

Exercise: library uncurried [★★]

Here is an uncurried version of List.nth:

```
let uncurried_nth (lst, n) = List.nth lst n
```

In a similar way, write uncurried versions of these library functions:

- List.append
- Char.compare
- Stdlib.max

Exercise: map composition $[\star \star \star]$

Show how to replace any expression of the form List.map f (List.map g lst) with an equivalent expression that calls List.map only once.

Exercise: more list fun $[\star \star \star]$

Write functions that perform the following computations. Each function that you write should use one of List.fold, List.map or List.filter. To choose which of those to use, think about what the computation is doing: combining, transforming, or filtering elements.

- Find those elements of a list of strings whose length is strictly greater than 3.
- Add 1.0 to every element of a list of floats.
- Given a list of strings strs and another string sep, produce the string that contains every element of strs separated by sep. For example, given inputs ["hi"; "bye"] and ", ", produce "hi, bye", being sure not to produce an extra comma either at the beginning or end of the result string.

Exercise: association list keys $[\star \star \star]$

Recall that an association list is an implementation of a dictionary in terms of a list of pairs, in which we treat the first component of each pair as a key and the second component as a value.

Write a function keys: ('a * 'b) list -> 'a list that returns a list of the unique keys in an association list. Since they must be unique, no value should appear more than once in the output list. The order of values output does not matter. How compact and efficient can you make your solution? Can you do it in one line and linearithmic space and time? *Hint:* List.sort_uniq.

Exercise: valid matrix [★★★]

A mathematical *matrix* can be represented with lists. In *row-major* representation, this matrix

$$\begin{bmatrix} 1 & 1 & 1 \\ 9 & 8 & 7 \end{bmatrix}$$

would be represented as the list [[1; 1; 1]; [9; 8; 7]]. Let's represent a *row vector* as an int list. For example, [9; 8; 7] is a row vector.

A *valid* matrix is an int list list that has at least one row, at least one column, and in which every column has the same number of rows. There are many values of type int list list that are invalid, for example,

- []
- [[1; 2]; [3]]

Implement a function is_valid_matrix: int list list -> bool that returns whether the input matrix is valid. Unit test the function.

Exercise: row vector add $[\star\star\star]$

Implement a function add_row_vectors: int list -> int list -> int list for the element-wise addition of two row vectors. For example, the addition of [1; 1; 1] and [9; 8; 7] is [10; 9; 8]. If the two vectors do not have the same number of entries, the behavior of your function is *unspecified*—that is, it may do whatever you like. *Hint: there is an elegant one-line solution using List.map2*. Unit test the function.

Exercise: matrix add $[\star \star \star]$

Implement a function add_matrices: int list list -> int list list -> int list list for matrix addition. If the two input matrices are not the same size, the behavior is unspecified. *Hint: there is an elegant one-line solution using List.map2 and add_row_vectors*. Unit test the function.

Exercise: matrix multiply $[\star \star \star \star]$

Implement a function multiply_matrices: int list list -> int list list -> int list list or matrix multiplication. If the two input matrices are not of sizes that can be multiplied together, the behavior is unspecified. Unit test the function. *Hint: define functions for matrix transposition and row vector dot product.*

4.9. Exercises 139

CHAPTER

FIVE

MODULAR PROGRAMMING

When a program is small enough, we can keep all of the details of the program in our heads at once. But real-world applications can be many order of magnitude larger than those we write in college classes. They are simply too large and complex to hold all their details in our heads. They are also written by many programmers. To build large software systems requires techniques we haven't talked about so far.

One key solution to managing complexity of large software is *modular programming*: the code is composed of many different code modules that are developed separately. This allows different developers to take on discrete pieces of the system and design and implement them without having to understand all the rest. But to build large programs out of modules effectively, we need to be able to write modules that we can convince ourselves are correct *in isolation* from the rest of the program. Rather than have to think about every other part of the program when developing a code module, we need to be able to use *local reasoning*: that is, reasoning about just the module and the contract it needs to satisfy with respect to the rest of the program. If everyone has done their job, separately developed code modules can be plugged together to form a working program without every developer needing to understand everything done by every other developer in the team. This is the key idea of modular programming.

Therefore, to build large programs that work, we must use *abstraction* to make it manageable to think about the program. Abstraction is simply the removal of detail. A well-written program has the property that we can think about its components (such as functions) abstractly, without concerning ourselves with all the details of how those components are implemented.

Modules are abstracted by giving *specifications* of what they are supposed to do. A good module specification is clear, understandable, and gives just enough information about what the module does for clients to successfully use it. This abstraction makes the programmer's job much easier; it is helpful even when there is only one programmer working on a moderately large program, and it is crucial when there is more than one programmer.

Industrial-strength languages contain mechanisms that support modular programming. In general (i.e. across programming languages), a module specification is known as an *interface*, which provides information to clients about the module's functionality while hiding the *implementation*. Object-oriented languages support modular programming with *classes*. The Java interface construct is one example of a mechanism for specifying the interface to a class. A Java interface informs clients of the available functionality in any class that implements it without revealing the details of the implementation. But even just the public methods of a class constitute an interface in the more general sense—an abstract description of what the module can do.

Developers working with a module take on distinct roles. Most developers are usually *clients* of the module who understand the interface but do not need to understand the implementation of the module. A developer who works on the module implementation is naturally called an *implementer*. The module interface is a *contract* between the client and the implementer, defining the responsibilities of both. Contracts are very important because they help us to isolate the source of the problem when something goes wrong—and to know who to blame!

It is good practice to involve both clients and implementers in the design of a module's interface. Interfaces designed solely by one or the other can be seriously deficient. Each side will have its own view of what the final product should look like, and these may not align! So mutual agreement on the contract is essential. It is also important to think hard about global module structure and interfaces *early*, because changing an interface becomes more and more difficult as the development proceeds and more of the code comes to depend on it.

Modules should be used only through their declared interfaces, which the language should help to enforce. This is true even when the client and the implementer are the same person. Modules decouple the system design and implementation problem into separate tasks that can be carried out largely independently. When a module is used only through its interface, the implementer has the flexibility to change the module as long as the module still satisfies its interface.

5.1 Module Systems

A programming language's *module system* is the set of features it provides in support of modular programming. Below are some common concerns of module systems. We focus on Java and OCaml in this discussion, mentioning some of the most related features in the two languages.

Namespaces. A namespace provides a set of names that are grouped together, are usually logically related, and are distinct from other namespaces. That enables a name foo in one namespace to have a distinct meaning from foo in another namespace. A namespace is thus a scoping mechanism. Namespaces are essential for modularity. Without them, the names that one programmer chooses could collide with the names another programmer chooses. In Java, classes (and packages) group names. In OCaml, *structures* (which we will soon study) are similar to classes in that they group names — but without any of the added complexity of object-oriented programming that usually accompanies classes (constructors, static vs. instance members, inheritance, overriding, this, etc.) Structures are the core of the OCaml module system; in fact, we've been using them all along without thinking too much about them.

Abstraction. An *abstraction* hides some information while revealing other information. Abstraction thus enables *encapsulation*, aka *information hiding*. Usually, abstraction mechanisms for modules allow revealing some names that exist inside the module, but hiding some others. Abstractions therefore describe relationships among modules: there might be many modules that could considered to satisfy a given abstraction. Abstraction is essential for modularity, because it enables implementers of a module to hide the details of the implementation from clients, thus preventing the clients from abusing those details. In a large team, the modules one programmer designs are thereby protected from abuse by another programmer. It also enables clients to be blissfully unaware of those details. So, in a large team, no programmer has to be aware of all the details of all the modules. In Java, interfaces and abstract classes provide abstraction. In OCaml, *signatures* are used to abstract structures by hiding some of the structure's names and definitions. Signatures are essentially the types of structures.

Code reuse. A module system enables *code reuse* by providing features that enable code from one module to be used as part of another module without having to copy that code. Code reuse thereby enables programmers to build on the work of others in a way that is maintainable: when the implementer of one module makes an improvement in that module, all the programmers who are reusing that code automatically get the benefit of that improvement. Code reuse is essential for modularity, because it enables "building blocks" that can be assembled and reassembled to form complex pieces of software. In Java, subtyping and inheritance provide code reuse. In OCaml, *functors* and *includes* enable code reuse. Functors are like functions, in that they produce new modules out of old modules. Includes are like an intelligent form of copy-paste: they include code from once part of a program in another.

Warning: These analogies between Java and OCaml are necessarily imperfect. You might naturally come away from the above discussion thinking either of the following:

- "Structures are like Java classes, and signatures are like interfaces."
- "Structures are like Java objects, and signatures are like classes."

Both are helpful to a degree, yet both are ultimately wrong. So it might be best to let go of object-oriented programming at this point and come to terms with the OCaml module system in and of itself. Compared to Java, it's just built different.

5.2 Modules

We begin with a couple of examples of the OCaml module system before diving into the details.

A *structure* is simply a collection of definitions, such as:

```
struct
  let inc x = x + 1
  type primary_color = Red | Green | Blue
  exception Oops
end
```

In a way, the structure is like a record: the structure has some distinct components with names. But unlike a record, it can define new types, exceptions, and so forth.

By itself the code above won't compile, because structures do not have the same first-class status as values like integers or functions. You can't just enter that code in utop, or pass that structure to a function, etc. What you can do is bind the structure to a name:

```
module MyModule = struct
let inc x = x + 1
  type primary_color = Red | Green | Blue
  exception Oops
end
```

```
module MyModule :
    sig
    val inc : int -> int
    type primary_color = Red | Green | Blue
    exception Oops
    end
```

The output from OCaml has the form:

```
module MyModule : sig ... end
```

This indicates that MyModule has been defined, and that it has been inferred to have the *module type* that appears to the right of the colon. That module type is written as *signature*:

```
sig
  val inc : int -> int
  type primary_color = Red | Green | Blue
  exception Oops
end
```

The signature itself is a collection of *specifications*. The specifications for variant types and exceptions are simply their original definitions, so primary_color and Oops are no different than they were in the original structure. The specification for inc though is written with the val keyword, exactly as the toplevel would respond if we defined inc in it.

Note: This use of the word "specification" is perhaps confusing, since many programmers would use that word to mean "the comments specifying the behavior of a function." But if we broaden our sight a little, we could allow that the type of a function is part of its specification. So it's at least a related sense of the word.

The definitions in a module are usually more closely related than those in MyModule. Often a module will implement some data structure. For example, here is a module for stacks implemented as linked lists:

```
module ListStack = struct
 (** [empty] is the empty stack. *)
 let empty = []
  (** [is_empty s] is whether [s] is empty. *)
 let is_empty = function [] -> true | _ -> false
  (** [push x s] pushes [x] onto the top of [s]. *)
 let push x s = x :: s
  (** [Empty] is raised when an operation cannot be applied
     to an empty stack. *)
 exception Empty
  (** [peek s] is the top element of [s].
     Raises [Empty] if [s] is empty. *)
 let peek = function
  | [] -> raise Empty
  | x :: _ -> x
  (** [pop s] is all but the top element of [s].
     Raises [Empty] if [s] is empty. *)
 let pop = function
   | [] -> raise Empty
   | _ :: s -> s
end
```

```
module ListStack :
   sig
   val empty : 'a list
   val is_empty : 'a list -> bool
   val push : 'a -> 'a list -> 'a list
   exception Empty
   val peek : 'a list -> 'a
   val pop : 'a list -> 'a list
   end
```

Important: The specification of pop might surprise you. Note that it does not return the top element. That's the job of peek. Instead, pop returns all but the top element.

We can then use that module to manipulate a stack:

```
ListStack.push 2 (ListStack.push 1 ListStack.empty)
```

```
- : int list = [2; 1]
```

Warning: There's a common confusion lurking here for those programmers coming from object-oriented languages. It's tempting to think of ListStack as being an object on which you invoke methods. Indeed ListStack.push vaguely looks like we're invoking a push method on a ListStack object. But that's not what is happening. In an OO language you could instantiate many stack objects. But here, there is only one ListStack. Moreover it is not an object, in large part because it has no notion of a this or self keyword to denote the receiving object of the method call.

That's admittedly rather verbose code. Soon we'll see several solutions to that problem, but for now here's one:

```
ListStack. (push 2 (push 1 empty))
```

```
- : int list = [2; 1]
```

By writing ListStack. (e), all the names from ListStack become usable in e without needing to write the prefix ListStack. each time. Another improvement could be using the pipeline operator:

```
ListStack. (empty |> push 1 |> push 2)
```

```
- : int list = [2; 1]
```

Now we can read the code left-to-right without having to parse parentheses. Nice.

Warning: There's another common OO confusion lurking here. It's tempting to think of ListStack as being a class from which objects are instantiated. That's not the case though. Notice how there is no new operator used to create a stack above, nor any constructors (in the OO sense of that word).

Modules are considerably more basic than classes. A module is just a collection of definitions in its own namespace. In ListStack, we have some definitions of functions—push, pop, etc.—and one value, empty.

So whereas in Java we might create a couple of stacks using code like this:

```
Stack s1 = new Stack();
s1.push(1);
s1.push(2);
Stack s2 = new Stack();
s2.push(3);
```

In OCaml the same stacks could be created as follows:

```
let s1 = ListStack.(empty |> push 1 |> push 2)
let s2 = ListStack.(empty |> push 3)
```

```
val s1 : int list = [2; 1]
```

```
val s2 : int list = [3]
```

5.2.1 Module Definitions

The module definition keyword is much like the let definition keyword that we learned before. (The OCaml designers hypothetically could have chosen to use let_module instead of module to emphasize the similarity.) The difference is just that:

- let binds a value to a name, whereas
- module binds a module value to a name.

Syntax.

The most common syntax for a module definition is simply:

```
module ModuleName = struct
  module_items
end
```

where module_items inside a structure can include let definitions, type definitions, and exception definitions, as well as nested module definitions. Module names must begin with an uppercase letter, and idiomatically they use CamelCase rather than Snake_case.

But a more accurate version of the syntax would be:

```
module ModuleName = module_expression
```

where a struct is just one sort of module_expression. Here's another: the name of an already defined module. For example, you can write module L = List if you'd like a short alias for the List module. We'll see other sorts of module expressions later in this section and chapter.

The definitions inside a structure can optionally be terminated by ;; as in the toplevel:

```
module M = struct
  let x = 0;;
  type t = int;;
end
```

```
module M : sig val x : int type t = int end
```

Sometimes that can be useful to add temporarily if you are trying to diagnose a syntax error. It will help OCaml understand that you want two definitions to be syntactically separate. After fixing whatever the underlying error is, though, you can remove the ;;.

One use case for ; ; is if you want to evaluate an expression as part of a module:

```
module M = struct
  let x = 0;;
  assert (x = 0);;
end
```

```
module M : sig val x : int end
```

But that can be rewritten without ;; as:

```
module M = struct
  let x = 0
  let _ = assert (x = 0)
end
```

```
module M : sig val x : int end
```

Structures can also be written on a single line, with optional;; between items for readability:

```
module N = struct let x = 0 let y = 1 end module O = struct let x = 0;; let y = 1 end
```

```
module N : sig val x : int val y : int end
```

```
module O : sig val x : int val y : int end
```

An empty structure is permitted:

```
module E = struct end
```

```
module E : sig end
```

Dynamic semantics.

We already know that expressions are evaluated to values. Similarly, a module expression is evaluated to a *module value* or just "module" for short. The only interesting kind of module expression we have so far, from the perspective of evaluation anyway, is the structure. Evaluation of structures is easy: just evaluate each definition in it, in the order they occur. Because of that, earlier definitions are therefore in scope in later definitions, but not vice versa. So this module is fine:

```
module M = struct
  let x = 0
  let y = x
end
```

```
module M : sig val x : int val y : int end
```

But this module is not, because at the time the let definition of x is being evaluated, y has not yet been bound:

```
module M = struct
let x = y
let y = 0
end
```

Of course, mutual recursion can be used if desired:

```
module M = struct
let rec even = function 0 -> true | n -> odd (n - 1)
and odd = function 1 -> true | n -> even (n - 1)
end
```

```
module M : sig val even : int -> bool val odd : int -> bool end
```

Static semantics.

A structure is well typed if all the definitions in it are themselves well-typed, according to all the typing rules we have already learned.

As we've seen in toplevel output, the module type of a structure is a signature. There's more to module types than that, though. Let's put that off for a moment to first talk about scope.

5.2.2 Scope and Open

After a module M has been defined, you can access the names within it using the dot operator. For example:

```
module M = struct let x = 42 end
```

```
module M : sig val x : int end
```

```
M.x
```

```
- : int = 42
```

Of course from outside the module the name x by itself is not meaningful:

```
x
```

But you can bring all of the definitions of a module into the current scope using open:

```
open M
```

```
X
```

```
- : int = 42
```

Opening a module is like writing a local definition for each name defined in the module. For example, open String brings all the definitions from the String module into scope, and has an effect similar to the following on the local namespace:

```
let length = String.length
let get = String.get
let lowercase_ascii = String.lowercase_ascii
...
```

If there are types, exceptions, or modules defined in a module, those also are brought into scope with open.

The Always-Open Module. There is a special module called Stdlib that is automatically opened in every OCaml program. It contains the "built-in" functions and operators. You therefore never need to prefix any of the names it defines with Stdlib., though you could do so if you ever needed to unambiguously identify a name from it. In earlier days, this module was named Pervasives, and you might still see that name in some code bases.

Open as a Module Item. An open is another sort of module_item. So we can open one module inside another:

```
module M = struct
  open List

  (** [uppercase_all lst] upper-cases all the elements of [lst]. *)
  let uppercase_all = map String.uppercase_ascii
end
```

```
module M : sig val uppercase_all : string list -> string list end
```

Since List is open, the name map from it is in scope. But what if we wanted to get rid of the String. as well?

```
module M = struct
  open List
  open String

  (** [uppercase_all lst] upper-cases all the elements of [lst]. *)
  let uppercase_all = map uppercase_ascii
end
```

Now we have a problem, because String also defines the name map, but with a different type than List. As usual a later definition shadows an earlier one, so it's String.map that gets chosen instead of List.map as we intended.

If you're using many modules inside your code, chances are you'll have at least one collision like this. Often it will be with a standard higher-order function like map that is defined in many library modules.

Tip: It is therefore generally good practice **not** to open all the modules you're going to use at the top of a .ml file or structure. This is perhaps different than how you're used to working with languages like Java, where you might import many packages with *. Instead, it's good to restrict the scope in which you open modules.

Limiting the Scope of Open. We've already seen one way of limiting the scope of an open: M. (e). Inside e all the names from module M are in scope. This is useful for briefly using M in a short expression:

```
(* remove surrounding whitespace from [s] and convert it to lower case *)
let s = "BigRed "
let s' = s |> String.trim |> String.lowercase_ascii (* long way *)
let s'' = String.(s |> trim |> lowercase_ascii) (* short way *)
```

```
val s : string = "BigRed "
```

```
val s' : string = "bigred"
```

```
val s'' : string = "bigred"
```

But what if you want to bring a module into scope for an entire function, or some other large block of code? The (admittedly strange) syntax for that is let open M in e. It makes all the names from M be in scope in e. For example:

```
(** [lower_trim s] is [s] in lower case with whitespace removed. *)
let lower_trim s =
  let open String in
  s |> trim |> lowercase_ascii
```

```
val lower_trim : string -> string = <fun>
```

Going back to our uppercase_all example, it might be best to eschew any kind of opening and simply to be explicit about which module we are using where:

```
module M = struct
   (** [uppercase_all lst] upper-cases all the elements of [lst]. *)
   let uppercase_all = List.map String.uppercase_ascii
end
```

```
module M : sig val uppercase_all : string list -> string list end
```

5.2.3 Module Type Definitions

We've already seen that OCaml will infer a signature as the type of a module. Let's now see how to write those modules types ourselves. As an example, here is a module type for our list-based stacks:

```
module type LIST_STACK = sig
  exception Empty
  val empty : 'a list
  val is_empty : 'a list -> bool
  val push : 'a -> 'a list -> 'a list
  val peek : 'a list -> 'a
  val pop : 'a list -> 'a list
end
```

```
module type LIST_STACK =
   sig
   exception Empty
   val empty : 'a list
   val is_empty : 'a list -> bool
   val push : 'a -> 'a list -> 'a list
   val peek : 'a list -> 'a
   val pop : 'a list -> 'a list
end
```

Now that we have both a module and a module type for list-based stacks, we should move the specification comments from the structure into the signature. Those comments are properly part of the specification of the names in the signature. They specify behavior, thus augmenting the specification of types provided by the val declarations.

```
module type LIST_STACK = sig
   (** [Empty] is raised when an operation cannot be applied
        to an empty stack. *)
   exception Empty

   (** [empty] is the empty stack. *)
   val empty : 'a list

   (** [is_empty s] is whether [s] is empty. *)
   val is_empty : 'a list -> bool

   (** [push x s] pushes [x] onto the top of [s]. *)
   val push : 'a -> 'a list -> 'a list

   (** [peek s] is the top element of [s].
        Raises [Empty] if [s] is empty. *)
   val peek : 'a list -> 'a
```

(continues on next page)

```
(** [pop s] is all but the top element of [s].
     Raises [Empty] if [s] is empty. *)
  val pop : 'a list -> 'a list
end
module ListStack = struct
  let empty = []
  let is_empty = function [] -> true | _ -> false
  let push x s = x :: s
  exception Empty
  let peek = function
   | [] -> raise Empty
    | x :: _ -> x
  let pop = function
    | [] -> raise Empty
    | _ :: s -> s
end
```

Nothing so far, however, tells OCaml that there is a relationship between LIST_STACK and ListStack. If we want OCaml to ensure that ListStack really does have the module type specified by LIST_STACK, we can add a type annotation in the first line of the module definition:

```
module ListStack : LIST_STACK = struct
  let empty = []

let is_empty = function [] -> true | _ -> false

let push x s = x :: s

exception Empty

let peek = function
  | [] -> raise Empty
  | x :: _ -> x

let pop = function
  | [] -> raise Empty
  | _ :: s -> s
end
```

```
module ListStack : LIST_STACK
```

The compiler agrees that the module ListStack does define all the items specified by LIST_STACK with appropriate types. If we had accidentally omitted some item, the type annotation would have been rejected:

```
module ListStack : LIST_STACK = struct
let empty = []
let is_empty = function [] -> true | _ -> false
```

(continues on next page)

```
File "[28]", lines 1-15, characters 32-3:
1 | .....struct
2. 1
    let empty = []
3 |
4 |
    let is_empty = function [] -> true | _ -> false
5 I
12 I
       | x :: _ -> x
13 I
14 | (* [pop] is missing *)
15 | end
Error: Signature mismatch:
      The value `pop' is required but not provided
      File "[26]", line 21, characters 2-30: Expected declaration
```

Syntax.

The most common syntax for a module type is simply:

```
module type ModuleTypeName = sig
   specifications
end
```

where specifications inside a signature can include val declarations, type definitions, exception definitions, and nested module type definitions. Like structures, a signature can be written on many lines or just one line, and the empty signature sig end is allowed.

But, as we saw with module definitions, a more accurate version of the syntax would be:

```
module type ModuleTypeName = module_type
```

where a signature is just one sort of module_type. Another would be the name of an already defined module type—e.g., module type LS = LIST_STACK. We'll see other module types later in this section and chapter.

By convention, module type names are usually CamelCase, like module names. So why did we use ALL_CAPS above for LIST_STACK? It was to avoid a possible point of confusion in that example, which we now illustrate. We could instead have used ListStack as the name of both the module and the module type:

```
module type ListStack = sig ... end
module ListStack : ListStack = struct ... end
```

In OCaml the namespaces for modules and module types are distinct, so it's perfectly valid to have a module named ListStack and and module type named ListStack. The compiler will not get confused about which you mean, because they occur in distinct syntactic contexts. But as a human you might well get confused by those seemingly overloaded names.

Note: The use of ALL_CAPS for module types was at one point common, and you might see it still. It's an older convention from Standard ML. But the social conventions of all caps have changed since those days. To modern readers, a name like LIST_STACK might feel like your code is impolitely shouting at you. That is a connotation that evolved in the 1980s. Older programming languages (e.g., Pascal, COBOL, FORTRAN) commonly used all caps for keywords and even their own names. Modern languages still idiomatically use all caps for constants—see, for example, Java's Math.PI or Python's style guide.

More Syntax.

We should also add syntax now for module type annotations. Module definitions may include an optional type annotation:

```
module ModuleName : module_type = module_expression
```

And module expressions may include manual type annotations:

```
(module_expression : module_type)
```

That syntax is analogous to how we can write (e : t) to manually specify the type t of an expression e.

Here are a few examples to show how that syntax can be used:

```
module ListStackAlias : LIST_STACK = ListStack
  (* equivalently *)
module ListStackAlias = (ListStack : LIST_STACK)

module M : sig val x : int end = struct let x = 42 end
  (* equivalently *)
module M = (struct let x = 42 end : sig val x : int end)
```

And, module types can include nested module specifications:

```
module type X = sig
  val x : int
end

module type T = sig
  module Inner : X
end

module M : T = struct
  module Inner : X = struct
  let x = 42
  end
end
```

In the example above, T specifies that there must be an inner module named Inner whose module type is X. Here, the type annotation is mandatory, because otherwise nothing would be known about Inner. In implementing T, module M therefore has to provide a module (i) with that name, which also (ii) meets the specifications of module type X.

Dynamic semantics.

Since modules types are in fact types, they are not evaluated. They have no dynamic semantics.

Static semantics.

Earlier in this section we delayed discussing the static semantics of module expressions. Now that we have learned about module types, we can return to that discussion. We do so, next, in its own section, because the discussion will be lengthy.

5.2.4 Module Type Semantics

The key question we have to answer is: what does a type annotation mean for modules? That is, what does it mean when we write the: T in module M: $T = \dots$?

The answer turns out to involve *subtyping*, which is a concept you've probably seen before in an object-oriented language. We're going to take a brief detour into that realm now, then come back to OCaml and modules.

In Java, the extends keyword creates subtype relationships between classes:

```
class C { }
class D extends C { }

D d = new D();
C c = d;
```

Subtyping is what permits the assignment of d to c on the last line of that example. Because D extends C, Java considers D to be a subtype of C, and therefore permits an object instantiated from D to be used any place where an object instantiated from C is expected. It's up to the programmer of D to ensure that doesn't lead to any run-time errors, of course. The methods of D have to ensure that class invariants of C hold, for example. So by writing D extends C, the programmer is taking on some responsibility, and in turn gaining some flexibility by being able to write such assignment statements.

So what is a "subtype"? That notion is in many ways dependent on the language. For a language-independent notion, we turn to Barbara Liskov. She won the Turing Award in 2008 in part for her work on object-oriented language design. Twenty years before that, she invented what is now called the *Liskov Substitution Principle* to explain subtyping. It says that if S is a subtype of T, then substituting an object of type T for an object of type T should not change any desirable behaviors of a program. You can see that at work in the Java example above, both in terms of what the language allows and what the programmer must guarantee.

The particular flavor of subtyping in Java is called *nominal subtyping*, which is to say, it is based on names. In our example, $\mathbb D$ is a subtype of $\mathbb C$ just because of the way the names were declared. The programmer decreed that subtype relationship, and the language accepted the decree without question. Indeed the *only* subtype relationships that exist are those that have been decreed by name through such uses of extends and implements.

Now it's time to return to OCaml. Its module system also uses subtyping, with the same underlying intuition about the Liskov Substitution Principle. But OCaml uses a different flavor called *structural subtyping*. That is, it is based on the structure of modules rather than their names. "Structure" here simply means the definitions contained in the module. That structure is —i.e., those definitions are—what is used to determine whether (M:T) is acceptable as a type annotation, where M is a module and T is a module type.

Let's play with this idea of structure through several examples, starting with this module:

```
module M = struct
  let x = 0
  let z = 2
end
```

```
module M : sig val x : int val z : int end
```

Module M contains two definitions. You can see those in the signature for the module that OCaml outputs: it contains x: int and z: int. Because of the former, the module type annotation below is accepted:

```
module type X = sig
  val x : int
end

module MX = (M : X)
```

```
module type X = sig val x : int end
```

```
module MX : X
```

Module type X requires a module item named x with type int. Module M does contain such an item. So (M : X) is valid. The same would work for z:

```
module type Z = sig
  val Z : int
end

module MZ = (M : Z)
```

```
module type Z = sig val z : int end
```

```
module MZ : Z
```

Or for both x and z:

```
module type XZ = sig
  val x : int
  val z : int
end

module MXZ = (M : XZ)
```

```
module type XZ = sig val x : int val z : int end
```

```
module MXZ : XZ
```

But not for y, because M contains no such item:

```
module type Y = sig
  val y : int
end
module MY = (M : Y)
```

```
module type Y = sig val y : int end
```

Take a close look at that error message. Learning to read such errors on small examples will help you when they appear in large bodies of code. OCaml is comparing two signatures, corresponding to the two expressions on either side of the colon in (M: Y). The line

```
sig val x : int val z : int end
```

is the signature that OCaml is using for M. Since M is a module, that signature is just the names and types as they were defined in M. OCaml compares that signature to Y, and discovers a mismatch:

```
The value `y' is required but not provided
```

That's because Y requires y but M provides no such definition.

Here's another error message to practice reading:

```
module type Xstring = sig
  val x : string
end

module MXstring = (M : Xstring)
```

```
module type Xstring = sig val x : string end
```

```
File "[36]", line 5, characters 19-20:
5 | module MXstring = (M : Xstring)

Error: Signature mismatch:
    Modules do not match:
        sig val x : int val z : int end
    is not included in
        Xstring
    Values do not match: val x : int is not included in val x : string
    File "[36]", line 2, characters 2-16: Expected declaration
    File "[31]", line 2, characters 6-7: Actual declaration
```

This time the error is

```
Values do not match: val x : int is not included in val x : string
```

The error changed, because M does provide a definition of x, but at a different type than Xstring requires. That's what "is not included in" means here. So why doesn't OCaml say something a little more straightforward, like "is not the same as"? It's because the types do not have to be exactly the same. If the provided value's type is polymorphic, it suffices for the required value's type to be an instantiation of that polymorphic type.

For example, if a signature requires a type int -> int, it suffices for a structure to provide a value of type 'a -> 'a:

```
module type IntFun = sig
  val f : int -> int
end

module IdFun = struct
  let f x = x
end

module Iid = (IdFun : IntFun)
```

```
module type IntFun = sig val f : int -> int end
```

```
module IdFun : sig val f : 'a -> 'a end
```

```
module Iid : IntFun
```

So far all these examples were just a matter of comparing the definitions required by a signature to the definitions provided by a structure. But here's an example that might be surprising:

```
module MXZ' = ((M : X) : Z)
```

Why does OCaml complain that z is required but not provided? We know from the definition of M that it indeed does have a value z: int. Yet the error message perhaps strangely claims:

```
The value `z' is required but not provided.
```

The reason for this error is that we've already supplied the type annotation X in the module expression (M: X). That causes the module expression to be known only at the module type X. In other words, we've forgotten irrevocably about the existence of Z after that annotation. All that is known is that the module has items required by X.

After all those examples, here are the static semantics of module type annotations:

- Module type annotation (M : T) is valid if the module type of M is a subtype of T. The module type of (M : T) is then T in any further type checking.
- Module type S is a subtype of T if the set of definitions in S is a superset of those in T. Definitions in T are permitted to instantiate type variables from S.

The "sub" vs. "super" in the second rule is not a typo. Think about T as a set of module values. There could be many subtypes S1, S2, etc. of T, each of which is a set of module values, and each of those sets would be a subset of T. But, the way those sets differ from T and one another is by adding new definitions of their own. For example, a list-based stack might differ from ListStack by adding a value to keep track of the number of items on the stack. Another implementation might add a value to keep track of how many items have ever been pushed on the stack in its history. Those additional values inside the module make its set of definitions bigger, hence the use of "superset" in the rule.

If M is just a plain old structure, its module type is whatever signature the compiler infers for it. But that can be restricted, as we saw above, by module type annotations. All of this is *static*, which is to say the decisions about validity are made at compile time rather than run time.

Important: Module type annotations therefore offer potential confusion to programmers accustomed to object-oriented languages, in which subtyping works differently.

Python programmers, for example, are accustomed to so-called "duck typing". They might expect ((M : X) : Z) to be valid, because z does exist at run-time in M. But in OCaml, the compile-time type of (M : X) has hidden z from view irrevocably.

Java programmers, on the other hand, might expect that module type annotations work like type casts. So it might seem valid to first "cast" \mathbb{M} to \mathbb{X} then to \mathbb{Z} . In Java such type casts are checked, as needed, at run time. But OCaml module type annotations are static. Once an annotation of \mathbb{X} is made, there is no way to check at compile time what other items might exist in the module—that would require a run-time check, which OCaml does not permit.

In both cases it might feel as though OCaml is being too restrictive. Maybe. But in return for that restrictiveness, OCaml is guaranteeing an **absence of run-time errors** of the kind that would occur in Java or Python, whether because of a run-time error from a cast, or a run-time error from a missing method.

In summary, if a module is given a module type, such as in module M : T = struct ... end, then there are two properties the compiler guarantees:

- 1. Signature matching: every name declared in T is defined in M at the same or a more general type.
- 2. Opacity: any name defined in M that does not appear in T is not visible to code outside of M.

5.2.5 First-Class Modules

Modules are not as first-class in OCaml as functions. But it is possible to *package* modules as first-class values. Briefly:

- (module M : T) packages module M with module type T into a value.
- (val e : T) un-packages e into a module with type T.

We won't cover this any further, but if you're curious you can have a look at the manual.

5.3 Modules and the Toplevel

There are several pragmatics involving modules and the toplevel that are important to master to use the two together effectively.

5.3.1 Loading Compiled Modules

Compiling an OCaml file produces a module having the same name as the file, but with the first letter capitalized. These compiled modules can be loaded into the toplevel using #load.

For example, suppose you create a file called mods.ml, and put the following code in it:

```
let b = "bigred"
let inc x = x + 1
module M = struct
let y = 42
end
```

Note that there is no module Mods = struct ... end around that. The code is at the topmost level of the file, as it were.

Then suppose you type ocamlbuild mods.byte to compile it. Inside the _build directory you will now find the files that ocamlbuild produced. One of them is mods.cmo: this is a compiled module object file, aka bytecode.

You can make this bytecode available for use in the toplevel with the following directives. Recall that the # character is required in front of a directive. It is not part of the prompt.

```
# #directory "_build";;
# #load "mods.cmo";;
```

The first directive tells utop to add the _build directory to the path in which it looks for compiled (and source) files. The second directive loads the bytecode found in mods.cmo, thus making a module named Mods available to be used. It exactly as if you had entered this code:

```
module Mods = struct
  let b = "bigred"
  let inc x = x + 1
  module M = struct
   let y = 42
  end
end
```

```
module Mods :
   sig val b : string val inc : int -> int module M : sig val y : int end end
```

Both of these expressions will therefore evaluate successfully:

```
Mods.b;;
Mods.M.y;;
```

```
- : string = "bigred"
```

```
- : int = 42
```

But this will fail:

```
inc
```

It fails because inc is in the namespace of Mods.

```
Mods.inc
```

```
- : int -> int = <fun>
```

Of course, if you open the module, you can directly name inc:

```
open Mods;;
inc;;
```

```
- : int -> int = <fun>
```

5.3.2 Initializing the Toplevel

If you are doing a lot of testing of a particular module, it can be annoying to have to type those directives (#directory and #load) every time you start utop. You really want to initialize the toplevel with some code as it launches, so that you don't have to keep typing that code.

The solution is to create a file in the working directory and call that file .ocamlinit. Note that the . at the front of that filename is required and makes it a hidden file that won't appear in directory listings unless explicitly requested (e.g., with ls -a). Everything in .ocamlinit will be processed by utop when it loads.

For example, suppose you create a file named .ocamlinit in the same directory as mods.ml, and in that file put the following code:

```
#directory "_build";;
#load "mods.cmo";;
open Mods
```

Now restart utop. All the names defined in Mods will already be in scope. For example, these will both succeed:

```
inc;;
M.y;;
```

```
- : int -> int = <fun>
```

```
-: int = 42
```

5.3.3 Requiring Libraries

Suppose you were to add the following lines to the end of mods.ml:

```
open OUnit2
let test = "testb" >:: (fun _ -> assert_equal "bigred" b)
```

If you try to recompile the module with <code>ocamlbuild</code> mods.byte, it will fail. The problem is that you need to tell the build system to include the third-party library OUnit. Recompiling with <code>ocamlbuild</code> <code>-pkg</code> <code>ounit2</code> mods.byte will, as usual, succeed.

But if you restart utop, there will be an error message:

```
File ".ocamlinit", line 1:
Error: Reference to undefined global `OUnit2'
```

The problem is that the OUnit library hasn't been loaded into utop yet. It can be with the following directive:

```
#require "ounit2";;
```

Now you can successfully load your own module without getting an error.

```
#load "mods.cmo";;
```

Moreover, if you add that #require directive to .ocamlinit anywhere before the #load directive, the "undefined global" error will go away.

5.3.4 Dependencies

When compiling a file, the build system automatically figures out which other files it depends on and recompiles those as necessary. The toplevel, however, is not as sophisticated: you have to make sure to load all the dependencies of a file.

Suppose you have a file named mods2.ml in the same directory as mods.ml from above, and mods2.ml contains this code:

```
open Mods
let x = inc 0
```

If you run ocambuild -pkg ounit2 mods2.byte, the compilation will succeed. You don't have to name mods.byte on the command line, even though mods2.ml depends on the module Mod. The build system is smart that way.

Also suppose that .ocamlinit contains exactly the following:

```
#directory "_build";;
#require "ounit2";;
```

If you restart utop and try to load mods 2. cmo, you will get an error:

```
# #load "mods2.cmo";;
Error: Reference to undefined global `Mods'
```

The problem is that the toplevel does not automatically load the modules that Mods2 depends upon. There are two ways to solve this problem. First, you can manually load the dependencies, like this:

```
# #load "mods.cmo";;
# #load "mods2.cmo";;
```

Second, you could instead tell the toplevel to load Mods2 and recursively to load everything it depends on:

```
# #load_rec "mods2.cmo";;
```

And that is probably the better solution.

5.3.5 Load vs Use

There is a big difference between #load-ing a compiled module file and #use-ing an uncompiled source file. The former loads bytecode and makes it available for use. For example, loading mods.cmo caused the Mod module to be available, and we could access its members with expressions like Mod.b. The latter (#use) is *textual inclusion*: it's like typing the contents of the file directly into the toplevel. So using mods.ml does **not** cause a Mod module to be available, and the definitions in the file can be accessed directly, e.g., b.

For example, in the following interaction, we can directly refer to b but cannot use the qualified name Mods.b:

```
# #use "mods.ml"

# b;;
val b : string = "bigred"

# Mods.b;;
Error: Unbound module Mods
```

Whereas in this interaction the situation is reversed:

```
# #directory "_build";;
# #load "mods.cmo";;

# Mods.b;;
- : string = "bigred"

# b;;
Error: Unbound value b
```

So when you're using the toplevel to experiment with your code, it's often better to work with #load rather than #use. The #load directive accurately reflects how your modules interact with each other and with the outside world.

5.4 Encapsulation

One of the main concerns of a module system is to provide *encapsulation*: the hiding of information about implementation behind an interface. OCaml's module system makes this possible with a feature we've already seen: the *opacity* that module type annotations create. One special use of opacity is the declaration of *abstract types*. We'll study both of those ideas in this section.

5.4.1 Opacity

When implementing a module, you might sometimes have helper functions that you don't want to expose to clients of the module. For example, maybe you're implementing a math module that provides a tail-recursive factorial function:

```
module Math = struct
  (** [fact_aux n acc] is [n! * acc]. *)
  let rec fact_aux n acc =
    if n = 0 then acc else fact_aux (n - 1) (n * acc)

  (** [fact n] is [n!]. *)
  let fact n = fact_aux n 1
end
```

```
module Math : sig val fact_aux : int -> int -> int val fact : int -> int end
```

You'd like to make fact usable by clients of Math, but you'd also like to keep fact_aux hidden. But in the code above, you can see that fact_aux is visible in the signature inferred for Math. One way to hide it is simply to nest fact_aux:

```
module Math = struct
  (** [fact n] is [n!]. *)
  let fact n =
     (** [fact_aux n acc] is [n! * acc]. *)
    let rec fact_aux n acc =
        if n = 0 then acc else fact_aux (n - 1) (n * acc)
     in
     fact_aux n 1
end
```

```
module Math : sig val fact : int -> int end
```

Look at the signature, and notice how fact_aux is gone. But, that nesting makes fact just a little harder to read. It also means fact_aux is not available for any other functions <code>inside</code> Math to use. In this case that's probably fine—there probably aren't any other functions in Math that need fact_aux. But if there were, we couldn't nest fact_aux.

So another way to hide fact_aux from clients of Math, while still leaving it available for implementers of Math, is to use a module type that exposes only those names that clients should see:

```
module type MATH = sig
    (** [fact n] is [n!]. *)
    val fact : int -> int
end

module Math : MATH = struct
    (** [fact_aux n acc] is [n! * acc]. *)
    let rec fact_aux n acc =
```

(continues on next page)

```
if n = 0 then acc else fact_aux (n - 1) (n * acc)
let fact n = fact_aux n 1
end
```

```
module type MATH = sig val fact : int -> int end
```

```
module Math : MATH
```

Now since MATH does not mention fact_aux, the module type annotation Math : MATH causes fact_aux to be hidden:

```
Math.fact_aux
```

In that sense, module type annotations are *opaque*: they can prevent visibility of module items. We say that the module type *seals* the module, making any components not named in the module type be inaccessible.

Important: Remember that module type annotations are therefore not *only* about checking to see whether a module defines certain items. The annotations also hide items.

What if you did want to just check the definitions, but not hide anything? Then don't supply the annotation at the time of module definition:

```
module type MATH = sig
    (** [fact n] is [n!]. *)
    val fact : int -> int
end

module Math = struct
    (** [fact_aux n acc] is [n! * acc]. *)
    let rec fact_aux n acc =
        if n = 0 then acc else fact_aux (n - 1) (n * acc)

let fact n = fact_aux n 1
end

module MathCheck : MATH = Math
```

```
module type MATH = sig val fact : int -> int end
```

```
module Math : sig val fact_aux : int -> int -> int val fact : int -> int end
```

```
module MathCheck : MATH
```

Now Math.fact_aux is visible, but MathCheck.fact_aux is not:

```
Math.fact_aux
```

```
- : int -> int -> int = <fun>
```

```
MathCheck.fact_aux
```

You wouldn't even have to give the "check" module a name since you probably never intend to access it; you could instead leave it anonymous:

```
module _ : MATH = Math
```

A Comparison to Visibility Modifiers. The use of sealing in OCaml is thus similar to the use of visibility modifiers such as private and public in Java. In fact one way to think about Java class definitions is that they simultaneously define multiple signatures.

For example, consider this Java class:

```
class C {
  private int x;
  public int y;
}
```

An analogy to it in OCaml would be the following modules and types:

```
module type C_PUBLIC = sig
  val y : int
end

module CPrivate = struct
  let x = 0
  let y = 0
end

module C : C_PUBLIC = CPrivate
```

```
module type C_PUBLIC = sig val y : int end
```

```
module CPrivate : sig val x : int val y : int end
```

```
module C : C_PUBLIC
```

With those definitions, any code that uses C will have access only to the names exposed in the C_PUBLIC module type.

That analogy can be extended to the other visibility modifiers, protected and default, as well. Which means that Java classes are effectively defining four related types, and the compiler is making sure the right type is used at each place in the code base C is named. No wonder it can be challenging to master visibility in OO languages at first.

5.4.2 Abstract Types

In an earlier section we implemented stacks as lists with the following module and type:

```
module type LIST STACK = sig
  (** [Empty] is raised when an operation cannot be applied
     to an empty stack. *)
 exception Empty
  (** [empty] is the empty stack. *)
 val empty : 'a list
  (** [is_empty s] is whether [s] is empty. *)
 val is_empty : 'a list -> bool
  (** [push x s] pushes [x] onto the top of [s]. *)
 val push : 'a -> 'a list -> 'a list
  (** [peek s] is the top element of [s].
     Raises [Empty] if [s] is empty. *)
 val peek : 'a list -> 'a
  (** [pop s] is all but the top element of [s].
      Raises [Empty] if [s] is empty. *)
 val pop : 'a list -> 'a list
module ListStack : LIST STACK = struct
 exception Empty
 let empty = []
 let is_empty = function [] -> true | _ -> false
 let push x s = x :: s
 let peek = function [] -> raise Empty | x :: _ -> x
 let pop = function [] -> raise Empty | _ :: s -> s
```

What if we wanted to modify that data structure to add an operation for the size of the stack? The easy way would be to implement it using List.length:

```
module type LIST_STACK = sig
...
  (** [size s] is the number of elements on the stack. *)
  val size : 'a list -> int
end

module ListStack : LIST_STACK = struct
...
  let size = List.length
end
```

That results in a linear-time implementation of size. What if we wanted a faster, constant-time implementation? At the cost of a little space, we could cache the size of the stack. Let's now represent the stack as a pair, where the first component of the pair is the same list as before, and the second component of the pair is the size of the stack:

5.4. Encapsulation 165

We have a big problem. ListStackCachedSize does not implement the LIST_STACK module type, because that module type specifies 'a list throughout it to represent the stack—not 'a list * int.

```
module CheckListStackCachedSize : LIST_STACK = ListStackCachedSize
```

Moreover, any code we previously wrote using ListStack now has to be modified to deal with the pair, which could mean revising pattern matches, function types, and so forth.

As you no doubt learned in earlier programming courses, the problem we are encountering here is a lack of encapsulation. We should have kept the type that implements ListStack hidden from clients. In Java, for example, we might have written:

```
class ListStack<T> {
  private List<T> stack;
  private int size;
  ...
}
```

That way clients of ListStack would be unaware of stack or size. In fact, they wouldn't be able to name those fields at all. Instead, they would just use ListStack as the type of the stack:

```
ListStack<Integer> s = new ListStack<>();
s.push(1);
```

So in OCaml, how can we keep the *representation type* of the stack hidden? What we learned about opacity and sealing thus far does not suffice. The problem is that the type 'a list * int literally appears in the signature of ListStackCachedSize, e.g. in push:

```
ListStackCachedSize.push
```

```
- : 'a -> 'a list * int -> 'a list * int = <fun>
```

A module type annotation could hide one of the values defined in ListStackCachedSize, e.g., push itself, but that doesn't solve the problem: we need to hide the type 'a list * int while exposing the operation push. So OCaml has a feature for doing exactly that: abstract types. Let's see an example of this feature.

We begin by modifying LIST_STACK, replacing 'a list with a new type 'a stack everywhere. We won't repeat the specification comments here, so as to keep the example shorter. And while we're at it, let's add the size operation.

```
module type LIST_STACK = sig
  type 'a stack
  exception Empty
  val empty : 'a stack
  val is_empty : 'a stack -> bool
  val push : 'a -> 'a stack -> 'a stack
  val peek : 'a stack -> 'a
  val pop : 'a stack -> 'a stack
  val size : 'a stack -> int
end
```

Note how 'a stack is not actually defined in that signature. We haven't said anything about what it is. It might be 'a list, or 'a list * int, or {stack : 'a list; size : int}, or anything else. That is what makes it an *abstract* type: we've declared its name but not specified its definition.

Now ListStackCachedSize can implement that module type with the addition of just one line of code: the first line of the structure, which defines 'a stack:

```
module ListStackCachedSize : LIST_STACK
```

Take a careful look at the output: nowhere does 'a list show up in it. In fact, only LIST_STACK does. And LIST_STACK mentions only 'a stack. So no one's going to know that internally a list is used. (Ok, they're going to know: the name suggests it. But the point is they can't take advantage of that, because the type is abstract.)

Likewise, our original implementation with linear-time size satisfies the module type. We just have to add a line to define 'a stack:

```
module ListStack : LIST_STACK = struct
  type 'a stack = 'a list
  exception Empty
  let empty = []
  let is_empty = function [] -> true | _ -> false
  let push x s = x :: s
  let peek = function [] -> raise Empty | x :: _ -> x
  let pop = function [] -> raise Empty | _ :: s -> s
  let size = List.length
end
```

```
module ListStack : LIST_STACK
```

Note that omitting that added line would result in an error, just as if we had failed to define push or any of the other operations from the module type:

```
module ListStack : LIST_STACK = struct
    (* type 'a stack = 'a list *)
    exception Empty
    let empty = []
    let is_empty = function [] -> true | _ -> false
    let push x s = x :: s
    let peek = function [] -> raise Empty | x :: _ -> x
    let pop = function [] -> raise Empty | _ :: s -> s
    let size = List.length
end
```

```
File "[17]", lines 1-10, characters 32-3:
1 | ......struct
2 | (* type 'a stack = 'a list *)
3 | exception Empty
4 | let empty = []
5 | let is_empty = function [] -> true | _ -> false
6 | let push x s = x :: s
7 | let peek = function [] -> raise Empty | x :: _ -> x
8 | let pop = function [] -> raise Empty | _ :: s -> s
9 | let size = List.length
10 | end
Error: Signature mismatch:
...
The type `stack' is required but not provided
File "[14]", line 2, characters 2-15: Expected declaration
```

Here is a third, custom implementation of LIST_STACK. This one is deliberately overly-complicated, in part to illustrate how the abstract type can hide implementation details that are better not revealed to clients:

```
module CustomStack : LIST_STACK = struct
   type 'a entry = {top : 'a; rest : 'a stack; size : int}
   and 'a stack = S of 'a entry option
   exception Empty
   let empty = S None
   let is_empty = function S None -> true | _ -> false
   let size = function S None -> 0 | S (Some {size}) -> size
   let push x s = S (Some {top = x; rest = s; size = size s + 1})
   let peek = function S None -> raise Empty | S (Some {top}) -> top
   let pop = function S None -> raise Empty | S (Some {rest}) -> rest
end
```

```
module CustomStack : LIST_STACK
```

Is that really a "list" stack? It satisfies the module type LIST_STACK. But upon reflection, that module type never really had anything to do with lists once we made the type 'a stack abstract. There's really no need to call it LIST_STACK. We'd be better off using just STACK, since it can be implemented with list or without. At that point, we could just go with Stack as its name, since there is no module named Stack we've written that would be confused with it. That avoids the all-caps look of our code shouting at us.

```
module type Stack = sig
  type 'a stack
  exception Empty
  val empty : 'a stack
  val is_empty : 'a stack -> bool
  val push : 'a -> 'a stack -> 'a stack
```

(continues on next page)

```
val peek : 'a stack -> 'a
val pop : 'a stack -> 'a stack
val size : 'a stack -> int
end

module ListStack : Stack = struct
  type 'a stack = 'a list
  exception Empty
  let empty = []
  let is_empty = function [] -> true | _ -> false
  let push x s = x :: s
  let peek = function [] -> raise Empty | x :: _ -> x
  let pop = function [] -> raise Empty | _ :: s -> s
  let size = List.length
end
```

There's one further naming improvement we could make. Notice the type of ListStack.empty (and don't worry about the abstr part for now; we'll come back to it):

```
ListStack.empty
```

```
- : 'a ListStack.stack = <abstr>
```

That type, 'a ListStack.stack, is rather unwieldy, because it conveys the word "stack" twice: once in the name of the module, and again in the name of the representation type inside that module. In places like this, OCaml programmers idiomatically use a standard name, t, in place of a longer representation type name:

```
module type Stack = sig
 type 'a t
 exception Empty
 val empty : 'a t
 val is_empty : 'a t -> bool
 val push : 'a -> 'a t -> 'a t
 val peek : 'a t -> 'a
 val pop : 'a t -> 'a t
 val size : 'a t -> int
end
module ListStack : Stack = struct
 type 'a t = 'a list
 exception Empty
 let empty = []
 let is_empty = function [] -> true | _ -> false
 let push x s = x :: s
 let peek = function [] -> raise Empty | x :: _ -> x
 let pop = function [] -> raise Empty | _ :: s -> s
 let size = List.length
end
module CustomStack : Stack = struct
 type 'a entry = {top : 'a; rest : 'a t; size : int}
 and 'a t = S of 'a entry option
 exception Empty
 let empty = S None
 let is_empty = function S None -> true | _ -> false
 let size = function S None -> 0 | S (Some {size}) -> size
```

(continues on next page)

```
let push x s = S (Some {top = x; rest = s; size = size s + 1})
let peek = function S None -> raise Empty | S (Some {top}) -> top
let pop = function S None -> raise Empty | S (Some {rest}) -> rest
end
```

Now the type of stacks is simpler:

```
ListStack.empty;;
CustomStack.empty;;
```

```
- : 'a ListStack.t = <abstr>
```

```
- : 'a CustomStack.t = <abstr>
```

That idiom is fairly common when there's a single representation type exposed by an interface to a data structure. You'll see it used throughout the standard library.

In informal conversation we would usually pronounce those types without the "dot t" part. For example, we might say "alpha ListStack", simply ignoring the t—though it does technically have to be there to be legal OCaml code.

Finally, abstract types are really just a special case of opacity. You actually can expose the definition of a type in a signature if you want to:

```
module type T = sig
  type t = int
  val x : t
end

module M : T = struct
  type t = int
  let x = 42
end

let a : int = M.x
```

```
module type T = sig type t = int val x : t end
```

```
module M : T
```

```
val a : int = 42
```

Note how we're able to use M.x at its type of int. That works because the equality of types t and int has been exposed in the module type. But if we kept t abstract, the same usage would fail:

```
module type T = sig
  type t (* = int *)
  val x : t
end

module M : T = struct
  type t = int
  let x = 42
end

let a : int = M.x
```

```
module type T = sig type t val x : t end
```

```
module M : T
```

We're not allowed to use M.x at type int outside of M, because its type M.t is abstract. This is encapsulation at work, keeping that implementation detail hidden.

5.4.3 Pretty Printing

In some output above, we observed something curious: the toplevel prints <abstr> in place of the actual contents of a value whose type is abstract:

```
ListStack.empty;;
ListStack.(empty |> push 1 |> push 2);;
```

```
- : 'a ListStack.t = <abstr>
```

```
- : int ListStack.t = <abstr>
```

Recall that the toplevel uses this angle-bracket convention to indicate an unprintable value. We've encountered that before with functions and <fun>:

```
fun x -> x
```

```
- : 'a -> 'a = <fun>
```

On the one hand, it's reasonable for the toplevel to behave this way. Once a type is abstract its implementation isn't meant to be revealed to clients. So actually printing out the list [] or [2; 1] as responses to the above inputs would be revealing more than is intended.

On the other hand, it's also reasonable for implementers to provide clients with a friendly way to view a value of an abstract type. Java programmers, for example, will often write toString() methods so that objects can be printed as output in the terminal or in JShell. To support that, the OCaml toplevel has a directive #install_printer, which registers a function to print values. Here's how it works.

- You write a *pretty printing* function of type Format.formatter -> t -> unit, for whatever type t you like. Let's suppose for sake of example that you name that function pp.
- You invoke #install printer pp in the toplevel.
- From now on, anytime the toplevel wants to print a value of type t it uses your function pp to do so.

It probably makes sense the pretty printing function needs to take in a value of type t (because that's what it needs to print) and returns unit (as other printing functions do). But why does it take the Format.formatter argument? It's because of a fairly high-powered feature that OCaml is attempting to provide here: automatic line breaking and indentation in the middle of very large outputs.

Consider the output from this expression, which creates nested lists:

5.4. Encapsulation 171

```
List.init 15 (fun n -> List.init n (Fun.const n))
```

Each inner list contains n copies of the number n. Note how the indentation and line breaks are somewhat sophisticated. All the inner lists are indented one space from the left-hand margin. Line breaks have been inserted to avoid splitting inner lists over multiple lines.

The Format module is what provides this functionality, and Format.formatter is an abstract type in it. You could think of a formatter as being a place to send output, like a file, and have it be automatically formatted along the way. The typical use of a formatter is as argument to a function such as Format.fprintf, which like Printf uses format specifiers.

For example, suppose you wanted to change how strings are printed by the toplevel and add "kupo" to the end of each string. Here's code that would do it:

```
let kupo_pp fmt s = Format.fprintf fmt "%s kupo" s;;
#install_printer kupo_pp;;
```

```
val kupo_pp : Format.formatter -> string -> unit = <fun>
```

Now you can see that the toplevel adds "kupo" to each string while printing it, even though it's not actual a part of the original string:

```
let h = "Hello"
let s = String.length h
```

```
val h : string = Hello kupo
```

```
val s : int = 5
```

To keep ourselves from getting confused about strings in the rest of this section, let's uninstall that pretty printer before going on:

```
#remove_printer kupo_pp;;
```

As a bigger example, let's add pretty printing to ListStack:

```
module type Stack = sig
  type 'a t
  exception Empty
  val empty : 'a t
  val is_empty : 'a t -> bool
  val push : 'a -> 'a t -> 'a t
  val peek : 'a t -> 'a
  val pop : 'a t -> 'a t
  val size : 'a t -> int
  val pp :
```

(continues on next page)

173

```
(Format.formatter -> 'a -> unit) -> Format.formatter -> 'a t -> unit
end
```

First, notice that we have to expose pp as part of the module type. Otherwise it would be encapsulated, hence we wouldn't be able to install it. Second, notice that the type of pp now takes an extra first argument of type Format.formatter -> 'a -> unit. That is itself a pretty printer for type 'a, on which t is parameterized. We need that argument in order to be able to pretty print the values of type 'a.

```
module ListStack : Stack = struct
 type 'a t = 'a list
 exception Empty
 let empty = []
 let is_empty = function [] -> true | _ -> false
 let push x s = x :: s
 let peek = function [] \rightarrow raise Empty | x :: _ \rightarrow x
 let pop = function [] -> raise Empty | _ :: s -> s
 let size = List.length
 let pp pp_val fmt s =
   let open Format in
   let pp_break fmt () = fprintf fmt "@," in
   fprintf fmt "@[<v 0>top of stack";
   if s <> [] then fprintf fmt "@,";
   pp_print_list ~pp_sep:pp_break pp_val fmt (List.rev s);
    fprintf fmt "@, bottom of stack@]"
end
```

In ListStack.pp, we use some of the advanced features of the Format module. Function Format. pp_print_list does the heavy lifting to print all the elements of the stack. The rest of the code handles the indentation and line breaks. Here's the result:

```
#install_printer ListStack.pp
```

```
ListStack.empty
```

```
- : 'a ListStack.t = top of stack

bottom of stack
```

```
ListStack. (empty |> push 1 |> push 2)
```

```
-: int ListStack.t = top of stack

1

2

bottom of stack
```

For more information, see the toplevel manual (search for #install_printer), the Format module, and this OCaml Github issue. The latter seems to be the only place that documents the use of extra arguments, as in pp_val above, to print values of polymorphic types.

5.4. Encapsulation

5.5 Functional Data Structures

A functional data structure is one that does not make use of any imperative features. That is, no operations of the data structure have any side effects. It's possible to build functional data structures both in functional languages and in imperative languages.

Functional data structures have the property of being *persistent*: updating the data structure with one of its operations does not change the existing version of the data structure but instead produces a new version. Both exist and both can still be accessed. A good language implementation will ensure that any parts of the data structure that are not changed by an operation will be *shared* between the old version and the new version. Any parts that do change will be *copied* so that the old version may persist. The opposite of a persistent data structure is an *ephemeral* data structure: changes are destructive, so that only one version exists at any time. Both persistent and ephemeral data structures can be built in both functional and imperative languages.

5.5.1 Lists

The built-in singly-linked list data structure in OCaml is functional. We know that, because we've seen how to implement it with algebraic data types. It's also persistent, which we can demonstrate:

```
let lst = [1; 2];;
let lst' = List.tl lst;;
lst';;
```

```
val lst : int list = [1; 2]
```

```
val lst' : int list = [2]
```

```
- : int list = [2]
```

Taking the tail of 1st does not change the list. Both 1st and 1st' coexist without affecting one another.

5.5.2 Stacks

We implemented stacks earlier in this chapter. Here's a terse variant of one of those implementations, in which add a to_list operation to make it easier to view the contents of the stack in examples:

```
module type Stack = sig
  type 'a t
  exception Empty
  val empty : 'a t
  val is_empty : 'a t -> bool
  val push : 'a -> 'a t -> 'a t
  val peek : 'a t -> 'a
  val pop : 'a t -> 'a t
  val size : 'a t -> int
  val to_list : 'a t -> 'a list
end

module ListStack : Stack = struct
  type 'a t = 'a list
  exception Empty
  let empty = []
```

(continues on next page)

```
let is_empty = function [] -> true | _ -> false
let push = List.cons
let peek = function [] -> raise Empty | x :: _ -> x
let pop = function [] -> raise Empty | _ :: s -> s
let size = List.length
let to_list = Fun.id
end
```

That implementation is functional, as can be seen above, and also persistent:

```
open ListStack;;
let s = empty |> push 1 |> push 2;;
let s' = pop s;;
to_list s;;
to_list s';;
```

```
val s : int ListStack.t = <abstr>
```

```
val s' : int ListStack.t = <abstr>
```

```
- : int list = [2; 1]
```

```
- : int list = [1]
```

The value s is unchanged by the pop operation on s. Both versions of the stack coexist.

The Stack module type gives us a strong hint that the data structure is persistent in the types it provides for push and pop:

```
val push : 'a -> 'a t -> 'a t
val pop : 'a t -> 'a t
```

Both of those take a stack as an argument and return a new stack as a result. An ephemeral data structure usually would not bother to return a stack. In Java, for example, similar methods might have a void return type; the equivalent in OCaml would be returning unit.

5.5.3 Options vs Exceptions

All of our stack implementations so far have raised an exception whenever peek or pop is applied to the empty stack. Another possibility would be to use an option for the return value. If the input stack is empty, then peek and pop return None; otherwise, they return Some.

```
module type Stack = sig
  type 'a t
  val empty : 'a t
  val is_empty : 'a t -> bool
  val push : 'a -> 'a t -> 'a t
  val peek : 'a t -> 'a option
  val pop : 'a t -> 'a t option
  val size : 'a t -> int
  val to_list : 'a t -> 'a list
end
```

```
module ListStack : Stack = struct
  type 'a t = 'a list
  exception Empty
  let empty = []
  let is_empty = function [] -> true | _ -> false
  let push = List.cons
  let peek = function [] -> None | x :: _ -> Some x
  let pop = function [] -> None | _ :: s -> Some s
  let size = List.length
  let to_list = Fun.id
end
```

But that makes it harder to pipeline:

```
ListStack. (empty |> push 1 |> pop |> peek)
```

```
File "[5]", line 1, characters 37-41:

1 | ListStack.(empty |> push 1 |> pop |> peek)

^^^^

Error: This expression has type 'a ListStack.t -> 'a option

but an expression was expected of type int ListStack.t option -> 'b

Type 'a ListStack.t is not compatible with type int ListStack.t option
```

The types break down for the pipeline right after the pop, because that now returns an 'a t option, but peek expects an input that is merely an 'a t.

It is possible to define some additional operators to help restore the ability to pipeline. In fact, these functions are already defined in the Option module in the standard library, though not as infix operators:

```
(* Option.map aka fmap *)
let ( >> | ) opt f =
   match opt with
   | None -> None
   | Some x -> Some (f x)

(* Option.bind *)
let ( >>= ) opt f =
   match opt with
   | None -> None
   | Some x -> f x
```

```
val ( >>| ) : 'a option -> ('a -> 'b) -> 'b option = <fun>
```

```
val ( >>= ) : 'a option -> ('a -> 'b option) -> 'b option = <fun>
```

We can use those as needed for pipelining:

```
ListStack.(empty |> push 1 |> pop >>| push 2 >>= pop >>| push 3 >>| to_list)
```

```
- : int list option = Some [3]
```

But it's not so pleasant to figure out which of the three operators to use where.

There is therefore a tradeoff in the interface design:

• Using options ensures that surprising exceptions regarding empty stacks never occur at run-time. The program is therefore more robust. But the convenient pipeline operator is lost.

• Using exceptions means that programmers don't have to write as much code. If they are sure that an exception can't occur, they can omit the code for handling it. The program is less robust, but writing it is more convenient.

There is thus a tradeoff between writing more code early (with options) or doing more debugging later (with exceptions). The OCaml standard library has recently begun providing both versions of the interface in a data structure, so that the client can make the choice of how they want to use it. For example, we could provide both peek and peek_opt, and the same for pop, for clients of our stack module:

```
module type Stack = sig
 type 'a t
 val empty : 'a t
 val is_empty : 'a t -> bool
 val push : 'a -> 'a t -> 'a t
 val peek : 'a t -> 'a
 val peek_opt : 'a t -> 'a option
 val pop : 'a t -> 'a t
 val pop_opt : 'a t -> 'a t option
 val size : 'a t -> int
 val to_list : 'a t -> 'a list
module ListStack : Stack = struct
 type 'a t = 'a list
 exception Empty
 let empty = []
 let is_empty = function [] -> true | _ -> false
 let push = List.cons
 let peek = function [] -> raise Empty | x :: _ -> x
 let peek_opt = function [] -> None | x :: _ -> Some x
 let pop = function [] -> raise Empty | _ :: s -> s
 let pop_opt = function [] -> None | _ :: s -> Some s
 let size = List.length
 let to_list = Fun.id
end
```

One nice thing about this implementation is that it is efficient. All the operations except for size are constant time. We saw earlier in the chapter that size could be made constant time as well, at the cost of some extra space — though just a constant factor more — by caching the size of the stack at each node in the list.

5.5.4 Queues

Queues and stacks are fairly similar interfaces. We'll stick with exceptions instead of options for now.

```
module type Queue = sig
   (** An ['a t] is a queue whose elements have type ['a]. *)
   type 'a t

   (** Raised if [front] or [dequeue] is applied to the empty queue. *)
   exception Empty

   (** [empty] is the empty queue. *)
   val empty : 'a t

   (** [is_empty q] is whether [q] is empty. *)
   val is_empty : 'a t -> bool
```

```
(** [enqueue x q] is the queue [q] with [x] added to the end. *)
val enqueue : 'a -> 'a t -> 'a t

(** [front q] is the element at the front of the queue. Raises [Empty]
    if [q] is empty. *)
val front : 'a t -> 'a

(** [dequeue q] is the queue containing all the elements of [q] except the
    front of [q]. Raises [Empty] is [q] is empty. *)
val dequeue : 'a t -> 'a t

(** [size q] is the number of elements in [q]. *)
val size : 'a t -> int

(** [to_list q] is a list containing the elements of [q] in order from
    front to back. *)
val to_list : 'a t -> 'a list
end
```

Important: Similarly to peek and pop, note how front and dequeue divide the responsibility of getting the first element vs. getting all the rest of the elements.

It's easy to implement queues with lists, just as it was for implementing stacks:

```
module ListQueue : Queue
```

But despite being as easy, this implementation is not as efficient as our list-based stacks. Dequeueing is a constant-time operation with this representation, but enqueueing is a linear-time operation. That's because dequeue does a single pattern match, whereas enqueue must traverse the entire list to append the new element at the end.

There's a very clever way to do better on efficiency. We can use two lists to represent a single queue. This representation was invented by Robert Melville as part of his PhD dissertation at Cornell (*Asymptotic Complexity of Iterative Computations*, Jan 1981), which was advised by Prof. David Gries. Chris Okasaki (*Purely Functional Data Structures*, Cambridge University Press, 1988) calls these *batched queues*. Sometimes you will see this same implementation referred to as "implementing a queue with two stacks". That's because stacks and lists are so similar (as we've already seen) that you could rewrite pop as List.tl, and so forth.

The core idea has a Part A and a Part B. Part A is: we use the two lists to split the queue into two pieces, the *inbox* and *outbox*. When new elements are enqueued, we put them in the inbox. Eventually (we'll soon come to how) elements are transferred from the inbox to the outbox. When a dequeue is requested, that element is removed from the outbox; or when the front element is requested, we check the outbox for it. For example, if the inbox currently had [3; 4; 5] and the

outbox had [1; 2], then the front element would be 1, which is the head of the outbox. Dequeuing would remove that element and leave the inbox with just [2], which is the tail of the outbox. Likewise, enqueuing 6 would make the inbox become [3; 4; 5; 6].

The efficiency of front and dequeue is very good so far. We just have to take the head or tail of the outbox, respectively, assuming it is non-empty. Those are constant-time operations. But the efficiency of enqueue is still bad. It's linear time, because we have to append the new element to the end of the list. It's too bad we have to use the append operator, which is inherently linear time. It would be much better if we could use cons, which is constant time.

So here's Part B of the core idea: let's keep the inbox in reverse order. For example, if we enqueued 3 then 4 then 5, the inbox would actually be [5; 4; 3], not [3; 4; 5]. Then if 6 were enqueued next, we could cons it onto the beginning of the inbox, which becomes [6; 5; 4; 3]. The queue represented by inbox i and outbox o is therefore o @ List.rev i. So enqueue can now always be a constant-time operation.

But what about dequeue (and front)? They're constant time too, as long as the outbox is not empty. If it's empty, we have a problem. We need to transfer whatever is in the inbox to the outbox at that point. For example, if the outbox is empty, and the inbox is [6; 5; 4; 3], then we need to switch them around, making the outbox be [3; 4; 5; 6] and the inbox be empty. That's actually easy: we just have to reverse the list.

Unfortunately, we just re-introduced a linear-time operation. But with one crucial difference: we don't have to do that linear-time reverse on every dequeue, whereas with ListQueue above we had to do the linear-time append on every enqueue. Instead, we only have to do the reverse on those rare occasions when the outbox becomes empty.

So even though in the worst case dequeue (and front) will be linear time, most of the time they will not be. In fact, later in this book when we study *amortized analysis* we will show that in the long run they can be understood as constant-time operations. For now, here's a piece of intuition to support that claim: every individual element enters the outbox once (with a cons), moves to the inbox once (with a pattern match then cons), and leaves the inbox once (with a pattern match). Each of those is constant time. So each element only ever experiences constant-time operations from its own perspective.

For now, let's move on to implementing these ideas. In the implementation, we'll add one more idea: the outbox always has to have an element in it, unless the queue is empty. In other words, if the outbox is empty, we're guaranteed the inbox is too. That requirement isn't necessary for batched queues, but it does keep the code simpler by reducing the number of times we have to check whether a list is empty. The tiny tradeoff is that if the queue is empty, enqueue now has to directly put an element into the outbox. No matter, that's still a constant-time operation.

```
module BatchedQueue : Queue = struct
  (** [{o; i]} represents the queue [o @ List.rev i]. For example,
      [\{o = [1; 2]; i = [5; 4; 3]\}] represents the queue [1, 2, 3, 4, 5],
      where [1] is the front element. To avoid ambiguity about emptiness,
      whenever only one of the lists is empty, it must be [i]. For example,
      [\{o = [1]; i = []\}] is a legal representation, but [\{o = []; i = [1]\}]
      is not. This implies that if [o] is empty, [i] must also be empty. *)
 type 'a t = {o : 'a list; i : 'a list}
 exception Empty
 let empty = {o = []; i = []}
 let is_empty = function
   | {o = []} -> true
   | _ -> false
 let enqueue x = function
   \{o = []\} \rightarrow \{o = [x]; i = []\}
   | \{o; i\} \rightarrow \{o; i = x :: i\}
 let front = function
```

```
| {o = []} -> raise Empty
| {o = h :: _} -> h

let dequeue = function
| {o = []} -> raise Empty
| {o = [_]; i} -> {o = List.rev i; i = []}
| {o = _ :: t; i} -> {o = t; i}

let size {o; i} = List.(length o + length i)

let to_list {o; i} = o @ List.rev i
end
```

```
module BatchedQueue : Queue
```

The efficiency of batched queues comes at a price in readability. If we compare ListQueue and BatchedQueue, it's hopefully clear that ListQueue is a simple and correct implementation of a queue data structure. It's probably far less clear that BatchedQueue is a correct implementation. Just look at how many paragraphs of writing it took to explain it above!

5.5.5 Maps

Recall that a *map* (aka *dictionary*) binds keys to values. Here is a module type for maps. There are many other operations a map might support, but these will suffice for now.

```
module type Map = sig
  (**[('k, 'v) t] is the type of maps that bind keys of type ['k] to
      values of type ['v]. *)
  type ('k, 'v) t
  (** [empty] does not bind any keys. *)
  val empty : ('k, 'v) t
  (** [insert k v m] is the map that binds [k] to [v], and also contains
      all the bindings of [m]. If [k] was already bound in [m], that old
      binding is superseded by the binding to [v] in the returned map. *)
  val insert : 'k \rightarrow 'v \rightarrow ('k, 'v) t \rightarrow ('k, 'v) t
  (**[lookup \ k \ m] \ is the value bound to [k] in [m]. Raises: [Not_found] if [k]
      is not bound in [m]. *)
  val lookup : 'k -> ('k, 'v) t -> 'v
  (** [bindings m] is an association list containing the same bindings as [m].
      The keys in the list are guaranteed to be unique. *)
  val bindings : ('k, 'v) t \rightarrow ('k * 'v) list
end
```

```
module type Map =
   sig
   type ('k, 'v) t
   val empty : ('k, 'v) t
   val insert : 'k -> 'v -> ('k, 'v) t -> ('k, 'v) t
   val lookup : 'k -> ('k, 'v) t -> 'v
```

```
val bindings : ('k, 'v) t -> ('k * 'v) list end
```

Note how Map.t is parameterized on two types, 'k and 'v, which are written in parentheses and separated by commas. Although ('k, 'v) might look like a pair of values, it is not: it is a syntax for writing multiple type variables.

Recall that association lists are lists of pairs, where the first element of each pair is a key, and the second element is the value it binds. For example, here is an association list that maps some well-known names to an approximation of their numeric value:

```
[("pi", 3.14); ("e", 2.718); ("phi", 1.618)]
```

Naturally we can implement the Map module type with association lists:

```
module AssocListMap : Map = struct
    (** The list [(k1, v1); ...; (kn, vn)] binds key [ki] to value [vi].
        If a key appears more than once in the list, it is bound to the
        the left-most occurrence in the list. *)

    type ('k, 'v) t = ('k * 'v) list
    let empty = []
    let insert k v m = (k, v) :: m
    let lookup k m = List.assoc k m
    let keys m = List.(m |> map fst |> sort_uniq Stdlib.compare)
    let bindings m = m |> keys |> List.map (fun k -> (k, lookup k m))
end
```

```
module AssocListMap : Map
```

This implementation of maps is persistent. For example, adding a new binding to the map m below does not change m itself:

```
open AssocListMap
let m = empty |> insert "pi" 3.14 |> insert "e" 2.718
let m' = m |> insert "phi" 1.618
let b = bindings m
let b' = bindings m'
```

```
val m : (string, float) AssocListMap.t = <abstr>
```

```
val m' : (string, float) AssocListMap.t = <abstr>
```

```
val b : (string * float) list = [("e", 2.718); ("pi", 3.14)]
```

```
val b' : (string * float) list = [("e", 2.718); ("phi", 1.618); ("pi", 3.14)]
```

The insert operation is constant time, which is great. But the lookup operation is linear time. It's possible to do much better than that. In a later chapter, we'll see how to do better. Logarithmic-time performance is achievable with balanced binary trees, and something like constant-time performance with hash tables. Neither of those, however, achieves the simplicity of the code above.

The bindings operation is complicated by potential duplicate keys in the list. It uses a keys helper function to extract the unique list of keys with the help of library function List.sort_uniq. That function sorts an input list and in the process discards duplicates. It requires a comparison function as input.

Note: A comparison function must return 0 if its arguments compare as equal, a positive integer if the first is greater, and a negative integer if the first is smaller.

Here we use the standard library's comparison function Stdlib.compare, which behaves essentially the same as the built-in comparison operators =, <, >, etc. Custom comparison functions are useful if you want to have a relaxed notion of what being a duplicate means. For example, maybe you'd like to ignore the case of strings, or the sign of a number, etc.

The running time of List.sort_uniq is linearithmic, and it produces a linear number of keys as output. For each of those keys, we do a linear-time lookup operation. So the total running time of bindings is $O(n \log n) + O(n) \cdot O(n)$, which is $O(n^2)$. We can definitely do better than that with more advanced data structures.

Actually we can have a constant-time bindings operation even with association lists, if we are willing to pay for a linear-time insert operation:

```
module UniqAssocListMap : Map = struct
    (** The list [(k1, v1); ...; (kn, vn)] binds key [ki] to value [vi].
         No duplicate keys may occur. *)
    type ('k, 'v) t = ('k * 'v) list
    let empty = []
    let insert k v m = (k, v) :: List.remove_assoc k m
    let lookup k m = List.assoc k m
    let bindings m = m
end
```

```
module UniqAssocListMap : Map
```

That implementation removes any duplicate binding of k before inserting a new binding.

5.5.6 Sets

Here is a module type for sets. There are many other operations a set data structure might be expected to support, but these will suffice for now.

```
module type Set = sig
  (** ['a t] is the type of sets whose elements are of type ['a]. *)
  type 'a t

  (** [empty] is the empty set *)
  val empty : 'a t

  (** [mem x s] is whether [x] is an element of [s]. *)
  val mem : 'a -> 'a t -> bool

  (** [add x s] is the set that contains [x] and all the elements of [s]. *)
  val add : 'a -> 'a t -> 'a t

  (** [elements s] is a list containing the elements of [s]. No guarantee is made about the ordering of that list, but each is guaranteed to be unique. *)
  val elements : 'a t -> 'a list end
```

```
module type Set =
   sig
   type 'a t
   val empty : 'a t
   val mem : 'a -> 'a t -> bool
   val add : 'a -> 'a t -> 'a t
   val elements : 'a t -> 'a list
   end
```

Here's an implementation of that interface using a list to represent the set. This implementation ensures that the list never contains any duplicate elements, since sets themselves do not:

```
module UniqListSet : Set = struct
  type 'a t = 'a list
  let empty = []
  let mem = List.mem
  let add x s = if mem x s then s else x :: s
  let elements = Fun.id
end
```

```
module UniqListSet : Set
```

Note how add ensures that the representation never contains any duplicates, so the implementation of elements is easy. Of course, that comes with the tradeoff of add being linear time.

Here's a second implementation, which permits duplicates in the list:

```
module ListSet : Set = struct
  type 'a t = 'a list
  let empty = []
  let mem = List.mem
  let add = List.cons
  let elements s = List.sort_uniq Stdlib.compare s
end
```

```
module ListSet : Set
```

In that implementation, the add operation is now constant time, and the elements operation is linearithmic time.

5.6 Module Type Constraints

We have extolled the virtues of encapsulation. Now we're going to do something that might seem counter-intuitive: selectively violate encapsulation.

As a motivating example, here is a module type that represents values that support the usual operations from arithmetic, or more precisely, a *field*:

```
module type Field = sig
  type t
  val zero : t
  val one : t
  val ( + ) : t -> t -> t
  val ( * ) : t -> t -> t
  val ( ~ - ) : t -> t
```

```
val to_string : t -> string
end
```

```
module type Field =
   sig
    type t
   val zero : t
   val one : t
   val ( + ) : t -> t -> t
   val ( * ) : t -> t -> t
   val ( ~- ) : t -> t
   val to_string : t -> string
   end
```

Recall that we must write (*) instead of (*) because the latter would be parsed as beginning a comment. And we write the \sim in (\sim -) to indicate a *unary* negation operator.

This is a bit weird of an example. We don't normally think of numbers as a data structure. But what is a data structure except for a set of values and operations on them? The Field module type makes it clear that's what we have.

Here is a module that implements that module type:

```
module IntField : Field = struct
  type t = int
  let zero = 0
  let one = 1
  let ( + ) = Stdlib.( + )
  let ( * ) = Stdlib.( * )
  let ( ~ - ) = Stdlib.( ~ - )
  let to_string = string_of_int
end
```

```
module IntField: Field
```

Because t is abstract, the toplevel can't give us good output about what the sum of one and one is:

```
IntField.(one + one)
```

```
- : IntField.t = <abstr>
```

But we could convert it to a string:

```
IntField.(one + one |> to_string)
```

```
- : string = "2"
```

We could even install a pretty printer to avoid having to manually call to_string:

```
let pp_intfield fmt i =
   Format.fprintf fmt "%s" (IntField.to_string i);;
#install_printer pp_intfield;;
IntField.(one + one)
```

```
val pp_intfield : Format.formatter -> IntField.t -> unit = <fun>
```

```
-: IntField.t = 2
```

We could implement other kinds of fields, too:

```
module FloatField : Field = struct
  type t = float
  let zero = 0.
  let one = 1.
  let ( + ) = Stdlib.( +. )
  let ( * ) = Stdlib.( *. )
  let ( ~- ) = Stdlib.( ~-. )
  let to_string = string_of_float
end
```

```
module FloatField : Field
```

Then we'd have to install a printer for it, too:

```
let pp_floatfield fmt f =
   Format.fprintf fmt "%s" (FloatField.to_string f);;
#install_printer pp_floatfield;;
FloatField.(one + one)
```

```
val pp_floatfield : Format.formatter -> FloatField.t -> unit = <fun>
```

```
- : FloatField.t = 2.
```

Was there really a need to make type t abstract in the field examples above? Arguably not. And if it were not abstract, we wouldn't have to go to the trouble of converting abstract values into strings, or installing printers. Let's pursue that idea, next.

5.6.1 Specializing Module Types

In the past, we've seen that we can leave off the module type annotation, then do a separate check to make sure the structure satisfies the signature:

```
module IntField = struct
  type t = int
  let zero = 0
  let one = 1
  let ( + ) = Stdlib.( + )
  let ( * ) = Stdlib.( * )
  let ( ~- ) = Stdlib.( ~- )
  let to_string = string_of_int
end

module _ : Field = IntField
```

```
IntField.(one + one)
```

```
- : int = 2
```

There's a more sophisticated way of accomplishing the same goal. We can specialize the Field module type to specify that t must be int or float. We do that by adding a *constraint* using the with keyword:

```
module type INT_FIELD = Field with type t = int
```

```
module type INT_FIELD =
    sig
      type t = int
    val zero : t
    val one : t
    val ( + ) : t -> t -> t
    val ( * ) : t -> t -> t
    val ( ~- ) : t -> t
    val to_string : t -> string
    end
```

Note how the INT_FIELD module type now specifies that t and int are the same type. It exposes or *shares* that fact with the world, so we could call these "sharing constraints."

Now IntField can be given that module type:

```
module IntField : INT_FIELD = struct
  type t = int
  let zero = 0
  let one = 1
  let ( + ) = Stdlib.( + )
  let ( * ) = Stdlib.( * )
  let ( ~- ) = Stdlib.( ~- )
  let to_string = string_of_int
end
```

```
module IntField : INT_FIELD
```

And since the equality of t and int is exposed, the toplevel can print values of type t without any help needed from a pretty printer:

```
IntField. (one + one)
```

```
-: IntField.t = 2
```

Programmers can even mix and match built-in int values with those provided by IntField:

```
IntField.(1 + one)
```

```
-: IntField.t = 2
```

The same can be done for floats:

```
module type FLOAT_FIELD = Field with type t = float
module FloatField : FLOAT_FIELD = struct
   type t = float
   let zero = 0.
   let one = 1.
```

```
let ( + ) = Stdlib.( +. )
let ( * ) = Stdlib.( *. )
let ( ~- ) = Stdlib.( ~-. )
let to_string = string_of_float
end
```

```
module type FLOAT_FIELD =
    sig
    type t = float
    val zero : t
    val one : t
    val ( + ) : t -> t -> t
    val ( * ) : t -> t -> t
    val ( ~- ) : t -> t
    val to_string : t -> string
    end
```

```
module FloatField : FLOAT_FIELD
```

It turns out there's no need to separately define INT_FIELD and FLOAT_FIELD. The with keyword can be used as part of the module definition, though the syntax becomes a little harder to read because of the proximity of the two = signs:

```
module FloatField : Field with type t = float = struct
  type t = float
  let zero = 0.
  let one = 1.
  let ( + ) = Stdlib.( +. )
  let ( * ) = Stdlib.( *. )
  let ( ~- ) = Stdlib.( ~-. )
  let to_string = string_of_float
end
```

```
module FloatField :
    sig
    type t = float
    val zero : t
    val one : t
    val ( + ) : t -> t -> t
    val ( * ) : t -> t -> t
    val ( ~- ) : t -> t
    val to_string : t -> string
    end
```

5.6.2 Constraints

Syntax.

There are two sorts of constraints. One is the sort we saw above, with type equations:

• T with type x = t, where T is a module type, x is a type name, and t is a type.

The other sort is a module equation, which is syntactic sugar for specifying the equality of all types in the two modules:

• T with module M = N, where M and N are module names.

Multiple constraints can be added with the and keyword:

• T with constraint1 and constraint2 and ... constraintN

Static semantics.

The constrained module type T with type x = t is the same as T, except that the declaration of type x inside T is replaced by type x = t. For example, compare the two signatures output below:

```
module type T = sig type t end
module type U = T with type t = int
```

```
module type T = sig type t end
```

```
module type U = sig type t = int end
```

Likewise, T with module M = N is the same as T, except that the any declaration type x inside the module type of M is replaced by type $x = N \cdot X$. (And the same recursively for any nested modules.) It takes more work to give and understand this example:

```
module type XY = sig
 type x
 type y
end
module type T = sig
 module A : XY
end
module B = struct
 type x = int
 type y = float
end
module type U = T with module A = B
module C : U = struct
 module A = struct
   type x = int
   type y = float
    let x = 42
  end
end
```

```
module type XY = sig type x type y end
```

```
module type T = sig module A : XY end
```

```
module B : sig type x = int type y = float end
```

```
module type U = sig module A : sig type x = int type y = float end end
```

```
module C : U
```

Focus on the output for module type U. Notice that the types of x and y in it have become int and float because of the module A = B constraint. Also notice how modules B and C.A are *not* the same module; the latter has an

extra item x in it. So the syntax module A = B is potentially confusing. The constraint is not specifying that the two *modules* are the same. Rather, it specifies that all their *types* are constrained to be equal.

Dynamic semantics.

There are no dynamic semantics for constraints, because they are only for type checking.

5.7 Compilation Units

A *compilation unit* is a pair of OCaml source files in the same directory. They share the same base name, call it x, but their extensions differ: one file is x.ml, the other is x.mli. The file x.mli is called the *implementation*, and x.mli is called the *interface*.

For example, suppose that foo.mli contains exactly the following:

```
val x : int
val f : int -> int -> int
```

and foo.ml, in the same directory, contains exactly the following:

```
let x = 0
let y = 12
let f x = x + y
```

Then compiling foo.ml will have the same effect as defining the module Foo as follows:

```
module Foo : sig
  val x : int
  val f : int -> int -> int
end = struct
  let x = 0
  let y = 12
  let f x = x + y
end
```

In general, when the compiler encounters a compilation unit, it treats it as defining a module and a signature like this:

```
module Foo
    : sig (* insert contents of foo.mli here *) end
= struct
    (* insert contents of foo.ml here *)
end
```

The *unit name* $F \circ \circ$ is derived from the base name $f \circ \circ$ by just capitalizing the first letter. Notice that there is no named module type being defined; the signature of $F \circ \circ$ is actually anonymous.

The standard library uses compilation units to implement most of the modules we have been using so far, like List and String. You can see that in the standard library source code.

5.7.1 Documentation Comments

Some documentation comments belong in the interface file, whereas others belong in the implementation file:

- Clients of an abstraction can be expected to read interface files, or rather the HTML documentation generated from them. So the comments in an interface file should be written with that audience in mind. These comments should describe how to use the abstraction, the preconditions for calling its functions, what exceptions they might raise, and perhaps some notes on what algorithms are used to implement operations. The standard library's List module contains many examples of these kinds of comments.
- Clients should not be expected to read implementation files. Those files will be read by creators and maintainers
 of the implementation. The documentation in the implementation file should provide information that explains
 the internal details of the abstraction, such as how the representation type is used, how the code works, important
 internal invariants it maintains, and so forth. Maintainers can also be expected to read the specifications in the
 interface files.

Documentation should **not** be duplicated between the files. In particular, the client-facing specification comments in the interface file should not be duplicated in the implementation file. One reason is that duplication inevitably leads to errors. Another reason is that OCamldoc has the ability to automatically inject the comments from the interface file into the generated HTML from the implementation file.

In interface files —that is, .mli files— OCamldoc comments can be placed either before or after an element of the interface. For example, both of these placements are possible:

```
(** The mathematical constant 3.14... *)
val pi : float
```

```
val pi : float
(** The mathematical constant 3.14... *)
```

Tip: The standard library developers apparently prefer the post-placement of the comment, and OCamlFormat seems to work better with that, too.

5.7.2 An Example with Stacks

Put this code in mystack.mli, noting that there is no sig..end around it or any module type:

```
type 'a t
exception Empty
val empty : 'a t
val is_empty : 'a t -> bool
val push : 'a -> 'a t -> 'a t
val peek : 'a t -> 'a
val pop : 'a t -> 'a t
```

We're using the name "mystack" because the standard library already has a Stack module. Re-using that name could lead to error messages that are somewhat hard to understand.

Also put this code in mystack.ml, nothing that there is no struct..end around it or any module:

```
type 'a t = 'a list
exception Empty
let empty = []
let is_empty = function [] -> true | _ -> false
```

```
let push = List.cons
let peek = function [] -> raise Empty | x :: _ -> x
let pop = function [] -> raise Empty | _ :: s -> s
```

Then from the command-line compile that source code. Note that all we need is the .cmo file, so we request it to be built instead of the .byte file:

```
$ ocambuild mystack.cmo
```

Launch utop and load your compilation unit for use:

```
# #directory "_build";;
# #load "mystack.cmo";;
# Mystack.empty;;
- : 'a Mystack.t = <abstr>
```

5.7.3 Incomplete Compilation Units

What if either the interface or implementation file is missing for a compilation unit?

Missing Interface Files. Actually this is exactly how we've normally been working up until this point. For example, you might have done some homework in a file named lab1.ml but never needed to worry about lab1.mli. There is no requirement that every .ml file have a corresponding .mli file, or in other words, that every compilation unit be complete.

If the .mli file is missing there is still a module that is created, as we saw back when we learned about #load and modules. It just doesn't have an automatically imposed signature. For example, the situation with lab1 above would lead to the following module being created during compilation:

```
module Lab1 = struct
   (* insert contents of lab1.ml here *)
end
```

Missing Implementation Files. This case is much rarer, and not one you are likely to encounter in everyday development. But be aware that there is a misuse case that Java or C++ programmers sometimes accidentally fall into. Suppose you have an interface for which there will be a few implementation. Thinking back to stacks earlier in this chapter, perhaps you have a module type Stack and two modules that implement it, ListStack and CustomStack:

```
module type Stack = sig
  type 'a t
  val empty : 'a t
  val push : 'a -> 'a t -> 'a t
  (* etc. *)
end

module type ListStack : Stack = sig
  type 'a t = 'a list
  let empty = []
  let push = List.cons
   (* etc. *)
end

module type CustomStack : Stack = sig
  (* omitted *)
end
```

It's tempting to divide that code up into files as follows:

The reason it's tempting is that in Java you might put the Stack interface into a Stack.java file, the ListStack class in a ListStack.java file, and so forth. In C++ something similar might be done with .hpp and .cpp files.

But the OCaml file organization shown above just won't work. To be a compilation unit, the interface for listStack. ml must be in listStack.mli. It can't be in a file with any other name. So there's no way with that code division to stipulate that ListStack: Stack.

Instead, the code could be divided like this:

```
(************
(* stack.ml *)
module type S = sig
 type 'a t
 val empty : 'a t
 val push : 'a -> 'a t -> 'a t
 (* etc. *)
end
(************
(* listStack.ml *)
module M : Stack.S = struct
 type 'a t = 'a list
 let empty = []
 let push = List.cons
 (* etc. *)
end
(************
(* customStack.ml *)
module M : Stack.S = struct
 (* omitted *)
end
```

Note the following about that division:

- The module type goes in a .ml file not a .mli, because we're not trying to create a compilation unit.
- We give short names to the modules and module types in the files, because they will already be inside a module based on their filename. It would be rather verbose, for example, to name S something longer like Stack. If we did, we'd have to write Stack. Stack in the module type annotations instead of Stack.S.

Another possibility for code division would be to put all the code in a single file stack.ml. That works if all the code is part of the same library, but not if (e.g.) ListStack and CustomStack are developed by separate organizations. If it is in a single file, then we could turn it into a compilation unit:

```
(************
(* stack.mli *)
module type S = sig
 type 'a t
 val empty : 'a t
 val push : 'a -> 'a t -> 'a t
  (* etc. *)
end
module ListStack : S
module CustomStack : S
(************
(* stack.ml *)
module type S = sig
 type 'a t
 val empty : 'a t
 val push : 'a -> 'a t -> 'a t
 (* etc. *)
end
module ListStack : S = struct
 type 'a t = 'a list
 let empty = []
 let push = List.cons
  (* etc. *)
end
module CustomStack : S = struct
 (* omitted *)
end
```

Unfortunately that does mean we've duplicated Stack.S in both the interface and implementation files. There's no way to automatically "import" an already declared module type from a .mli file into the corresponding .ml file.

Code duplication naturally makes us unhappy. Is there a way to eliminate it? Keep reading the next section to find out.

5.8 Includes

Copying and pasting code is almost always a bad idea. Duplication of code causes duplication and proliferation of errors. So why are we so prone to making this mistake? Maybe because it always seems like the easier option — easier and quicker than applying the Abstraction Principle as we should to factor out common code.

The OCaml module system provides a neat feature called *includes* that is like a principled copy-and-paste that is quick and easy to use, but avoids actual duplication. It can be used to solve some of the same problems as *inheritance* in object-oriented languages.

Let's start with an example. Recall this implementation of sets as lists:

```
module type Set = sig
type 'a t

(continues on next page)
```

5.8. Includes 193

```
val empty : 'a t
val mem : 'a -> 'a t -> bool
val add : 'a -> 'a t -> 'a t
val elements : 'a t -> 'a list
end

module ListSet : Set = struct
  type 'a t = 'a list
  let empty = []
  let mem = List.mem
  let add = List.cons
  let elements s = List.sort_uniq Stdlib.compare s
end
```

Suppose we wanted to add a function of_list: 'a list -> 'a t that could construct a set out of a list. If we had access to the source code of both ListSet and Set, and if we were permitted to modify it, this wouldn't be hard. But what if they were third-party libraries for which we didn't have source code?

In Java, we might use inheritance to solve this problem:

```
interface Set<T> { ... }
class ListSet<T> implements Set<T> { ... }
class ListSetExtended<T> extends ListSet<T> {
   Set<T> ofList(List<T> lst) { ... }
}
```

That helps us to reuse code, because the subclass inherits all the methods of its superclass.

OCaml *includes* are similar. They enable a module to include all the items defined by another module, or a module type to include all the specifications of another module type.

Here's how we can use includes to solve the problem of adding of_list to ListSet:

```
module ListSetExtended = struct
  include ListSet
  let of_list lst = List.fold_right add lst empty
end
```

```
module ListSetExtended :
    sig
    type 'a t = 'a ListSet.t
    val empty : 'a t
    val mem : 'a -> 'a t -> bool
    val add : 'a -> 'a t -> 'a t
    val elements : 'a t -> 'a list
    val of_list : 'a list -> 'a t
    end
```

This code says that ListSetExtended is a module that includes all the definitions of the ListSet module, as well as a definition of of_list. We don't have to know the source code implementing ListSet to make this happen.

Note: You might wonder we why can't simply implement of_list as the identity function. See the section below on encapsulation for the answer.

5.8.1 Semantics of Includes

Includes can be used inside of structures and signatures. When we include inside a signature, we must be including another signature. And when we include inside a structure, we must be including another structure.

Including a structure is effectively just syntactic sugar for writing a local definition for each name defined in the module. Writing include ListSet as we did above, for example, has an effect similar to writing the following:

```
module ListSetExtended = struct
    (* BEGIN all the includes *)
    type 'a t = 'a ListSet.t
    let empty = ListSet.empty
    let mem = ListSet.mem
    let add = ListSet.add
    let elements = ListSet.elements
    (* END all the includes *)
    let of_list lst = List.fold_right add lst empty
end
```

```
module ListSetExtended :
    sig
    type 'a t = 'a ListSet.t
    val empty : 'a ListSet.t -> bool
    val add : 'a -> 'a ListSet.t -> 'a ListSet.t
    val elements : 'a ListSet.t -> 'a list
    val of_list : 'a list -> 'a ListSet.t
end
```

None of that is actually copying the source code of ListSet. Rather, the include just creates a new definition in ListSetExtended with the same name as each definition in ListSet. But if the set of names defined inside ListSet ever changed, the include would reflect that change, whereas a copy-paste job would not.

Including a signature is much the same. For example, we could write:

```
module type SetExtended = sig
  include Set
  val of_list : 'a list -> 'a t
end
```

```
module type SetExtended =
   sig
    type 'a t
   val empty : 'a t
   val mem : 'a -> 'a t -> bool
   val add : 'a -> 'a t -> 'a t
   val elements : 'a t -> 'a list
   val of_list : 'a list -> 'a t
   end
```

Which would have an effect similar to writing the following:

```
module type SetExtended = sig
  (* BEGIN all the includes *)
  type 'a t
  val empty : 'a t
  val mem : 'a -> 'a t -> bool
```

(continues on next page)

5.8. Includes 195

```
val add : 'a -> 'a t -> 'a t
val elements : 'a t -> 'a list
  (* END all the includes *)
val of_list : 'a list -> 'a t
end
```

```
module type SetExtended =
  sig
  type 'a t
  val empty : 'a t
  val mem : 'a -> 'a t -> bool
  val add : 'a -> 'a t -> 'a t
  val elements : 'a t -> 'a list
  val of_list : 'a list -> 'a t
  end
```

That module type would be suitable for ListSetExtended:

```
module ListSetExtended : SetExtended = struct
  include ListSet
  let of_list lst = List.fold_right add lst empty
end
```

```
module ListSetExtended : SetExtended
```

5.8.2 Encapsulation and Includes

We mentioned above that you might wonder why we didn't write this simpler definition of of_list:

```
module ListSetExtended : SetExtended = struct
  include ListSet
  let of_list lst = lst
end
```

Check out that error message. It looks like of_list doesn't have the right type. What if we try adding some type annotations?

```
module ListSetExtended : SetExtended = struct
include ListSet
```

```
let of_list (lst : 'a list) : 'a t = lst
end
```

Ah, now the problem is clearer: in the body of of_list, the equality of 'a t and 'a list isn't known. In ListSetExtended, we do know that 'a t = 'a ListSet.t, because that's what the include gave us. But the fact that 'a ListSet.t = 'a list was hidden when ListSet was sealed at module type Set. So, includes must obey encapsulation, just like the rest of the module system.

One workaround is to rewrite the definitions as follows:

```
module ListSetImpl = struct
 type 'a t = 'a list
 let empty = []
 let mem = List.mem
  let add = List.cons
  let elements s = List.sort_uniq Stdlib.compare s
module ListSet : Set = ListSetImpl
module type SetExtended = sig
 include Set
 val of_list : 'a list -> 'a t
end
module ListSetExtendedImpl = struct
 include ListSetImpl
  let of_list lst = lst
end
module ListSetExtended : SetExtended = ListSetExtendedImpl
```

```
module ListSetImpl :
    sig
    type 'a t = 'a list
    val empty : 'a list
    val mem : 'a -> 'a list -> bool
    val add : 'a -> 'a list -> 'a list
    val elements : 'a list -> 'a list
    end
```

```
module ListSet : Set
```

```
module type SetExtended =
   sig
   type 'a t
   val empty : 'a t
   val mem : 'a -> 'a t -> bool
   val add : 'a -> 'a t -> 'a t
```

(continues on next page)

5.8. Includes 197

```
val elements : 'a t -> 'a list
val of_list : 'a list -> 'a t
end
```

```
module ListSetExtendedImpl :
    sig
    type 'a t = 'a list
    val empty : 'a list
    val mem : 'a -> 'a list -> bool
    val add : 'a -> 'a list -> 'a list
    val elements : 'a list -> 'a list
    val of_list : 'a -> 'a
    end
```

```
module ListSetExtended : SetExtended
```

The important change is that ListSetImpl is not sealed, so its type 'a t is not abstract. When we include it in ListSetExtended, we can therefore exploit the fact that it's a synonym for 'a list.

What we just did is effectively the same as what Java does to handle the visibility modifiers public, private, etc. The "private version" of a class is like the Impl version above: anyone who can see that version gets to see all the exposed items (fields in Java, types in OCaml), without any encapsulation. The "public version" of a class is like the sealed version above: anyone who can see that version is forced to treat the items as abstract, hence encapsulated.

With that technique, if we want to provide a new implementation of one of the included functions we *could* do that too:

```
module ListSetExtendedImpl :
    sig
    type 'a t = 'a list
    val empty : 'a list
    val mem : 'a -> 'a list -> bool
    val add : 'a -> 'a list -> 'a list
    val of_list : 'a list -> 'a list
    val elements : 'a list -> 'a list
    end
```

But that's a bad idea. First, it's actually a quadratic implementation of elements instead of linearithmic. Second, it does not *replace* the original implementation of elements. Remember the semantics of modules: all definitions are evaluated from top to bottom, in order. So the new definition of elements above won't come into use until the very end of evaluation. If any earlier functions had happened to use elements as a helper, they would use the original linearithmic version, not the new quadratic version.

Warning: This differs from what you might expect from Java, which uses a language feature called dynamic dispatch to figure out which method implementation to invoke. Dynamic dispatch is arguably *the* defining feature of object-oriented languages. OCaml functions are not methods, and they do not use dynamic dispatch.

5.8.3 Include vs. Open

The include and open statements are quite similar, but they have a subtly different effect on a structure. Consider this code:

```
module M = struct
  let x = 0
end

module N = struct
  include M
  let y = x + 1
end

module O = struct
  open M
  let y = x + 1
end
```

```
module M : sig val x : int end
```

```
module N : sig val x : int val y : int end
```

```
module O : sig val y : int end
```

Look closely at the values contained in each structure. N has both an x and y, whereas O has only a y. The reason is that include M causes all the definitions of M to also be included in N, so the definition of x from M is present in N. But open M only made those definitions available in the *scope* of O; it doesn't actually make them part of the *structure*. So O does not contain a definition of x, even though x is in scope during the evaluation of O's definition of y.

A metaphor for understanding this difference might be: open M imports definitions from M and makes them available for local consumption, but they aren't exported to the outside world. Whereas include M imports definitions from M, makes them available for local consumption, and additionally exports them to the outside world.

5.8.4 Including Code in Multiple Modules

Recall that we also had an implementation of sets that made sure every element of the underlying list was unique:

```
module UniqListSet : Set = struct
   (** All values in the list must be unique. *)
   type 'a t = 'a list
   let empty = []
   let mem = List.mem
   let add x s = if mem x s then s else x :: s
   let elements = Fun.id
end
```

```
module UniqListSet : Set
```

Suppose we wanted add of_list to that module too. One possibility would be to copy and paste that function from ListSet into UniqListSet. But that's poor software engineering. So let's rule that out right away as a non-solution.

Instead, suppose we try to define the function outside of either module:

5.8. Includes 199

```
let of_list lst = List.fold_right add lst empty
```

The problem is we either need to choose which module's add and empty we want. But as soon as we do, the function becomes useful only with that one module:

```
let of_list lst = List.fold_right ListSet.add lst ListSet.empty
```

```
val of_list : 'a list -> 'a ListSet.t = <fun>
```

We could make add and empty be parameters instead:

```
let of_list' add empty lst = List.fold_right add lst empty
let of_list lst = of_list' ListSet.add ListSet.empty lst
let of_list_uniq lst = of_list' UniqListSet.add UniqListSet.empty lst
```

```
val of_list' : ('a -> 'b -> 'b) -> 'a list -> 'b = <fun>
```

```
val of_list : 'a list -> 'a ListSet.t = <fun>
```

```
val of_list_uniq : 'a list -> 'a UniqListSet.t = <fun>
```

But this is annoying in a couple of ways. First, we have to remember which function name to call, whereas all the other operations that are part of those modules have the same name, regardless of which module they're in. Second, the of_list functions live outside either module, so clients who open one of the modules won't automatically get the ability to name those functions.

Let's try to use includes to solve this problem. First, we write a module that contains the parameterized implementation:

```
module SetOfList = struct
  let of_list' add empty lst = List.fold_right add lst empty
end
```

```
module SetOfList :
   sig val of_list' : ('a -> 'b -> 'b) -> 'b -> 'a list -> 'b end
```

Then we include that module to get the helper function:

```
module UniqListSetExtended : SetExtended = struct
  include UniqListSet
  include SetOfList
  let of_list lst = of_list' add empty lst
end

module ListSetExtended : SetExtended = struct
  include ListSet
  include SetOfList
  let of_list lst = of_list' add empty lst
end
```

```
module UniqListSetExtended : SetExtended
```

```
module ListSetExtended : SetExtended
```

That works, but we've only partially succeeded in achieving code reuse:

- On the positive side, the code that implements of_list' has been factored out into a single location and reused in the two structures.
- But on the negative side, we still had to write an implementation of of_list in both modules. Worse yet, those implementations are identical. So there's still code duplication occurring.

Could we do better? Yes. And that leads us to functors, next.

5.9 Functors

The problem we were having in the previous section was that we wanted to add code to two different modules, but that code needed to be parameterized on the details of the module to which it was being added. It's that kind of parameterization that is enabled by an OCaml language feature called *functors*.

Note: Why the name "functor"? In category theory, a *category* contains *morphisms*, which are a generalization of functions as we known them, and a *functor* is map between categories. Likewise, OCaml modules contain functions, and OCaml functors map from modules to modules.

The name is unfortunately intimidating, but a functor is simply a "function" from modules to modules. The word "function" is in quotation marks in that sentence only because it's a kind of function that's not interchangeable with the rest of the functions we've already seen. OCaml's type system is *stratified*: module values are distinct from other values, so functions from modules to modules cannot be written or used in the same way as functions from values to values. But conceptually, functors really are just functions.

Here's a tiny example of a functor:

```
module type X = sig
  val x : int
end

module IncX (M : X) = struct
  let x = M.x + 1
end
```

```
module type X = sig val x : int end
```

```
module IncX : functor (M : X) -> sig val x : int end
```

The functor's name is IncX. It's essentially a function from modules to modules. As a function, it takes an input and produces an output. Its input is named M, and the type of its input is X. Its output is the structure that appears on the right-hand side of the equals sign: struct let x = M.x + 1.

Another way to think about IncX is that it's a *parameterized structure*. The parameter that it takes is named M and has type X. The structure itself has a single value named x in it. The value that x has will depend on the parameter M.

Since functors are essentially functions, we *apply* them. Here's an example of applying IncX:

5.9. Functors 201

```
module A = struct let x = 0 end

module A : sig val x : int end

A.x

- : int = 0

module B = IncX (A)

module B : sig val x : int end

B.x

- : int = 1

module C = IncX (B)

c.x

- : int = 2
```

Each time, we pass IncX a module. When we pass it the module bound to the name A, the input to IncX is struct let x = 0 end. Functor IncX takes that input and produces an output struct let x = A.x + 1 end. Since A.x is 0, the result is struct let x = 1 end. So B is bound to struct let x = 1 end. Similarly, C ends up being bound to struct let x = 2 end.

Although the functor IncX returns a module that is quite similar to its input module, that need not be the case. In fact, a functor can return any module it likes, perhaps something very different than its input structure:

```
module AddX (M : X) = struct
let add y = M.x + y
end
```

```
module AddX : functor (M : X) -> sig val add : int -> int end
```

Let's apply that functor to a module. The module doesn't even have to be bound to a name; we can just write an anonymous structure:

```
module Add42 = AddX (struct let x = 42 end)
```

```
module Add42 : sig val add : int -> int end
```

```
Add42.add 1
```

```
- : int = 43
```

Note that the input module to AddX contains a value named x, but the output module from AddX does not:

Add42.x

Warning: It's tempting to think that a functor is the same as extends in Java, and that the functor therefore extends the input module with new definitions while keeping the old definitions around too. The example above shows that is not the case. A functor is essentially just a function, and that function can return whatever the programmer wants. In fact the output of the functor could be arbitrarily different than the input.

5.9.1 Functor Syntax and Semantics

In the functor syntax we've been using:

the type annotation: S and the parentheses around it, (M : S) are required. The reason why is that OCaml needs the type information about S to be provided in order to do a good job with type inference for F itself.

Much like functions, functors can be written anonymously. The following two syntaxes for functors are equivalent:

```
module F (M : S) = ...
module F = functor (M : S) -> ...
```

The second form uses the functor keyword to create an anonymous functor, like how the fun keyword creates an anonymous function.

And functors can be parameterized on multiple structures:

```
module F (M1 : S1) ... (Mn : Sn) = ...
```

Of course, that's just syntactic sugar for a *higher-order functor* that takes a structure as input and returns an anonymous functor:

```
module F = functor (M1 : S1) -> ... -> functor (Mn : Sn) -> ...
```

If you want to specify the output type of a functor, the syntax is again similar to functions:

```
module F (M : Si) : So = ...
```

As usual, it's also possible to write the output type annotation on the module expression:

```
module F (M : Si) = (... : So)
```

To evaluate an application $module_expression1$ ($module_expression2$), the first module expression is evaluated and must produce a functor F. The second module expression is then evaluated to a module M. The functor is then applied to the module. The functor will be producing a new module N as part of that application. That new module is evaluated as always, in order of definition from top to bottom, with the definitions of M available for use.

5.9. Functors 203

5.9.2 Functor Type Syntax and Semantics

The simplest syntax for functor types is actually the same as for functions:

```
module_type -> module_type
```

For example, X -> Add below is a functor type, and it works for the AddX module we defined earlier in this section:

```
module type Add = sig val add : int -> int end
module CheckAddX : X -> Add = AddX
```

```
module type Add = sig val add : int -> int end
```

```
module CheckAddX : X -> Add
```

Functor type syntax becomes more complicated if the output module type is dependent upon the input module type. For example, suppose we wanted to create a functor that pairs up a value from one module with another value:

```
module type T = sig
  type t
  val x : t
end

module Pair1 (M : T) = struct
  let p = (M.x, 1)
end
```

```
module type T = sig type t val x : t end
```

```
module Pair1 : functor (M : T) -> sig val p : M.t * int end
```

The type of Pair1 turns out to be:

```
functor (M : T) -> sig val p : M.t * int end
```

So we could also write:

```
module type P1 = functor (M : T) -> sig val p : M.t * int end
module Pair1 : P1 = functor (M : T) -> struct
  let p = (M.x, 1)
end
```

```
module type P1 = functor (M : T) -> sig val p : M.t * int end
```

```
module Pair1 : P1
```

Module type P1 is the type of a functor that takes an input module named M of module type T, and returns an output module whose module type is given by the signature sig..end. Inside the signature, the name M is in scope. That's why we can write M.t in it, thereby ensuring that the type of the first component of pair p is the type from the *specific* module M that is passed into Pair1, not any *other* module. For example, here are two different instantiations:

```
module P0 = Pair1 (struct type t = int let x = 0 end)
module PA = Pair1 (struct type t = char let x = 'a' end)
```

```
module P0 : sig val p : int * int end
```

```
module PA : sig val p : char * int end
```

Note the difference between int and char in the resulting module types. It's important that the output module type of Pair1 can distinguish those. And that's why M has to be nameable on the right-hand side of the arrow in P1.

Note: Functor types are an example of an advanced programming language feature called *dependent types*, with which the **type** of an output is determined by the **value** of an input. That's different than the normal case of a function, where it's the output **value** that's determined by the input value, and the output **type** is independent of the input value.

Dependent types enable type systems to express much more about the correctness of a program, but type checking and inference for dependent types is much more challenging. Practical dependent type systems are an active area of research. Perhaps someday they will become popular in mainstream languages.

The module type of a functor's actual argument need not be identical to the formal declared module type of the argument; it's fine to be a subtype. For example, it's fine to apply F below to either X or Z. The extra item in Z won't cause any difficulty.

```
module F (M : sig val x : int end) = struct let y = M.x end
module X = struct let x = 0 end
module Z = struct let x = 0;; let z = 0 end
module FX = F (X)
module FZ = F (Z)
```

```
module F : functor (M : sig val x : int end) -> sig val y : int end
```

```
module X : sig val x : int end
```

```
module Z : sig val x : int val z : int end
```

```
module FX : sig val y : int end
```

```
module FZ : sig val y : int end
```

5.9.3 The Map Module

The standard library's Map module implements a map (a binding from keys to values) using balanced binary trees. It uses functors in an important way. In this section, we study how to use it. You can see the implementation of that module on GitHub as well as its interface.

The Map module defines a functor Make that creates a structure implementing a map over a particular type of keys. That type is the input structure to Make. The type of that input structure is Map.OrderedType, which are types that support a compare operation:

```
module type OrderedType = sig
  type t
  val compare : t -> t -> int
end
```

5.9. Functors 205

```
module type OrderedType = sig type t val compare : t -> t -> int end
```

The Map module needs ordering, because balanced binary trees need to be able to compare keys to determine whether one is greater than another. The compare function's specification is the same as that for the comparison argument to List.sort_uniq, which we previously discussed:

- The comparison should return 0 if two keys are equal.
- The comparison should return a strictly negative number if the first key is lesser than the second.
- The comparison should return a strictly positive number if the first key is greater than the second.

Note: Does that specification seem a little strange? Does it seem hard to remember when to return a negative vs. positive number? Why not define a variant instead?

```
type order = LT | EQ | GT
val compare : t -> t -> order
```

Alas, historically many languages have used comparison functions with similar specifications, such as the C standard library's stromp function. When comparing two integers, it does make the comparison easy: just perform a subtraction. It's not necessarily so easy for other data types.

The output of Map. Make supports all the usual operations we would expect from a dictionary:

```
module type S = sig
  type key
  type 'a t
  val empty: 'a t
  val mem: key -> 'a t -> bool
  val add: key -> 'a -> 'a t -> 'a t
  val find: key -> 'a t -> 'a
  ...
end
```

The type variable 'a is the type of values in the map. So any particular map module created by Map. Make can handle only one type of key, but is not restricted to any particular type of value.

An Example Map

Here's an example of using the Map. Make functor:

```
module IntMap = Map.Make(Int)
```

If you show that output, you'll see the long module type of IntMap. The Int module is part of the standard library. Conveniently, it already defines the two items required by OrderedType, which are t and compare, with appropriate behaviors. The standard library also already defines modules for the other primitive types (String, etc.) that make it convenient to use any primitive type as a key.

Now let's try out that map by mapping an int to a string:

```
open IntMap;;
let m1 = add 1 "one" empty
```

```
val m1 : string IntMap.t = <abstr>
```

```
find 1 m1
```

```
- : string = "one"
```

```
mem 42 m1
```

```
- : bool = false
```

```
find 42 m1
```

```
Exception: Not_found.

Raised at Stdlib__map.Make.find in file "map.ml", line 136, characters 10-25

Called from Toploop.load_lambda in file "toplevel/toploop.ml", line 212, characters_

417-27
```

```
bindings m1
```

```
- : (IntMap.key * string) list = [(1, "one")]
```

The same IntMap module allows us to map an int to a float:

```
let m2 = add 1 1. empty
```

```
val m2 : float IntMap.t = <abstr>
```

```
bindings m2
```

```
- : (IntMap.key * float) list = [(1, 1.)]
```

But the keys must be int, not any other type:

```
let m3 = add true "one" empty
```

That's because the IntMap module was specifically created for keys that are integers and ordered accordingly. Again, order is crucial, because the underlying data structure is a binary search tree, which requires key comparisons to figure out where in the tree to store a key. You can even see that in the standard library source code (v4.12), of which the following is a lightly-edited extract:

(continues on next page)

5.9. Functors 207

```
let empty = Empty

let rec mem x = function
    | Empty -> false
    | Node {1, v, r} ->
        let c = Ord.compare x v in
        c = 0 || mem x (if c < 0 then 1 else r)
    ...
end</pre>
```

The key type is defined to be a synonym for the type t inside Ord, so key values are comparable using Ord.compare. The mem function uses that to compare keys and decide whether to recurse on the left subtree or right subtree.

Note how the implementor of Map had a tricky problem to solve: balanced binary search trees require a way to compare keys, but the implementor can't know in advance all the different types of keys that a client of the data structure will want to use. And each type of key might need its own comparison function. Although Stdlib.compare can be used to compare any two values of the same type, the result it returns isn't necessarily what a client will want. For example, it's not guaranteed to sort names in the way we wanted above.

So the implementor of Map used a functor to solve their problem. They parameterized on a module that bundles together the type of keys with a function that can be used to compare them. It's the client's responsibility to implement that module.

The Java Collections Framework solves a similar problem in the TreeMap class, which has a constructor that takes a Comparator. There, the client has the responsibility of implementing a class for comparisons, rather than a structure. Though the language features are different, the idea is the same.

Maps with Custom Key Types

When the type of a key becomes complicated, we might want to write our own custom comparison function. For example, suppose we want a map in which keys are records representing names, and in which names are sorted alphabetically by last name then by first name. In the code below, we provide a module Name that can compare records that way:

```
type name = { first : string; last : string; }
```

```
module Name : sig type t = name val compare : name -> name -> int end
```

The Name module can be used as input to Map. Make because it satisfies the Map. OrderedType signature:

```
module NameMap = Map.Make (Name)
```

Now we could use that map to associate names with birth years:

```
let k1 = {last = "Kardashian"; first = "Kourtney"}
let k2 = {last = "Kardashian"; first = "Kimberly"}
```

```
let k3 = {last = "Kardashian"; first = "Khloe"}
let k4 = {last = "West"; first = "Kanye"}

let nm =
    NameMap.(empty |> add k1 1979 |> add k2 1980 |> add k3 1984 |> add k4 1977)

let lst = NameMap.bindings nm
```

```
val k1 : name = {first = "Kourtney"; last = "Kardashian"}
```

```
val k2 : name = {first = "Kimberly"; last = "Kardashian"}
```

```
val k3 : name = {first = "Khloe"; last = "Kardashian"}
```

```
val k4 : name = {first = "Kanye"; last = "West"}
```

```
val nm : int NameMap.t = <abstr>
```

```
val lst : (NameMap.key * int) list =
  [({first = "Khloe"; last = "Kardashian"}, 1984);
  ({first = "Kimberly"; last = "Kardashian"}, 1980);
  ({first = "Kourtney"; last = "Kardashian"}, 1979);
  ({first = "Kanye"; last = "West"}, 1977)]
```

Note how the order of keys in that list is not the same as the order in which we added them. The list is sorted according to the Name.compare function we wrote. Several of the other functions in the Map.S signature will also process map bindings in that sorted order—for example, map, fold, and iter.

How Map Uses Module Type Constraints

In the standard library's map.mli interface, the specification for Map.Make is:

```
module Make (Ord : OrderedType) : S with type key = Ord.t
```

The with constraint there is crucial. Recall that type constraints specialize a module type. Here, S with type key = Ord.t specializes S to expose the equality of S.key and Ord.t. In other words, the type of keys is the ordered type.

You can see the effect of that sharing constraint by looking at the module type of our IntMap example from before. The sharing constraint is what caused the = Int.t to be present:

```
module IntMap : sig
  type key = Int.t
  ...
end
```

And the Int module contains this line:

```
type t = int
```

So IntMap.key = Int.t = int, which is exactly why we're allowed to pass an int to the add and mem functions of IntMap.

5.9. Functors 209

Without the type constraint, type key would have remained abstract. We can simulate that by adding a module type annotation of Map. S, thereby resealing the module at that type without exposing the equality:

```
module UnusableMap = (IntMap : Map.S);;
```

```
module UnusableMap : Map.S
```

Now it's impossible to add a binding to the map:

```
let m = UnusableMap.(empty |> add 0 "zero")
```

```
File "[31]", line 1, characters 34-35:

1 | let m = UnusableMap.(empty |> add 0 "zero")

Error: This expression has type int but an expression was expected of type

UnusableMap.key
```

This kind of use case is why module type constraints are quite important in effective programming with the OCaml module system. Often it is necessary to specialize the output type of a functor to show a relationship between a type in it and a type in one of the functor's inputs. Thinking through exactly what constraint is necessary can challenging, though!

5.9.4 Using Functors

With Map we saw one use case for functors: producing a data structure that was parameterized on a client-provided ordering. Here are two more use cases.

Test Suites

Here are two implementations of a stack:

```
exception Empty
module type Stack = sig
 type 'a t
 val empty : 'a t
 val push : 'a -> 'a t -> 'a t
 val peek : 'a t -> 'a
 val pop : 'a t -> 'a t
end
module ListStack = struct
 type 'a t = 'a list
 let empty = []
 let push = List.cons
 let peek = function [] -> raise Empty | x :: _ -> x
 let pop = function [] -> raise Empty | _ :: s -> s
module VariantStack = struct
 type 'a t = E | S of 'a * 'a t
 let empty = E
 let push x s = S (x, s)
 let peek = function E -> raise Empty | S (x, _) -> x
 let pop = function E -> raise Empty | S (_, s) -> s
end
```

Suppose we wanted to write an OUnit test for ListStack:

```
let test = "peek (push x empty) = x" >:: fun _ ->
    assert_equal 1 ListStack.(empty |> push 1 |> peek)
```

Unfortunately, to test a Variant Stack, we'd have to duplicate that code:

```
let test' = "peek (push x empty) = x" >:: fun _ ->
    assert_equal 1 VariantStack.(empty |> push 1 |> peek)
```

And if we had other stack implementations, we'd have to duplicate the test for them, too. That's not so horrible to contemplate if it's just one test case for a couple implementations, but if it's hundreds of tests for even a couple implementations, that's just too much duplication to be good software engineering.

Functors offer a better solution. We can write a functor that is parameterized on the stack implementation, and produces the test for that implementation:

```
module StackTester (S : Stack) = struct
  let tests = [
    "peek (push x empty) = x" >:: fun _ ->
        assert_equal 1 S.(empty |> push 1 |> peek)
    ]
end

module ListStackTester = StackTester (ListStack)
module VariantStackTester = StackTester (VariantStack)

let all_tests = List.flatten [
    ListStackTester.tests;
    VariantStackTester.tests
]
```

Now whenever we invent a new test we add it to StackTester, and it automatically gets run on both stack implementations. Nice!

There is still some objectionable code duplication, though, in that we have to write two lines of code per implementation. We can eliminate that duplication through the use of first-class modules:

```
let stacks = [ (module ListStack : Stack); (module VariantStack) ]

let all_tests =
    let tests m =
    let module S = (val m : Stack) in
    let module T = StackTester (S) in
    T.tests
    in
    let open List in
    stacks |> map tests |> flatten
```

Now it suffices just to add the newest stack implementation to the stacks list. Nicer!

5.9. Functors 211

Extending Multiple Modules

Earlier, we tried to add a function of_list to both ListSet and UniqListSet without having any duplicated code, but we didn't totally succeed. Now let's really do it right.

The problem we had earlier was that we needed to parameterize the implementation of of_list on the add function and empty value in the set module. We can accomplish that parameterization with a functor:

```
module type Set = sig
  type 'a t
  val empty : 'a t
  val mem : 'a -> 'a t -> bool
  val add : 'a -> 'a t -> 'a t
  val elements : 'a t -> 'a list
end

module SetOfList (S : Set) = struct
  let of_list lst = List.fold_right S.add lst S.empty
end
```

Notice how the functor, in its body, uses S.add. It takes the implementation of add from S and uses it to implement of_list (and the same for empty), thus solving the exact problem we had before when we tried to use includes.

When we apply SetOfList to our set implementations, we get modules containing an of_list function for each implementation:

```
module ListSet : Set = struct
    type 'a t = 'a list
    let empty = []
    let mem = List.mem
    let add = List.cons
    let elements s = List.sort_uniq Stdlib.compare s
end

module UniqListSet : Set = struct
    (** All values in the list must be unique. *)
    type 'a t = 'a list
    let empty = []
    let mem = List.mem
    let add x s = if mem x s then s else x :: s
    let elements = Fun.id
end
```

```
module OfList = SetOfList (ListSet)
module UniqOfList = SetOfList (UniqListSet)
```

```
module OfList : sig val of_list : 'a list -> 'a ListSet.t end
```

```
module UniqOfList : sig val of_list : 'a list -> 'a UniqListSet.t end
```

The functor has enabled the code reuse we couldn't get before: we now can implement a single of_list function and from it derive implementations for two different sets.

But that's the **only** function those two modules contain. Really what we want is a full set implementation that also contains the of_list function. We can get that by combining includes with functors:

```
module SetWithOfList (S : Set) = struct
  include S
  let of_list lst = List.fold_right S.add lst S.empty
end
```

```
module SetWithOfList :
  functor (S : Set) ->
    sig
    type 'a t = 'a S.t
    val empty : 'a t
    val mem : 'a -> 'a t -> bool
    val add : 'a -> 'a t -> 'a t
    val elements : 'a t -> 'a list
    val of_list : 'a list -> 'a S.t
    end
```

That functor takes a set as input, and produces a module that contains everything from that set (because of the include) as well as a new function of_list.

When we apply the functor, we get a very nice set module:

```
module SetL = SetWithOfList (ListSet)
module UniqSetL = SetWithOfList (UniqListSet)
```

```
module SetL :
    sig
    type 'a t = 'a ListSet.t
    val empty : 'a t
    val mem : 'a -> 'a t -> bool
    val add : 'a -> 'a t -> 'a t
    val elements : 'a t -> 'a list
    val of_list : 'a list -> 'a ListSet.t
    end
```

```
module UniqSetL :
    sig
    type 'a t = 'a UniqListSet.t
    val empty : 'a t
    val mem : 'a -> 'a t -> bool
    val add : 'a -> 'a t -> 'a t
    val elements : 'a t -> 'a list
    val of_list : 'a list -> 'a UniqListSet.t
    end
```

Notice how the output structure records the fact that its type t is the same type as the type t in its input structure. They share it because of the include.

Stepping back, what we just did bears more than a passing resemblance to class extension in Java. We created a base module and extended its functionality with new code while preserving its old functionality. But whereas class extension necessitates that the newly extended class is a subtype of the old, and that it still has all the old functionality, OCaml functors are more fine-grained in what they can accomplish. We can choose whether they include the old functionality. And no subtyping relationships are necessarily involved. Moreover, the functor we wrote can be used to extend **any** set implementation with of_list, whereas class extension applies to just a **single** base class. There are ways of achieving something similar in Java with *mixins*, which were added in Java 1.5.

5.9. Functors 213

5.10 Summary

The OCaml module system provides mechanisms for modularity that provide the similar capabilities as mechanisms you will have seen in other languages. But seeing those mechanisms appear in different ways is hopefully helping you understand them better. OCaml abstract types and signatures, for example, provide a mechanism for abstraction that resembles Java visibility modifiers and interfaces. Seeing the same idea embodied in two different languages, but expressed in rather different ways, will hopefully help you recognize that idea when you encounter it in other languages in the future.

Moreover, the idea that a type could be abstract is a foundational notion in programming language design. The OCaml module system makes that idea brutally apparent. Other languages like Java obscure it a bit by coupling it together with many other features all at once. There's a sense in which every Java class implicitly defines an abstract type (actually, four abstract types that are related by subtyping, one for each visibility modifier [public, protected, private, and default]), and all the methods of the class are functions on that abstract type.

Functors are an advanced language feature in OCaml that might seem mysterious at first. If so, keep in mind: they're really just a kind of function that takes a structure as input and returns a structure as output. The reason they don't behave quite like normal OCaml functions is that structures are not first-class values in OCaml: you can't write regular functions that take a structure as input or return a structure as output. But functors can do just that.

Functors and includes enable code reuse. The kinds of code reuse that object-oriented features enable can also be achieved with functors and include. That's not to say that functors and includes are exactly equivalent to those object-oriented features: some kinds of code reuse might be easier to achieve with one set of features than the other.

One way to think about this might be that class extension is a very limited, but very useful, combination of functors and includes. Extending a class is like writing a functor that takes the base class as input, includes it, then adds new functions. But functors provide more general capability than class extension, because they can compute arbitrary functions of their input structure, rather than being limited to just certain kinds of extension.

Perhaps the most important idea to get out of studying the OCaml module system is an appreciation for the aspects of modularity that transcend any given language: namespaces, abstraction, and code reuse. Having seen those ideas in a couple very different languages, you're equipped to recognize them more clearly in the next language you learn.

5.10.1 Terms and concepts

- · abstract type
- · abstraction
- · client
- · code reuse
- · compilation unit
- · declaration
- · definition
- · encapsulation
- · ephemeral data structure
- · functional data structure
- functor
- · implementation
- implementer
- include

- · information hiding
- interface
- · local reasoning
- · maintainability
- maps
- · modular programming
- modularity
- module
- module type
- namespace
- open
- · parameterized structure
- · persistent data structure
- · representation type
- scope
- · sealed
- · set representations
- · sharing constraints
- signature
- · signature matching
- specification
- structure

5.10.2 Further reading

- Introduction to Objective Caml, chapters 11, 12, and 13
- OCaml from the Very Beginning, chapter 16
- Real World OCaml, chapters 4, 9, and 10
- Purely Functional Data Structures, chapters 1 and 2, by Chris Okasaki.
- "Design Considerations for ML-Style Module Systems" by Robert Harper and Benjamin C. Pierce, chapter 8 of *Advanced Topics in Types and Programming Languages*, ed. Benjamin C. Pierce, MIT Press, 2005. An advanced treatment of the static semantics of modules.

5.10. Summary 215

5.11 Exercises

Solutions to exercises are available to students in Cornell's CS 3110. Instructors at other institutions are welcome to contact Michael Clarkson for access.

Exercise: complex synonym [★]

Here is a module type for complex numbers, which have a real and imaginary component:

```
module type ComplexSig = sig
  val zero : float * float
  val add : float * float -> float * float * float
end
```

Improve that code by adding type t = float * float. Show how the signature can be written more tersely because of the type synonym.

Exercise: complex encapsulation [★★]

Here is a module for the module type from the previous exercise:

```
module Complex : ComplexSig = struct
  type t = float * float
  let zero = (0., 0.)
  let add (r1, i1) (r2, i2) = r1 +. r2, i1 +. i2
end
```

Investigate what happens if you make the following changes (each independently), and explain why any errors arise:

- remove zero from the structure
- remove add from the signature
- change zero in the structure to let zero = 0, 0

Exercise: big list queue $[\star \star]$

Use the following code to create ListQueue of exponentially increasing length: 10, 100, 1000, etc. How big of a queue can you create before there is a noticeable delay? How big until there's a delay of at least 10 seconds? (Note: you can abort utop computations with Ctrl-C.)

```
(** Creates a ListQueue filled with [n] elements. *)
let fill_listqueue n =
  let rec loop n q =
    if n = 0 then q
    else loop (n - 1) (ListQueue.enqueue n q) in
  loop n ListQueue.empty
```

Exercise: big batched queue $[\star \star]$

Use the following function to create BatchedQueue of exponentially increasing length:

```
let fill_batchedqueue n =
  let rec loop n q =
   if n = 0 then q
    else loop (n - 1) (BatchedQueue.enqueue n q) in
  loop n BatchedQueue.empty
```

Now how big of a queue can you create before there's a delay of at least 10 seconds?

Exercise: queue efficiency $[\star \star \star]$

Compare the implementations of enqueue in ListQueue vs. BatchedQueue. Explain in your own words why the efficiency of ListQueue enqueue is linear time in the length of the queue. Hint: consider the @ operator. Then explain why adding n elements to the queue takes time that is quadratic in n.

Now consider BatchedQueue.enqueue. Suppose that the queue is in a state where it has never had any elements dequeued. Explain in your own words why BatchedQueue.enqueue is constant time. Then explain why adding n elements to the queue takes time that is linear in n.

Exercise: binary search tree dictionary $[\star \star \star]$

Write a module BstDict that implements the Dictionary module type using the a binary search tree type.

Exercise: fraction $[\star \star \star]$

Write a module that implements the Fraction module type below:

```
module type Fraction = sig
    (* A fraction is a rational number p/q, where q != 0.*)
    type t

    (** [make n d] is n/d. Requires d != 0. *)
    val make : int -> int -> t

    val numerator : t -> int
    val denominator : t -> int
    val to_string : t -> string
    val to_float : t -> float

    val add : t -> t -> t
    val mul : t -> t -> t
end
```

Exercise: fraction reduced $[\star\star\star]$

Modify your implementation of Fraction to ensure these invariants hold of every value v of type t that is returned from make, add, and mul:

- 1. v is in reduced form
- 2. the denominator of v is positive

For the first invariant, you might find this implementation of Euclid's algorithm to be helpful:

5.11. Exercises 217

```
(** [gcd x y] is the greatest common divisor of [x] and [y].
   Requires: [x] and [y] are positive. *)
let rec gcd x y =
   if x = 0 then y
   else if (x < y) then gcd (y - x) x
   else gcd y (x - y)</pre>
```

Exercise: make char map [★]

To create a standard library map, we first have to use the Map. Make functor to produce a module that is specialized for the type of keys we want. Type the following in utop:

```
# module CharMap = Map.Make(Char);;
```

The output tells you that a new module named CharMap has been defined, and it gives you a signature for it. Find the values empty, add, and remove in that signature. Explain their types in your own words.

Exercise: char ordered [★]

The Map. Make functor requires its input module to match the Map.OrderedType signature. Look at that signature as well as the signature for the Char module. Explain in your own words why we are allowed to pass Char as an argument to Map.Make.

Exercise: use char map $[\star \star]$

Using the CharMap you just made, create a map that contains the following bindings:

```
'A' maps to "Alpha"'E' maps to "Echo"
```

- 'S' maps to "Sierra"
- 'V' maps to "Victor"

Use CharMap.find to find the binding for 'E'.

Now remove the binding for 'A'. Use CharMap.mem to find whether 'A' is still bound.

Use the function CharMap.bindings to convert your map into an association list.

Exercise: bindings $[\star\star]$

Investigate the documentation of the Map. S signature to find the specification of bindings. Which of these expressions will return the same association list?

```
1. CharMap.(empty |> add 'x' 0 |> add 'y' 1 |> bindings)
```

```
2. CharMap.(empty \mid > add 'y' 1 \mid > add 'x' 0 \mid > bindings)
```

```
3. CharMap.(empty \mid > add 'x' 2 \mid > add 'y' 1 \mid > remove 'x' \mid > add 'x' 0 \mid > bindings)
```

Check your answer in utop.

Exercise: date order [★★]

Here is a type for dates:

```
type date = {month : int; day : int}
```

For example, March 31st would be represented as $\{month = 3; day = 31\}$. Our goal in the next few exercises is to implement a map whose keys have type date.

Obviously it's possible to represent invalid dates with type date—for example, { month=6; day=50 } would be June 50th, which is not a real date. The behavior of your code in the exercises below is unspecified for invalid dates.

To create a map over dates, we need a module that we can pass as input to Map. Make. That module will need to match the Map.OrderedType signature. Create such a module. Here is some code to get you started:

```
module Date = struct
  type t = date
  let compare ...
end
```

Recall the specification of compare in Map. Ordered Type as you write your Date. compare function.

Exercise: calendar [★★]

Use the Map. Make functor with your Date module to create a DateMap module. Then define a calendar type as follows:

```
type calendar = string DateMap.t
```

The idea is that calendar maps a date to the name of an event occurring on that date.

Using the functions in the DateMap module, create a calendar with a few entries in it, such as birthdays or anniversaries.

Exercise: print calendar $[\star \star]$

Write a function print_calendar : calendar -> unit that prints each entry in a calendar in a format similar the inspiring examples in the previous exercise. Hint: use DateMap.iter, which is documented in the Map.S signature.

Exercise: is for $[\star \star \star]$

Write a function is_for: string CharMap.t -> string CharMap.t that given an input map with bindings from k_1 to v_1, \ldots, k_n to v_n , produces an output map with the same keys, but where each key k_i is now bound to the string " k_i is for v_i ". For example, if m maps 'a' to "apple", then is_for m would map 'a' to "a is for apple". Hint: there is a one-line solution that uses a function from the Map.S signature. To convert a character to a string, you could use String.make. An even function way would be to use Printf.sprintf.

Exercise: first after $[\star \star \star]$

Write a function first_after: calendar -> Date.t -> string that returns the name of the first event that occurs strictly after the given date. If there is no such event, the function should raise Not_found, which is an

5.11. Exercises 219

exception already defined in the standard library. *Hint: there is a one-line solution that uses two functions from the Map . S signature.*

Exercise: sets $[\star \star \star]$

The standard library Set module is quite similar to the Map module. Use it to create a module that represents sets of *case-insensitive strings*. Strings that differ only in their case should be considered equal by the set. For example, the sets {"grr", "argh"} and {"aRgh", "GRR"} should be considered the same, and adding "gRr" to either set should not change the set.

Exercise: ToString [★★]

Write a module type ToString that specifies a signature with an abstract type t and a function to_string : t -> string.

Exercise: Print $[\star \star]$

Write a functor Print that takes as input a module named M of type ToString. The nidyke returned by your functor should have exactly one value in it, print, which is a function that takes a value of type M.t and prints a string representation of that value.

Exercise: Print Int $[\star \star]$

Create a module named PrintInt that is the result of applying the functor Print to a new module Int. You will need to write Int yourself. The type Int.t should be int. *Hint: do not seal Int*.

Experiment with PrintInt in utop. Use it to print the value of an integer.

Exercise: Print String [★★]

Create a module named PrintString that is the result of applying the functor Print to a new module MyString. You will need to write MyString yourself. *Hint: do not seal MyString*.

Experiment with PrintString in utop. Use it to print the value of a string.

Exercise: Print Reuse [★]

Explain in your own words how Print has achieved code reuse, albeit a very small amount.

Exercise: Print String reuse revisited $[\star \star]$

The PrintString module you created above supports just one operation: print. It would be great to have a module that supports all the String module functions in addition to that print operation, and it would be super great to derive such a module without having to copy any code.

Define a module StringWithPrint. It should have all the values of the built-in String module. It should also have the print operation, which should be derived from the Print functor rather than being copied code. *Hint: use two include statements*.

Exercise: implementation without interface [★]

Create a file named date.ml. In it put the following code:

```
type date = {month : int; day : int}
let make_date month day = {month; day}
let get_month d = d.month
let get_day d = d.day
let to_string d = (string_of_int d.month) ^ "/" ^ (string_of_int d.day)
```

Compile that file to bytecode:

```
$ ocamlbuild date.cmo
```

Now start utop and type the following to use the module you've just created:

```
# #directory "_build";;
# #load "date.cmo";;
```

Finally, create a date, access its day, and convert it to a string.

Exercise: implementation with interface [★]

After doing the previous exercise, also create a file named date.mli. In it put the following code:

```
type date = {month : int; day : int}
val make_date : int -> int -> date
val get_month : date -> int
val get_day : date -> int
val to_string : date -> string
```

Recompile date.ml to bytecode:

```
$ ocamlbuild date.cmo
```

Restart utop and re-issue the same phrases as before.

Exercise: implementation with abstracted interface [★]

After doing the previous two exercises, edit date.mli and change the first declaration in it to the following:

```
type date
```

The type date is now abstract. Recompile date.ml to bytecode, restart utop, and re-issue the same phrases as before. The responses to two of them will change. Explain in your own words those changes.

Exercise: printer for dat $[\star\star\star]$

Add a declaration to date.mli:

```
val format : Format.formatter -> date -> unit
```

And add a definition of format to date.ml. *Hint: use Format.fprintf and Date.to_string.*

Now recompile, load utop, and after loading date. cmo install the printer by issuing the directive

```
#install_printer Date.format;;
```

5.11. Exercises 221

Reissue the other phrases to utop as you did in the exercises above. The response from one phrase will change in a helpful way. Explain why.

Exercise: refactor arith $[\star \star \star \star]$

Download this file: algebra.ml. It contains two signatures and four structures:

- Ring is signature that describes the algebraic structure called a *ring*, which is an abstraction of the addition and multiplication operators.
- Field is a signature that describes the algebraic structure called a *field*, which is like a ring but also has an abstraction of the division operation.
- IntRing and FloatRing are structures that implement rings in terms of int and float.
- IntField and FloatField are structures that implement fields in terms of int and float.
- IntRational and FloatRational are structures that implement fields in terms of ratios (aka fractions)—that is, pairs of int and pairs of float.

Note: Dear fans of abstract algebra: of course these representations don't necessarily obey all the axioms of rings and fields because of the limitations of machine arithmetic. Also, the division operation in IntField is ill-defined on zero. Try not to worry about that.

Refactor the code to improve the amount of code reuse it exhibits. To do that, use include, functors, and introduce additional structures and signatures as needed. There isn't necessarily a right answer here, but here's some advice:

- No name should be *directly declared* in more than one signature. For example, (+) should not be directly declared in Field; it should be reused from an earlier signature. By "directly declared" we mean a declaration of the form val name: An indirect declaration would be one that results from an include.
- You need only three *direct definitions* of the algebraic operations and numbers (plus, minus, times, divide, zero, one): once for int, once for float, and once for ratios. For example, IntField. (+) should not be directly defined as Stdlib. (+); rather, it should be reused from elsewhere. By "directly defined" we mean a definition of the form let name = An indirect definition would be one that results from an include or a functor application.
- The rational structures can both be produced by a single functor that is applied once to IntField and once to FloatField.
- It's possible to eliminate all duplication of of_int, such that it is directly defined exactly once, and all structures reuse that definition; and such that it is directly declared in only one signature. This will require the use of functors. It will also require inventing an algorithm that can convert an integer to an arbitrary Ring representation, regardless of what the representation type of that Ring is.

When you're done, the types of all the modules should remain unchanged. You can easily see those types by running ocamlc -i algebra.ml.

Part III Correctness and Efficiency

CORRECTNESS

When we write code, we always hope that we get it right. We *hope* that our code is correct. But how can we *know* it's correct? In this chapter, we'll study three possible answers: documentation, testing, and proof.

Let's be honest: we all at one time or another have thought that documentation or testing was a boring, tedious, and altogether postponable task. But with maturity programmers come to realize that both are essential to writing correct code. Both get at the *truth* of what code really does.

Documentation is the ground truth of what a programmer intended, as opposed to what they actually wrote. It communicates to other humans the ideas the author had in their head. No small amount of the time (even in this book!), we fail at communicating ideas as we intended. Maybe the failure occurs in the code, or maybe in the documentation. But writing documentation forces us to think a second (er, *hopefully* second) time about our intentions. The cognitive task of explaining our ideas to other humans is certainly different than explaining our ideas to the computer. That can expose failures in our thinking.

More importantly, documentation is a message in a time capsule. Imagine this: someone far away and now unreachable has sent that message to you, the programmer. You need that message to interpret the archeological evidence now in front of you—i.e., the otherwise unintelligible source code you have inherited. Your only hope is that the original author, long ago, had enough empathy to commit their thoughts to the written word.

And now imagine this: that author from the distant past? **What if they were YOU?** It might be you from two weeks ago, two months ago, or two years ago. Human memory is fleeting. If you've only been programming for a couple of years yourself, this can be difficult to understand, but give it a generous try: Someday, you're going to come back to the code you're writing today and have no clue what it means. Your only hope is to leave yourself some breadcrumbs at the time you write it. Otherwise, you'll be lost when you circle back.

Testing is the ground truth of what a program actually does, as opposed to what the programmer intended. It provides evidence that the programmer got it right. Good scientists demand evidence. That demand comes not out of arrogance but humility. We human beings are so amazingly good at deluding ourselves. (Consider the echo chamber of modern social media.) You can write a piece of code that you *think* is right. But then you can write a test case that *demonstrates* it's right. Then you can write ten more. The evidence accumulates, and eventually it's enough to be convincing. Is it absolute? Of course not. Maybe there's some test case you weren't clever enough to invent. That's science: new ideas come along to challenge the old.

Even more importantly, testing is *repeatable* science. The ability to replicate experiments is crucial to the truth they establish. By capturing tests as automatically repeatable experiments as unit test suites, we can demonstrate to ourselves and other, now and in the future, that our code is correct.

The challenge of documentation and testing is discipline. It's so tempting, so easy, to care only about writing the code. "That's the fun part", right? But it's like leaving out a third of the letter we intended to write. One part of the letter is to the machine, regarding how to compute. But another part is to other humans, about what we wanted to compute. And another part is to both machines and humans, about what we really did manage to compute. Your job isn't done until all three parts have been written.

If you're not yet convinced about the importance of documentation and testing, no worries. You will be in the future, if you stick with the craft of programming long enough. Meanwhile, let's proceed with learning about how to do it better.

In this chapter, we're going to learn about some successful (and hopefully new-to-you) techniques for both.

Finally, beyond documentation and testing, there is mathematical **proof** of correctness. Techniques from logic and discrete math can be used to formally prove that a program is correct according to a specification. Such proofs aren't necessarily easy—in fact they take even more human discipline and training than documentation and testing do. But they can make sense to apply when programs are used for safety critical tasks where human lives are on the line.

6.1 Specifications

A *specification* is a contract between a *client* of some unit of code and the *implementer* of that code. The most common place we find specifications is as comments in the interface (.mli) files for a module. There, the implementer of the module spells out what the client may and may not assume about the module's behavior. This contract makes it clear who to blame if something goes wrong: Did the client misuse the module? Or did the implementer fail to deliver the promised functionality?

Specifications usually involve preconditions and postconditions. The preconditions inform what the client must guarantee about inputs they pass in, and what the implementer may assume about those inputs. The postconditions inform what they client may assume about outputs they receive, and what the implementer must guarantee about those outputs.

An implementation *satisfies* a specification if it provides the behavior described by the specification. There may be many possible implementations of a given specification that are feasible. The client may not assume anything about which of those implementations is actually provided. The implementer, on the other hand, gets to provide one of their choice.

Clear specifications serve many important functions in software development teams. One important one is when something goes wrong, everyone can agree on whose job it is to fix the problem: either the implementer has not met the specification and needs to fix the implementation, or the client has written code that assumes something not guaranteed by the spec, and therefore needs to fix the using code. Or, perhaps the spec is wrong, and then the client and implementer need to decide on a new spec. This ability to decide whose problem a bug is prevents problems from slipping through the cracks.

Writing Specifications. Good specifications have to balance two conflicting goals; they must be

- sufficiently restrictive, ruling out implementations that would be useless to clients, as well as
- sufficiently general, not ruling out implementations that would be useful to clients.

Some common mistakes include not stating enough in preconditions, failing to identify when exceptions will be thrown, failing to specify behavior at boundary cases, writing operational specifications instead of definitional and stating too much in postconditions.

Writing good specifications is hard because the language and compiler do nothing to check the correctness of a specification: there's no type system for them, no warnings, etc. (Though there is ongoing research on how to improve specifications and the writing of them.) The specifications you write will be read by other people, and with that reading can come misunderstanding. Reading specifications requires close attention to detail.

Specifications should be written quite early. As soon as a design decision is made, document it in a specification. Specifications should continue to be updated throughout implementation. A specification becomes obsolete only when the code it specifies becomes obsolete and is removed from the code base.

Abstraction by Specification. Abstraction enables modular programming by hiding the details of implementations. Specifications are a part of that kind of abstraction: they reveal certain information about the behavior of a module without disclosing all the details of the module's implementation.

Locality is one of the benefits of abstraction by specification. A module can be understood without needing to examine its implementation. This locality is critical in implementing large programs, and even in in implementing smaller programs in teams. No one person can keep the entire system in their head at a time.

Modifiability is another benefit. Modules can be reimplemented without changing the implementation of other modules or functions. Software libraries depend upon this to improve their functionality without forcing all their clients to rewrite

code every time the library is upgraded. Modifiability also enables performance enhancements: we can write simple, slow implementations first, then improve bottlenecks as necessary.

The client should not assume more about the implementation than is given in the spec because that allows the implementation to change. The specification forms an *abstraction barrier* that protects the implementer from the client and vice versa. Making assumptions about the implementation that are not guaranteed by the specification is known as *violating the abstraction barrier*. The abstraction barrier enforces local reasoning. Further, it promotes *loose coupling* between different code modules. If one module changes, other modules are less likely to have to change to match.

6.2 Function Documentation

This section continues the discussion of documentation, which we began in chapter 2.

A specification is written for humans to read, not machines. "Specs" can take time to write well, and it is time well spent. The main goal is clarity. It is also important to be concise, because client programmers will not always take the effort to read a long spec. As with anything we write, we need to be aware of your audience when writing specifications. Some readers may need a more verbose specification than others.

A well-written specification usually has several parts communicating different kinds of information about the thing specified. If we know what the usual ingredients of a specification are, we are less likely to forget to write down something important. Let's now look at a recipe for writing specifications.

6.2.1 Returns Clause

How might we specify sqr, a square-root function? First, we need to describe its result. We will call this description the *returns clause* because it is a part of the specification that describes the result of a function call. It is also known as a *postcondition*: it describes a condition that holds after the function is called. Here is an example of a returns clause:

```
(** returns: [sqr x] is the square root of [x]. *)
```

But we would typically leave out the returns:, and simply write the returns clause as the first sentence of the comment:

```
(** [sqr x] is the square root of [<math>x]. *)
```

For numerical programming, we should probably add some information about how accurate it is.

```
(** [sqr \ x] is the square root of [x]. Its relative accuracy is no worse than [1.0e-6]. *)
```

Similarly, here's how we might write a returns clause for a find function:

```
(** [find lst x] is the index of [x] in [lst], starting from zero. *)
```

A good specification is concise but clear—it should say enough that the reader understands what the function does, but without extra verbiage to plow through and possibly cause the reader to miss the point. Sometimes there is a balance to be struck between brevity and clarity.

These two specifications use a useful trick to make them more concise: they talk about the result of applying the function being specified to some arbitrary arguments. Implicitly we understand that the stated postcondition holds for all possible values of any unbound variables (the argument variables).

6.2.2 Requires Clause

The specification for sqr doesn't completely make sense because the square root does not exist for some x of type real. The mathematical square root function is a *partial* function that is defined over only part of its domain. A good function specification is complete with respect to the possible inputs; it provides the client with an understanding of what inputs are allowed and what the results will be for allowed inputs.

We have several ways to deal with partial functions. A straightforward approach is to restrict the domain so that it is clear the function cannot be legitimately used on some inputs. The specification rules out bad inputs with a *requires clause* establishing when the function may be called. This clause is also called a *precondition* because it describes a condition that must hold before the function is called. Here is a requires clause for sqr:

```
(** [sqr \ x] is the square root of [x]. Its relative accuracy is no worse than [1.0e-6]. Requires: [x >= 0] *)
```

This specification doesn't say what happens when x < 0, nor does it have to. Remember that the specification is a contract. This contract happens to push the burden of showing that the square root exists onto the client. If the requires clause is not satisfied, the implementation is permitted to do anything it likes: for example, go into an infinite loop or throw an exception. The advantage of this approach is that the implementer is free to design an algorithm without the constraint of having to check for invalid input parameters, which can be tedious and slow down the program. The disadvantage is that it may be difficult to debug if the function is called improperly, because the function can misbehave and the client has no understanding of how it might misbehave.

6.2.3 Raises Clause

Another way to deal with partial functions is to convert them into total functions (functions defined over their entire domain). This approach is arguably easier for the client to deal with because the function's behavior is always defined; it has no precondition. However, it pushes work onto the implementer and may lead to a slower implementation.

How can we convert sqr into a total function? One approach that is (too) often followed is to define some value that is returned in the cases that the requires clause would have ruled; for example:

```
(** [sqr x] is the square root of [x] if [x >= 0],
with relative accuracy no worse than 1.0e-6.
Otherwise, a negative number is returned. *)
```

This practice is not recommended because it tends to encourage broken, hard-to-read client code. Almost any correct client of this abstraction will write code like this if the precondition cannot be argued to hold:

```
if sqr(a) < 0.0 then ... else ...</pre>
```

The error must still be handled in the <code>if</code> expression, so the job of the client of this abstraction isn't any easier than with a requires clause: the client still needs to wrap an explicit test around the call in cases where it might fail. If the test is omitted, the compiler won't complain, and the negative number result will be silently treated as if it were a valid square root, likely causing errors later during program execution. This coding style has been the source of innumerable bugs and security problems in the Unix operating systems and its descendents (e.g., Linux).

A better way to make functions total is to have them raise an exception when the expected input condition is not met. Exceptions avoid the necessity of distracting error-handling logic in the client's code. If the function is to be total, the specification must say what exception is raised and when. For example, we might make our square root function total as follows:

```
(** [sqr x] is the square root of [x], with relative accuracy no worse than 1.0e-6. Raises: [Negative] if [x < 0]. *)
```

Note that the implementation of this sqr function must check whether x >= 0, even in the production version of the code, because some client may be relying on the exception to be raised.

6.2.4 Examples Clause

It can be useful to provide an illustrative example as part of a specification. No matter how clear and well written the specification is, an example is often useful to clients.

```
(** [find lst x] is the index of [x] in [lst], starting from zero.
Example: [find ["b","a","c"] "a" = 1]. *)
```

6.2.5 The Specification Game

When evaluating specifications, it can be useful to imagine that a game is being played between two people: a *specifier* and a *devious programmer*.

Suppose that the specifier writes the following specification:

```
(** returns a list *)
val reverse : 'a list -> 'a list
```

This spec is clearly incomplete. For example, a devious programmer could meet the spec with an implementation that gives the following output:

```
# reverse [1; 2; 3];;
- : int list = []
```

The specifier, upon realizing this, refines the spec as follows:

```
(** [reverse lst] returns a list that is the same length as [lst] *)
val reverse : 'a list -> 'a list
```

But the devious specifier discovers that the spec still allows broken implementations:

```
# reverse [1; 2; 3];;
- : int list = [0; 0; 0]
```

Finally, the specifier settles on a third spec:

```
(** [reverse lst] returns a list [m] satisfying the following conditions:
    - [length lst = length m]
    - for all [i], [nth m i = nth lst (n - i - 1)],
        where [n] is the length of [lst].
    For example, [reverse [1; 2; 3]] is [3; 2; 1], and [reverse []] is []. *)
val reverse : 'a list -> 'a list
```

With this spec, the devious programmer is forced to provide a working implementation to meet the spec, so the specifier has successfully written her spec.

The point of playing this game is to improve your ability to write specifications. Obviously we're not advocating that you deliberately try to violate the intent of a specification and get away with it. When reading someone else's specification, read as generously as possible. But be ruthless about improving your own specifications.

6.2.6 Comments

In addition to specifying functions, programmers need to provide comments in the body of the functions. In fact, programmers usually do not write enough comments in their code. (For a classic example, check out the actual comment on line 561 of the Quake 3 Arena game engine.)

But this doesn't mean that adding more comments is always better. The wrong comments will simply obscure the code further. Shoveling as many comments into code as possible usually makes the code worse! Both code and comments are precise tools for communication (with the computer and with other programmers) that should be wielded carefully.

It is particularly annoying to read code that contains many interspersed comments (typically of questionable value), e.g.:

```
let y = x + 1 (* make y one greater than x *)
```

For complex algorithms, some comments may be necessary to explain how the code implementing the algorithm works. Programmers are often tempted to write comments about the algorithm interspersed through the code. But someone reading the code will often find these comments confusing because they don't have a high-level picture of the algorithm. It is usually better to write a paragraph-style comment at the beginning of the function explaining how its implementation works. Explicit points in the code that need to be related to that paragraph can then be marked with very brief comments, like (* case 1 *).

Another common but well-intentioned mistake is giving variables long, descriptive names, as in the following verbose code:

```
let number_of_zeros the_list =
   List.fold_left (fun (accumulator : int) (list_element : int) ->
   accumulator + (if list_element = 0 then 1 else 0)) 0 the_list
```

```
val number_of_zeros : int list -> int = <fun>
```

Code using such long names is verbose and hard to read. Instead of trying to embed a complete description of a variable in its name, use a short and suggestive name (e.g., zeros), and if necessary, add a comment at its declaration explaining the purpose of the variable.

```
let zeros lst =
  let is0 = function 0 -> 1 | _ -> 0 in
  List.fold_left (fun zs x -> zs + is0 x) 0 lst
```

```
val zeros : int list -> int = <fun>
```

A similarly bad practice is to encode the type of the variable in its name, e.g. naming a variable i_count to show that it's an integer. The type system is going to guarantee that for you, and your editor can provide a hover-over to show the type. If you really want to emphasize the type in the code, add a type annotation at the point where the variable comes into scope.

6.3 Module Documentation

The specification of functions provided by a module can be found in its interface, which is what clients will consult. But what about *internal* documentation, which is relevant to those who implement and maintain a module? The purpose of such implementation comments is to explain to the reader how the implementation correctly implements its interface.

Reminder

It is inappropriate to copy the specifications of functions found in the module interface into the module implementation. Copying runs the risk of introducing inconsistency as the program evolves, because programmers don't keep the copies

in sync. Copying code and specifications is a major source (if not *the* major source) of program bugs. In any case, implementers can always look at the interface for the specification.

Implementation comments fall into two categories. The first category arises because a module implementation may define new types and functions that are purely internal to the module. If their significance is not obvious, these types and functions should be documented in much the same style that we have suggested for documenting interfaces. Often, as the code is written, it becomes apparent that the new types and functions defined in the module form an internal data abstraction or at least a collection of functionality that makes sense as a module in its own right. This is a signal that the internal data abstraction might be moved to a separate module and manipulated only through its operations.

The second category of implementation comments is associated with the use of *data abstraction*. Suppose we are implementing an abstraction for a set of items of type 'a. The interface might look something like this:

```
(** A set is an unordered collection in which multiplicity is ignored. *)
module type Set = sig
  (** ['a t] represents a set whose elements are of type ['a] *)
 type 'a t
  (** [empty] is the set containing no elements *)
 val empty : 'a t
  (** [mem x s] is whether [x] is a member of set [s] *)
 val mem : 'a -> 'a t -> bool
  (** [add x s] is the set containing all the elements of [s]
     as well as [x]. *)
 val add : 'a -> 'a t -> 'a t
  (** [rem x s] is the set containing all the elements of [s],
     minus [x]. *)
 val rem : 'a -> 'a t -> 'a t
  (** [size s] is the cardinality of [s] *)
 val size: 'a t -> int
  (** [union s1 s2] is the set containing all the elements that
     are in either [s1] or [s2]. *)
 val union: 'a t -> 'a t -> 'a t
  (** [inter s1\ s2] is the set containing all the elements that
      are in both [s1] and [s2]. *)
 val inter: 'a t -> 'a t -> 'a t
```

In a real signature for sets, we'd want operations such as map and fold as well, but let's omit these for now for simplicity. There are many ways to implement this abstraction.

As we've seen before, one easy way is as a list:

```
(** Implementation of sets as lists with duplicates *)
module ListSet : Set = struct
  type 'a t = 'a list
  let empty = []
  let mem = List.mem
  let add = List.cons
  let rem x = List.filter (( <> ) x)
```

(continues on next page)

(continued from previous page)

```
let size lst = List.(lst |> sort_uniq Stdlib.compare |> length)
let union lst1 lst2 = lst1 @ lst2
let inter lst1 lst2 = List.filter (fun h -> mem h lst2) lst1
end
```

```
module ListSet : Set
```

This implementation has the advantage of simplicity. For small sets that tend not to have duplicate elements, it will be a fine choice. Its performance will be poor for large sets or applications with many duplicates but for some applications that's not an issue.

Notice that the types of the functions do not need to be written down in the implementation. They aren't needed because they're already present in the signature, just like the specifications that are also in the signature don't need to be replicated in the structure.

Here is another implementation of Set that also uses 'a list but requires the lists to contain no duplicates. This implementation is also correct (and also slow for large sets). Notice that we are using the same representation type, yet some important aspects of the implementation (add, size, union) are quite different.


```
(** Implementation of sets as lists without duplicates *)
module UniqListSet : Set = struct
  type 'a t = 'a list
  let empty = []
  let mem = List.mem
  let add x lst = if mem x lst then lst else x :: lst
  let rem x = List.filter (( <> ) x)
  let size = List.length
  let union lst1 lst2 = lst1 @ lst2 |> List.sort_uniq Stdlib.compare
  let inter lst1 lst2 = List.filter (fun h -> mem h lst2) lst1
end
```

```
module UniqListSet : Set
```

An important reason why we introduced the writing of function specifications was to enable *local reasoning*: once a function has a spec, we can judge whether the function does what it is supposed to without looking at the rest of the program. We can also judge whether the rest of the program works without looking at the code of the function. However, we cannot reason locally about the individual functions in the three module implementations just given. The problem is that we don't have enough information about the relationship between the concrete type (int list) and the corresponding abstract type (set). This lack of information can be addressed by adding two new kinds of comments to the implementation: the *abstraction function* and the *representation invariant* for the abstract data type. We turn to discussion of those, next.

6.3.1 Abstraction Functions

The client of any Set implementation should not be able to distinguish it from any other implementation based on its functional behavior. As far as the client can tell, the operations act like operations on the mathematical ideal of a set. In the first implementation, the lists [3; 1], [1; 3], and [1; 1; 3] are distinguishable to the implementer, but not to the client. To the client, they all represent the abstract set {1, 3} and cannot be distinguished by any of the operations of the Set signature. From the point of view of the client, the abstract data type describes a set of abstract values and associated operations. The implementer knows that these abstract values are represented by concrete values that may contain additional information invisible from the client's view. This loss of information is described by the *abstraction function*, which is a mapping from the space of concrete values to the abstract space. The abstraction function for the implementation ListSet looks like this:



Notice that several concrete values may map to a single abstract value; that is, the abstraction function may be *many-to-one*. It is also possible that some concrete values do not map to any abstract value; the abstraction function may be *partial*. That is not the case with ListSet, but it might be with other implementations.

The abstraction function is important for deciding whether an implementation is correct, therefore it belongs as a comment in the implementation of any abstract data type. For example, in the ListSet module, we could document the abstraction function as follows:

```
module ListSet : Set = struct
  (** Abstraction function: The list [[a1; ...; an]] represents the
    set [{b1, ..., bm}], where [[b1; ...; bm]] is the same list as
    [[a1; ...; an]] but with any duplicates removed. The empty list
    [[]] represents the empty set [{}]. *)
    type 'a t = 'a list
    ...
end
```

This comment explicitly points out that the list may contain duplicates, which is helpful as a reinforcement of the first sentence. Similarly, the case of an empty list is mentioned explicitly for clarity, although some might consider it to be redundant.

The abstraction function for the second implementation, which does not allow duplicates, hints at an important difference. We can write the abstraction function for this second representation a bit more simply because we know that the elements are distinct.

```
module UniqListSet : Set = struct
  (** Abstraction function: The list [[a1; ...; an]] represents the set
      [{a1, ..., an}]. The empty list [[]] represents the empty set [{}]. *)
   type 'a t = 'a list
   ...
end
```

6.3.2 Implementing the Abstraction Function

What would it mean to implement the abstraction function for ListSet? We'd want a function that took an input of type 'a ListSet.t. But what should its output type be? The abstract values are mathematical sets, not OCaml types. If we did hypothetically have a type 'a set that our abstraction function could return, there would have been little point in developing ListSet; we could have just used that 'a set type without doing any work of our own.

On the other hand, we might implement something close to the abstraction function by converting an input of type 'a ListSet.t to a built-in OCaml type or standard library type:

- We could convert to a string. That would have the advantage of being easily readable by humans in the toplevel or in debug output. Java programmers use toString() for similar purposes.
- We could convert to 'a list. (Actually there's little conversion to be done). For data collections this is a convenient choice, since lists can at least approximately represent many data structures: stacks, queues, dictionaries, sets, heaps, etc.

The following functions implement those ideas. Note that to_string has to take an additional argument string_of_val from the client to convert 'a to string.

```
module ListSet : Set = struct
    ...

let uniq lst = List.sort_uniq Stdlib.compare lst


let to_string string_of_val lst =
    let interior =
        lst |> uniq |> List.map string_of_val |> String.concat ", "
    in
        "{" ^ interior ^ "}"

let to_list = uniq
end
```

Installing a custom formatter, as discussed in the *section on encapsulation*, could also be understood as implementing the abstraction function. But in that case it's usable only by humans at the toplevel rather than other code, programatically.

6.3.3 Commutative Diagrams

Using the abstraction function, we can now talk about what it means for an implementation of an abstraction to be *correct*. It is correct exactly when every operation that takes place in the concrete space makes sense when mapped by the abstraction function into the abstract space. This can be visualized as a *commutative diagram*:



A commutative diagram means that if we take the two paths around the diagram, we have to get to the same place. Suppose that we start from a concrete value and apply the actual implementation of some operation to it to obtain a new concrete value or values. When viewed abstractly, a concrete result should be an abstract value that is a possible result of applying the function *as described in its specification* to the abstract view of the actual inputs. For example, consider the union function from the implementation of sets as lists with repeated elements covered last time. When this function is applied to the concrete pair [1; 3], [2; 2], it corresponds to the lower-left corner of the diagram. The result of this operation is the list [2; 2; 1; 3], whose corresponding abstract value is the list $\{1, 2, 3\}$. Note that if we apply the abstraction function AF to the input lists [1; 3] and [2; 2], we have the sets $\{1, 3\}$ and $\{2\}$. The commutative diagram requires that in this instance the union of $\{1, 3\}$ and $\{2\}$ is $\{1, 2, 3\}$, which is of course true.

6.3.4 Representation Invariants

The abstraction function explains how information within the module is viewed abstractly by module clients. But that is not all we need to know to ensure correctness of the implementation. Consider the size function in each of the two implementations. For ListSet, which allows duplicates, we need to be sure not to double-count duplicate elements:

```
let size lst = List.(lst |> sort_uniq Stdlib.compare |> length)
```

But for UniqListSet, in which the lists have no duplicates, the size is just the length of the list:

```
let size = List.length
```

How do wo know that latter implementation is correct? That is, how do we know that "lists have no duplicates"? It's hinted at by the name of the module, and it can be deduced from the implementation of add, but we've never carefully documented it. Right now, the code does not explicitly say that there are no duplicates.

In the UniqListSet representation, not all concrete data items represent abstract data items. That is, the *domain* of the abstraction function does not include all possible lists. There are some lists, such as [1; 1; 2], that contain duplicates and must never occur in the representation of a set in the UniqListSet implementation; the abstraction function is undefined on such lists. We need to include a second piece of information, the *representation invariant* (or *rep invariant*, or *RI*), to determine which concrete data items are valid representations of abstract data items. For sets represented as lists without duplicates, we write this as part of the comment together with the abstraction function:

```
module UniqListSet : Set = struct
   (** Abstraction function: the list [[a1; ...; an]] represents the set
      [{a1, ..., an}]. The empty list [[]] represents the empty set [{}].
      Representation invariant: the list contains no duplicates. *)
```

(continues on next page)

(continued from previous page)

```
type 'a t = 'a list
   ...
end
```

If we think about this issue in terms of the commutative diagram, we see that there is a crucial property that is necessary to ensure correctness: namely, that all concrete operations preserve the representation invariant. If this constraint is broken, functions such as size will not return the correct answer. The relationship between the representation invariant and the abstraction function is depicted in this figure:



We can use the rep invariant and abstraction function to judge whether the implementation of a single operation is correct in isolation from the rest of the functions in the module. A function is correct if these conditions:

- 1. The function's preconditions hold of the argument values.
- 2. The concrete representations of the arguments satisfy the rep invariant.

imply these conditions:

- 1. All new representation values created satisfy the rep invariant.
- 2. The commutative diagram holds.

The rep invariant makes it easier to write code that is provably correct, because it means that we don't have to write code that works for all possible incoming concrete representations—only those that satisfy the rep invariant. For example, in the implementation UniqListSet, we do not care what the code does on lists that contain duplicate elements. However, we do need to be concerned that on return, we only produce values that satisfy the rep invariant. As suggested in the figure above, if the rep invariant holds for the input values, then it should hold for the output values, which is why we call it an *invariant*.

6.3.5 Implementing the Representation Invariant

When implementing a complex abstract data type, it is often helpful to write an internal function that can be used to check that the rep invariant holds of a given data item. By convention we will call this function rep_ok . If the module accepts values of the abstract type that are created outside the module, say by exposing the implementation of the type in the signature, then rep_ok should be applied to these to ensure the representation invariant is satisfied. In addition, if the implementation creates any new values of the abstract type, rep_ok can be applied to them as a sanity check. With this approach, bugs are caught early, and a bug in one function is less likely to create the appearance of a bug in another.

A convenient way to write rep_ok is to make it an identity function that just returns the input value if the rep invariant holds and raises an exception if it fails.

```
(* Checks whether x satisfies the representation invariant. *)
let rep_ok x =
  if (* check the RI holds of x *) then x else failwith "RI violated"
```

Here is an implementation of Set that uses the same data representation as UniqListSet, but includes copious rep_ok checks. Note that rep_ok is applied to all input sets and to any set that is ever created. This ensures that if a bad set representation is created, it will be detected immediately. In case we somehow missed a check on creation, we also apply rep_ok to incoming set arguments. If there is a bug, these checks will help us quickly figure out where the rep invariant is being broken.

```
(** Implementation of sets as lists without duplicates. *)
module UniqListSet : Set = struct
  (** Abstraction function: The list [[a1; ...; an]] represents the
      set {a1, ..., an}. The empty list [[]] represents the empty set [{}].
      Representation invariant: the list contains no duplicates. *)
 type 'a t = 'a list
 let rep_ok lst =
   let u = List.sort_uniq Stdlib.compare lst in
   match List.compare_lengths lst u with 0 -> lst | _ -> failwith "RI"
 let empty = []
 let mem x lst = List.mem x (rep_ok lst)
 let add x lst = rep_ok (if mem x (rep_ok lst) then lst else x :: lst)
 let rem x lst = rep_ok (List.filter (( <> ) x) (rep_ok lst))
 let size lst = List.length (rep_ok lst)
 let union lst1 lst2 =
   rep_ok
      (List.fold left
         (fun u x -> if mem x 1st2 then u else x :: u)
         (rep_ok lst2) (rep_ok lst1))
 let inter lst1 lst2 = rep_ok (List.filter (fun h -> mem h lst2) (rep_ok lst1))
end
```

```
module UniqListSet : Set
```

Calling rep_ok on every argument can be too expensive for the production version of a program. The rep_ok above, for example, requires linearithmic time, which destroys the efficiency of all the previously constant time or linear time

operations. For production code, it may be more appropriate to use a version of rep_ok that only checks the parts of the rep invariant that are cheap to check. When there is a requirement that there be no run-time cost, rep_ok can be changed to an identity function (or macro) so the compiler optimizes away the calls to it. However, it is a good idea to keep around the full code of rep_ok so it can be easily reinstated during future debugging:

```
let rep_ok lst = lst

let rep_ok_expensive =
   let u = List.sort_uniq Stdlib.compare lst in
   match List.compare_lengths lst u with 0 -> lst | _ -> failwith "RI"
```

Some languages provide support for *conditional compilation*, which provides some kind of support for compiling some parts of the codebase but not others. The OCaml compiler supports a flag noassert that disables assertion checking. So you could implement rep invariant checking with assert, and turn it off with noassert. The problem with that is that some portions of your codebase might *require* assertion checking to be turned on to work correctly.

6.4 Testing and Debugging

Correct programs behave as we intend them to behave. *Validation* is the process of building our confidence in correct program behavior.

6.4.1 Validation

There are many ways to increase that confidence. Social methods, formal methods, and testing are three. The latter is our main focus, but let's first consider the other two.

Social methods involve developing programs with other people, relying on their assistance to improve correctness. Some good techniques include the following:

- Code walkthrough. In the walkthrough approach, the programmer presents the documentation and code to a reviewing team, and the team gives comments. This is an informal process. The focus is on the code rather than the coder, so hurt feelings are easier to avoid. However, the team may not get as much assurance that the code is correct
- *Code inspection*. Here, the review team drives the code review process. Some, though not necessarily very much, team preparation beforehand is useful. They define goals for the review process and interact with the coder(s) to understand where there may be quality problems. Again, making the process as blameless as possible is important.
- Pair programming. The most informal approach to code review is through pair programming, in which code is developed by a pair of engineers: the driver who writes the code, and the observer who watches. The role of the observer is be a critic, to think about potential errors, and to help navigate larger design issues. It's usually better to have the observer be the engineer with the greater experience with the coding task at hand. The observer reviews the code, serving as the devil's advocate that the driver must convince. When the pair is developing specifications, the observer thinks about how to make specs clearer or shorter. Pair programming has other benefits. It is often more fun and educational to work with a partner, and it helps focus both partners on the task. If you are just starting to work with another programmer, pair programming is a good way to understand how your partner thinks and to establish common vocabulary. It is a good idea for partners to trade off roles, too.

These social techniques for *code review* can be remarkably effective. In one study conducted at IBM (Jones, 1991), code inspection found 65% of the known coding errors and 25% of the known documentation errors, whereas testing found only 20% of the coding errors and none of the documentation errors. The code inspection process may be more effective than walkthroughs. One study (Fagan, 1976) found that code inspections resulted in code with 38% fewer failures, compared to code walkthroughs.

Thorough code review can be expensive, however. Jones found that preparing for code inspection took one hour per 150 lines of code, and the actual inspection covered 75 lines of code per hour. Having up to three people on the inspection team improves the quality of inspection; beyond that, more inspectors doesn't seem to help. Spending a lot of time preparing for inspection did not seem to be useful, either. Perhaps this is because much of the value of inspection lies in the interaction with the coders.

Formal methods use the power of mathematics and logic to validate program behavior. *Verification* uses the program code and its specifications to construct a proof that the program behaves correctly on all possible inputs. There are research tools available to help with program verification, often based on automated theorem provers, as well as research languages that are designed for program verification. Verification tends to be expensive and to require thinking carefully about and deeply understanding the code to be verified. So in practice, it tends to be applied to code that is important and relatively short. Verification is particularly valuable for critical systems where testing is less effective. Because their execution is not deterministic, concurrent programs are hard to test, and sometimes subtle bugs can only be found by attempting to verify the code formally. In fact, tools to help prove programs correct have been getting increasingly effective and some large systems have been fully verified, including compilers, processors and processor emulators, and key pieces of operating systems.

Testing involves actually executing the program on sample inputs to see whether the behavior is as expected. By comparing the actual results of the program with the expected results, we find out whether the program really works on the particular inputs we try it on. Testing can never provide the absolute guarantees that formal methods do, but it is significantly easier and cheaper to do. It is also the validation methodology with which you are probably most familiar. Testing is a good, cost-effective way of building confidence in correct program behavior.

6.4.2 Debugging

When testing reveals an error, we usually say that the program is "buggy". But the word "bug" suggests something that wandered into a program. Better terminology would be that there are

- faults, which are the result of human errors in software systems, and
- failures, which are violations of requirements.

Some faults might never appear to an end user of a system, but failures are those faults that do. A fault might result because an implementation doesn't match design, or a design doesn't match the requirements.

Debugging is the process of discovering and fixing faults. Testing clearly is the "discovery" part, but fixing can be more complicated. Debugging can be a task that takes even more time than an original implementation itself! So you would do well to make it easy to debug your programs from the start. Write good specifications for each function. Document the AF and RI for each data abstraction. Keep modules small, and test them independently.

Inevitably, though, you will discover faults in your programs. When you do, approach them as a scientist by employing the *scientific method:*

- evaluate the data that are available;
- formula a hypothesis that might explain the data;
- · design a repeatable experiment to test that hypothesis; and
- use the result of that experiment to refine or refute your hypothesis.

Often the crux of this process is finding the simplest, smallest input that triggers a fault. That's not usually the original input for which we discover a fault. So some initial experimentation might be needed to find a *minimal test case*.

Never be afraid to write additional code, even a lot of additional code, to help you find faults. Functions like to_string or format can be invaluable in understanding computations, so writing them up front before any faults are detected is completely worthwhile.

When you do discover the source of a fault, be extra careful in fixing it. It is tempting to slap a quick fix into the code and move on. This is quite dangerous. Far too often, fixing a fault just introduces a new (and unknown) fault! If a bug is

difficult to find, it is often because the program logic is complex and hard to reason about. You should think carefully about why the bug happened in the first place and what the right solution to the problem is. *Regression testing* (i.e., recording only test cases that originally failed but now pass) is important whenever a bug fix is introduced, but nothing can replace careful thinking about the code.

6.5 Black-box and Glass-box Testing

We would like to know that a program works on all possible inputs. The problem with testing is that it is usually infeasible to try all the possible inputs. For example, suppose that we are implementing a module that provides an abstract data type for rational numbers. One of its operations might be an addition function plus, e.g.:

```
module type RATIONAL = sig
  (** A [t] is a rational. *)
 type t
  (** [create p q] is the rational number [p/q].
     Raises: [Invalid_argument "0"] if [q] is 0. *)
 val create : int -> int -> t
  (** [plus r1 r2] is [r1 + r2] *)
 val plus : t -> t -> t
module Rational : RATIONAL = struct
  (** AF: [(p, q)] represents the rational number [p/q]
     RI: [q] is not 0. *)
 type t = int * int
 let create p q =
   if q = 0 then invalid_arg "0" else (p, q)
 let plus (p1, q1) (p2, q2) =
    (p1 * q2 + p2 * q1, q1 * q2)
end
```

What would it take to exhaustively test just this one function? We'd want to try all possible rationals as both the r1 and r2 arguments. A rational is formed from two ints, and there are 2^{63} ints on a modern OCaml implementation. Therefore there are approximately $(2^{63})^4 = 2^{252}$ possible inputs to the plus function. Even if we test one addition every nanosecond, it will take about 10^{59} years to finish testing this one function.

Clearly we can't test software exhaustively. But that doesn't mean we should give up on testing. It just means that we need to think carefully about what our test cases should be so that they are as effective as possible at convincing us that the code works.

Consider our create function, above. It takes in two integers p and q as arguments. How should we go about selecting a relatively small number of test cases that will convince us that the function works correctly on all possible inputs? We can visualize the space of all possible inputs as a large square:



(minint, minint)

There are about 2^{126} points in this square, so we can't afford to test them all. And testing them all is going to mostly be a waste of time—most of the possible inputs provide nothing new. We need a way to find a set of points in this space to test that are interesting and will give a good sense of the behavior of the program across the whole space.

Input spaces generally comprise a number of subsets in which the behavior of the code is similar in some essential fashion across the entire subset. We don't get any additional information by testing more than one input from each such subset.

If we test all the interesting regions of the input space, we have achieved good *coverage*. We want tests that in some useful sense cover the space of possible program inputs.

Two good ways of achieving coverage are black-box testing and glass-box testing. We discuss those, next.

6.5.1 Black-box Testing

In selecting our test cases for good coverage, we might want to consider both the specification and the implementation of the program or module being tested. It turns out that we can often do a pretty good job of picking test cases by just looking at the specification and ignoring the implementation. This is known as **black-box testing**. The idea is that we think of the code as a black box about which all we can see is its surface: its specification. We pick test cases by looking at how the specification implicitly introduces boundaries that divide the space of possible inputs into different regions.

When writing black-box test cases, we ask ourselves what set of test cases that will produce distinctive behavior as predicted by the specification. It is important to try out both *typical inputs* and inputs that are *boundary cases* aka *corner cases* or *edge cases*. A common error is to only test typical inputs, with the result that the program usually works but fails in less frequent situations. It's also important to identify ways in which the specification creates classes of inputs that should elicit similar behavior from the function, and to test on those *paths through the specification*. Here are some examples.

Example 1.

Here are some ideas for how to test the create function:

• Looking at the square above, we see that it has boundaries at min_int and max_int. We want to try to construct rationals at the corners and along the sides of the square, e.g. create min_int min_int, create

```
max int 2, etc.
```

- The line p=0 is important because p/q is zero all along it. We should try (0, q) for various values of q.
- We should try some typical (p, q) pairs in all four quadrants of the space.
- We should try both (p, q) pairs in which q divides evenly into p, and pairs in which q does not divide into p.
- Pairs of the form (1, q), (-1, q), (p, 1), (p, -1) for various p and q also may be interesting given the properties of rational numbers.

The specification also says that the code will check that q is not zero. We should construct some test cases to ensure this checking is done as advertised. Trying (1, 0), $(\max_int, 0)$, $(\min_int, 0)$, (-1, 0), (0, 0) to see that they all raise the specified exception would probably be an adequate set of black-box tests.

Example 2.

Consider a function list max:

```
(** Return the maximum element in the list. *)
val list_max: int list -> int
```

What is a good set of black-box test cases? Here the input space is the set of all possible lists of ints. We need to try some typical inputs and also consider boundary cases. Based on this spec, boundary cases include the following:

- A list containing one element. In fact, an empty list is probably the first boundary case we think of. Looking at the spec above, we realize that it doesn't specify what happens in the case of an empty list. Thus, thinking about boundary cases is also useful in identifying errors in the specification.
- A list containing two elements.
- A list in which the maximum is the first element. Or the last element. Or somewhere in the middle of the list.
- A list in which every element is equal.
- A list in which the elements are arranged in ascending sorted order, and one in which they are arranged in descending sorted order.
- A list in which the maximum element is max_int, and a list in which the maximum element is min_int.

Example 3.

Consider the function sqrt:

```
(** [sqrt x n] is the square root of [x] computed to an accuracy of [n]
    significant digits.
    Requires: [x >= 0] and [n >= 1]. *)
val sqrt : float -> int -> float
```

The precondition identifies two possibilities for x (either it is 0 or greater) and two possibilities for n (either it is 1 or greater). That leads to four "paths through the specification", i.e., representative and boundary cases for satisfying the precondition, which we should test:

- x is 0 and n is 1
- x is greater than 0 and n is 1
- x is 0 and n is greater than 1
- x is greater than 0 and n is greater than 1.

6.5.2 Black-box Testing of Data Abstractions

So far we've been thinking about testing just one function at a time. But data abstractions usually have many operations, and we need to test how those operations interact with one another. It's useful to distinguish *consumer* and *producers* of the data abstraction:

- A consumer is an operation that takes a value of the data abstraction as input.
- A producer is an operation that a value of the data abstraction.

For example, consider this set abstraction:

```
module type Set = sig

  (** ['a t] is the type of a set whose elements have type ['a]. *)
  type 'a t

  (** [empty] is the empty set. *)
  val empty : 'a t

  (** [size s] is the number of elements in [s]. *
      [size empty] is [0]. *)

  val size : 'a t -> int

  (** [add x s] is a set containing all the elements of
      [s] as well as element [x]. *)

  val add : 'a -> 'a t -> 'a t

      (** [mem x s] is [true] iff [x] is an element of [s]. *)
  val mem : 'a -> 'a t -> bool

end
```

The empty and add functions are producers; and the size, add and mem functions are consumers.

When black-box testing a data abstraction, we should test how each consumer of the data abstraction handles every path through each producer of it. In the Set example, that means testing the following:

- how size handles the empty set;
- how size handles a set produced by add, both when add leaves the set unchanged as well as when it increases
 the set;
- how add handles sets produced by empty as well as add itself;
- how mem handles sets produced by empty as well as add, including paths where mem is invoked on elements that have been added as well as elements that have not.

6.5.3 Glass-box Testing

Black-box testing is a good place to start when writing test cases, but ultimately it is not enough. In particular, it's not possible to determine how much coverage of the implementation a black-box test suite actually achieves—we actually need to know the implementation source code. Testing based on that code is known as *glass box* or *white box* testing. Glass-box testing can improve on black-box by testing *execution paths* through the implementation code: the series of expressions that is conditionally evaluated based on if-expressions, match-expressions, and function applications. Test cases that collectively exercise all paths are said to be *path-complete*. At a minimum, path-completeness requires that for every line of code, and even for every expression in the program, there should be a test case that causes it to be executed. Any unexecuted code could contain a bug if has never been tested.

For true path completeness we must consider all possible execution paths from start to finish of each function, and try to exercise every distinct path. In general this is infeasible, because there are too many paths. A good approach is to think of the set of paths as the space that we are trying to explore, and to identify boundary cases within this space that are worth testing.

For example, consider the following implementation of a function that finds the maximum of its three arguments:

```
let max3 x y z =
  if x > y then
   if x > z then x else z
  else
   if y > z then y else z
```

```
val max3 : 'a -> 'a -> 'a -> 'a = <fun>
```

Black-box testing might lead us to invent many tests, but looking at the implementation reveals that there are only four paths through the code—the paths that return x, z, y, or z (again). We could test each of those paths with representative inputs such as: $max3 \ 3 \ 2 \ 1$, $max3 \ 3 \ 2 \ 4$, $max3 \ 1 \ 2 \ 1$, $max3 \ 1 \ 2 \ 3$.

When doing glass box testing, we should include test cases for each branch of each (nested) if expression, and each branch of each (nested) pattern match. If there are recursive functions, we should include test cases for the base cases as well as each recursive call. Also, we should include test cases to trigger each place where an exception might be raised.

Of course, path complete testing does not guarantee an absence of errors. We still need to test against the specification, i.e., do black-box testing. For example, here is a broken implementation of max3:

```
let max3 x y z = x
```

The test max 2 1 1 is path complete, but doesn't reveal the error.

6.5.4 Glass-box Testing of Data Abstractions

Look at the abstraction function and representation invariant for hints about what boundaries may exist in the space of values manipulated by a data abstraction. The rep invariant is a particularly effective tool for constructing useful test cases. Looking at the rep invariant of the Rational data abstraction above, we see that it requires that q is non-zero. Therefore we should construct test cases to see whether it's possible to cause that invariant to be violated.

6.5.5 Black-box vs. Glass-box

Black-box testing has some important advantages:

- It doesn't require that we see the code we are testing. Sometimes code will not be available in source code form, yet we can still construct useful test cases without it. The person writing the test cases does not need to understand the implementation.
- The test cases do not depend on the implementation. They can be written in parallel with or before the implementation. Further, good black-box test cases do not need to be changed, even if the implementation is completely rewritten.
- Constructing black-box test cases causes the programmer to think carefully about the specification and its implications. Many specification errors are caught this way.

The disadvantage of black box testing is that its coverage may not be as high as we'd like, because it has to work without the implementation.

6.5.6 Bisect

Glass-box testing can be aided by *code-coverage tools* that assess how much of the code has been exercised by a test suite. The bisect_ppx tool for OCaml can tell you which expressions in your program have been tested, and which have not. Here's how it works:

- You compile your code using Bisect_ppx (henceforth, just Bisect for short) as part of the compilation process. It *instruments* your code, mainly by inserting additional expressions to be evaluated.
- You run your code. The instrumentation that Bisect inserted causes your program to do something in addition to
 whatever functionality you programmed yourself: the program will now record which expressions from the source
 code actually get executed at run time, and which do not. Also, the program will now produce an output file that
 contains that information.
- You run a tool called bisect-ppx-report on that output file. It produces HTML showing you which parts of your code got executed, and which did not.

How does that help with computing coverage of a test suite? If you run your OUnit test suite, the test cases in it will cause the code in whatever functions they test to be executed. If you don't have enough test cases, some code in your functions will never be executed. The report produced by Bisect will show you exactly what code that is. You can then design new glass-box test cases to cause that code to execute, add them to your OUnit suite, and create a new Bisect report to confirm that the code really did get executed.

Bisect Tutorial.

Download the file sorts.ml. You will find an implementation of insertion sort and merge sort.

Create a file in the same directory called myocamlbuild.ml. That file name is actually mandatory, despite the customary use of "my" in CS demos to indicate a name that you could choose yourself. Put this code in it:

```
open Ocamlbuild_plugin
let () = dispatch Bisect_ppx_plugin.dispatch
```

Create a tags file in the same directory, and put the following in it:

```
<sorts.ml>: coverage
<test_sorts.{byte,native}>: coverage
true: package(ounit2), package(bisect_ppx)
```

Download the file test_sorts.ml. It has the skeleton for an OUnit test suite.

Run

to build the test suite, and

```
$ ./test_sorts.byte
```

to run it. OUnit will by default attempt to run some of the tests in parallel, which reduces the time it takes to run a large test suite, at the tradeoff of making it nondeterministic in what order the tests run. If you want you can instead add a flag:

```
$ ./test_sorts.byte -runner sequential
```

That makes the tests run one at a time in order.

Running the suite will cause a file named bisectNNNN.coverage to be produced. Next run

```
$ bisect-ppx-report html
```

to generate the Bisect report from your test suite execution.

Open the file _coverage/index.html in a web browser. Look at the per-file coverage; you'll see we've managed to test only 10% of sorts.ml with our test suite so far. Click on the link in that report for sorts.ml. You'll see that we've managed to cover a couple lines of the source code so far with our test suite.

There are some additional tests in the test file. Try uncommenting those, as documented in the test file, and increasing your code coverage. Between each run, you will need to delete the report file, recompile, rerun OUnit, and rerun the Bisect report tool. (Obviously, a Makefile would be a good thing to construct.)

By the time you're done uncommenting the provided tests, you should be at 30% coverage, including all of the insertion sort implementation. For fun, try adding more tests to get 100% coverage of merge sort.

6.6 Randomized Testing with QCheck

Randomized testing aka fuzz testing is the process of generating random inputs and feeding them to a program or a function to see whether the program behaves correctly. The immediate issue is how to determine what the correct output is for a given input. If a reference implementation is available—that is, an implementation that is believed to be correct but in some other way does not suffice (e.g., its performance is too slow, or it is in a different language)—then the outputs of the two implementations can be compared. Otherwise, perhaps some property of the output could be checked. For example,

- "not crashing" is a property of interest in user interfaces;
- adding n elements to a data collection then removing those elements, and ending up with an empty collection, is a
 property of interest in data structures; and
- encrypting a string under a key then decrypting it under that key and getting back the original string is a property
 of interest in an encryption scheme like Enigma.

Randomized testing is an incredibly powerful technique. It is often used in testing programs for security vulnerabilities. The qcheck package for OCaml supports randomized testing. We'll look at it, next, after we discuss random number generation.

6.6.1 Random Number Generation

To understand randomized testing, we need to take a brief digression into random number generation.

Most languages provide the facility to generate random numbers. In truth, these generators are usually not truly random (in the sense that they are completely unpredictable) but in fact are *pseudorandom*: the sequence of numbers they generate pass good statistical tests to ensure there is no discernible pattern in them, but the sequence itself is a deterministic function of an initial *seed* value. (Recall that the prefix *pseudo* is from the Greek *pseudēs* meaning "false".) Java and Python both provide pseudorandom number generators (PRNGs). So does OCaml in the standard library's Random module.

An Experiment. Start a new session of utop and enter the following:

```
# Random.int 100;;
# Random.int 100;;
# Random.int 100;;
```

Each response will be an integer i such that $0 \le i < 100$.

Now quit utop and start another new session. Enter the same phrases again. You will get the same responses as last time. In fact, unless your OCaml installation is somehow different than that used to produce this book, you will get the same numbers as those below:

```
Random.int 100;;
Random.int 100;;
Random.int 100;;
```

```
- : int = 44
```

```
- : int = 85
```

```
- : int = 82
```

Not exactly unpredictable, eh?

PRNGs. Although for purposes of security and cryptography a PRNG leads to terrible vulnerabilities, for other purposes—including testing and simulation—PRNGs are just fine. Their predictability can even be useful: given the same initial seed, a PRNG will always produce the same sequence of pseudorandom numbers, leading to the ability to repeat a particular sequence of tests or a particular simulation.

The way a PRNG works in general is that it initializes a *state* that it keeps internally from the initial seed. From then on, each time the PRNG generates a new value, it imperatively updates that state. The Random module makes it possible to manipulate that state in limited ways. For example, you can

- get the current state with Random.get_state,
- duplicate the current state with Random. State.copy,
- request a random int generated from a particular state with Random. State.int, and
- initialize the state yourself. The functions Random.self_init and Random.State.make_self_init will choose a "random" seed to initialize the state. They do so by sampling from a special Unix file named /dev/urandom, which is meant to provide as close to true randomness as a computer can.

Repeating the Experiment. Start a new session of utop. Enter the following:

```
# Random.self_init ();;
# Random.int 100;;
# Random.int 100;;
# Random.int 100;;
```

Now do that a second time (it doesn't matter whether you exit utop or not in between). You will notice that you get a different sequence of values. With high probability, what you get will be different than the values below:

```
Random.self_init ();;
Random.int 100;;
Random.int 100;;
Random.int 100;;
```

```
- : unit = ()
```

```
-: int = 33
```

6.6.2 QCheck Abstractions

QCheck has three abstractions we need to cover before using it for testing: generators, properties, and arbitraries. If you want to follow along in utop, load QCheck with this directive:

```
#require "qcheck";;
```

Generators. One of the key pieces of functionality provided by QCheck is the ability to generate pseudorandom values of various types. Here is some of the signature of the module that does that:

```
module QCheck : sig
...
module Gen :
sig
   type 'a t = Random.State.t -> 'a
   val int : int t
   val generate : ?rand:Random.State.t -> n:int -> 'a t -> 'a list
   val generate1 : ?rand:Random.State.t -> 'a t -> 'a
   ...
end
...
end
...
end
```

An 'a QCheck.Gen.t is a function that takes in a PRNG state and uses it to produce a pseudorandom value of type 'a. So QCheck.Gen.int produces pseudorandom integers. The function generate1 actually does the generation of one pseudorandom value. It takes an optional argument that is a PRNG state; if that argument is not supplied, it uses the default PRNG state. The function generate produces a list of n pseduorandom values.

QCheck implements many producers of pseudorandom values. Here are a few more of them:

```
module QCheck : sig
...
module Gen :
sig
    val int : int t
    val small_int : int t
    val int_range : int -> int -> int t
    val list : 'a t -> 'a list t
    val list_size : int t -> 'a t -> 'a list t
    val string : ?gen:char t -> string t
    val small_string : ?gen:char t -> string t
    ...
end
...
end
```

You can read the documentation of those and many others.

Properties. It's tempting to think that QCheck would enable us to test a function by generating many pseudorandom inputs to the function, running the function on them, then checking that the outputs are correct. But there's immediately a problem: how can QCheck know what the correct output is for each of those inputs? Since they're randomly generated, the test engineer can't hardcode the right outputs.

So instead, QCheck allows us to check whether a *property* of each output holds. A property is a function of type t -> bool, for some type t, that tells use whether the value of type t exhibits some desired characteristic. Here, for example, here are two properties; one that determines whether an integer is even, and another that determines whether a list is sorted in non-decreasing order according to the built-in <= operator:

```
let is_even n = n mod 2 = 0

let rec is_sorted = function
    | [] -> true
    | [ h ] -> true
    | h1 :: (h2 :: t as t') -> h1 <= h2 && is_sorted t'</pre>
```

```
val is_even : int -> bool = <fun>
```

```
val is_sorted : 'a list -> bool = <fun>
```

Arbitraries. The way we present to QCheck the outputs to be checked is with a value of type 'a QCheck. Arbitrary. This type represents an "arbitrary" value of type 'a—that is, it has been pseudorandomly chosen as a value that we want to check, and more specifically, to check whether it satisfies a property.

We can create *arbitraries* out of generators using the function <code>QCheck.make</code>: 'a <code>QCheck.Gen.t</code> -> 'a <code>QCheck.arbitrary</code>. (Actually that function takes some optional arguments that we elide here.) This isn't actually the normal way to create arbitraries, but it's a simple way that will help us understand them; we'll get to the normal way in a little while. For example, the following expression represents an arbitrary integer:

```
QCheck.make QCheck.Gen.int
```

6.6.3 Testing Properties

To construct a QCheck test, we create an arbitrary and a property, and pass them to QCheck. Test.make, whose type can be simplified to:

```
QCheck.Test.make : 'a QCheck.arbitrary -> ('a -> bool) -> QCheck.Test.t
```

In reality, that function also takes several optional arguments that we elide here. The test will generate some number of arbitraries and check whether the property holds of each of them. For example, the following code creates a QCheck test that checks whether an arbitrary integer is even:

```
let t = QCheck.Test.make (QCheck.make QCheck.Gen.int) is_even
```

```
val t : QCheck.Test.t = QCheck.Test.Test <abstr>
```

If we want to change the number of arbitraries that are checked, we can pass an optional integer argument ~count to QCheck.Test.make.

We can run that test with QCheck_runner.run_tests: QCheck.Test.t list -> int. (Once more, that function takes some optional arguments that we elide here.) The integer it returns is 0 if all the tests in the list pass, and 1 otherwise. For the test above, running it will output 1 with high probability, because it will generate at least one odd integer.

```
QCheck_runner.run_tests [t]
```

```
random seed: 85138665
```

```
Test anon_test_1 failed (0 shrink steps):

<no printer>
failure (1 tests failed, 0 tests errored, ran 1 tests)

- : int = 1
```

Unfortunately, that output isn't very informative; it doesn't tell us what particular values failed to satisfy the property! We'll fix that problem in a little while.

If you want to make an OCaml program that runs QCheck tests and prints the results, there is a function QCheck_runner.run_tests_main that works much like OUnit2.run_test_tt_main: just invoke it as the final expression in a test file. For example:

```
let tests = (* code that constructs a [QCheck.Test.t list] *)
let _ = QCheck_runner.run_tests_main tests
```

If that code is in a file test_x.ml, you can compile and run it as follows:

```
$ ocamlbuild -pkg qcheck test_x.byte
$ ./test_x.byte
```

QCheck tests can be converted to OUnit tests and included in the usual kind of OUnit test suite we've been writing all along. The function that does this is:

```
Check_runner.to_ounit2_test
- : ?rand:Random.State.t -> QCheck.Test.t -> OUnit2.test = <fun>
```

6.6.4 Informative Output from QCheck

We noted above that the output of QCheck so far has told us only *whether* some arbitraries satisfied a property, but not *which* arbitraries failed to satisfy it. Let's fix that problem.

The issue is with how we constructed an arbitrary directly out of a generator. An arbitrary is properly more than just a generator. The QCheck library needs to know how to print values of the generator, and a few other things as well. You can see that in the definition of 'a QCheck.arbitrary:

```
#show QCheck.arbitrary;;

type nonrec 'a arbitrary = private {
```

```
gen : 'a QCheck.Gen.t;

print : ('a -> string) option;

small : ('a -> int) option;

shrink : 'a QCheck.Shrink.t option;

collect : ('a -> string) option;

stats : 'a QCheck.stat list;

}
```

In addition to the generator field gen, there is a field containing an optional function to print values from the generator, and a few other optional fields as well. Luckily, we don't usually have to find a way to complete those fields ourselves; the QCheck module provides many arbitraries that correspond to the generators found in QCheck. Gen:

```
module QCheck :
    sig
    ...
    val int : int arbitrary
    val small_int : int arbitrary
    val int_range : int -> int -> int arbitrary
    val list : 'a arbitrary -> 'a list arbitrary
    val list_of_size : int Gen.t -> 'a arbitrary -> 'a list arbitrary
    val string : string arbitrary
    val small_string : string arbitrary
    ...
    end
```

Using those arbitraries, we can get improved error messages:

```
let t = QCheck.Test.make ~name:"my_test" QCheck.int is_even;;
QCheck_runner.run_tests [t];;

val t : QCheck.Test.t = QCheck.Test.Test <abstr>

--- Failure

Test my_test failed (76 shrink steps):
```

```
failure (1 tests failed, 0 tests errored, ran 1 tests)
```

```
- : int = 1
```

The output tells us the my_test failed, and shows us the input that caused the failure.

6.6.5 Testing Functions with QCheck

The final piece of the QCheck puzzle is to use a randomly generated input to test whether a function's output satisfies some property. For example, here is a QCheck test to see whether the output of double is correct:

```
let double x = 2 * x;;
let double_check x = double x = x + x;;
let t = QCheck.Test.make ~count:1000 QCheck.int double_check;;
QCheck_runner.run_tests [t];;
```

```
val double : int -> int = <fun>
```

```
val double_check : int -> bool = <fun>
```

```
val t : QCheck.Test.t = QCheck.Test.Test <abstr>
```

```
-----
```

```
success (ran 1 tests)
```

```
- : int = 0
```

Above, double is the function we are testing. The property we're testing double_check, is that double x is always x + x. We do that by having QCheck create 1000 arbitrary integers and test that the property holds of each of them.

Here are a couple more examples, drawn from QCheck's own documentation. The first checks that List.rev is an *involution*, meaning that applying it twice brings you back to the original list. That is a property that should hold of a correct implementation of list reversal.

```
let rev_involutive lst = List.(lst |> rev |> rev = lst);;
let t = QCheck.(Test.make ~count:1000 (list int) rev_involutive);;
QCheck_runner.run_tests [t];;
```

```
val rev_involutive : 'a list -> bool = <fun>
```

```
val t : QCheck.Test.t = QCheck.Test.Test <abstr>
```

```
_____
```

```
success (ran 1 tests)
```

```
- : int = 0
```

Indeed, running 1000 random tests reveals that none of them fails. The int generator used above generates integers uniformly over the entire range of OCaml integers. The list generator creates lists whose elements are individual

generated by int. According to the documentation of list, the length of each list is randomly generated by another generator nat, which generates "small natural numbers." What does that mean? It isn't specified. But if we read the current source code, we see that those are integers from 0 to 10,000, and biased toward being smaller numbers in that range.

The second example checks that all lists are sorted. Of course, not all lists *are* sorted! So we should expect this test to fail.

The output shows an example of a list that is not sorted, hence violates the property. Generator small_nat is like nat but ranges from 0 to 100.

6.7 Proving Correctness

Testing provides evidence of correctness, but not full assurance. Even after extensive black-box and glass-box testing, maybe there's still some test case the programmer failed to invent, and that test case would reveal a fault in the program.

Program testing can be used to show the presence of bugs, but never to show their absence.

---Edsger W. Dijkstra

The point is not that testing is useless! It can be quite effective. But it is a kind of *inductive reasoning*, in which evidence (i.e., passing tests) accumulates in support of a conclusion (i.e., correctness of the program) without absolutely guaranteeing the validity of that conclusion. (Note that the word "inductive" here is being used in a different sense than the proof technique known as induction.) To get that guarantee, we turn to *deductive reasoning*, in which we proceed from premises and rules about logic to a valid conclusion. In other words, we prove the correctness of the program. Our goal, next, is to learn some techniques for such correctness proofs. These techniques are known as *formal methods* because of their use of logical formalism.

Correctness here means that the program produces the right output according to a specification. Specifications are usually provided in the documentation of a function (hence the name "specification comment"): they describe the program's precondition and postcondition. Postconditions, as we have been writing them, have the form $[f \ x]$ is "...a description of the output in terms of the input [x]...". For example, the specification of a factorial function could be:

```
(** [fact n] is [n!]. Requires: [n >= 0]. *)
let rec fact n = ...
```

The postcondition is asserting an equality between the output of the function and some English description of a computation on the input. *Formal verification* is the task for proving that the implementation of the function satisfies its specification.

Equalities are one of the fundamental ways we think about correctness of functional programs. The absence of mutable state makes it possible to reason straightforwardly about whether two expressions are equal. It's difficult to do that in an imperative language, because those expressions might have side effects that change the state.

6.7.1 Equality

When are two expressions equal? Two possible answers are:

- When they are syntactically identical.
- When they are semantically equivalent: they produce the same value.

For example, are 42 and 41+1 equal? The syntactic answer would say they are not, because they involve different tokens. The semantic answer would say they are: they both produce the value 42.

What about functions: are fun $x \to x$ and fun $y \to y$ equal? Syntactically they are different. But semantically, they both produce a value that is the identity function: when they are applied to an input, they will both produce the same output. That is, $(fun \times x \to x) \times z = z$, and $(fun \times y \to y) \times z = z$. If it is the case that for all inputs two functions produce the same output, we will consider the functions to be equal:

```
if (forall x, f x = g x), then f = g.
```

That definition of equality for functions is known as the *Axiom of Extensionality* in some branches of mathematics; henceforth we'll refer to it simply as "extensionality".

Here we will adopt the semantic approach. If e1 and e2 evaluate to the same value v, then we write e1 = e2. We are using = here in a mathematical sense of equality, not as the OCaml polymorphic equality operator. For example, we allow (fun $x \rightarrow x$) = (fun $y \rightarrow y$), even though OCaml's operator would raise an exception and refuse to compare functions.

We're also going to restrict ourselves to expressions that are well typed, pure (meaning they have no side effects), and total (meaning they don't have exceptions or infinite loops).

6.7.2 Equational Reasoning

Consider these functions:

```
let twice f x = f (f x)
let compose f g x = f (g x)
```

```
val twice : ('a -> 'a) -> 'a -> 'a = <fun>
```

```
val compose : ('a -> 'b) -> ('c -> 'a) -> 'c -> 'b = <fun>
```

We know from the rules of OCaml evaluation that twice h x = h (h x), and likewise, compose h h x = h (h x). Thus we have:

```
twice h x = h (h x) = compose h h x
```

Therefore can conclude that twice $h \times = compose h h \times And$ by extensionality we can simplify that equality: Since twice $h \times = compose h h \times holds$ for all x, we can conclude twice h = compose h h.

As another example, suppose we define an infix operator for function composition:

```
let ( << ) = compose
```

Then we can prove that composition is associative, using equational reasoning:

```
Theorem: (f << g) << h = f << (g << h)

Proof: By extensionality, we need to show
  ((f << g) << h) x = (f << (g << h)) x

for an arbitrary x.

  ((f << g) << h) x
= (f << g) (h x)
= f (g (h x))

and

  (f << (g << h)) x
= f ((g << h)) x
= f ((g << h) x)
= f (g (h x))

So ((f << g) << h) x = f (g (h x)) = (f << (g << h)) x.

QED
```

All of the steps in the equational proof above follow from evaluation. Another format for writing the proof would provide hints as to why each step is valid:

```
((f << g) << h) x
= { evaluation of << }
    (f << g) (h x)
= { evaluation of << }
    f (g (h x))

and

(f << (g << h)) x
= { evaluation of << }
    f ((g << h) x)
= { evaluation of << }
    f ((g (h x)))</pre>
```

6.7.3 Induction on Natural Numbers

The following function sums the non-negative integers up to n:

```
let rec sumto n =
  if n = 0 then 0 else n + sumto (n - 1)
```

```
val sumto : int -> int = <fun>
```

You might recall that the same summation can be expressed in closed form as n * (n + 1) / 2. To prove that for all $n \ge 0$, sum to n = n * (n + 1) / 2, we will need mathematical induction.

Recall that induction on the natural numbers (i.e., the non-negative integers) is formulated as follows:

```
forall properties P, if P(0), and if forall k, P(k) implies P(k+1), then forall n, P(n)
```

That is called the *induction principle* for natural numbers. The *base case* is to prove P(0), and the *inductive case* is to prove that P(k + 1) holds under the assumption of the *inductive hypothesis* P(k).

Let's use induction to prove the correctness of sumto.

```
Claim: sumto n = n * (n + 1) / 2
Proof: by induction on n.
P(n) = sumto n = n * (n + 1) / 2
Base case: n = 0
Show: sumto 0 = 0 * (n + 1) / 2
 sumto 0
  { evaluation }
 0
  { algebra }
 0 * (n + 1) / 2
Inductive case: n = k + 1
Show: sumto (k + 1) = (k + 1) * ((k + 1) + 1) / 2
IH: sumto k = k * (k + 1) / 2
 sumto (k + 1)
  { evaluation }
 k + 1 + sumto k
 { IH }
 k + 1 + k * (k + 1) / 2
 { algebra }
  (k + 1) * (k + 2) / 2
QED
```

Note that we have been careful in each of the cases to write out what we need to show, as well as to write down the inductive hypothesis. It is important to show all this work.

Suppose we now define:

```
let sumto_closed n = n * (n + 1) / 2
```

```
val sumto_closed : int -> int = <fun>
```

Then as a corollary to our previous claim, by extensionality we can conclude

```
sumto_closed = sumto
```

Technically that equality holds only inputs that are natural numbers. But since all our examples henceforth will be for naturals, not integers per se, we will elide stating any preconditions or restrictions regarding natural numbers.

6.7.4 Programs as Specifications

We have just proved the correctness of an efficient implementation relative to an inefficient implementation. The inefficient implementation, sumto_closed.

That technique is common in verifying functional programs: write an obviously correct implementation that is lacking in some desired property, such as efficiency, then prove that a better implementation is equal to the original.

Let's do another example of this kind of verification. This time, well use the factorial function.

The simple, obviously correct implementation of factorial would be:

```
let rec fact n =
  if n = 0 then 1 else n * fact (n - 1)
```

```
val fact : int -> int = <fun>
```

A tail-recursive implementation would be more efficient about stack space:

```
let rec facti acc n =
  if n = 0 then acc else facti (acc * n) (n - 1)
let fact_tr n = facti 1 n
```

```
val facti : int -> int -> int = <fun>
```

```
val fact_tr : int -> int = <fun>
```

The i in the name fact i stands for *iterative*. We call this an iterative implementation because it strongly resembles how the same computation would be expressed using a loop (that is, an iteration construct) in an imperative language. For example, in Java we might write:

```
int facti (int n) {
  int acc = 1;
  while (n != 0) {
    acc *= n;
    n--;
  }
  return acc;
}
```

Both the OCaml and Java implementation of facti share these features:

- they start acc at 1
- they check whether n is 0
- they multiply acc by n

- · they decrement n
- they return the accumulator, acc

Let's try to prove that fact_tr correctly implements the same computation as fact.

```
Claim: forall n, fact n = fact_tr n
Since fact_tr n = facti 1 n, it suffices to show fact n = facti 1 n.
Proof: by induction on n.
P(n) = fact n = facti 1 n
Base case: n = 0
Show: fact 0 = facti 1 0
 fact 0
  { evaluation }
 1
  { evaluation }
 facti 1 0
Inductive case: n = k + 1
Show: fact (k + 1) = facti 1 (k + 1)
IH: fact k = facti 1 k
 fact (k + 1)
  { evaluation }
  (k + 1) * fact k
  { IH }
 (k + 1) * facti 1 k
 facti 1 (k + 1)
  { evaluation }
 facti (1 * (k + 1)) k
  { evaluation }
 facti (k + 1) k
Unfortunately, we're stuck. Neither side of what we want to show
can be manipulated any further.
ABORT
```

We know that facti (k + 1) k and (k + 1) * facti 1 k should yield the same value. But the IH allows us only to use 1 as the second argument to facti, instead of a bigger argument like k + 1. So our proof went astray the moment we used the IH. We need a stronger inductive hypothesis!

So let's strengthen the claim we are making. Instead of showing that fact n = facti 1 n, we'll try to show forall p, p * fact n = facti p n. That generalizes the k + 1 we were stuck on to an arbitrary quantity p.

```
Claim: forall n, forall p . p * fact n = facti p n

Proof: by induction on n.

P(n) = forall p, p * fact n = facti p n

Base case: n = 0

Show: forall p, p * fact 0 = facti p 0
```

```
p * fact 0
  { evaluation and algebra }
 р
   { evaluation }
 facti p 0
Inductive case: n = k + 1
Show: forall p, p * fact (k + 1) = facti p (k + 1)
IH: forall p, p * fact k = facti p k
 p * fact (k + 1)
  { evaluation }
 p * (k + 1) * fact k
 { IH, instantiating its p as p * (k + 1) }
 facti (p * (k + 1)) k
 facti p (k + 1)
  { evaluation }
 facti (p * (k + 1)) k
Claim: forall n, fact n = fact_tr n
Proof:
 fact n
   { algebra }
 1 * fact n
   { previous claim }
 facti 1 n
  { evaluation }
 fact_tr n
QED
```

That finishes our proof that the efficient, tail-recursive function fact_tr is equivalent to the simple, recursive function fact. In essence, we have proved the correctness of fact_tr using fact as its specification.

6.7.5 Recursion vs. Iteration

We added an accumulator as an extra argument to make the factorial function be tail recursive. That's a trick we've seen before. Let's abstract and see how to do it in general.

Suppose we have a recursive function over integers:

```
let rec f_r n =
  if n = 0 then i else op n (f_r (n - 1))
```

Here, the r in f_r is meant to suggest that f_r is a recursive function. The i and op are pieces of the function that are meant to be replaced by some concrete value i and operator op. For example, with the factorial function, we have:

```
f_r = fact
i = 1
op = ( * )
```

Such a function can be made tail recursive by rewriting it as follows:

```
let f_i acc n =
   if n = 0 then acc
   else f_i (op acc n) (n - 1)
let f_tr = f_i i
```

Here, the i in f_i is meant to suggest that f_i is an iterative function, and i and op are the same as in the recursive version of the function. For example, with factorial we have:

```
f_i = fact_i
i = 1
op = ( * )
f_tr = fact_tr
```

We can prove that f_r and f_t compute the same function. During the proof, next, we will discover certain conditions that must hold of i and op to make the transformation to tail recursion be correct.

```
Theorem: f_r = f_tr
Proof: By extensionality, it suffices to show that forall n, f_r = f_t = n.
As in the previous proof for fact, we will need a strengthened induction
hypothesis. So we first prove this lemma, which quantifies over all accumulators
that could be input to f_i, rather than only i:
 Lemma: forall n, forall acc, op acc (f_r n) = f_i acc n
 Proof of Lemma: by induction on n.
 P(n) = forall acc, op acc (f_r n) = f_i acc n
 Base: n = 0
 Show: forall acc, op acc (f_r 0) = f_i acc 0
   op acc (f_r 0)
    { evaluation }
   op acc i
    { if we assume forall x, op x i = x }
   acc
   f_i acc 0
   acc
 Inductive case: n = k + 1
 Show: forall acc, op acc (f_r (k + 1)) = f_i acc (k + 1)
 IH: forall acc, op acc (f_r k) = f_i acc k
   op acc (f_r (k + 1))
 = { evaluation }
   op acc (op (k + 1) (f_r k))
    { if we assume forall x y z, op x (op y z) = op (op x y) z }
   op (op acc (k + 1)) (f_r k)
   f_i acc (k + 1)
 = { evaluation }
   f_i (op acc (k + 1)) k
```

```
= { IH, instantiating acc as op acc (k + 1)}
    op (op acc (k + 1)) (f_r k)

QED

The proof then follows almost immediately from the lemma:

    f_r n
= { if we assume forall x, op i x = x }
    op i (f_r n)
= { lemma, instantiating acc as i }
    f_i i n
= { evaluation }
    f_tr n
```

Along the way we made three assumptions about i and op:

```
1. forall x, op x i = x

2. op x (op y z) = op (op x y) z

3. forall x, op i x = x
```

The first and third say that i is an *identity* of op: using it on the left or right side leaves the other argument x unchanged. The second says that op is *associative*. Both those assumptions held for the values we used in the factorial functions:

- op is multiplication, which is associative.
- i is 1, which is an identity of multiplication: multiplication by 1 leaves the other argument unchanged.

So our transformation from a recursive to a tail-recursive function is valid as long as the operator applied in the recursive call is associative, and the value returned in the base case is an identity of that operator.

Returning to the sumto function, we can apply the theorem we just proved to immediately get a tail-recursive version:

```
let rec sumto_r n =
if n = 0 then 0 else n + sumto_r (n - 1)
```

```
val sumto_r : int -> int = <fun>
```

Here, the operator is addition, which is associative; and the base case is zero, which is an identity of addition. Therefore our theorem applies, and we can use it to produce the tail-recursive version without even having to think about it:

```
let rec sumto_i acc n =
  if n = 0 then acc else sumto_i (acc + n) (n - 1)
let sumto_tr = sumto_i 0
```

```
val sumto_i : int -> int -> int = <fun>
```

```
val sumto_tr : int -> int = <fun>
```

We already know that sumto_tr is correct, thanks to our theorem.

6.7.6 Termination

Sometimes correctness of programs is further divided into:

- partial correctness: meaning that if a program terminates, then its output is correct; and
- total correctness: meaning that a program *does* terminate, *and* its output is correct.

Total correctness is therefore the conjunction of partial correctness and termination. Thus far, we have been proving partial correctness.

To prove that a program terminates is difficult. Indeed, it is impossible in general for an algorithm to do so: a computer can't precisely decide whether a program will terminate. (Look up the "halting problem" for more details.) But, a smart human sometimes can do so.

There is a simple heuristic that can be used to show that a recursive function terminates:

- · All recursive calls are on a "smaller" input, and
- all base cases are terminating.

For example, consider the factorial function:

```
let rec fact n =
  if n = 0 then 1
  else n * fact (n - 1)
```

```
val fact : int -> int = <fun>
```

The base case, 1, obviously terminates. The recursive call is on n-1, which is a smaller input than the original n. So fact always terminates (as long as its input is a natural number).

The same reasoning applies to all the other functions we've discussed above.

To make this more precise, we need a notion of what it means to be smaller. Suppose we have a binary relation < on inputs. Despite the notation, this relation need not be the less-than relation on integers—although that will work for fact. Also suppose that it is never possible to create an infinite sequence x0 > x1 > x2 > x3 . . . of elements using this relation. (Where of course a > b iff b < a.) That is, there are no infinite descending chains of elements: once you pick a starting element x0, there can be only a finite number of "descents" according to the < relation before you bottom out and hit a base case. This property of < makes it a well-founded relation.

So, a recursive function terminates if all its recursive calls are on elements that are smaller according to <. Why? Because there can be only a finite number of calls before a base case is reached, and base cases must terminate.

The usual < relation is well-founded on the natural numbers, because eventually any chain must reach the base case of 0. But it is not well-founded on the integers, which can get just keep getting smaller: $-1 > -2 > -3 > \dots$

Here's an interesting function for which the usual < relation doesn't suffice to prove termination:

```
let rec ack = function
| (0, n) -> n + 1
| (m, 0) -> ack (m - 1, 1)
| (m, n) -> ack (m - 1, ack (m, n - 1))
```

```
val ack : int * int -> int = <fun>
```

This is known as *Ackermann's function*. It grows faster than any exponential function. Try running ack (1, 1), ack (2, 1), ack (3, 1), then ack (4, 1) to get a sense of that. It also is a famous example of a function that can be implemented with while loops but not with for loops. Nonetheless, it does terminate.

To show that, the base case is easy: when the input is (0, _), the function terminates. But in other cases, it makes a recursive call, and we need to define an appropriate < relation. It turns out *lexicograpic ordering* on pairs works. Define (a, b) < (c, d) if:

```
• a < c, or
```

```
• a = c and b < d.
```

The < order in those two cases is the usual < on natural numbers.

In the first recursive call, (m - 1, 1) < (m, 0) by the first case of the definition of <, because m - 1 < m. In the nested recursive call ack (m - 1, ack (m, n - 1)), both cases are needed:

```
• (m, n - 1) < (m, n) because m = m and n - 1 < n
```

```
• (m - 1, _) < (m, _n) because m - 1 < m.
```

6.8 Structural Induction

So far we've proved the correctness of recursive functions on natural numbers. We can do correctness proofs about recursive functions on variant types, too. That requires us to figure out how induction works on variants. We'll do that, next, starting with a variant type for representing natural numbers, then generalizing to lists, trees, and other variants. This inductive proof technique is sometimes known as *structural induction* instead of *mathematical induction*. But that's just a piece of vocabulary; don't get hung up on it. The core idea is completely the same.

6.8.1 Induction on Naturals

We used OCaml's int type as a representation of the naturals. Of course, that type is somewhat of a mismatch: negative int values don't represent naturals, and there is an upper bound to what natural numbers we can represent with int.

Let's fix those problems by defining our own variant to represent natural numbers:

```
type nat = Z | S of nat
```

```
type nat = Z | S of nat
```

The constructor Z represents zero; and the constructor S represents the successor of another natural number. So,

- 0 is represented by Z,
- 1 by S Z,
- 2 by S (S Z),
- 3 by S (S (S Z)),

and so forth. This variant is thus a *unary* (as opposed to binary or decimal) representation of the natural numbers: the number of times S occurs in a value n : nat is the natural number that n represents.

We can define addition on natural numbers with the following function:

```
let rec plus a b =
  match a with
  | Z -> b
  | S k -> S (plus k b)
```

```
val plus : nat -> nat -> nat = <fun>
```

Immediately we can prove the following rather trivial claim:

```
Claim: plus Z n = n
Proof:
   plus Z n
= { evaluation }
   n
```

But suppose we want to prove this also trivial-seeming claim:

```
Claim: plus n Z = n
Proof:
   plus n Z
=
   ???
```

We can't just evaluate plus n Z, because plus matches against its first argument, not second. One possibility would be to do a case analysis: what if n is Z, vs. S k for some k? Let's attempt that.

```
Proof:
By case analysis on n, which must be either Z or S k.

Case: n = Z

plus Z Z

= { evaluation }
   Z

Case: n = S k

plus (S k) Z

= { evaluation }
   S (plus k Z)

=
   ???
```

We are again stuck, and for the same reason: once more plus can't be evaluated any further.

When you find yourself needing to solve the same subproblem in programming, you use recursion. When it happens in a proof, you use induction!

We'll need an induction principle for nat. Here it is:

```
forall properties P, if P(Z), and if forall k, P(k) implies P(S|k), then forall n, P(n)
```

Compare that to the induction principle we used for natural numbers before, when we were using int in place of natural numbers:

```
forall properties P, if P(0), and if forall k, P(k) implies P(k+1), then forall n, P(n)
```

There's no essential difference between the two: we just use Z in place of 0, and S k in place of k + 1.

Using that induction principle, we can carry out the proof:

```
Claim: plus n Z = n
Proof: by induction on n.
P(n) = plus n Z = n
Base case: n = Z
Show: plus Z Z = Z
 plus Z Z
 { evaluation }
 7.
Inductive case: n = S k
IH: plus k Z = k
Show: plus (S k) Z = S k
 plus (S k) Z
  { evaluation }
 S (plus k Z)
  { IH }
 S k
OED
```

6.8.2 Induction on Lists

It turns out that natural numbers and lists are quite similar, when viewed as data types. Here are the definitions of both, aligned for comparison:

Both types have a constructor representing a concept of "nothing". Both types also have a constructor representing "one more" than another value of the type: S_n is one more than n, and h:: t is a list with one more element than t.

The induction principle for lists is likewise quite similar to the induction principle for natural numbers. Here is the principle for lists:

```
forall properties P,
  if P([]),
  and if forall h t, P(t) implies P(h :: t),
  then forall lst, P(lst)
```

An inductive proof for lists therefore has the following structure:

```
Proof: by induction on lst.
P(lst) = ...
```

```
Base case: lst = []
Show: P([])

Inductive case: lst = h :: t
IH: P(t)
Show: P(h :: t)
```

Let's try an example of this kind of proof. Recall the definition of the append operator:

```
let rec append lst1 lst2 =
   match lst1 with
   | [] -> lst2
   | h :: t -> h :: append t lst2

let (@) = append
```

```
val append : 'a list -> 'a list -> 'a list = <fun>
```

```
val (@): 'a list -> 'a list = <fun>
```

We'll prove that append is associative.

```
Theorem: forall xs ys zs, xs @ (ys @ zs) = (xs @ ys) @ zs
Proof: by induction on xs.
P(xs) = forall ys zs, xs @ (ys @ zs) = (xs @ ys) @ zs
Base case: xs = []
Show: forall ys zs, [] @ (ys @ zs) = ([] @ ys) @ zs
 [] @ (ys @ zs)
  { evaluation }
ys @ zs
  { evaluation }
 ([] @ ys) @ zs
Inductive case: xs = h :: t
IH: forall ys zs, t @ (ys @ zs) = (t @ ys) @ zs
Show: forall ys zs, (h :: t) @ (ys @ zs) = ((h :: t) @ ys) @ zs
 (h :: t) @ (ys @ zs)
 { evaluation }
h :: (t @ (ys @ zs))
= { IH }
 h :: ((t @ ys) @ zs)
 ((h :: t) @ ys) @ zs
  { evaluation of inner @ }
 (h :: (t @ ys)) @ zs
  { evaluation of outer @ }
h :: ((t @ ys) @ zs)
OED
```

6.8.3 A Theorem about Folding

When we studied List.fold_left and List.fold_right, we discussed how they sometimes compute the same function, but in general do not. For example,

```
List.fold_left ( + ) 0 [1; 2; 3]

= (((0 + 1) + 2) + 3

= 6

= 1 + (2 + (3 + 0))

= List.fold_right ( + ) lst [1; 2; 3]
```

but

```
List.fold_left ( - ) 0 [1; 2; 3]

= (((0 - 1) - 2) - 3

= -6

<> 2

= 1 - (2 - (3 - 0))

= List.fold_right ( - ) lst [1; 2; 3]
```

Based on the equations above, it looks like the fact that + is commutative and associative, whereas – is not, explains this difference between when the two fold functions get the same answer. Let's prove it!

First, recall the definitions of the fold functions:

```
let rec fold_left f acc lst =
   match lst with
   | [] -> acc
   | h :: t -> fold_left f (f acc h) t

let rec fold_right f lst acc =
   match lst with
   | [] -> acc
   | h :: t -> f h (fold_right f t acc)
```

```
val fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a = <fun>
```

```
val fold_right : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b = <fun>
```

Second, recall what it means for a function f: 'a -> 'a to be commutative and associative:

```
Commutative: forall x y, f x y = f y x Associative: forall x y z, f x (f y z) = f (f x y) z
```

Those might look a little different than the normal formulations of those properties, because we are using f as a prefix operator. If we were to write f instead as an infix operator op, they would look more familiar:

```
Commutative: forall x y, x op y = y op x
Associative: forall x y z, x op (y 	ext{ op } z) = (x 	ext{ op } y) 	ext{ op } z
```

When f is both commutative and associative we have this little interchange lemma that lets us swap two arguments around:

```
Lemma (interchange): f \times (f y z) = f y (f \times z)

Proof:
f \times (f y z)
```

```
= { associativity }
  f (f x y) z
= { commutativity }
  f (f y x) z
= { associativity }
  f y (f z x)
```

Now we're ready to state and prove the theorem.

```
Theorem: If f is commutative and associative, then
 forall 1st acc,
    fold_left f acc lst = fold_right f lst acc.
Proof: by induction on 1st.
P(lst) = forall acc,
 fold_left f acc lst = fold_right f lst acc
Base case: 1st = []
Show: forall acc,
 fold_left f acc [] = fold_right f [] acc
 fold_left f acc []
  { evaluation }
 acc
   { evaluation }
 fold_right f [] acc
Inductive case: lst = h :: t
IH: forall acc,
 fold_left f acc t = fold_right f t acc
Show: forall acc,
 fold_left f acc (h :: t) = fold_right f (h :: t) acc
 fold_left f acc (h :: t)
  { evaluation }
 fold_left f (f acc h) t
= { IH with acc := f acc h }
 fold_right f t (f acc h)
 fold_right f (h :: t) acc
  { evaluation }
 f h (fold_right f t acc)
```

Now, it might seem as though we are stuck: the left and right sides of the equality we want to show have failed to "meet in the middle." But we're actually in a similar situation to when we proved the correctness of facti earlier: there's something (applying f to h and another argument) that we want to push into the accumulator of that last line (so that we have f acc h).

Let's try proving that with its own lemma:

```
Lemma: forall lst acc x,
   f x (fold_right f lst acc) = fold_right f lst (f acc x)

Proof: by induction on lst.
P(lst) = forall acc x,
```

```
f x (fold_right f lst acc) = fold_right f lst (f acc x)
Base case: lst = []
Show: forall acc x,
 f x (fold_right f [] acc) = fold_right f [] (f acc x)
 f x (fold_right f [] acc)
  { evaluation }
 f x acc
 fold_right f [] (f acc x)
  { evaluation }
 f acc x
 { commutativity of f }
 f x acc
Inductive case: lst = h :: t
IH: forall acc x,
 f x (fold_right f t acc) = fold_right f t (f acc x)
Show: forall acc x,
 f x (fold_right f (h :: t) acc) = fold_right f (h :: t) (f acc x)
 f x (fold_right f (h :: t) acc)
= { evaluation }
 f x (f h (fold_right f t acc))
  { interchange lemma }
 f h (f x (fold_right f t acc))
 { IH }
 f h (fold_right f t (f acc x))
 fold_right f (h :: t) (f acc x)
  { evaluation }
 f h (fold_right f t (f acc x))
QED
```

Now that the lemma is completed, we can resume the proof of the theorem. We'll restart at the beginning of the inductive case:

```
Inductive case: lst = h :: t
IH: forall acc,
  fold_left f acc t = fold_right f t acc
Show: forall acc,
  fold_left f acc (h :: t) = fold_right f (h :: t) acc

fold_left f acc (h :: t)
= { evaluation }
  fold_left f (f acc h) t
= { IH with acc := f acc h }
  fold_right f t (f acc h)

fold_right f (h :: t) acc
= { evaluation }
  f h (fold_right f t acc)
= { lemma with x := h and lst := t }
  fold_right f t (f acc h)
```

```
QED
```

It took two inductions to prove the theorem, but we succeeded! Now we know that the behavior we observed with + wasn't a fluke: any commutative and associative operator causes fold_left and fold_right to get the same answer.

6.8.4 Induction on Trees

Lists and binary trees are similar when viewed as data types. Here are the definitions of both, aligned for comparison:

```
type 'a list = [] | (::) of 'a * 'a list
```

```
type 'a tree = Leaf | Node of 'a tree * 'a * 'a tree
```

Both have a constructor that represents "empty", and both have a constructor that combines a value of type 'a together with another instance of the data type. The only real difference is that (::) takes just *one* list, whereas Node takes *two* trees.

The induction principle for binary trees is therefore very similar to the induction principle for lists, except that with binary trees we get *two* inductive hypotheses, one for each subtree:

```
forall properties P, if P(Leaf), and if forall 1 v r, (P(1) and P(r)) implies P(Node\ (l,\ v,\ r)), then forall t, P(t)
```

An inductive proof for binary trees therefore has the following structure:

```
Proof: by induction on t.
P(t) = ...

Base case: t = Leaf
Show: P(Leaf)

Inductive case: t = Node (l, v, r)
IH1: P(l)
IH2: P(r)
Show: P(Node (l, v, r))
```

Let's try an example of this kind of proof. Here is a function that creates the mirror image of a tree, swapping its left and right subtrees at all levels:

```
val reflect : 'a tree -> 'a tree = <fun>
```

For example, these two trees are reflections of each other:



If you take the mirror image of a mirror image, you should get the original back. That means reflection is an *involution*, which is any function f such that f(fx) = x. Another example of an involution is multiplication by negative one on the integers.

Let's prove that reflect is an involution.

```
Claim: forall t, reflect (reflect t) = t
Proof: by induction on t.
P(t) = reflect (reflect t) = t
Base case: t = Leaf
Show: reflect (reflect Leaf) = Leaf
 reflect (reflect Leaf)
   { evaluation }
 reflect Leaf
   { evaluation }
 Leaf
Inductive case: t = Node (1, v, r)
IH1: reflect (reflect 1) = 1
IH2: reflect (reflect r) = r
Show: reflect (reflect (Node (1, v, r))) = Node (1, v, r)
 reflect (reflect (Node (1, v, r)))
  { evaluation }
 reflect (Node (reflect r, v, reflect 1))
   { evaluation }
 Node (reflect (reflect 1), v, reflect (reflect r))
   { IH1 }
 Node (1, v, reflect (reflect r))
  { IH2 }
 Node (1, v, r)
QED
```

Induction on trees is really no more difficult than induction on lists or natural numbers. Just keep track of the inductive hypotheses, using our stylized proof notation, and it isn't hard at all.

6.8.5 Induction Principles for All Variants

We've now seen induction principles for nat, list, and tree. Generalizing from what we've seen, each constructor of a variant either generates a base case for the inductive proof, or an inductive case. And, if a constructor itself carries values of that data type, each of those values generates in inductive hypothesis. For example:

- Z, [], and Leaf all generated base cases.
- S, ::, and Node all generated inductive cases.
- S and : : each generated one IH, because each carries one value of the data type.

Node generated two IHs, because it carries two values of the data type.

Suppose we have these types to represent the AST for expressions in a simple language with integers, Booleans, unary operators, and binary operators:

The induction principle for expr is:

```
forall properties P,
  if forall i, P(Int i)
  and forall b, P(Bool b)
  and forall u e, P(e) implies P(Unop (u, e))
  and forall b e1 e2, (P(e1) and P(e2)) implies P(Binop (e1, b, e2))
  then forall e, P(e)
```

There are two base cases, corresponding to the two constructors that don't carry an expr. There are two inductive cases, corresponding to the two constructors that do carry exprs. Unop gets one IH, whereas Binop gets two IHs, because of the number of exprs that each carries.

6.8.6 Induction and Recursion

Inductive proofs and recursive programs bear a striking similarity. In a sense, an inductive proof *is* a recursive program that shows how to construct evidence for a theorem involving an algebraic data type (ADT). The **structure** of an ADT determines the structure of proofs and programs:

- The **constructors** of an ADT are the organizational principle of both proofs and programs. In a proof, we have a base or inductive case for each constructor. In a program, we have a pattern-matching case for each constructor.
- The use of **recursive types** in an ADT determine where recursion occurs in both proofs and programs. By "recursive type", we mean the occurrence of the type in its own definition, such as the second 'a list in type 'a list = [] | (::) 'a * 'a list. Such occurrences lead to "smaller" values of a type occurring inside larger values. In a proof, we apply the inductive hypothesis upon reaching such a smaller value. In a program, we recurse on the smaller value.

6.9 Algebraic Specification

Next let's tackle a bigger challenge: proving the correctness of a data structure, such as a stack, queue, or set.

Correctness proofs always need specifications. In proving the correctness of iterative factorial, we used recursive factorial as a specification. By analogy, we could provide two implementations of a data structure—one simple, the other complex and efficient—and prove that the two are equivalent. That would require us to introduce ways to translate between the two implementations. For example, we could prove the correctness of a map implemented with an efficient balanced binary search tree relative to an implementation as an inefficient association list, by defining functions to convert trees to lists. Such an approach is certainly valid, but it doesn't lead to new ideas about verification for us to study.

Instead, we will pursue a different approach based on equational specifications, aka algebraic specifications. The idea with these is to

- define the types of the data structure operations, and
- to write a set of equations that define how the operations interact with one another.

The reason the word "algebra" shows up here is (in part) that this type-and-equation based approach is something we learned in high-school algebra. For example, here is a specification for some operators:

```
0: int
1: int
-: int -> int
+: int -> int -> int
*: int -> int -> int

(a + b) + c = a + (b + c)
a + b = b + a
a + 0 = a
a + (-a) = 0
(a * b) * c = a * (b * c)
a * b = b * a
a * 1 = a
a * 0 = 0
a * (b + c) = a * b + a * c
```

The types of those operators, and the associated equations, are facts learned when studying algebra.

Our goal is now to write similar specifications for data structures, and use them to reason about the correctness of implementations.

6.9.1 Example: Stacks

Here are a few familiar operations on stacks along with their types.

```
module type Stack = sig
  type 'a t
  val empty : 'a t
  val is_empty : 'a t -> bool
  val peek : 'a t -> 'a
  val push : 'a -> 'a t -> 'a t
  val pop : 'a t -> 'a t
end
```

As usual, there is a design choice to be made with peek etc. about what to do with empty stacks. Here we have not used option, which suggests that peek will raise an exception on the empty stack. So we are cautiously relaxing our prohibition on exceptions.

In the past we've given these operations specifications in English, e.g.,

```
(** [push x s] is the stack [s] with [x] pushed on the top. *)
val push : 'a -> 'a stack -> 'a stack
```

But now, we'll write some equations to describe how the operations work:

```
1. is_empty empty = true
2. is_empty (push x s) = false
3. peek (push x s) = x
4. pop (push x s) = s
```

(Later we'll return to the question of *how* to design such equations.) The variables appearing in these equations are implicitly universally quantified. Here's how to read each equation:

- 1. is_empty empty = true. The empty stack is empty.
- 2. is_{empty} (push x s) = false. A stack that has just been pushed is non-empty.
- 3. peek (push x s) = x. Pushing then immediately peeking yields whatever value was pushed.
- 4. pop (push x s) = s. Pushing then immediately popping yields the original stack.

Just with these equations alone, we already can deduce a lot about how any sequence of stack operations must work. For example,

```
peek (pop (push 1 (push 2 empty)))
= { equation 4 }
  peek (push 2 empty)
= { equation 3 }
2
```

And peek empty doesn't equal any value according to the equations, since there is no equation of the form peek empty = All that is true regardless of the stack implementation that is chosen: any correct implementation must cause the equations to hold.

Suppose we implemented stacks as lists, as follows:

```
module ListStack = struct
  type 'a t = 'a list
  let empty = []
  let is_empty = function [] -> true | _ -> false
  let peek = List.hd
  let push = List.cons
  let pop = List.tl
end
```

Next we could *prove* that each equation holds of the implementation. All these proofs are quite easy by now, and proceed entirely by evaluation. For example, here's a proof of equation 3:

```
peek (push x s)
= { evaluation }
  peek (x :: s)
= { evaluation }
  x
```

6.9.2 Example: Queues

Stacks were easy. How about queues? Here is the specification:

```
module type Queue = sig
  type 'a t
  val empty : 'a t
  val is_empty : 'a t -> bool
  val front : 'a t -> 'a
  val enq : 'a -> 'a t -> 'a t
  val deq : 'a t -> 'a t
end
```

The types of the queue operations are actually identical to the types of the stack operations. Here they are, side-by-side for comparison:

```
module type Stack = sig
                                  module type Queue = sig
                                   type 'a t
 type 'a t
 val empty : 'a t
                                   val empty : 'a t
 val is_empty : 'a t -> bool
                                   val is_empty : 'a t -> bool
 val peek : 'a t -> 'a
                                   val front : 'a t -> 'a
 val push : 'a -> 'a t -> 'a t
                                  val eng : 'a -> 'a t -> 'a t
                                   val deq : 'a t -> 'a t
 val pop : 'a t -> 'a t
end
                                  end
```

Look at each line: though the operation may have a different name, its type is the same. Obviously, the types alone don't tell us enough about the operations. But the equations do! Here's how to read each equation:

- 1. The empty queue is empty.
- 2. Enqueueing makes a queue non-empty.
- 3. Enqueueing x on an empty queue makes x the front element. But if the queue isn't empty, enqueueing doesn't change the front element.
- 4. Enqueueing then dequeueing on an empty queue leaves the queue empty. But if the queue isn't empty, the enqueue and dequeue operations can be swapped.

For example,

```
front (deq (enq 1 (enq 2 empty)))
= { equation 4b }
  front (enq 1 (deq (enq 2 empty)))
= { equation 4a }
  front (enq 1 empty)
= { equation 3a }
1
```

And front empty doesn't equal any value according to the equations.

Implementing a queue as a list results in an implementation that is easy to verify just with evaluation.

```
module ListQueue : Queue = struct
  type 'a t = 'a list
  let empty = []
  let is_empty q = q = []
  let front = List.hd
  let eng x q = q @ [x]
  let deq = List.tl
end
```

```
module ListQueue : Queue
```

For example, 4a can be verified as follows:

```
deq (enq x empty)
= { evaluation of empty and enq}
  deq ([] @ [x])
= { evaluation of @ }
  deq [x]
= { evaluation of deq }
  []
= { evaluation of empty }
  empty
```

And 4b, as follows:

```
deq (enq x q)
= { evaluation of enq and deq }
  List.tl (q @ [x])
= { lemma, below, and q <> [] }
  (List.tl q) @ [x]

enq x (deq q)
= { evaluation }
  (List.tl q) @ [x]
```

Here is the lemma:

276

```
Lemma: if xs <> [], then List.tl (xs @ ys) = (List.tl xs) @ ys.
Proof: if xs <> [], then xs = h :: t for some h and t.

List.tl ((h :: t) @ ys)
= { evaluation of @ }
List.tl (h :: (t @ ys))
= { evaluation of tl }
t @ ys

(List.tl (h :: t)) @ ys
= { evaluation of tl }
t @ ys
QED
```

Note how the precondition in 3b and 4b of q not being empty ensures that we never have to deal with an exception being raised in the equational proofs.

6.9.3 Example: Batched Queues

Recall that batched queues represent a queue with two lists:

```
module BatchedQueue = struct
    (* AF: [(o, i)] represents the queue [o @ (List.rev i)].
        RI: if [o] is empty then [i] is empty. *)
    type 'a t = 'a list * 'a list

let empty = ([], [])

let is_empty (o, i) = if o = [] then true else false

let enq x (o, i) = if o = [] then ([x], []) else (o, x :: i)

let front (o, _) = List.hd o

let deq (o, i) =
    match List.tl o with
    | [] -> (List.rev i, [])
    | t -> (t, i)
end
```

This implementation is superficially different from the earlier implementation we gave, in that it uses pairs instead of records, and it raises the built-in exception Failure instead of a custom exception Empty.

Is this implementation correct? We need only verify the equations to find out.

First, a lemma:

```
Lemma: if is_empty q = true, then q = empty.

Proof: Since is_empty q = true, it must be that q = (f, b) and f = [].

By the RI, it must also be that b = []. Thus q = ([], []) = empty.

QED
```

Verifying equation 1:

```
is_empty empty
= { eval empty }
  is_empty ([], [])
= { eval is_empty }
  [] = []
= { eval = }
  true
```

Verifying equation 2:

```
is_empty (enq x q) = false
= { eval enq }
  is_empty (if f = [] then [x], [] else f, x :: b)

case analysis: f = []
  is_empty (if f = [] then [x], [] else f, x :: b)
= { eval if, f = [] }
  is_empty ([x], [])
= { eval is_empty }
  [x] = []
```

```
= { eval = }
  false

case analysis: f = h :: t

  is_empty (if f = [] then [x], [] else f, x :: b)
= { eval if, f = h :: t }
  is_empty (h :: t, x :: b)
= { eval is_empty }
  h :: t = []
= { eval = }
  false
```

Verifying equation 3a:

```
front (enq x q) = x
= { emptiness lemma }
  front (enq x ([], []))
= { eval enq }
  front ([x], [])
= { eval front }
  x
```

Verifying equation 3b:

```
front (enq x q)
= { rewrite q as (h :: t, b), because q is not empty }
  front (enq x (h :: t, b))
= { eval enq }
  front (h :: t, x :: b)
= { eval front }
  h

front q
= { rewrite q as (h :: t, b), because q is not empty }
  front (h :: t, b)
= { eval front }
  h
```

Verifying equation 4a:

```
deq (enq x q)
= { emptiness lemma }
  deq (enq x ([], []))
= { eval enq }
  deq ([x], [])
= { eval deq }
  List.rev [], []
= { eval rev }
  [], []
= { eval empty }
  empty
```

Verifying equation 4b:

```
Show: deq (enq x q) = enq x (deq q) assuming is_empty q = false.

Proof: Since is_empty q = false, q must be (h :: t, b).
```

```
Case analysis: t = [], b = []
 deq (enq x q)
   { rewriting q as ([h], []) }
 deq (enq x ([h], []))
   { eval enq }
 deq ([h], [x])
  { eval deq }
 List.rev [x], []
= { eval rev }
 [x], []
 eng x (deg g)
= { rewriting q as ([h], []) }
 enq x (deq ([h], []))
= { eval deq }
 enq x (List.rev [], [])
= { eval rev }
 eng x ([], [])
= { eval enq }
 [x], []
Case analysis: t = [], b = h' :: t'
 deq (enq x q)
= { rewriting q as ([h], h' :: t') }
 deq (enq x ([h], h' :: t'))
  { eval enq }
 deq ([h], x :: h' :: t')
  { eval deq }
 List.rev (x :: h' :: t'), []
 enq x (deq q)
= { rewriting q as ([h], h' :: t') }
 enq x (deq ([h], h' :: t'))
= { eval deq }
 enq x (List.rev (h' :: t'), [])
= { eval enq }
  (List.rev (h' :: t'), [x])
STUCK
```

Wait, we just got stuck! List.rev (x :: h' :: t'), [] and (List.rev (h' :: t'), [x]) are not the same. But, abstractly, they do represent the same queue: (List.rev t') @ [h'; x]. We need to allow an additional equation for the representation type:

```
e = e' if AF(e) = AF(e')
```

Using that additional equation, we can continue:

```
(List.rev (h' :: t'), [x])
= { AF equation }
  List.rev (x :: h' :: t'), []
The AF equation holds because:
```

```
List.rev (h' :: t') @ [x]
  { eval rev }
 List.rev (h' :: t') @ List.rev [x]
   { rev distributes over @, an exercise in the previous lecture }
 List.rev ([x] @ (h' :: t'))
   { eval @ }
 List.rev (x :: h' :: t'))
   { lst @ [] = lst, an exercise in the previous lecture }
 List.rev (x :: h' :: t') @ []
Case analysis: t = h' :: t'
 deg (eng x g)
  { rewriting q as (h :: h' :: t', b) }
 deq (enq x (h :: h' :: t', b))
 { eval enq }
 deq (h :: h' :: t, x :: b)
 { eval deq }
 h' :: t, x :: b
 enq x (deq q)
  { rewriting q as (h :: h' :: t', b) }
 enq x (deq (h :: h' :: t', b))
  { eval deq }
 enq x (h' :: t', b)
  { eval eng }
 h' :: t', x :: b
OED
```

That concludes our verification of the batched queue. Note that we had to add the extra equation involving the abstraction function to get the proofs to go through:

```
e = e' if AF(e) = AF(e')
```

and that we made use of the RI during the proof. The AF and RI really are important!

6.9.4 Designing Algebraic Specifications

For both stacks and queues we provided some equations as the specification. Designing those equations is, in part, a matter of thinking hard about the data structure. But there's more to it than that.

Every value of the data structure is constructed with some operations. For a stack, those operations are empty and push. There might be some pop operations involved, but those can be eliminated. For example, pop (push 1 (push 2 empty)) is really the same stack as push 2 empty. The latter is the *canonical form* of that stack: there are many other ways to construct it, but that is the simplest. Indeed, every possible stack value can be constructed just with empty and push. Similarly, every possible queue value can be constructed just with empty and enq: if there are deq operations involved, those can be eliminated.

Let's categorize the operations of a data structure as follows:

- **Generators** are those operations involved in creating a canonical form. They return a value of the data structure type. For example, empty, push, enq.
- Manipulators are operations that create a value of the data structure type, but are not needed to create canonical forms. For example, pop, deq.

• Queries do not return a value of the data structure type. For example, is_empty, peek, front.

Given such a categorization, we can design the equational specification of a data structure by applying non-generators to generators. For example: What does is_empty return on empty? on push? What does front return on enq? What does deq return on enq? etc.

So if there are n generators and m non-generators of a data structure, we would begin by trying to create n * m equations, one for each pair of a generator and non-generator. Each equation would show how to simplify an expression. In some cases we might need a couple equations, depending on the result of some comparison. For example, in the queue specification, we have the following equations:

- 1. is_empty empty = true: this is a non-generator is_empty applied to a generator empty. It reduces just to a Boolean value, which doesn't involve the data structure type (queues) at all.
- 2. is_empty (enq x q) = false: a non-generator is_empty applied to a generator enq. Again it reduces simply to a Boolean value.
- 3. There are two subcases.
 - front (enq x q) = x, if is_empty q = true. A non-generator front applied to a generator enq. It reduces to x, which is a smaller expression than the original front (enq x q).
 - front (enq x q) = front q, if is_empty q = false. This similarly reduces to a smaller expression.
- 4. Again, there are two subcases.
 - deq (enq x q) = empty, if is_empty q = true. This simplifies the original expression by reducing it to empty.
 - deq (enq x q) = enq x (deq q), if is_empty q = false. This simplifies the original expression by reducing it to an generator applied to a smaller argument, deq q instead of deq (enq x q).

We don't usually design equations involving pairs of non-generators. Sometimes pairs of generators are needed, though, as we will see in the next example.

Example: Sets. Here is a small interface for sets:

```
module type Set = sig
  type 'a t
  val empty : 'a t
  val is_empty : 'a t -> bool
  val add : 'a -> 'a t -> 'a t
  val mem : 'a -> 'a t -> bool
  val remove : 'a -> 'a t -> 'a t
end
```

The generators are empty and add. The only manipulator is remove. Finally, is_empty and mem are queries. So we should expect at least 2 * 3 = 6 equations, one for each pair of generator and non-generator. Here is an equational specification:

Consider, though, these two sets:

- add 0 (add 1 empty)
- add 1 (add 0 empty)

They both intuitively represent the set $\{0,1\}$. Yet, we cannot prove that those two sets are equal using the above specification. We are missing an equation involving two generators:

```
7. add x (add y s) = add y (add x s)
```

6.10 Summary

Documentation and testing are crucial to establishing the truth of what a correct program does. Documentation communicates to other humans the intent of the programmer. Testing communicates evidence about the success of the programmer.

Good documentation provides several pieces: a summary, preconditions, postconditions (including errors), and examples. Documentation is written for two different audiences, clients and maintainers. The latter needs to know about abstraction functions and representation invariants.

Testing methodologies include black-box, glass-box, and randomized tests. These are complementary, not orthogonal, approaches to developing correct code.

Formal methods is an important link between mathematics and computer science. We can use techniques from discrete math, such as induction, to prove the correctness of functional programs. Equational reasoning makes the proofs relatively pleasant.

Proving the correctness of imperative programs can be more challenging, because of the need to reason about mutable state. That can break equational reasoning. Instead, *Hoare logic*, named for Tony Hoare, is a common formal method for imperative programs. Dijkstra's *weakest precondition* calculus is another.

6.10.1 Terms and Concepts

- · abstract value
- · abstraction by specification
- · abstraction function
- algebraic specification
- · asserting
- · associative
- · base case
- · black box
- · boundary case
- bug
- · canonical form
- client
- code inspection
- · code review
- code walkthrough

- · comments
- · commutative
- commutative diagram
- · concrete value
- conditional compilation
- consumer
- correctness
- · data abstraction
- · debugging by scientific method
- defensive programming
- equation
- · equational reasoning
- example clause
- · extensionality
- · failure
- fault
- · formal methods
- · formal methods
- generator
- glass box
- identity
- implementer
- induction
- induction hypothesis
- induction principle
- · inductive case
- inputs for classes of output
- inputs that satisfy precondition
- inputs that trigger exceptions
- iterative
- locality
- · manipulator
- many to one
- · minimal test case
- modifiability
- · natural numbers

6.10. Summary 283

OCaml Programming: Correct + Efficient + Beautiful

- pair programming
- · partial
- partial correctness
- · partial function
- path coverage
- paths through implementation
- paths through specification
- postcondition
- · postcondition
- precondition
- precondition
- producer
- query
- raises clause
- · randomized testing
- · regression testing
- rely
- · rep ok
- representation invariant
- · representation type
- · representative inputs
- · requires clause
- · returns clause
- satisfaction
- · social methods
- specification
- · specification
- testing
- · total correctness
- total function
- · typical input
- validation
- · verification
- well-founded

6.10.2 Further Reading

- Program Development in Java: Abstraction, Specification, and Object-Oriented Design, chapters 3, 5, and 9, by Barbara Liskov with John Guttag.
- The Functional Approach to Programming, section 3.4. Guy Cousineau and Michel Mauny. Cambridge, 1998.
- ML for the Working Programmer, second edition, chapter 6. L.C. Paulson. Cambridge, 1996.
- Thinking Functionally with Haskell, chapter 6. Richard Bird. Cambridge, 2015.
- Software Foundations, volume 1, chapters Basic, Induction, Lists, Poly. Benjamin Pierce et al. https://softwarefoundations.cis.upenn.edu/
- "Algebraic Specifications", Robert McCloskey, https://www.cs.scranton.edu/~mccloske/courses/se507/alg_specs_lec.html.
- Software Engineering: Theory and Practice, third edition, section 4.5. Shari Lawrence Pfleeger and Joanne M. Atlee. Prentice Hall, 2006.
- "Algebraic Semantics", chapter 12 of Formal Syntax and Semantics of Programming Languages, Kenneth Slonneger and Barry L. Kurtz, Addison-Wesley, 1995.
- "Algebraic Semantics", Muffy Thomas. Chapter 6 in *Programming Language Syntax and Semantics*, David Watt, Prentice Hall, 1991.
- Fundamentals of Algebraic Specification 1: Equations and Initial Semantics. H. Ehrig and B. Mahr. Springer-Verlag, 1985.

6.10.3 Acknowledgments

Our treatment of formal methods is inspired by and indebted to course materials for Princeton COS 326 by David Walker et al.

Our example algebraic specifications are based on McCloskey's. The terminology of "generator", "manipulator", and "query" is based on Pfleeger and Atlee.

Many of our exercises on formal methods are inspired by Software Foundations, volume 1.

6.11 Exercises

Solutions to exercises are available to students in Cornell's CS 3110. Instructors at other institutions are welcome to contact Michael Clarkson for access.

Exercise: spec game $[\star\star\star]$

Pair up with another programmer and play the specification game with them. Take turns being the specifier and the devious programmer. Here are some suggested functions you could use:

```
• num_vowels : string -> int
• is_sorted : 'a list -> bool
• sort : 'a list -> 'a list
• max : 'a list -> 'a
• is_prime : int -> bool
```

6.11. Exercises 285

```
is_palindrome : string -> boolsecond_largest : int list -> intdepth : 'a tree -> int
```

Exercise: poly spec $[\star\star\star]$

Let's create a data abstraction for single-variable integer polynomials of the form

$$c_n x^n + \dots + c_1 x + c_0$$
.

Let's assume that the polynomials are *dense*, meaning that they contain very few coefficients that are zero. Here is an incomplete interface for polynomials:

Finish the design of Poly by adding more operations to the interface. Consider what operations would be useful to a client of the abstraction:

- How would they create polynomials?
- How would they combine polynomials to get new polynomials?
- How would they query a polynomial to find out what it represents?

Write specification comments for the operations that you invent. Keep in mind the spec game as you write them: could a devious programmer subvert your intentions?

Exercise: poly impl $[\star \star \star]$

Implement your specification of Poly. As part of your implementation, you will need to choose a representation type t. Hint: recalling that our polynomials are dense might guide you in choosing a representation type that makes for an easier implementation.

Exercise: interval arithmetic $[\star \star \star \star]$

Specify and implement a data abstraction for interval arithmetic. Be sure to include the abstraction function, representation invariant, and rep_ok. Also implement a to_string function and a format that can be installed in the top level with #install_printer.

Exercise: function maps $[\star \star \star \star]$

Implement a map (aka dictionary) data structure with abstract type ('k, 'v) t. As the representation type, use 'k -> 'v. That is, a map is represented as an OCaml function from keys to values. Document the AF. You do not need an RI. Your solution will make heavy use of higher-order functions. Provide at least these values and operations: empty, mem, find, add, remove.

Exercise: set black box $[\star \star \star]$

Go back to the implementation of sets with lists in the previous chapter. Based on the specification comments of Set, write an OUnit test suite for ListSet that does black-box testing of all its operations.

Exercise: set glass box $[\star \star \star]$

Achieve as close to 100% code coverage with Bisect as you can for ListSet and UniqListSet.

Exercise: random lists $[\star \star \star]$

Use QCheck.Gen.generate1 to generate a list whose length is between 5 and 10, and whose elements are integers between 0 and 100. Then use QCheck.Gen.generate to generate a 3-element list, each element of which is a list of the kind you just created with generate1.

Then use QCheck.make to create an arbitrary that represents a list whose length is between 5 and 10, and whose elements are integers between 0 and 100. The type of your arbitrary should be int list QCheck.arbitrary.

Finally create and run a QCheck test that checks whether at least one element of an arbitrary list (of 5 to 10 elements, each between 0 and 100) is even. You'll need to "upgrade" the is_even property to work on a list of integers rather than a single integer.

Each time you run the test, recall that it will generate 100 lists and check the property of them. If you run the test many times, you'll likely see some successes and some failures.

Exercise: gcheck odd divisor $[\star \star \star]$

Here is a buggy function:

```
(** [odd_divisor x] is an odd divisor of [x].
    Requires: [x >= 0]. *)
let odd_divisor x =
    if x < 3 then 1 else
    let rec search y =
        if y >= x then y (* exceeded upper bound *)
        else if x mod y = 0 then y (* found a divisor! *)
        else search (y + 2) (* skip evens *)
    in search 3
```

Write a QCheck test to determine whether the output of that function (on a positive integer, per its precondition; *hint:* there is an arbitrary that generates positive integers) is both odd and is a divisor of the input. You will discover that there is a bug in the function. What is the smallest integer that triggers that bug?

Exercise: qcheck avg $[\star\star\star\star]$

Here is a buggy function:

```
(** [avg [x1; ...; xn]] is [(x1 + ... + xn) / n].
    Requires: the input list is not empty. *)
let avg lst =
    let rec loop (s, n) = function
    | [] -> (s, n)
```

(continues on next page)

6.11. Exercises 287

```
| [ h ] -> (s + h, n + 1)
| h1 :: h2 :: t -> if h1 = h2 then loop (s + h1, n + 1) t
    else loop (s + h1 + h2, n + 2) t
in
let (s, n) = loop (0, 0) lst
in float_of_int s /. float_of_int n
```

Write a QCheck test that detects the bug. For the property that you check, construct your own *reference implementation* of average—that is, a less optimized version of avg that is obviously correct.

Exercise: $\exp \left[\star \star \right]$

Prove that exp x (m + n) = exp x m * exp x n, where

```
let rec exp x n =
  if n = 0 then 1 else x * exp x (n - 1)
```

Proceed by induction on m.

Exercise: fibi [★★★]

Prove that forall $n \ge 1$, fib n = fibi n (0, 1), where

```
let rec fib n =
  if n = 1 then 1
  else if n = 2 then 1
  else fib (n - 2) + fib (n - 1)

let rec fibi n (prev, curr) =
  if n = 1 then curr
  else fibi (n - 1) (curr, prev + curr)
```

Proceed by induction on n, rather than trying to apply the theorem about converting recursion into iteration.

Exercise: expsq $[\star \star \star]$

Prove that expsq x n = exp x n, where

```
let rec expsq x n =
  if n = 0 then 1
  else if n = 1 then x
  else (if n mod 2 = 0 then 1 else x) * expsq (x * x) (n / 2)
```

Proceed by *strong induction* on n. Function expsq implements *exponentiation by repeated squaring*, which results in more efficient computation than exp.

Exercise: mult $[\star \star]$

Prove that for all n, mult n Z = Z by induction on n, where:

```
let rec mult a b =
  match a with
  | Z -> Z
  | S k -> plus b (mult k b)
```

Exercise: append nil $[\star \star]$

Prove that forall lst, lst @[] = lst by induction on lst.

Exercise: rev dist append $[\star\star\star]$

Prove that reverse distributes over append, i.e., that forall 1st1 1st2, rev (1st1 @ 1st2) = rev 1st2 @ rev 1st1, where:

```
let rec rev = function
    | [] -> []
    | h :: t -> rev t @ [h]
```

(That is, of course, an inefficient implementation of rev.) You will need to choose which list to induct over. You will need the previous exercise as a lemma, as well as the associativity of append, which was proved in the notes above.

Exercise: rev involutive $[\star \star \star]$

Prove that reverse is an involution, i.e., that forall lst, rev (rev lst) = lst. Proceed by induction on lst. You will need the previous exercise as a lemma.

Exercise: reflect size $[\star \star \star]$

Prove that forall t, size (reflect t) = size t by induction on t, where:

Exercise: fold theorem $2 \left[\star \star \star \star \right]$

```
let concat_l lst = List.fold_left ( ^ ) "" lst
let concat_r lst = List.fold_right ( ^ ) lst ""
```

Formulate and prove a new theorem about when fold_left and fold_right yield the same results, under the relaxed assumption that their function argument is associative but not necessarily commutative. *Hint: make a new assumption about the initial value of the accumulator.*

Exercise: propositions $[\star \star \star \star]$

6.11. Exercises 289

In propositional logic, we have atomic propositions, negation, conjunction, disjunction, and implication. For example, raining /\ snowing /\ cold is a proposition stating that it is simultaneously raining and snowing and cold (a weather condition known as *Ithacating*).

Define an OCaml type to represent propositions. Then state the induction principle for that type.

Exercise: list spec $[\star\star\star]$

Design an OCaml interface for lists that has nil, cons, append, and length operations. Design the equational specification. Hint: the equations will look strikingly like the OCaml implementations of @ and List.length.

Exercise: bag spec $[\star\star\star\star]$

A bag or multiset is like a blend of a list and a set: like a set, order does not matter; like a list, elements may occur more than once. The number of times an element occurs is its multiplicity. An element that does not occur in the bag has multiplicity 0. Here is an OCaml signature for bags:

```
module type Bag = sig
  type 'a t
  val empty : 'a t
  val is_empty : 'a t -> bool
  val insert : 'a -> 'a t -> 'a t
  val mult : 'a -> 'a t -> int
  val remove : 'a -> 'a t -> 'a t
end
```

Categorize the operations in the Bag interface as generators, manipulators, or queries. Then design an equational specification for bags. For the remove operation, your specification should cause at most one occurrence of an element to be removed. That is, the multiplicity of that value should decrease by at most one.

CHAPTER

SEVEN

MUTABILITY

OCaml is not a *pure* language: it does admit side effects. We have seen that already with I/O, especially printing. But up till now we have limited ourself to the subset of the language that is *immutable*: values could not change.

Mutability is neither good nor bad. It enables new functionality that we couldn't implement (at least not easily) before, and it enables us to create certain data structures that are asymptotically more efficient than their purely functional analogues. But mutability does make code more difficult to reason about, hence it is a source of many faults in code. One reason for that might be that humans are not good at thinking about change. With immutable values, we're guaranteed that any fact we might establish about them can never change. But with mutable values, that's no longer true. "Change is hard," as they say.

7.1 Refs

A *ref* is like a pointer or reference in an imperative language. It is a location in memory whose contents may change. Refs are also called *ref cells*, the idea being that there's a cell in memory that can change.

Here's an example of creating a ref, getting the value from inside it, changing its contents, and observing the changed contents:

```
let x = ref 0;;

val x : int ref = {contents = 0}

!x;;

- : int = 0

x := 1;;

- : unit = ()
!x;;

- : int = 1
```

The first phrase, let x = ref 0, creates a reference using the ref keyword. That's a location in memory whose contents are initialized to 0. Think of the location itself as being an address—for example, 0x3110bae0—even though there's no way to write down such an address in an OCaml program. The keyword ref is what causes the memory location to be allocated and initialized.

The first part of the response from OCaml, val x : int ref, indicates that x is a variable whose type is int ref. We have a new type constructor here. Much like list and option are type constructors, so is ref. At ref, for any type t, is a reference to a memory location that is guaranteed to contain a value of type t. As usual we should read a type from right to left: t ref means a reference to a t. The second part of the response shows us the contents of the memory location. Indeed, the contents have been initialized to 0.

The second phrase, !x, dereferences x and returns the contents of the memory location. Note that ! is the dereference operator in OCaml, not Boolean negation.

The third phrase, x := 1, is an assignment. It mutates the contents x to be 1. Note that x itself still points to the same location (i.e., address) in memory. Memory is mutable; variable bindings are not. What changes is the contents. The response from OCaml is simply (), meaning that the assignment took place—much like printing functions return () to indicate that the printing did happen.

The fourth phrase, ! x again dereferences x to demonstrate that the contents of the memory location did indeed change.

7.1.1 Aliasing

Now that we have refs, we have *aliasing*: two refs could point to the same memory location, hence updating through one causes the other to also be updated. For example,

```
let x = ref 42;;
let y = ref 42;;
let z = x;;
x := 43;;
let w = !y + !z;;
```

```
val x : int ref = {contents = 42}
```

```
val y : int ref = {contents = 42}
```

```
val z : int ref = {contents = 42}
```

```
- : unit = ()
```

```
val w : int = 85
```

The result of executing that code is that w is bound to 85, because let z = x causes z and x to become aliases, hence updating x to be 43 also causes z to be 43.

7.1.2 Syntax and Semantics

The semantics of refs is based on *locations* in memory. Locations are values that can be passed to and returned from functions. But unlike other values (e.g., integers, variants), there is no way to directly write a location in an OCaml program. That's different than languages like C, in which programmers can directly write memory addresses and do arithmetic on pointers. C programmers want that kind of low-level access to do things like interface with hardware and build operating systems. Higher-level programmers are willing to forego it to get *memory safety*. That's a hard term to define, but according to Hicks 2014 it intuitively means that

- pointers are only created in a safe way that defines their legal memory region,
- pointers can only be dereferenced if they point to their allotted memory region,
- that region is (still) defined.

Syntax.

```
Ref creation: ref e
Ref assignment: e1 := e2
Dereference: !e
```

Dynamic semantics.

- To evaluate ref e,
 - Evaluate e to a value ∨
 - Allocate a new location loc in memory to hold v
 - Store v in loc
 - Return loc
- To evaluate e1 := e2,
 - Evaluate e2 to a value v, and e1 to a location loc.
 - Store v in loc.
 - Return (), i.e., unit.
- To evaluate !e,
 - Evaluate e to a location loc.
 - Return the contents of loc.

Static semantics.

We have a new type constructor, ref, such that t ref is a type for any type t. Note that the ref keyword is used in two ways: as a type constructor, and as an expression that constructs refs.

```
ref e : t ref if e : t.
e1 := e2 : unit if e1 : t ref and e2 : t.
!e : t if e : t ref.
```

7.1.3 Sequencing of Effects

The semicolon operator is used to sequence effects, such as mutating refs. We've seen semicolon occur previously with printing. Now that we're studying mutability, it's time to treat it formally.

```
• Syntax: e1; e2
```

- **Dynamic semantics:** To evaluate e1; e2,
 - First evaluate e1 to a value v1.
 - Then evaluate e2 to a value v2.
 - Return v2. (v1 is not used at all.)
 - If there are multiple expressions in a sequence, e.g., e1; e2; ...; en, then evaluate each one in order from left to right, returning only vn. Another way to think about this is that semicolon is right associative—for example e1; e2; e3 is the same as e1; (e2; e3)).
- Static semantics: e1; e2: tife1: unit and e2: t. Similarly, e1; e2; ...; en: tife1: unit, e2: unit, ... (i.e., all expressions except en have type unit), and en: t.

7.1. Refs 293

The typing rule for semicolon is designed to prevent programmer mistakes. For example, a programmer who writes 2+3; 7 probably didn't mean to: there's no reason to evaluate 2+3 then throw away the result and instead return 7. The compiler will give you a warning if you violate this particular typing rule.

To get rid of the warning (if you're sure that's what you need to do), there's a function ignore: 'a -> unit in the standard library. Using it, ignore (2+3); 7 will compile without a warning. Of course, you could code up ignore yourself: let ignore _ = ().

7.1.4 Example: Mutable Counter

Here is code that implements a *counter*. Every time next_val is called, it returns one more than the previous time.

```
let counter = ref 0

let next_val =
  fun () ->
    counter := !counter + 1;
    !counter
```

```
val counter : int ref = {contents = 0}
```

```
val next_val : unit -> int = <fun>
```

```
next_val ()
```

```
- : int = 1
```

```
next_val ()
```

```
- : int = 2
```

```
next_val ()
```

```
-: int = 3
```

In the implementation of next_val, there are two expressions separated by semi-colon. The first expression, counter: = !counter + 1, is an assignment that increments counter by 1. The second expression, !counter, returns the newly incremented contents of counter.

The next_val function is unusual in that every time we call it, it returns a different value. That's quite different than any of the functions we've implemented ourselves so far, which have always been *deterministic*: for a given input, they always produced the same output. On the other hand, we've seen some library functions that are *nondeterministic*, for example, functions in the Random module, and Stdlib.read_line. It's no coincidence that those happen to be implemented using mutable features.

We could improve our counter in a couple ways. First, there is a library function incr: int ref -> unit that increments an int ref by 1. Thus it is like the ++ operator that is familiar from many languages in the C family. Using it, we could write incr counter instead of counter:=!counter + 1. (There's also a decr function that decrements by 1.)

Second, the way we coded the counter currently exposes the counter variable to the outside world. Maybe we're prefer to hide it so that clients of next_val can't directly change it. We could do so by nesting counter inside the scope of next_val:

```
let next_val =
  let counter = ref 0 in
  fun () ->
   incr counter;
  !counter
```

```
val next_val : unit -> int = <fun>
```

Now counter is in scope inside of next val, but not accessible outside that scope.

When we gave the dynamic semantics of let expressions before, we talked about substitution. One way to think about the definition of next_val is as follows.

- First, the expression ref 0 is evaluated. That returns a location loc, which is an address in memory. The contents of that address are initialized to 0.
- Second, everywhere in the body of the let expression that counter occurs, we substitute for it that location. So we get:

```
fun () -> incr loc; !loc
```

• Third, that anonymous function is bound to next_val.

So any time next_val is called, it increments and returns the contents of that one memory location loc.

Now imagine that we instead had written the following (broken) code:

```
let next_val_broken = fun () ->
  let counter = ref 0 in
  incr counter;
  !counter
```

```
val next_val_broken : unit -> int = <fun>
```

It's only a little different: the binding of counter occurs after the fun () -> instead of before. But it makes a huge difference:

```
next_val_broken ();;
next_val_broken ();;
next_val_broken ();;
```

```
- : int = 1
```

```
- : int = 1
```

```
-: int = 1
```

Every time we call next_val_broken, it returns 1: we no longer have a counter. What's going wrong here?

The problem is that every time <code>next_val_broken</code> is called, the first thing it does is to evaluate <code>ref 0</code> to a new location that is initialized to 0. That location is then incremented to 1, and 1 is returned. *Every* call to <code>next_val_broken</code> is thus allocating a new ref cell, whereas <code>next_val</code> allocates just *one* new ref cell.

7.1. Refs 295

7.1.5 Example: Pointers

In languages like C, pointers combine two features: they can be null, and they can be changed. (Java has a similar construct with object references, but that term is confusing in our OCaml context since "reference" currently means a ref cell. So we'll stick with the word "pointer".) Let's code up pointers using OCaml ref cells.

```
type 'a pointer = 'a ref option
```

```
type 'a pointer = 'a ref option
```

As usual, read that type right to left. The option part of it encodes the fact that a pointer might be null. We're using None to represent that possibility.

```
let null : 'a pointer = None
```

```
val null : 'a pointer = None
```

The ref part of the type encodes the fact that the contents are mutable. We can create a helper function to allocate and initialize the contents of a new pointer:

```
let malloc (x : 'a) : 'a pointer = Some (ref x)
```

```
val malloc : 'a -> 'a pointer = <fun>
```

Now we could create a pointer to any value we like:

```
let p = malloc 42
```

```
val p : int pointer = Some {contents = 42}
```

Dereferencing a pointer is the * prefix operator in C. It returns the contents of the pointer, and raises an exception if the pointer is null:

```
exception Segfault

let deref (ptr : 'a pointer) : 'a =
   match ptr with None -> raise Segfault | Some r -> !r
```

```
exception Segfault
```

```
val deref : 'a pointer -> 'a = <fun>
```

```
deref p
```

```
- : int = 42
```

```
deref null
```

```
Exception: Segfault.

Raised at deref in file "[17]", line 4, characters 25-39

Called from Toploop.load_lambda in file "toplevel/toploop.ml", line 212, characters_417-27
```

We could even introduce our own OCaml operator for dereference. We have to put \sim in front of it to make it parse as a prefix operator, though.

```
val ( ~* ) : 'a pointer -> 'a = <fun>
```

```
- : int = 42
```

In C, an assignment through a pointer is written p = x. That changes the memory to which p points, making it contain x. We can code up that operator as follows:

```
let assign (ptr : 'a pointer) (x : 'a) : unit =
  match ptr with None -> raise Segfault | Some r -> r := x
```

```
val assign : 'a pointer -> 'a -> unit = <fun>
```

```
assign p 2;
deref p
```

```
- : int = 2
```

```
assign null 0
```

```
Exception: Segfault.
Raised at assign in file "[21]", line 2, characters 25-39
Called from Toploop.load_lambda in file "toplevel/toploop.ml", line 212, characters_-
-17-27
```

Again, we could introduce our own OCaml operator for that, though it's hard to pick a good symbol involving * and = that won't be misunderstood as involving multiplication:

```
let ( =* ) = assign;;
p =* 3;;
~*p
```

```
val ( =* ) : 'a pointer -> 'a -> unit = <fun>
```

```
- : unit = ()
```

```
-: int = 3
```

The one thing we can't do is treat a pointer as an integer. C allows that, including taking the address of a variable, which enables *pointer arithmetic*. That's great for efficiency, but also terrible because it leads to all kinds of program errors and security vulnerabilities.

Evil Secret

Okay that wasn't actually true what we just said, but this is dangerous knowledge that you really shouldn't even read. There is an undocumented function Obj.magic that we could use to get a memory address of a ref:

7.1. Refs 297

```
let address (ptr : 'a pointer) : int =
  match ptr with None -> 0 | Some r -> Obj.magic r
let ( ~& ) = address
```

But you have to promise to never, ever use that function yourself, because it completely circumvents the safety of the OCaml type system. All bets are off if you do.

None of this pointer encoding is part of the OCaml standard library, because you don't need it. You can always use refs and options yourself as you need to. Coding as we just did above is not particularly idiomatic. The reason we did it was to illustrate the relationship between OCaml refs and C pointers (equivalently, Java references).

7.1.6 Example: Recursion Without Rec

Here's a neat trick that's possible with refs: we can build recursive functions without ever using the keyword rec. Suppose we want to define a recursive function such as fact, which we would normally write as follows:

```
let rec fact_rec n = if n = 0 then 1 else n * fact_rec (n - 1)
```

```
val fact_rec : int -> int = <fun>
```

We want to define that function without using rec. We can begin by defining a ref to an obviously incorrect version of the function:

```
let fact0 = ref (\mathbf{fun} \times -> \times + 0)
```

```
val fact0 : (int -> int) ref = {contents = <fun>}
```

The way in which fact 0 is incorrect is actually irrelevant. We just need it to have the right type. We could just as well have used fun $x \rightarrow x$ instead of fun $x \rightarrow x + 0$.

At this point, fact 0 clearly doesn't compute the factorial function. For example, 5! ought to be 120, but that's not what fact 0 computes:

```
!fact0 5
```

```
-: int = 5
```

Next, we write fact as usual, but without rec. At the place where we need to make the recursive call, we instead invoke the function stored inside fact 0:

```
let fact n = if n = 0 then 1 else n * !fact0 (n - 1)
```

```
val fact : int -> int = <fun>
```

Now fact does actually get the right answer for 0, but not for 5:

```
fact 0;;
fact 5;;
```

```
- : int = 1
```

```
- : int = 20
```

The reason it's not right for 5 is that the recursive call isn't actually to the right function. We want the recursive call to go to fact, not to fact0. **So here's the trick:** we mutate fact0 to point to fact:

```
fact0 := fact
```

```
- : unit = ()
```

Now when fact makes its recursive call and dereferences fact0, it gets back itself! That makes the computation correct:

```
fact 5
```

```
- : int = 120
```

Abstracting a little, here's what we did. We started with a function that is recursive:

```
let rec f x = ... f y ...
```

We rewrote it as follows:

```
let f0 = ref (fun x -> x)
let f x = ... !f0 y ...
f0 := f
```

Now f will compute the same result as it did in the version where we defined it with rec.

What's happening here is sometimes called "tying the recursive knot": we update the reference to f0 to point to f, such that when f dereferences f0, it gets itself back. The initial function to which we made f0 point (in this case the identity function) doesn't really matter; it's just there as a placeholder until we tie the knot.

7.1.7 Weak Type Variables

Perhaps you have already tried using the identity function to define fact0, as we mentioned above. If so, you will have encountered this rather puzzling output:

```
let fact0 = ref (fun x → x)
```

```
val fact0 : ('_weak1 -> '_weak1) ref = {contents = <fun>}
```

What is this strange type for the identity function, '_weak1 -> '_weak1? Why isn't it the usual 'a -> 'a?

The answer has to do with a particularly tricky interaction between polymorphism and mutability. In a later chapter on interpreters, we'll learn how type inference works, and at that point we'll be able to explain the problem in detail. In short, allowing the type 'a -> 'a for that ref would lead the possibility of programs that crash at run time because of type errors.

For now, think about it this way: although the *value* stored in a ref cell is permitted to change, the *type* of that value is not. And if OCaml gave ref (fun x \rightarrow x) the type ('a \rightarrow 'a) ref, then that cell could first store fun x \rightarrow x + 1: int \rightarrow int but later store fun x \rightarrow s $^$ "!": string \rightarrow string. That would be the kind of change in type that is not allowed.

7.1. Refs 299

So OCaml uses *weak type variables* to stand for unknown but not polymorphic types. These variables always start with _weak. Essentially, type inference for these is just not finished yet. Once you give OCaml enough information, it will finish type inference and replace the weak type variable with the actual type:

```
!fact0

- : '_weak1 -> '_weak1 = <fun>
!fact0 1

- : int = 1
!fact0
```

After the application of !fact0 to 1, OCaml now knows that the function is meant to have type int -> int. So from then on, that's the only type at which it can be used. It can't, for example, be applied to a string.

If you would like to learn more about weak type variables right now, take a look at Section 2 of *Relaxing the value restriction* by Jacques Garrigue, or this section of the OCaml manual.

7.1.8 Physical Equality

- : int -> int = <fun>

OCaml has two equality operators, physical equality and structural equality. The documentation of Stdlib. (==) explains physical equality:

e1 == e2 tests for physical equality of e1 and e2. On mutable types such as references, arrays, byte sequences, records with mutable fields and objects with mutable instance variables, e1 == e2 is true if and only if physical modification of e1 also affects e2. On non-mutable types, the behavior of (==) is implementation-dependent; however, it is guaranteed that e1 == e2 implies compare e1 e2 = 0.

One interpretation could be that == should be used only when comparing refs (and other mutable data types) to see whether they point to the same location in memory. Otherwise, don't use ==.

Structural equality is also explained in the documentation of Stdlib. (=):

e1 = e2 tests for structural equality of e1 and e2. Mutable structures (e.g. references and arrays) are equal if and only if their current contents are structurally equal, even if the two mutable objects are not the same physical object. Equality between functional values raises Invalid_argument. Equality between cyclic data structures may not terminate.

Structural equality is usually what you want to test. For refs, it checks whether the contents of the memory location are equal, regardless of whether they are the same location.

The negation of physical equality is !=, and the negation of structural equality is <>. This can be hard to remember.

Here are some examples involving equality and refs to illustrate the difference between structural equality (=) and physical equality (==):

```
let r1 = ref 42
let r2 = ref 42
```

```
val r1 : int ref = {contents = 42}
```

```
val r2 : int ref = {contents = 42}
```

A ref is physically equal to itself, but not to another ref that is a different location in memory:

```
r1 == r1
```

```
- : bool = true
```

```
r1 == r2
```

```
- : bool = false
```

```
r1 != r2
```

```
- : bool = true
```

Two refs that are at different locations in memory but store structurally equal values are themselves structurally equal:

```
r1 = r1
```

```
- : bool = true
```

```
r1 = r2
```

```
- : bool = true
```

```
r1 <> r2
```

```
- : bool = false
```

Two refs that store structurally unequal values are themselves structurally unequal:

```
ref 42 <> ref 43
```

```
- : bool = true
```

7.1.9 Example: Singly-linked Lists

OCaml's built-in singly-linked lists are functional, not imperative. But we can code up imperative singly-linked lists, of course, with refs. (We could also use the pointers we invented above, but that only makes the code more complicated.)

We start by defining a type 'a node for nodes of a list that contains values of type 'a. The next field of a node is itself another list.

7.1. Refs 301

```
(** An ['a node] is a node of a mutable singly-linked list. It contains a value
    of type ['a] and a link to the [next] node. *)

type 'a node = { next : 'a mlist; value : 'a }

(** An ['a mlist] is a mutable singly-linked list with elements of type ['a].
    The [option] represents the possibility that the list is empty.
    RI: The list does not contain any cycles. *)
and 'a mlist = 'a node option ref
```

```
type 'a node = { next : 'a mlist; value : 'a; }
and 'a mlist = 'a node option ref
```

To create an empty list, we simply return a ref to None:

```
(** [empty ()] is an empty singly-linked list. *)
let empty () : 'a mlist = ref None
```

```
val empty : unit -> 'a mlist = <fun>
```

Note the type of empty: instead of being a value, it is now a function. This is typical of functions that create mutable data structures. At the end of this section, we'll return to why empty *has* to be a function.

Inserting a new first element just requires creating a new node, linking from it to the original list, and mutating the list:

```
(** [insert_first lst n] mutates mlist [lst] by inserting value [v] as the
   first value in the list. *)
let insert_first (lst : 'a mlist) (v : 'a) : unit =
   lst := Some { next = ref !lst; value = v }
```

```
val insert_first : 'a mlist -> 'a -> unit = <fun>
```

Again, note the type of insert_first. Rather than returning an 'a mlist, it returns unit. This again is typical of functions that modify mutable data structures.

In both empty and insert_first, the use of unit makes the functions more like their equivalents in an imperative language. The constructor for an empty list in Java, for example, might not take any arguments (which is equivalent to taking unit). And the insert_first operation for a Java linked list might return void, which is equivalent to returning unit.

Finally, here's a conversion function from our new mutable lists to OCaml's built-in lists:

```
(** [to_list lst] is an OCaml list containing the same values as [lst]
  in the same order. Not tail recursive. *)
let rec to_list (lst : 'a mlist) : 'a list =
  match !lst with None -> [] | Some { next; value } -> value :: to_list next
```

```
val to_list : 'a mlist -> 'a list = <fun>
```

Now we can see mutability in action:

```
let lst0 = empty ();;
let lst1 = lst0;;
insert_first lst0 1;;
to_list lst1;;
```

```
val lst0 : '_weak2 mlist = {contents = None}

val lst1 : '_weak2 mlist = {contents = None}

- : unit = ()
```

```
- : int list = [1]
```

The change to lst0 mutates lst1, because they are aliases.

The type of empty. Returning to empty, why must it be a function? It might seem as though we could define it more simply as follows:

```
let empty = ref None
```

```
val empty : '_weak3 option ref = {contents = None}
```

But that would be committing the same mistake as when we broke the counter above. Now there is only ever *one* ref that gets created, hence there is only one list ever in existence:

```
let lst2 = empty;;
let lst3 = empty;;
insert_first lst2 2;;
insert_first lst3 3;;
to_list lst2;;
to_list lst3;;
```

```
val lst2 : '_weak3 option ref = {contents = None}
```

```
val 1st3 : '_weak3 option ref = {contents = None}
```

```
- : unit = ()
```

```
- : unit = ()
```

```
- : int list = [3; 2]
```

```
-: int list = [3; 2]
```

Note how the mutations affect both lists, because they are both aliases for the same ref.

By correctly making empty a function, we guarantee that a new ref is returned every time an empty list is created.

```
val empty () = ref None
val empty : unit -> 'a option ref = <fun>
```

It really doesn't matter what argument that function takes, since it will never use it. We could define it as any of these in principle:

```
let empty _ = ref None
let empty (b : bool) = ref None
let empty (n : int) = ref None
  (* etc. *)
```

7.1. Refs 303

```
val empty : 'a -> 'b option ref = <fun>
```

```
val empty : bool -> 'a option ref = <fun>
```

```
val empty : int -> 'a option ref = <fun>
```

But the reason we prefer unit as the argument type is to indicate to the client that the argument value is not going to be used. After all, there's nothing interesting that the function can do with the unit value. Another way to think about that would be that a function whose input type is unit is like a function or method in an imperative language that takes in no arguments. For example, in Java a linked list class could have a constructor that takes no arguments and creates an empty list:

```
class LinkedList {
  /** Returns an empty list. */
  LinkedList() { ... }
}
```

Mutable values. In mlist, the nodes of the list are mutable, but the values are not. If we wanted the values also to be mutable, we can make them refs too:

```
type 'a node = { next : 'a mlist; value : 'a ref }
and 'a mlist = 'a node option ref

let empty () : 'a mlist = ref None

let insert_first (lst : 'a mlist) (v : 'a) : unit =
    lst := Some { next = ref !lst; value = ref v }

let rec set (lst : 'a mlist) (n : int) (v : 'a) : unit =
    match (!lst, n) with
    | None, _ -> invalid_arg "out of bounds"
    | Some { value }, 0 -> value := v
    | Some { next }, _ -> set next (n - 1) v

let rec to_list (lst : 'a mlist) : 'a list =
    match !lst with None -> [] | Some { next; value } -> !value :: to_list next
```

Now rather than having to create new nodes if we want to change a value, we can directly mutate the value in a node:

```
let lst = empty ();;
insert_first lst 42;;
insert_first lst 41;;
to_list lst;;
set lst 1 43;;
to_list lst;;
```

```
val lst : '_weak4 mlist = {contents = None}
```

```
- : unit = ()
```

```
- : unit = ()
```

```
- : int list = [41; 42]
```

```
- : unit = ()
```

```
- : int list = [41; 43]
```

7.2 Mutable Fields

The fields of a record can be declared as mutable, meaning their contents can be updated without constructing a new record. For example, here is a record type for two-dimensional colored points whose color field c is mutable:

```
type point = {x : int; y : int; mutable c : string}
```

```
type point = { x : int; y : int; mutable c : string; }
```

Note that mutable is a property of the field, rather than the type of the field. In particular, we write mutable field: type, not field: mutable type.

The operator to update a mutable field is <- which is meant to look like a left arrow.

```
let p = {x = 0; y = 0; c = "red"}
```

```
val p : point = {x = 0; y = 0; c = "red"}
```

```
p.c <- "white"</pre>
```

```
- : unit = ()
```

```
p
```

```
- : point = {x = 0; y = 0; c = "white"}
```

Non-mutable fields cannot be updated that way:

```
p.x <- 3;;
```

```
File "[5]", line 1, characters 0-8:

1 | p.x <- 3;;

    ^^^^^^^

Error: The record field x is not mutable
```

- **Syntax:** e1.f <- e2
- Dynamic semantics: To evaluate e1.f <- e2, evaluate e2 to a value v2, and e1 to a value v1, which must have a field named f. Update v1.f to v2. Return ().
- Static semantics: e1.f <- e2 : unit if e1 : t1 and t1 = $\{\ldots; \text{ mutable } f : t2; \ldots\}$, and e2 : t2.

7.2. Mutable Fields 305

7.2.1 Refs Are Mutable Fields

It turns out that refs are actually implemented as mutable fields. In Stdlib we find the following declaration:

```
type 'a ref = { mutable contents : 'a }
```

And that's why when the toplevel outputs a ref it looks like a record: it is a record with a single mutable field named contents!

```
let r = ref 42
```

```
val r : int ref = {contents = 42}
```

The other syntax we've seen for records is in fact equivalent to simple OCaml functions:

```
let ref x = {contents = x}
```

```
val ref : 'a -> 'a ref = <fun>
```

```
let ( ! ) r = r.contents
```

```
val (!): 'a ref -> 'a = <fun>
```

```
let ( := ) r x = r.contents <- x</pre>
```

```
val ( := ) : 'a ref -> 'a -> unit = <fun>
```

The reason we say "equivalent" is that those functions are actually implemented not in OCaml itself but in the OCaml run-time, which is implemented mostly in C. Nonetheless the functions do behave the same as the OCaml source given above.

7.2.2 Example: Mutable Singly-Linked Lists

Using mutable fields, we can implement singly-linked lists almost the same as we did with references. The types for nodes and lists are simplified:

```
(** An ['a node] is a node of a mutable singly-linked list. It contains a value
    of type ['a] and optionally has a pointer to the next node. *)

type 'a node = {
    mutable next : 'a node option;
    value : 'a
}

(** An ['a mlist] is a mutable singly-linked list with elements of type ['a].
    RI: The list does not contain any cycles. *)

type 'a mlist = {
    mutable first : 'a node option;
}
```

```
type 'a node = { mutable next : 'a node option; value : 'a; }
```

```
type 'a mlist = { mutable first : 'a node option; }
```

And there is no essential difference in the algorithms for implementing the operations, but the code is slightly simplified because we don't have to use reference operations:

```
(** [insert_first lst n] mutates mlist [lst] by inserting value [v] as the
    first value in the list. *)
let insert_first (lst : 'a mlist) (v : 'a) =
    lst.first <- Some {value = v; next = lst.first}

(** [empty ()] is an empty singly-linked list. *)
let empty () : 'a mlist = {
    first = None
}

(** [to_list lst] is an OCaml list containing the same values as [lst]
    in the same order. Not tail recursive. *)
let to_list (lst : 'a mlist) : 'a list =
    let rec helper = function
    | None -> []
    | Some {next; value} -> value :: helper next
    in
    helper lst.first
```

```
val insert_first : 'a mlist -> 'a -> unit = <fun>
```

```
val empty : unit -> 'a mlist = <fun>
```

```
val to_list : 'a mlist -> 'a list = <fun>
```

7.2.3 Example: Mutable Stacks

We already know that lists and stacks can be implemented in quite similar ways. Let's use what we've learned from mutable linked lists to implement mutable stacks. Here is an interface:

```
module type MutableStack = sig
  (** ['a t] is the type of mutable stacks whose elements have type ['a].
     The stack is mutable not in the sense that its elements can
     be changed, but in the sense that it is not persistent:
      the operations [push] and [pop] destructively modify the stack. *)
 type 'a t
  (** Raised if [peek] or [pop] encounter the empty stack. *)
 exception Empty
  (** [empty ()] is the empty stack *)
 val empty : unit -> 'a t
  (** [push x s] modifies [s] to make [x] its top element.
     The rest of the elements are unchanged. *)
 val push : 'a -> 'a t -> unit
  (**[peek s] is the top element of [s].
    Raises: [Empty] is [s] is empty. *)
 val peek : 'a t -> 'a
  (** [pop s] removes the top element of [s].
```

(continues on next page)

7.2. Mutable Fields 307

```
Raises: [Empty] is [s] is empty. *)
val pop : 'a t -> unit
end
```

```
module type MutableStack =
   sig
   type 'a t
   exception Empty
   val empty : unit -> 'a t
   val push : 'a -> 'a t -> unit
   val peek : 'a t -> 'a
   val pop : 'a t -> unit
   end
```

Now let's implement the mutable stack with a mutable linked list.

```
module MutableRecordStack : MutableStack = struct
  (** An ['a node] is a node of a mutable linked list. It has
    a field [value] that contains the node's value, and
    a mutable field [next] that is [Null] if the node has
    no successor, or [Some n] if the successor is [n]. *)
 type 'a node = {value : 'a; mutable next : 'a node option}
 (** AF: An ['a t] is a stack represented by a mutable linked list.
    The mutable field [top] is the first node of the list,
    which is the top of the stack. The empty stack is represented
    by \{top = None\}. The node \{top = Some n\} represents the
    stack whose top is [n], and whose remaining elements are
    the successors of [n]. *)
 type 'a t = {mutable top : 'a node option}
 exception Empty
 let empty () = {top = None}
 let push x s = s.top <- Some {value = x; next = s.top}</pre>
 let peek s =
   match s.top with
   | None -> raise Empty
    | Some {value} -> value
 let pop s =
   match s.top with
    | None -> raise Empty
    | Some {next} -> s.top <- next
end
```

```
module MutableRecordStack : MutableStack
```

7.3 Arrays and Loops

Arrays are fixed-length mutable sequences with constant-time access and update. So they are similar in various ways to refs, lists, and tuples. Like refs, they are mutable. Like lists, they are (finite) sequences. Like tuples, their length is fixed in advance and cannot be resized.

The syntax for arrays is similar to lists:

```
let v = [|0.; 1.|]
```

```
val v : float array = [|0.; 1.|]
```

That code creates an array whose length is fixed to be 2 and whose contents are initialized to 0. and 1. The keyword array is a type constructor, much like list.

Later those contents can be changed using the <- operator:

```
v.(0) <- 5.
```

```
- : unit = ()
```

```
v
```

```
- : float array = [|5.; 1.|]
```

As you can see in that example, indexing into an array uses the syntax array. (index), where the parentheses are mandatory.

The Array module has many useful functions on arrays.

Syntax.

- Array creation: [|e0; e1; ...; en|]
- Array indexing: e1. (e2)
- Array assignment: e1. (e2) <- e3

Dynamic semantics.

- To evaluate [|e0; e1; ...; en|], evaluate each ei to a value vi, create a new array of length n+1, and store each value in the array at its index.
- To evaluate e1. (e2), evaluate e1 to an array value v1, and e2 to an integer v2. If v2 is not within the bounds of the array (i.e., 0 to n-1, where n is the length of the array), raise Invalid_argument. Otherwise, index into v1 to get the value v at index v2, and return v.
- To evaluate e1. (e2) <- e3, evaluate each expression ei to a value vi. Check that v2 is within bounds, as in the semantics of indexing. Mutate the element of v1 at index v2 to be v3.

Static semantics.

- [|e0; e1; ...; en|] : t array if ei : t for all the ei.
- e1.(e2) : tife1 : t array and e2 : int.
- e1.(e2) <- e3 : unit if e1 : t array and e2 : int and e3 : t.

Loops.

OCaml has while loops and for loops. Their syntax is as follows:

```
while e1 do e2 done
for x=e1 to e2 do e3 done
for x=e1 downto e2 do e3 done
```

Each of these three expressions evaluates the expression between do and done for each iteration of the loop; while loops terminate when e1 becomes false; for loops execute once for each integer from e1 to e2; for..to loops evaluate starting at e1 and incrementing x each iteration; for..downto loops evaluate starting at e1 and decrementing x each iteration. All three expressions evaluate to () after the termination of the loop. Because they always evaluate to (), they are less general than folds, maps, or recursive functions.

Loops are themselves not inherently mutable, but they are most often used in conjunction with mutable features like arrays—typically, the body of the loop causes side effects. We can also use functions like Array.iter, Array.map, and Array.fold_left instead of loops.

7.4 Hash Tables

The *hash table* is a widely used data structure whose performance relies upon mutability. The implementation of a hash table is quite involved compared to other data structures we've implemented so far. We'll build it up slowly, so that the need for and use of each piece can be appreciated.

7.4.1 Maps

Hash tables implement the *map* data abstraction. A map binds *keys* to *values*. This abstraction is so useful that it goes by many other names, among them *associative array*, *dictionary*, and *symbol table*. We'll write maps abstractly (i.e, mathematically; not actually OCaml syntax) as $\{k_1:v_1,k_2:v_2,\ldots,k_n:v_n\}$. Each k:v is a *binding* of key k to value v. Here are a couple of examples:

- A map binding a course number to something about it: {3110 : "Fun", 2110 : "OO"}.
- A map binding a university name to the year it was chartered: {"Harvard": 1636, "Princeton": 1746, "Penn": 1740, "Cornell": 1865}.

The order in which the bindings are abstractly written does not matter, so the first example might also be written {2110 : "OO", 3110 : "Fun"}. That's why we use set braces—they suggest that the bindings are a set, with no ordering implied.

Note: As that notation suggests, maps and sets are very similar. Data structures that can implement a set can also implement a map, and vice-versa:

- Given a map data structure, we can treat the keys as elements of a set, and simply ignore the values which the keys are bound to. This admittedly wastes a little space, because we never need the values.
- Given a set data structure, we can store key-value pairs as the elements. Searching for elements (hence insertion
 and removal) might become more expensive, because the set abstraction is unlikely to support searching for keys
 by themselves.

Here is an interface for maps:

```
module type Map = sig

(** [('k, 'v) t] is the type of maps that bind keys of type
     ['k] to values of type ['v]. *)

type ('k, 'v) t
```

(continues on next page)

```
(** [insert k v m] is the same map as [m], but with an additional
      binding from [k] to [v]. If [k] was already bound in [m],
      that binding is replaced by the binding to [v] in the new map. *)
 val insert : 'k -> 'v -> ('k, 'v) t -> ('k, 'v) t
  (** [find k m] is [Some v] if [k] is bound to [v] in [m],
      and [None] if not. *)
 val find : 'k -> ('k, 'v) t -> 'v option
  (** [remove k m] is the same map as [m], but without any binding of [k].
      If [k] was not bound in [m], then the map is unchanged. *)
 val remove : 'k -> ('k, 'v) t -> ('k, 'v) t
  (** [empty] is the empty map. *)
 val empty : ('k, 'v) t
  (** [of_list lst] is a map containing the same bindings as
      association list [lst].
      Requires: [1st] does not contain any duplicate keys. *)
 val of_list : ('k * 'v) list -> ('k, 'v) t
  (** [bindings m] is an association list containing the same
      bindings as [m]. There are no duplicates in the list. *)
 val bindings : ('k, 'v) t -> ('k * 'v) list
end
```

Next, we're going to examine three implementations of maps based on

- · association lists,
- arrays, and
- a combination of the above known as a *hash table with chaining*.

Each implementation will need a slightly different interface, because of constraints resulting from the underlying representation type. In each case we'll pay close attention to the AF, RI, and efficiency of the operations.

7.4.2 Maps as Association Lists

The simplest implementation of a map in OCaml is as an association list. We've seen that representation twice so far [1] [2]. Here is an implementation of Map using it:

(continues on next page)

7.4. Hash Tables 311

```
(** Efficiency: O(n). *)
  let remove k lst = List.filter (fun (k', _) -> k <> k') lst
  (** Efficiency: O(1). *)
  let empty = []
  (** Efficiency: 0(1). *)
  let of_list lst = lst
  (** [keys m] is a list of the keys in [m], without
      any duplicates.
      Efficiency: O(n log n). *)
  let keys m = m |> List.map fst |> List.sort_uniq Stdlib.compare
  (** [binding m k] is [(k, v)], where [v] is the value that [k]
       binds in [m].
       Requires: [k] is a key in [m].
       Efficiency: O(n). *)
  let binding m k = (k, List.assoc k m)
  (** Efficiency: O(n \log n) + O(n) * O(n), which is O(n^2). *)
  let bindings m = List.map (binding m) (keys m)
end
```

7.4.3 Maps as Arrays

Mutable maps are maps whose bindings may be mutated. The interface for a mutable map therefore differs from a immutable map. Insertion and removal operations for a mutable map therefore return unit, because they do not produce a new map but instead mutate an existing map.

An array can be used to represent a mutable map whose keys are integers. A binding from a key to a value is stored by using the key as an index into the array, and storing the binding at that index. For example, we could use an array to map office numbers to their occupants:

Office	Occupant
459	Fan
460	Gries
461	Clarkson
462	Muhlberger
463	does not exist

This kind of map is called a *direct address table*. Since arrays have a fixed size, the implementer now needs to know the client's desire for the *capacity* of the table (i.e., the number of bindings that can be stored in it) whenever an empty table is created. That leads to the following interface:

```
module type DirectAddressMap = sig
  (** [t] is the type of maps that bind keys of type int to values of
          type ['v]. *)
  type 'v t

  (** [insert k v m] mutates map [m] to bind [k] to [v]. If [k] was
          already bound in [m], that binding is replaced by the binding to
```

(continues on next page)

```
[v] in the new map. Requires: [k] is in bounds for [m]. *)
 val insert : int -> 'v -> 'v t -> unit
  (** [find k m] is [Some v] if [k] is bound to [v] in [m], and [None]
      if not. Requires: [k] is in bounds for [m]. *)
 val find : int -> 'v t -> 'v option
  (** [remove k m] mutates [m] to remove any binding of [k]. If [k] was
     not bound in [m], then the map is unchanged. Requires: [k] is in
     bounds for [m]. *)
 val remove : int -> 'v t -> unit
  (** [create c] creates a map with capacity [c]. Keys [0] through [c-1]
     are _in bounds_ for the map. *)
 val create : int -> 'v t
  (** [of_list c lst] is a map containing the same bindings as
      association list [1st] and with capacity [c]. Requires: [1st] does
     not contain any duplicate keys, and every key in [1st] is in
     bounds for capacity [c]. *)
 val of_list : int -> (int * 'v) list -> 'v t
  (** [bindings m] is an association list containing the same bindings
      as [m]. There are no duplicate keys in the list. *)
 val bindings : 'v t -> (int * 'v) list
end
```

Here is an implementation of that interface:

```
module ArrayMap : DirectAddressMap = struct
  (** AF: [[|Some v0; Some v1; ... |]] represents {0 : v0, 1 : v1, ...}.
     If element [i] of the array is instead [None], then [i] is not
     bound in the map.
     RI: None. *)
 type 'v t = 'v option array
  (** Efficiency: 0(1) *)
 let insert k v a = a.(k) <- Some v</pre>
  (** Efficiency: 0(1) *)
 let find k = a.(k)
  (** Efficiency: 0(1) *)
 let remove k = a.(k) < - None
  (** Efficiency: 0(c) *)
 let create c = Array.make c None
  (** Efficiency: O(c) *)
 let of_list c lst =
   (* O(c) *)
   let a = create c in
   (* O(c) * O(1) = O(c) *)
   List.iter (fun (k, v) -> insert k v a) lst;
  (** Efficiency: 0(c) *)
```

(continues on next page)

7.4. Hash Tables 313

Its efficiency is great! The insert, find, and remove operations are constant time. But that comes at the expense of forcing keys to be integers. Moreover, they need to be small integers (or at least integers from a small range), otherwise the arrays we use will need to be huge.

7.4.4 Maps as Hash Tables

Arrays offer constant time performance, but come with severe restrictions on keys. Association lists don't place those restrictions on keys, but they also don't offer constant time performance. Is there a way to get the best of both worlds? Yes (more or less)! *Hash tables* are the solution.

The key idea is that we assume the existence of a hash function hash: 'a -> int that can convert any key to a non-negative integer. Then we can use that function to index into an array, as we did with direct address tables. Of course, we want the hash function itself to run in constant time, otherwise the operations that use it would not be efficient.

That leads to the following interface, in which the client of the hash table has to pass in a hash function when a table is created:

```
module type TableMap = sig
  (** [('k, 'v) t] is the type of mutable table-based maps that bind
     keys of type ['k] to values of type ['v]. *)
 type ('k, 'v) t
  (** [insert k v m] mutates map [m] to bind [k] to [v]. If [k] was
      already bound in [m], that binding is replaced by the binding to
      [v]. *)
 val insert : 'k -> 'v -> ('k, 'v) t -> unit
  (** [find k m] is [Some v] if [m] binds [k] to [v], and [None] if [m]
     does not bind [k]. *)
 val find : 'k -> ('k, 'v) t -> 'v option
  (** [remove k m] mutates [m] to remove any binding of [k]. If [k] was
     not bound in [m], the map is unchanged. *)
 val remove : 'k -> ('k, 'v) t -> unit
  (** [create hash c] creates a new table map with capacity [c] that
     will use [hash] as the function to convert keys to integers.
     Requires: The output of [hash] is always non-negative, and [hash]
     runs in constant time. *)
 val create : ('k -> int) -> int -> ('k, 'v) t
  (** [bindings m] is an association list containing the same bindings
      as [m]. *)
 val bindings : ('k, 'v) t -> ('k * 'v) list
```

(continues on next page)

```
(** [of_list hash lst] creates a map with the same bindings as [lst],
    using [hash] as the hash function. Requires: [lst] does not
    contain any duplicate keys. *)
val of_list : ('k -> int) -> ('k * 'v) list -> ('k, 'v) t
end
```

```
module type TableMap =
    sig
    type ('k, 'v) t
    val insert : 'k -> 'v -> ('k, 'v) t -> unit
    val find : 'k -> ('k, 'v) t -> 'v option
    val remove : 'k -> ('k, 'v) t -> unit
    val create : ('k -> int) -> int -> ('k, 'v) t
    val bindings : ('k, 'v) t -> ('k * 'v) list
    val of_list : ('k -> int) -> ('k * 'v) list -> ('k, 'v) t
    end
```

One immediate problem with this idea is what to do if the output of the hash is not within the bounds of the array. It's easy to solve this: if a is the length of the array then computing (hash k) mod a will return an index that is within bounds.

Another problem is what to do if the hash function is not *injective*, meaning that it is not one-to-one. Then multiple keys could *collide* and need to be stored at the same index in the array. That's okay! We deliberately allow that. But it does mean we need a strategy for what to do when keys collide.

There are two well-known strategies for dealing with collisions. One is to store multiple bindings at each array index. The array elements are called *buckets*. Typically, the bucket is implemented as a linked list. This strategy is known by many names, including *chaining*, *closed addressing*, and *open hashing*. We'll use **chaining** as the name. To check whether an element is in the hash table, the key is first hashed to find the correct bucket to look in. Then, the linked list is scanned to see if the desired element is present. If the linked list is short, this scan is very quick. An element is added or removed by hashing it to find the correct bucket. Then, the bucket is checked to see if the element is there, and finally the element is added or removed appropriately from the bucket in the usual way for linked lists.

The other strategy is to store bindings at places other than their proper location according to the hash. When adding a new binding to the hash table would create a collision, the insert operation instead finds an empty location in the array to put the binding. This strategy is (confusingly) known as *probing*, *open addressing*, and *closed hashing*. We'll use **probing** as the name. A simple way to find an empty location is to search ahead through the array indices with a fixed stride (often 1), looking for an unused entry; this *linear probing* strategy tends to produce a lot of clustering of elements in the table, leading to bad performance. A better strategy is to use a second hash function to compute the probing interval; this strategy is called *double hashing*. Regardless of how probing is implemented, however, the time required to search for or add an element grows rapidly as the hash table fills up.

Chaining has often been preferred over probing in software implementations, because it's easy to implement the linked lists in software. Hardware implementations have often used probing, when the size of the table is fixed by circuitry. But some modern software implementations are re-examining the performance benefits of probing.

7.4. Hash Tables 315

Chaining Representation

Here is a representation type for a hash table that uses chaining:

```
type ('k, 'v) t = {
  hash : 'k -> int;
  mutable size : int;
  mutable buckets : ('k * 'v) list array
}
```

The buckets array has elements that are association lists, which store the bindings. The hash function is used to determine which bucket a key goes into. The size is used to keep track of the number of bindings currently in the table, since that would be expensive to compute by iterating over buckets.

Here are the AF and RI:

```
(** AF: If [buckets] is
  [| [(k11,v11); (k12,v12); ...];
       [(k21,v21); (k22,v22); ...];
       ... |]
  that represents the map
       {k11:v11, k12:v12, ...,
       k21:v21, k22:v22, ..., ...}.
  RI: No key appears more than once in array (so, no
       duplicate keys in association lists). All keys are
    in the right buckets: if [k] is in [buckets] at index
    [b] then [hash(k) = b]. The output of [hash] must always
    be non-negative. [hash] must run in constant time.*)
```

What would the efficiency of insert, find, and remove be for this rep type? All require

- hashing the key (constant time),
- indexing into the appropriate bucket (constant time), and
- finding out whether the key is already in the association list (linear in the number of elements in that list).

So the efficiency of the hash table depends on the number of elements in each bucket. That, in turn, is determined by how well the hash function distributes keys across all the buckets.

A terrible hash function, such as the constant function $fun \ k \ -> \ 42$, would put all keys into same bucket. Then every operation would be linear in the number n of bindings in the map—that is, O(n). We definitely don't want that.

Instead, we want hash functions that distribute keys more or less randomly across the buckets. Then the expected length of every bucket will be about the same. If we could arrange that, on average, the bucket length were a constant L, then insert, find, and remove would all in expectation run in time O(L).

Resizing

How could we arrange for buckets to have expected constant length? To answer that, let's think about the number of bindings and buckets in the table. Define the *load factor* of the table to be

```
# bindings # buckets
```

So a table with 20 bindings and 10 buckets has a load factor of 2, and a table with 10 bindings and 20 buckets has a load factor of 0.5. The load factor is therefore the average number of bindings in a bucket. So if we could keep the load factor constant, we could keep L constant, thereby keeping the performance to (expected) constant time.

Toward that end, note that the number of bindings is not under the control of the hash table implementer—but the number of buckets is. So by changing the number of buckets, the implementer can change the load factor. A common strategy is to keep the load factor from approximately 1/2 to 2. Then each bucket contains only a couple bindings, and expected constant-time performance is guaranteed.

There's no way for the implementer to know in advance, though, exactly how many buckets will be needed. So instead, the implementer will have to *resize* the bucket array whenever the load factor gets too high. Typically the newly allocated bucket will be of a size to restore the load factor to about 1.

Putting those two ideas together, if the load factor reaches 2, then there are twice as many bindings as buckets in the table. So by doubling the size of the array, we can restore the load factor to 1. Similarly, if the load factor reaches 1/2, then there are twice as many buckets as bindings, and halving the size of the array will restore the load factor to 1.

Resizing the bucket array to become larger is an essential technique for hash tables. Resizing it to become smaller, though, is not essential. As long as the load factor is bounded by a constant from above, we can achieve expected constant bucket length. So not all implementations will reduce the size of the array. Although doing so would recover some space, it might not be worth the effort. That's especially true if the size of the hash table cycles over time: although sometimes it becomes smaller, eventually it becomes bigger again.

Unfortunately, resizing would seem to ruin our expected constant-time performance though. Insertion of a binding might cause the load factor to go over 2, thus causing a resize. When the resize occurs, all the existing bindings must be rehashed and added to the new bucket array. Thus, insertion has become a worst-case linear time operation! The same is true for removal, if we resize the array to become smaller when the load factor is too low.

Implementation

The implementation of a hash table, below, puts together all the pieces we discussed above.

```
module HashMap : TableMap = struct
  (** AF and RI: above *)
 type ('k, 'v) t = {
   hash : 'k -> int;
   mutable size : int;
   mutable buckets : ('k * 'v) list array
  (** [capacity tab] is the number of buckets in [tab].
     Efficiency: O(1) *)
 let capacity {buckets} =
   Array.length buckets
  (** [load_factor tab] is the load factor of [tab], i.e., the number of
      bindings divided by the number of buckets. *)
 let load_factor tab =
   float_of_int tab.size /. float_of_int (capacity tab)
  (** Efficiency: O(n) *)
 let create hash n =
    {hash; size = 0; buckets = Array.make n []}
  (** [index k tab] is the index at which key [k] should be stored in the
     buckets of [tab].
      Efficiency: O(1) *)
 let index k tab =
    (tab.hash k) mod (capacity tab)
```

(continues on next page)

7.4. Hash Tables 317

```
(** [insert_no_resize k v tab] inserts a binding from [k] to [v] in [tab]
   and does not resize the table, regardless of what happens to the
   load factor.
   Efficiency: expected O(L) *)
let insert_no_resize k v tab =
 let b = index k tab in (* O(1) *)
 let old_bucket = tab.buckets.(b) in
 tab.buckets.(b) <- (k,v) :: List.remove_assoc k old_bucket; (* O(L) *)
 if not (List.mem_assoc k old_bucket) then
   tab.size <- tab.size + 1;</pre>
  ()
(** [rehash tab new_capacity] replaces the buckets array of [tab] with a new
   array of size [new_capacity], and re-inserts all the bindings of [tab]
   into the new array. The keys are re-hashed, so the bindings will
   likely land in different buckets.
   Efficiency: O(n), where n is the number of bindings. *)
let rehash tab new_capacity =
 (* insert (k, v) into tab *)
 let rehash_binding (k, v) =
    insert_no_resize k v tab
 in
 (* insert all bindings of bucket into tab *)
 let rehash_bucket bucket =
   List.iter rehash_binding bucket
 let old_buckets = tab.buckets in
 tab.buckets <- Array.make new_capacity []; (* O(n) *)
 tab.size <- 0;
  (* [rehash_binding] is called by [rehash_bucket] once for every binding *)
 Array.iter rehash_bucket old_buckets (* expected O(n) *)
(* [resize_if_needed tab] resizes and rehashes [tab] if the load factor
  is too big or too small. Load factors are allowed to range from
  1/2 to 2. *)
let resize_if_needed tab =
 let lf = load_factor tab in
 if 1f > 2.0 then
   rehash tab (capacity tab * 2)
 else if lf < 0.5 then</pre>
   rehash tab (capacity tab / 2)
 else ()
(** Efficiency: O(n) *)
let insert k v tab =
 insert_no_resize k v tab; (* O(L) *)
 resize_if_needed tab (* O(n) *)
(** Efficiency: expected O(L) *)
let find k tab =
 List.assoc_opt k tab.buckets.(index k tab)
(** [remove_no_resize k tab] removes [k] from [tab] and does not trigger
   a resize, regardless of what happens to the load factor.
    Efficiency: expected O(L) *)
let remove no resize k tab =
 let b = index k tab in
```

(continues on next page)

(continued from previous page)

```
let old_bucket = tab.buckets.(b) in
    tab.buckets.(b) <- List.remove assoc k tab.buckets.(b);
    if List.mem_assoc k old_bucket then
      tab.size <- tab.size - 1;
  (** Efficiency: O(n) *)
  let remove k tab =
    remove_no_resize k tab; (* O(L) *)
    resize_if_needed tab (* O(n) *)
  (** Efficiency: O(n) *)
  let bindings tab =
    Array.fold_left
      (fun acc bucket ->
         List.fold_left
            (* 1 cons for every binding, which is O(n) *)
            (fun acc (k,v) \rightarrow (k,v) :: acc)
           acc bucket)
      [] tab.buckets
  (** Efficiency: O(n^2) *)
  let of_list hash lst =
    let m = create hash (List.length lst) in (* O(n) *)
    List. iter (fun (k, v) \rightarrow insert k v m) lst; <math>(*n * O(n) is O(n^2) *)
end
```

An optimization of rehash is possible. When it calls insert_no_resize to re-insert a binding, extra work is being done: there's no need for that insertion to call remove_assoc or mem_assoc, because we are guaranteed the binding does not contain a duplicate key. We could omit that work. If the hash function is good, it's only a constant amount of work that we save. But if the hash function is bad and doesn't distribute keys uniformly, that could be an important optimization.

7.4.5 Hash Functions

Hash tables are one of the most useful data structures ever invented. Unfortunately, they are also one of the most misused. Code built using hash tables often falls far short of achievable performance. There are two reasons for this:

- Clients choose poor hash functions that do not distribute keys randomly over buckets.
- Hash table abstractions do not adequately specify what is required of the hash function, or make it difficult to provide
 a good hash function.

Clearly, a bad hash function can destroy our attempts at a constant running time. A lot of obvious hash function choices are bad. For example, if we're mapping names to phone numbers, then hashing each name to its length would be a very poor function, as would a hash function that used only the first name, or only the last name. We want our hash function to use all of the information in the key. This is a bit of an art. While hash tables are extremely effective when used well, all too often poor hash functions are used that sabotage performance.

Hash tables work well when the hash function looks random. If it is to look random, this means that any change to a key, even a small one, should change the bucket index in an apparently random way. If we imagine writing the bucket index as a binary number, a small change to the key should randomly flip the bits in the bucket index. This is called information *diffusion*. For example, a one-bit change to the key should cause every bit in the index to flip with 1/2 probability.

Client vs. implementer. As we've described it, the hash function is a single function that maps from the key type to a

7.4. Hash Tables 319

bucket index. In practice, the hash function is the composition of *two* functions, one provided by the client and one by the implementer. This is because the implementer doesn't understand the element type, the client doesn't know how many buckets there are, and the implementer probably doesn't trust the client to achieve diffusion.

The client function hash_c first converts the key into an integer hash code, and the implementation function hash_i converts the hash code into a bucket index. The actual hash function is the composition of these two functions. As a hash table designer, you need to figure out which of the client hash function and the implementation hash function is going to provide diffusion. If clients are sufficiently savvy, it makes sense to push the diffusion onto them, leaving the hash table implementation as simple and fast as possible. The easy way to accomplish this is to break the computation of the bucket index into three steps.

- 1. Serialization: Transform the key into a stream of bytes that contains all of the information in the original key. Two equal keys must result in the same byte stream. Two byte streams should be equal only if the keys are actually equal. How to do this depends on the form of the key. If the key is a string, then the stream of bytes would simply be the characters of the string.
- 2. Diffusion: Map the stream of bytes into a large integer *x* in a way that causes every change in the stream to affect the bits of *x* apparently randomly. There is a tradeoff in performance versus randomness (and security) here.
- 3. Compression: Reduce that large integer to be within the range of the buckets. For example, compute the hash bucket index as $x \mod m$. This is particularly cheap if m is a power of two.

Unfortunately, hash table implementations are rarely forthcoming about what they assume of client hash functions. So it can be hard to know, as a client, how to get good performance from a table. The more information the implementation can provide to a client about how well distributed keys are in buckets, the better.

7.4.6 Standard Library Hashtbl

Although it's great to know how to implement a hash table, and to see how mutability is used in doing so, it's also great *not* to have to implement a data structure yourself in your own projects. Fortunately the OCaml standard library does provide a module <code>Hashtbl</code> [sic] that implements hash tables. You can think of this module as the imperative equivalent of the functional <code>Map</code> module.

Hash function. The function <code>Hashtbl.hash: 'a -> int</code> takes responsibility for serialization and diffusion. It is capable of hashing any type of value. That includes not just integers but strings, lists, trees, and so forth. So how does it run in constant time, if the length of a tree or size of a tree can be arbitrarily large? It looks only at a predetermined number of *meaningful nodes* of the structure it is hashing. By default, that number is 10. A meaningful node is an integer, floating-point number, string, character, booleans or constant constructor. You can see that as we hash these lists:

```
Hashtbl.hash [1; 2; 3; 4; 5; 6; 7; 8; 9];;
Hashtbl.hash [1; 2; 3; 4; 5; 6; 7; 8; 9; 10];;
Hashtbl.hash [1; 2; 3; 4; 5; 6; 7; 8; 9; 10; 11];;
Hashtbl.hash [1; 2; 3; 4; 5; 6; 7; 8; 9; 10; 11; 12];;
```

```
- : int = 635296333
```

```
- : int = 822221246
```

```
- : int = 822221246
```

```
- : int = 822221246
```

The hash values stop changing after the list goes beyond 10 elements. That has implications for how we use this built-in hash function: it will not necessarily provide good diffusion for large data structures, which means performance could degrade as collisions become common. To support clients who want to hash such structures, Hashtble provides another function hash_param which can be configured to examine more nodes.

Hash table. Here's an abstract of the hash table interface:

```
module type Hashtbl = struct
  type ('a, 'b) t
  val create : int -> ('a, 'b) t
  val add : ('a, 'b) t -> 'a -> 'b -> unit
  val find : ('a, 'b) t -> 'a -> 'b
  val remove : ('a, 'b) t -> 'a -> unit
  ...
end
```

The representation type ('a, 'b) Hashtbl.t maps keys of type 'a to values of type 'b. The create function initializes a hash table to have a given capacity, as our implementation above did. But rather than requiring the client to provide a hash function, the module uses Hashtbl.hash.

Resizing occurs when the load factor exceeds 2. Let's see that happen. First, we'll create a table and fill it up:

```
open Hashtbl;;
let t = create 16;;
for i = 1 to 16 do
   add t i (string_of_int i)
done;;
```

```
val t : ('_weak1, '_weak2) Hashtbl.t = <abstr>
```

```
- : unit = ()
```

We can query the hash table to find out how the bindings are distributed over buckets with Hashtbl.stats:

```
stats t
```

```
- : Hashtbl.statistics = {num_bindings = 16; num_buckets = 16; max_bucket_length = 3; bucket_histogram = [|6; 5; 4; 1|]}
```

The number of bindings and number of buckets are equal, so the load factor is 1. The bucket histogram is an array a in which a. (i) is the number of buckets whose size is i.

Let's pump up the load factor to 2:

```
for i = 17 to 32 do
  add t i (string_of_int i)
done;;
stats t;;
```

```
- : unit = ()
```

```
- : Hashtbl.statistics = {num_bindings = 32; num_buckets = 16; max_bucket_length = 4; bucket_histogram = [|3; 3; 3; 5; 2|]}
```

Now adding one more binding will trigger a resize, which doubles the number of buckets:

```
add t 33 "33";;
stats t;;
```

7.4. Hash Tables 321

```
- : unit = ()
```

```
-: Hashtbl.statistics = {num_bindings = 33; num_buckets = 32; max_bucket_length = 3; bucket_histogram = [|11; 11; 8; 2|]}
```

But Hashtbl does not implement resize on removal:

```
for i = 1 to 33 do
  remove t i
done;;
stats t;;
```

```
- : unit = ()
```

```
-: Hashtbl.statistics = {num_bindings = 0; num_buckets = 32; max_bucket_length = 0; bucket_histogram = [|32|]}
```

The number of buckets is still 32, even though all bindings have been removed.

Note: Java's HashMap has a default constructor HashMap () that creates an empty hash table with a capacity of 16 that resizes when the load factor exceeds 0.75 rather than 2. So Java hash tables would tend to have a shorter bucket length than OCaml hash tables, but also would tend to take more space to store because of empty buckets.

Client-provided hash functions. What if a client of Hashtbl found that the default hash function was leading to collisions, hence poor performance? Then it would make sense to change to a different hash function. To support that, Hashtbl provides a functorial interface similar to Map. The functor is Hashtbl.Make, and it requires an input of the following module type:

```
module type HashedType = sig
  type t
  val equal : t -> t -> bool
  val hash : t -> int
end
```

Type t is the key type for the table, and the two functions equal and hash say how to compare keys for equality and how to hash them. If two keys are equal according to equal, they must have the same hash value according to hash. If that requirement were violated, the hash table would no longer operate correctly. For example, suppose that equal k1 k2 holds but hash k1 <> hash k2. Then k1 and k2 would be stored in different buckets. So if a client added a binding of k1 to v, then looked up k2, they would not get v back.

Note: That final requirement might sound familiar from Java. There, if you override Object.equals() and Object.hashCode() you must ensure the same correspondence.

Part IV

Appendix

BIG-OH NOTATION

What does it mean to be *efficient*? Cornell professors Bobby Kleinberg and Eva Tardos have a wonderful explanation in chapter 2 of their textbook, *Algorithm Design* (2006). This appendix is a summary and reinterpretation of that explanation from a functional programming perspective. The ultimate answer will be that an algorithm is *efficient* if its worst-case running time on input size n is $O(n^d)$ for some constant d. But it will take us several steps to build up to that definition.

8.1 Algorithms and Efficiency, Attempt 1

A naive attempt at defining efficiency might proceed as follows:

Attempt 1: An algorithm is efficient if, when implemented, it runs in a small amount of time on particular input instances.

But there are many problems with that definition, such as:

- Inefficient algorithms can run quickly on small test cases.
- Fast processors and optimizing compilers can make inefficient algorithms run quickly.
- Efficient algorithms can run slowly when coded sloppily.
- Some input instances are harder than others.
- Efficiency on small inputs doesn't imply efficiency on large inputs.
- Some clients can afford to be more patient than others; quick for me might be slow for you.

Lesson 1: One lesson learned from that attempt is: time measured by a clock is not the right metric for algorithm efficiency. We need a metric that is reasonably independent of the hardware, compiler, other software that is running, etc. Perhaps a good metric would be to give up on time and instead count the number of *steps* taken during evaluation.

But, now we have a new problem: how should we define a "step"? It needs to be something machine independent. It ought to somehow represent a primitive unit of computation. There's a lot of flexibility. Here are some common choices:

- In pseudocode, we might think of a step as being a single line.
- In imperative languages, we might count assignments, array indexes, pointer dereferences, and arithmetic operations as steps.
- In OCaml, we could count function or operator application, let bindings, and choosing a branch of an if or match as steps.

In reality, all of those "steps" could really take a different amount of time. But in practice, that doesn't really matter much.

Lesson 2: Another lesson we learned from attempt 1 was: running time on a particular input instance is not the right metric. We need a metric that can predict running time on any input instance. So instead of using the particular input (e.g., a number, or a matrix, or a text document), it might be better to use the *size* of the input (e.g., the number of bits it takes to represent the number, or the number of rows and columns in the matrix, or the number of bytes in the document) as the metric.

But again we have a new problem: how to define "size"? As in the examples we just gave, size should be some measure of how big an input is compared to other inputs. Perhaps the most common representation of size in the context of data structures is just the number of elements maintained by the data structure: the number of nodes in a list, or the number of nodes and edges in a graph, etc.

Could an algorithm run in a different amount of time on two inputs of the same "size"? Sure. For example, multiplying a matrix by all zeroes might be faster than multiplying by arbitrary numbers. But in practice, size matters more than exact inputs.

Lesson 3: A third lesson we learned from attempt 1 was that "small amount of time" is too relative a term. We want a metric that is reasonably objective, rather than relying on subjective notions of what constitutes "small".

One sort-of-okay idea would be that an efficient algorithm needs to beat *brute-force search*. That means enumerating all answers one-by-one, checking each to see whether it's right. For example, a brute-force sorting algorithm would enumerate every possible permutation of a list, checking to see whether it's a sorted version of the input list. That's a terrible sorting algorithm! Certainly quicksort beats it.

Brute-force search is the simple, dumb solution to nearly any algorithmic problem. But it requires enumeration of a huge space. In fact, an exponentially-sized space. So a better idea would be doing less than exponential work. What's less than exponential (e.g., 2^n)? One possibility is polynomial (e.g., n^2).

An immediate objection might be that polynomials come in all sizes. For example, n^{100} is way bigger than n^2 . And some non-polynomials, such as $n^{1+.02(\log n)}$, might do an adequate job of beating exponentials. But in practice, polynomials do seem to work fine.

8.2 Algorithms and Efficiency, Attempt 2

Combining lessons 1 through 3 from Attempt 1, we have a second attempt at defining efficiency:

Attempt 2: An algorithm is efficient if its maximum number of execution steps is polynomial in the size of its input.

Note how all three ideas come together there: steps, size, polynomial.

But if we try to put that definition to use, it still isn't perfect. Coming up with an exact formula for the maximum number of execution steps can be insanely tedious. For example, in one other algorithm textbook, the authors develop this following polynomial for the number of execution steps taken by a pseudo-code implementation of insertion sort:

$$c_1 n + c_2 (n-1) + c_4 (n-1) + c_5 \sum_{j=2}^n t_j + c_6 \sum_{j=2}^n (t_j - 1) + c_7 \sum_{j=2}^n (t_j - 1) + c_8 (n-1)$$

No need for us to explain what all the variables mean. It's too complicated. Our hearts go out to the poor grad student who had to work out that one!

Note: That formula for running time of insertion sort is from *Introduction to Algorithms*, 3rd edition, 2009, by Cormen, Leiserson, Rivest, and Stein. We aren't making fun of them. They would also tell you that such formulas are too complicated.

Precise execution bounds like that are exhausting to find and somewhat meaningless. If it takes 25 steps in Java pseudocode, but compiled down to RISC-V would take 250 steps, is the precision useful?

In some cases, yes. If you're building code that flies an airplane or controls a nuclear reactor, you might actually care about precise, real-time guarantees.

But otherwise, it would be better for us to identify broad classes of algorithms with similar performance. Instead of saying that an algorithm runs in

$$1.62n^2 + 3.5n + 8$$

steps, how about just saying it runs in n^2 steps? That is, we could ignore the *low-order terms* and the *constant factor* of the highest-order term.

We ignore low-order terms because we want to THINK BIG. Algorithm efficiency is all about explaining the performance of algorithms when inputs get really big. We don't care so much about small inputs. And low-order terms don't matter when we think big. The following table shows the number of steps as a function of input size N, assuming each step takes 1 microsecond. "Very long" means more than the estimated number of atoms in the universe.

N	N^2	N^3	2^N
N=10	< 1 sec	< 1 sec	< 1 sec
N=100	< 1 sec	< 1 sec	1 sec
N=1,000	< 1 sec	1 sec	18 min
N=10,000	< 1 sec	2 min	12 days
N=100,000	< 1 sec	3 hours	32 years
N=1,000,000	1 sec	12 days	104 years

As you can see, when inputs get big, there's a serious difference between N^3 and N^2 and N. We might as well ignore low-order terms, because they are completely dominated by the highest-order term when we think big.

What about constant factors? My current laptop might be 2x faster (that is, a constant factor of 2) than the one I bought several years ago, but that's not an interesting property of the algorithm. Likewise, $1.62n^2$ steps in pseduocode might be $1620n^2$ steps in assembly (that is, a constant factor of 1000), but it's again not an interesting property of the algorithm. So, should we really care if one algorithm takes 2x or 1000x longer than another, if it's just a constant factor?

The answer is: maybe. Performance tuning in real-world code is about getting the constants to be small. Your employer might be really happy if you make something run twice as fast! But that's not about the **algorithm.** When we're measuring algorithm efficiency, in practice the constant factors just don't matter much.

So all that argues for having an **imprecise abstraction** to measure running time. Instead of $1.62n^2 + 3.5n + 8$, we can just write n^2 . Imprecise abstractions are nothing new to you. You might write ± 1 to imprecisely abstract a quantity within 1. In computer science, you already know that we use Big-Oh notation as an imprecise abstraction: $1.62n^2 + 3.5n + 8$ is $O(n^2)$.

8.3 Big-Ell Notation

Before reviewing Big-Oh notation, let's start with something simpler that you might not have seen before: Big-Ell notation.

Big-Ell is an imprecise abstraction of natural numbers less than or equal to another number, hence the L. It's defined as follows:

$$L(n) = \{m \mid 0 \leq m \leq n\}$$

where m and n are natural numbers. That is, L(n) represents all the natural numbers less than or equal to n. For example, $L(5) = \{0, 1, 2, 3, 4, 5\}$.

Could you do arithmetic with Big-Ell? For example, what would 1+L(5) be? It's not a well-posed question, to be honest: addition is an operation we think of being defined on integers, not sets of integers. But a reasonable interpretation of $1+\{0,1,2,3,4,5\}$ could be doing the addition for each element in the set, yielding $\{1,2,3,4,5,6\}$. Note that $\{1,2,3,4,5,6\}$ is a proper subset of $\{0,1,2,3,4,5,6\}$, and the latter is L(6). So we could say that $1+L(5)\subseteq L(6)$. We could even say that $1+L(5)\subseteq L(7)$, but it's not *tight*: the former subset relation included the fewest possible extra elements, whereas the latter was *loose* by needlessly including extra.

For more about Big Ell, see Concrete Mathematics, chapter 9, 1989, by Graham, Knuth, and Patashnik.

8.3. Big-Ell Notation

8.4 Big-Oh Notation

If you understand Big-Ell, and you understand functional programming, here's some good news: you can easily understand Big-Oh.

Let's build up the definition of Big-Oh in a few steps. We'll start with version 1, which we'll write as O_1 . It's based on L:

- L(n) represents any **natural number** that is less than or equal to a **natural number** n.
- $O_1(g)$ represents any **natural function** that is less than or equal to a **natural function** g.

A *natural function* is just a function on natural numbers; that is, its type is $\mathbb{N} \to \mathbb{N}$.

All we do with O_1 is upgrade from **natural numbers** to **natural functions**. So Big-Oh version 1 is just the *higher-order* version of Big-Ell. How about that!

Of course, we need to work out what it means for a function to be less than another function. Here's a reasonable formalization:

Big-Oh Version 1:
$$O_1(g) = \{ f \mid \forall n : f(n) \leq g(n) \}$$

For example, consider the function that doubles its input. In math textbooks, that function might be written as g(n)=2n. In OCaml we would write let g n = 2 * n or let g = fun n -> 2 * n or just anonymously as fun n -> 2 * n. In math that same anonymous function would be written with lambda notation as $\lambda n.2n$. Proceeding with lambda notation, we have:

$$O_1(\lambda n.2n) = \{ f \mid \forall n.f(n) \le 2n \}$$

and therefore

- $(\lambda n.n) \in O_1(\lambda n.2n)$,
- $(\lambda n.\frac{n}{2}) \in O_1(\lambda n.2n)$, but
- $(\lambda n.3n) \notin O_1(\lambda n.2n)$.

Next, recall that in defining algorithmic efficiency, we wanted to ignore constant factors. O_1 does not help us with that. We'd really like for all these functions:

- $(\lambda n.n)$
- $(\lambda n.2n)$
- $(\lambda n.3n)$

to be in $O(\lambda n.n)$.

Toward that end, let's define O_2 to ignore constant factors:

Big-Oh Version 2:
$$O_2(g) = \{ f \mid \exists c > 0 \forall n : f(n) \le cg(n) \}$$

That existentially-quantified positive constant c lets us "bump up" the function g to whatever constant factor we need. For example,

$$O_2(\lambda n.n^3) = \{ f \mid \exists c > 0 \forall n. f(n) \le cn^3 \}$$

and therefore $(\lambda n.3n^3) \in O_2(\lambda n.n^3)$, because $3n^3 \le cn^3$ if we take c=3, or c=4, or any larger c.

Finally, recall that we don't care about small inputs: we want to THINK BIG when we analyze algorithmic efficiency. It doesn't matter whether the running time of an algorithm happens to be a little faster or a little slower for small inputs. In fact, we could just hardcode a lookup table for those small inputs if the algorithm is too slow on them! What matters really is the performance on big-sized inputs.

Toward that end, let's define O_3 to ignore small inputs:

Big-Oh Version 3:
$$O_3(g) = \{ f \mid \exists c > 0 \exists n_0 > 0 \forall n \geq n_0 : f(n) \leq cg(n) \}$$

That existentially quantified positive constant n_0 lets us "ignore" all inputs of that size or smaller. For example,

$$O_{3}(\lambda n.n^{2})=\{f\mid \exists c>0 \exists n_{0}>0 \forall n\geq n_{0}\;.\; f(n)\leq cn^{2}\}$$

and therefore $(\lambda n.2n) \in O_3(\lambda n.n^2)$, because $2n \le cn^2$ if we take c=2 and $n_0=2$. Note how we get to ignore the fact that $\lambda n.2n$ is temporarily a little too big at n=1 by picking $n_0=2$. That's the power of ignoring "small" inputs.

8.5 Big-Oh, Finished

Version 3 is the right definition of Big-Oh. We repeat it here, for real:

Big-Oh:
$$O(g) = \{ f \mid \exists c > 0 \exists n_0 > 0 \forall n \ge n_0 : f(n) \le cg(n) \}$$

That's the final, important version you should know. But don't just memorize it. If you understand the derivation we gave here, you'll be able to recreate it from scratch anytime you need it.

Big-Oh is called an asymptotic upper bound. If $f \in O(g)$, then f is at least as efficient as g, and might be more efficient.

8.6 Big-Oh Notation Warnings

Warning 1. Because it's an upper bound, we can always inflate a Big-Oh statement: for example, if $f \in O(n^2)$, then also $f \in O(n^3)$, and $f \in O(2^n)$, etc. But our goal is always to give *tight* upper bounds, whether we explicitly say that or not. So when asked what the running time of an algorithm is, you must always give the tightest bound you can with Big-Oh.

Warning 2. Instead of $O(g) = \{f \mid ...\}$, most authors instead write $O(g(n)) = \{f(n) \mid ...\}$. They don't really mean g applied to n. They mean a function g parameterized on input n but not yet applied. This is badly misleading and generally a result of not understanding anonymous functions. Moral of that story: more people need to study functional programming.

Warning 3. Instead of $\lambda n.2n \in O(\lambda n.n^2)$ nearly all authors write $2n = O(n^2)$. This is a hideous and inexcusable abuse of notation that should never have been allowed and yet has permanently infected the computer science consciousness. The standard defense is that = here should be read as "is" not as "equals". That is patently ridiculous, and even those who make that defense usually have the good grace to admit it's nonsense. Sometimes we become stuck with the mistakes of our ancestors. This is one of those times. Be careful of this "one-directional equality" and, if you ever have a chance, teach your (intellectual) children to do better.

8.7 Algorithms and Efficiency, Attempt 3

Let's review. Our first attempt at defining efficiency was:

Attempt 1: An algorithm is efficient if, when implemented, it runs in a small amount of time on particular input instances.

By replacing time with steps, particular instances with input size, and small with polynomial, we improved that to:

Attempt 2: An algorithm is efficient if its maximum number of execution steps is polynomial in the size of its input.

And that's really a pretty good definition. But using Big-Oh notation to make it a little more concrete, we can produce our third and final attempt:

Attempt 3: An algorithm is efficient if its worst-case running time on input size n is $O(n^d)$ for some constant d.

By "worst-case running time" we mean the same thing as "maximum number of execution steps", just expressed in different and probably more common words. The worst-case is when execution takes the longest. "Time" is a common euphemism here for execution steps, and is used to emphasize we're thinking about how long a computation takes.

Space is the most common other feature of efficiency to consider. Algorithms can be more or less efficient at requiring constant or linear space, for example. You're already familiar with that from tail recursion and lists in OCaml.